

On Proportional Braking in H Bridges

Robert Atkinson
31 August 2018

Most H bridge designs use one of two common switching modalities, which are usually termed 'lock anti-phase' and 'sign-magnitude' respectively (Ref: <http://www.modularcircuits.com/blog/articles/h-bridge-secrets>). However, the STMicroelectronics VN5050E H bridge, used in FTC motor controllers for nearly a decade, employs neither of these modalities, instead using what can be called an 'asynchronous sign-magnitude' switching modality. In addition to its driving modes, as was observed by Eric & Douglas Chin, this modality admits a mode wherein dynamic-braking is applied in proportional to the PWM duty cycle, which we term 'proportional braking'. Here, we explore why using proportional braking might be advantageous, and the details of doing so.

1. Introduction

1.1. Administrative


Before we begin, we load in some previously computed logic.

```
Get[NotebookDirectory[] <> "Utilities.m"]
inputDirectory = FileNameJoin[{NotebookDirectory[], "MotorPhysicsGearsInitialConditions.Output"}] <> $PathnameSeparator;
Get[inputDirectory <> "ParametersUnitsAndAssumptions.m"];
Get[inputDirectory <> "MotorModels.m"];
Get[inputDirectory <> "MotorTimeDomainFunctions.m"];
Get[inputDirectory <> "Misc.m"];
```

```
SetOptions[Plot, LabelStyle → Directive[Background → None]];
SetOptions[ListPlot, LabelStyle → Directive[Background → None]];
prettyPrintFontSize = 20;
framed[expr_] := Framed[expr, FrameStyle → Darker[Green]]
```

1.2. H Bridge Architecture

In order to appreciate the design task explored in this document, it is important to have a comfortable, detailed working understanding of the operation of H bridges. From a simplified perspective, an H Bridge can abstractly be thought of as a circuit involving a motor, a voltage source, and four switches (Ref: https://en.wikipedia.org/wiki/H_bridge):




In this document, we need to be able to discuss with some precision the state of these switches. As there are four switches, there are 16 possible states. We will number those states by assigning weights to each switch, and summing over the weights of the switches which are closed in any particular state. We assign weights as follows: $S1 \rightarrow 8$, $S2 \rightarrow 4$, $S3 \rightarrow 2$, and $S4 \rightarrow 1$. With that nomenclature in hand, the motor's response in each of the 16 states is as follows:

State #	S1	S2	S3	S4	Instantaneous Velocity	
					CW	CCW
9	1	0	0	1	Drive CW	Drive CW
6	0	1	1	0	Drive CCW	Drive CCW
0	0	0	0	0	Coast	Coast
8	1	0	0	0	Coast	Brake to VCC
4	0	1	0	0	Brake to GND	Coast
2	0	0	1	0	Brake to VCC	Coast
1	0	0	0	1	Coast	Brake to GND
5	0	1	0	1	Brake to GND	Brake to GND
10	1	0	1	0	Brake to VCC	Brake to VCC
12	0	0	0	0	Short Circuit	Short Circuit
3	0	0	0	0	Short Circuit	Short Circuit
7	0	0	0	0	Short Circuit	Short Circuit
11	0	0	0	0	Short Circuit	Short Circuit
13	0	0	0	0	Short Circuit	Short Circuit
14	0	0	0	0	Short Circuit	Short Circuit
15	0	0	0	0	Short Circuit	Short Circuit

Note that in a design where an H bridge is driven by pulse width modulation, we can talk with precision about the operation of the bridge by naming the state used in the “on” part of the PWM cycle, the state used in the “off” part of the cycle, and the duty-cycle percentage of the “on” state as a fraction of the PWM framing period (which is contextually herein taken as fixed). Thus, for example, we might talk about an H bridge using a 6/2 mode being driven at a 75% duty cycle, or a bridge in 5/0 mode with a 30% duty cycle, etc.

States 9 and 6 are the driving states. In state 9, the motor turns one direction, which we have arbitrarily labeled as CW (or, more precisely, it attempts to increase its angular velocity in the CW direction). In state 6, the motor turns CCW. In state 0 it coasts, while in states 5 and 10 rheostatic dynamic braking is employed (Ref: https://en.wikipedia.org/wiki/Dynamic_braking), either through the ground rail or the power rail. Seven of the states are short circuits, and should be avoided.

States 8, 4, 2, and 1 deserve special mention. Their behavior is dependent not only on the state of the switches, but also on the sign of the instantaneous angular velocity. This is due to consequences of the fact that a more complete picture of an H bridge must include the so-called ‘catch diodes’ whose presence is necessary to avoid inductive voltage spikes during switching transitions (Ref: <http://modularcircuits.tantosonline.com/blog/articles/h-bridge-secrets/h-bridges-the-basics/>).



Let us consider state 8, where S1 (aka Q1) is closed, and the remaining switches are open. If we are driving CW, and, say, switching between state 9 and state 8 over a PWM frame (which we will denote as the 9/8 mode), then as switch S4 is opened, the inductive motor current briefly discharges through D3 (in FTC motors, ‘briefly’ is about 1ms). Because the shaft is turning CW, the polarity of the motor’s back EMF is such that it cannot sustain current through that same route. All is well: we drive forward.

However, consider the case where we had been driving in the CCW direction, using, say, the 6/2 mode, so that the motor shaft is currently rotating CCW. Then, were we to switch to 9/8 mode, we find that the polarity of the motor’s back EMF is the other way around; it can now also drive current through the diode, and we find ourselves in a dynamic braking situation.

States 4, 2, and 1 have analogous behavior.

1.3. Forward Driving vs Proportional Braking vs Reverse Driving

The STMicroelectronics VNH5050E H bridges used in FTC use 9/8 and 6/2 driving modes. Thus, the behaviors of states 8 & 2 mentioned above are pertinent.

Suppose we are driving CW, cruising along, forward driving using mode 9/8 (the 6/2 case is analogous), having developed some non-zero velocity in the shaft. Then, suppose further that our PID loop discerns that we're going a little too fast. At first, it will ask us to take our foot off the gas a little, which we do by reducing the 9/8 duty cycle. But if after that we're *still* going too fast, the PID loop will ask us to take stronger measures. What the current (August 2018) firmware does in response to that request is to start driving in the other direction (i.e. CCW, using mode 6/2). This we'll call reverse driving (reverse in the sense that it's opposite to the direction of the shaft rotation). Reverse driving brings in to play the 'brake to VCC' behavior of state 2.

The problem is that *it's impossible to decelerate gently using reverse driving*. This is because even 6/2 reverse driving with only the minimum possible 0% duty cycle is effectively (except for a diode voltage drop) a fully dynamic braking situation; as we increase the duty cycle percentage, this braking is only augmented by additional motor torque actively driving backwards. The transition from 0% forward driving to 0% reverse driving is thus a very abrupt one.

With asynchronous sign-magnitude H bridges such as the VNH5050E there is a third mode we can avail ourselves of to help ease this transition: mode 5/0, which we call proportional braking. In state 5, S2 and S4 are closed, causing dynamic braking; in state 0, all switches are open. A 0% duty cycle proportional braking thus is a (zero-torque) open circuit where we coast, just as we do in 0% 9/8 forward driving. At the other extreme, 100% proportional braking is full dynamic braking, just as 0% reverse driving is (again, ignoring the diode drop for the moment). In between, the balance of braking vs coasting is controlled by the PWM duty cycle. Thus, proportional braking provides a transitional regime between forward driving and reverse driving that we can exploit to respond smoothly to the PID loop's request for increasing amounts of deceleration.

The design challenge is to select the points in the control signal coming from the PID loop where we transition from one regime to the other. There are four points of relevance, which are as follows.

- S, the point of 100% reverse driving. It is clear that this occurs at the negative extreme of the PID control signal range.
- G, the point of 100% forward driving. This occurs at the positive extreme of the PID control signal range.
- Z, the point of 0% proportional braking and 0% forward driving.
- T, the point of 0% reverse driving and 100% proportional braking (we again ignore the diode for now).

A question arises as to the behavior of the driving regimes between these four points, where the PWM duty cycle is neither 0% nor 100%. As we presently lack any information to the contrary, we will in the below depict the regimes as linear in the duty cycle. This is certainly at least a roughly-good approximation at the time scale that is seen by the motor shaft, though it is only an approximation. Influences such as the ripple current LR time constant being several PWM cycles and thus never fully decaying might, for example, cause an offset bias in the linearity. Detailed investigation of the shape of these curves awaits future analysis (Update: the curves are not linear, but are close enough so as not to materially affect the analysis below. See [RippleCurrent.pdf](#)).

As a general scheme, the four points S, T, Z, and G are structurally related relative to each in the order illustrated below. Forward driving is shown in green, proportional braking in blue, and reverse driving in red.

```

Clear[abscissa, ordinate, line, points, pointNames, plotPoints];
abscissa[pt_] := pt[[1]]
ordinate[pt_] := pt[[2]]
line[{x1_, y1_}, {x2_, y2_}] := Function[{x}, y1 + (y2 - y1) / (x2 - x1) * (x - x1)]
points = {S → {-controlMax, yReverse100},
  T → {controlProp100, yProp100}, Z → {controlProp0, yProp0}, G → {controlMax, yForward100}};
pointNames = {S, T, Z, G};
Options[plotPoints] = {xOver → 1.1, yOver → 1.2} ~Join~ Options[Plot] ~Join~ Options[ListPlot];
plotPoints[pointsN_, title_, ylabel_, opts: OptionsPattern[]] := plotPoints[pointsN, title, ylabel, 32767, opts];
plotPoints[pointsN_, title_, ylabel_, controlMax_, opts: OptionsPattern[]] := Module[
  {zg, tz, st, ymax, clampMag, range, frame, listPlot, reverse, braking, forward, plot},
  zg = line[Z, G] /. pointsN;
  tz = line[T, Z] /. pointsN;
  st = line[S, T] /. pointsN;
  ymax = Max[st[-controlMax] // Abs, zg[controlMax] // Abs];
  clampMag[control_] := Max[Min[control, controlMax], -controlMax];
  range[x_, p1_, p2_] := {x, (p1 /. pointsN)[[1]], (p2 /. pointsN)[[1]]};
  frame =
    ListPlot[{{-controlMax * OptionValue[xOver], -ymax}, {controlMax * OptionValue[xOver], ymax}}, PlotStyle → PointSize[0],
    PlotRange → {-ymax * OptionValue[yOver], ymax * OptionValue[yOver]},
    AxesLabel → {"control", ylabel},
    PlotLabel → title,
    FilterRules[{opts}, Options[ListPlot]] // Evaluate
  ];
  listPlot = ListPlot[{S → "S", T → "T", Z → "Z", G → "G"} /. pointsN,
    PlotStyle → Black,
    FilterRules[{opts}, Options[ListPlot]] // Evaluate];
  plot[p1_, p2_, style_] := Plot[Evaluate[line[p1, p2] /. pointsN][x], Evaluate[range[x, p1, p2]],
    PlotStyle → style,
    FilterRules[{opts}, Options[Plot]] // Evaluate];
  reverse = plot[S, T, Red];
  braking = plot[T, Z, Blue];
  forward = plot[Z, G, Darker[Green]];
  Show[frame, listPlot, reverse, braking, forward]
]


```

```

examplePoints = {S → {-32767, -100}, T → {-16384, -70}, Z → {8192, 30}, G → {32767, 92}}
plotPoints[examplePoints, "Relative Structure of Points", "y"]

```

```
{S → {-32767, -100}, T → {-16384, -70}, Z → {8192, 30}, G → {32767, 92}}
```



1.3.1. Interpreting S, T, Z, and G

Our job here in this present document is to figure out how to locate the coordinates of the four points S, T, Z, and G for real driving situations based on sound, cogent principles.

Each time we run the PID control loop (currently every 50ms), we will calculate the coordinates of each of the four points according to the set of principles we have chosen. From those coordinates, the driving regime (forward, braking, reverse) is determined according to the position of the PID output control signal relative to the abscissas of the four points. The PWM duty cycle is then determined by the fractional location of the PID output control signal within the regime so located. The H bridge operating mode corresponding to the driving regime and the calculated duty cycle are then

passed to the H bridge.

For example, suppose that during a PID loop run, the four points happen to be as in the above figure, and that the control output from the PID loop is currently -10000. Then, because $\text{abscissa}[T] < -10000 < \text{abscissa}[Z]$, we will use proportional braking, and the duty cycle used will be 74%:

```
(pidOutput - (Z /. examplePoints // abscissa)) / ((T /. examplePoints // abscissa) - (Z /. examplePoints // abscissa)) /.
pidOutput -> -10000 // N
0.740234
```

1.4. Roadmap

In what follows, the location of the abscissas of S & G and either Z or T in the PID control range will be determined a priori based on external considerations, and the location of the remaining point then determined from the other three using arguments of linearity. We will examine four different schemes, Scheme A, Scheme B, Scheme C, and Scheme D.

2. Scheme A

2.1. Introduction: Points S, T, and G


The signature characteristic of the first scheme we examine is that it attempts to map in a linear manner the control signal coming from the PID loop to an “instantaneous equivalent applied DC voltage”. That is, for any H bridge switching mode and duty cycle, this scheme attempts to determine the applied voltage that would instantaneously produce the same motor response in a dual system that lacked the H bridge but instead had a DC voltage source directly connected to the motor. (This not to be confused with what one might call the “steady state equivalent applied DC voltage”, which varies (nearly) linearly with battery voltage and duty cycle: if battery voltage is 12V, then 50% duty cycle is equivalent to 6V, etc.)

For some switching modes and duty cycle combinations, the instantaneous equivalent applied DC voltage is clear.

- At G (100% forward driving; e.g. 100% state 9), the motor is connected between battery and ground 100% of the time. The equivalent applied voltage is thus clearly the battery voltage.
- Similarly, at S (100% reverse driving), the equivalent applied voltage is the negative battery voltage
- At T (100% proportional braking, ie: 100% state 5), the motor leads are shorted together; the equivalent applied voltage is thus zero volts. The same is also true (ignoring the diode) for 0% reverse driving (e.g.: 100% state 2). Further, given the equivalent voltages of G and S just discussed, if the overall scheme is to map the control signal to equivalent voltage *linearly*, then it follows that T must have a control value of zero.

With these in hand, and recognizing that Z must lie between T and G, we can visualize Scheme A as having the following overall shape. The particular value of Z shown is for illustrative purposes only.

```
Clear[plotScheme];
schemeA = {controlProp100 -> 0, yProp100 -> 0, yReverse100 -> -yMax, yForward100 -> yMax};
schemeAExample = schemeA ~Join~ {controlProp0 -> 2 controlMax / 3, yProp0 -> 2 yMax / 3};
plotScheme[scheme_, title_, yAxis_, yMaxVal_] := Module[
  {scale = {controlMax -> 32767, yMax -> yMaxVal}, pointsN},
  pointsN = points /. scheme /. scale;
  plotPoints[pointsN, title, yAxis]
]
plotScheme[schemeAExample, "Scheme A", "Instantaneous Equivalent Applied DC Volts", 12]
```



2.2. Point Z

The situation at Z is a bit trickier. Here (0% proportional braking/100% state 0, and 0% forward driving, 100% state 8) we have an open circuit, either literally, or nearly so (because of the diode). How can we map this open circuit into a closed circuit equivalence with some applied voltage?

The approach we examine is to recognize that in an open circuit, the current is zero (and thus the torque is zero as well, but we'll focus on current). We then ask ourselves, given the instantaneous state of the system (ie: some instantaneous velocity Ω), in the H-bridgeless-dual is there a DC voltage we could apply that would yield zero current? The answer is, messily, "sort of".

2.2.1. $v_g(t) = v_{app}(t)$ as an approximation to the "zero current voltage"

Recall (see <https://github.com/rgatkinson/RobotPhysics/blob/master/MotorPhysicsGearsInitialConditions.pdf>) the differential equation in our model that links applied DC voltage $v_{app}(t)$, the back EMF voltage $v_g(t)$, and the current $i(t)$ is the following:

```
(voltageEqn = diffEqns[[3]]) // prettyPrint
```

$$L \dot{i}(t) + R i(t) + v_g(t) = v_{app}(t)$$

An approximate argument goes something like this. If the current is zero, then the second term on the left disappears. Further, while initially there's a transient, the current soon stabilizes, and thus the first term will also disappear. That leaves us with the following:

```
voltageEqn /. {i[t] -> 0, i'[t] -> 0} // prettyPrint
```

$$v_g(t) = v_{app}(t)$$

That is, if the applied voltage matches the generated EMF, there should be no current. However intuitive that equivalence might be (or not), it is clearly only an approximation, for (at least) two reasons:

- The current isn't instantaneously zero, but instead must adjust through a transient
- After the voltage is dropped from the voltage that produced steady state velocity Ω to that of the back EMF, the velocity necessarily declines, which produces less back EMF. Thus, whatever voltage balance there was to achieve zero necessarily degrades.

The important question, then, is "how good of an approximation is it?" Let us examine that in some detail.

2.2.2. Analyzing $v_g(t) = v_{app}(t)$

It turns out that in the modelling work referenced above, we have a complete, closed form, analytic result for the current as a function of time in a motor with gearbox as the voltage steps at time zero from steady state applied voltage v_{app0} to some subsequent applied voltage $v_{app0} + \Delta v_{appConst}$. This result is also parameterized on the motor model parameters (R, L, etc), the motor load after the gearbox(if any), and an optional externally applied torque (if any). We illustrate that result here in its entirety in order that the reader may have some understanding of its complexity and completeness; however, in future we will refrain from directly illustrating such complex functions, as they are very difficult to appreciate in detail.

```
curStepGeneric
```

$$\left(\frac{2 K_e L \Delta v_{appConst} N (J a_{fter} + J \eta N^2)}{\left(\left(-1 + e^{-\frac{t \left(B a_{fter} L - J a_{fter} R + B L \eta N^2 + 2 R \eta N^2 + \sqrt{-4 L (J a_{fter} + J \eta N^2) (B a_{fter} R + (K_e K_t + B R) \eta N^2) + (B a_{fter} L - J a_{fter} R + (B L + J R) \eta N^2)^2}}}{2 L (J a_{fter} + J \eta N^2)} \right) \right) / (B a_{fter} L + J a_{fter} R + B L \eta N^2 + J R \eta N^2 - \sqrt{-4 L (J a_{fter} + J \eta N^2) (B a_{fter} R + (K_e K_t + B R) \eta N^2) + (B a_{fter} L + J a_{fter} R + (B L + J R) \eta N^2)^2}) - \left(-1 + e^{-\frac{t \left(B a_{fter} L - J a_{fter} R + B L \eta N^2 + 2 R \eta N^2 + \sqrt{-4 L (J a_{fter} + J \eta N^2) (B a_{fter} R + (K_e K_t + B R) \eta N^2) + (B a_{fter} L - J a_{fter} R + (B L + J R) \eta N^2)^2}}}{2 L (J a_{fter} + J \eta N^2)} \right) \right) / (B a_{fter} L + J a_{fter} R + B L \eta N^2 + J R \eta N^2 + \sqrt{-4 L (J a_{fter} + J \eta N^2) (B a_{fter} R + (K_e K_t + B R) \eta N^2) + (B a_{fter} L + J a_{fter} R + (B L + J R) \eta N^2)^2}) \right) \right)$$

[illegible]

$$\frac{\sqrt{(-4 L (J a f t e r + J \eta N^2) (B a f t e r R + (K e K t + B R) \eta N^2) + (B a f t e r L + J a f t e r R + (B L + J R) \eta N^2)^2)} \left((B a f t e r L + J a f t e r R + B L \eta N^2 + J R \eta N^2 + \sqrt{(-4 L (J a f t e r + J \eta N^2) (B a f t e r R + (K e K t + B R) \eta N^2) + (B a f t e r L + J a f t e r R + (B L + J R) \eta N^2)^2)} \right)}{2 L \sqrt{(-4 (J a f t e r L + J L \eta N^2) (B a f t e r R + K e K t \eta N^2 + B R \eta N^2) + (B a f t e r L + J a f t e r R + B L \eta N^2 + J R \eta N^2)^2)} + B a f t e r v a p p 0 + N (B v a p p 0 \eta N - K e \tau a p p 0)} \frac{B a f t e r R + (K e K t + B R) \eta N^2}{\eta N^2}$$

If we forget about the externally-applied torque, the result does become simpler, but is still significantly complex:

$$\frac{1}{B a f t e r R + (K e K t + B R) \eta N^2} \left((v a p p 0 + \Delta v a p p C o n s t) (B a f t e r + B \eta N^2) - e^{-\frac{\tau (B a f t e r L + J a f t e r R + (B L + J R) \eta N^2)}{2 L (J a f t e r + J \eta N^2)}} \Delta v a p p C o n s t (B a f t e r + B \eta N^2) \right. \\ \left. \frac{\cosh \left[\left(\sqrt{(-4 L (J a f t e r + J \eta N^2) (B a f t e r R + (K e K t + B R) \eta N^2) + (B a f t e r L + J a f t e r R + (B L + J R) \eta N^2)^2)} \right) / (2 L (J a f t e r + J \eta N^2)) \right]}{e^{-\frac{\tau (B a f t e r L + J a f t e r R + (B L + J R) \eta N^2)}{2 L (J a f t e r + J \eta N^2)}} \Delta v a p p C o n s t} \right. \\ \left. \frac{\sinh \left[\left(\sqrt{(-4 L (J a f t e r + J \eta N^2) (B a f t e r R + (K e K t + B R) \eta N^2) + (B a f t e r L + J a f t e r R + (B L + J R) \eta N^2)^2)} \right) / (2 L (J a f t e r + J \eta N^2)) \right]}{\left(\sqrt{(-4 L (J a f t e r + J \eta N^2) (B a f t e r R + (K e K t + B R) \eta N^2) + (B a f t e r L + J a f t e r R + (B L + J R) \eta N^2)^2)} \right)} \right)$$

Because of this complexity, in order to explore the effect on current of the drop of applied voltage to the back EMF we choose (initially, at least) to do so in the context of a concrete motor and particular loads. Here, we use an Andy Mark NeveRest 60 motor with a 10kg flywheel attached:

```
aMotor = motorParameters["AM 60 A"];
aFlywheel = flywheel[Quantity[10, "kg"], Quantity[10, "cm"]] // siUnits;
aMotorWithFlywheel = addMotorLoad[aMotor, aFlywheel] // siUnits;
unitlessMotorWithFlywheel = aMotorWithFlywheel // siUnits // clearUnits;
aMotorWithFlywheel
```

$$\left\{ \begin{array}{l} R \rightarrow \frac{33}{10} \text{ kg m}^2 / (\text{s}^3 \text{ A}^2), L \rightarrow \frac{347}{500000} \text{ kg m}^2 / (\text{s}^2 \text{ A}^2), N \rightarrow 60, \eta \rightarrow \frac{9}{10}, K e \rightarrow \frac{533}{30000} \text{ kg m}^2 / (\text{s}^2 \text{ A rad}), \\ K t \rightarrow \frac{533}{30000} \text{ kg m}^2 / (\text{s}^2 \text{ A rad}), B \rightarrow \frac{11}{1080000} \text{ kg m}^2 / (\text{s rad}^2), J \rightarrow \frac{347}{10800000000} \text{ kg m}^2 / \text{rad}^2, \\ J a f t e r \rightarrow \frac{1}{20} \text{ kg m}^2 / \text{rad}^2, B a f t e r \rightarrow 0 \text{ kg m}^2 / (\text{s rad}^2), \Delta \tau a p p C o n s t \rightarrow 0 \text{ kg m}^2 / (\text{s}^2 \text{ rad}) \end{array} \right\}$$

We want to explore the current as a function of the steady state pre-gearbox velocity Ω . To do that, we first construct a function that gives us the applied voltage that must be in effect to achieve the steady state velocity Ω .

```
vappFromVelocity[Ω_, motor_] := Module[{eqn},
  eqn = {0'[ss0] /. (initialConditions /. Equal -> Rule) /. τapp0 -> 0 /. motor} == Ω;
  vapp0 /. (uniqueSolve[eqn, vapp0] // First)]
vappFromVelocity[Ω, unitlessMotorWithFlywheel]
```

$$\frac{314339 \Omega}{15990000}$$

We also construct a function that gives us the back EMF as a function of Ω .


```

vgFromVelocity[Ω_, motor_] := Module[{v1, eqn},
  v1 = vappFromVelocity[Ω, motor];
  (vg[ss0] /. (initialConditions /. Equal → Rule) /. τapp0 → 0 /. motor) /. vapp0 → v1]
vgFromVelocity[Ω, unitlessMotorWithFlywheel]

```

$$\frac{533 \Omega}{30000}$$

This should, of course, come out as $K_e \Omega$, and, gladly, it does (remembering we're dealing here with pre-gearbox K_e and Ω). With these in hand, we can define a function that pairs up the two voltages in question as an input to our general current result.

```

dropToVg[Ω_, motor_] := Module[{v1 = vappFromVelocity[Ω, motor], v2 = vgFromVelocity[Ω, motor]},
  Association[vapp0 → v1, ΔvappConst → v2 - v1]
]
dropToVg[2, unitlessMotorWithFlywheel]

```

$$\left\langle \left| \begin{array}{l} vapp0 \rightarrow \frac{314339}{7995000}, \Delta vappConst \rightarrow -\frac{121}{31980} \end{array} \right| \right\rangle$$

Let's put that all together, illustrating the result for some reasonable initial velocity. The maximum velocity of the motor and load example we've been working with is about 610.4 radians / second (before the 60:1 gearbox) when on a 12 volt battery:

```

ssSimple = ss /. {vapp0 → 0, τapp0 → 0, ΔτappConst → 0};
maxVelExample = ssVel /. ssSimple /. unitlessMotorWithFlywheel /. ΔvappConst → 12 // N

```



610.424

We arbitrarily plot at 10% of the maximum velocity, and for 50ms, as that is the PID update interval in the FTC expansion hub firmware.

```

Clear[fn]
fn[Ω_] :=
  curStepGenericNoExternalTorque /. unitlessMotorWithFlywheel /. dropToVg[Ω, unitlessMotorWithFlywheel] // N // FullSimplify
Plot[fn[maxVelExample / 10], {t, 0, .050}, AxesLabel → {"Seconds", HoldForm[Amps]}, PlotRange → All, AxesOrigin → {0, 0}]
Plot[fn[maxVelExample / 10], {t, 0, .050}, AxesLabel → {"Seconds", HoldForm[Amps]}, PlotRange → {0, 0.005}, AxesOrigin → {0, 0}]

```



(Note: the shape of these plots is characteristic; the vertical axis scales linearly with initial velocity.)

```

fn[n Ω] / fn[Ω] // FullSimplify
n

```

It is worthy of note that the current never does, in fact, reach zero. Further, over the course of the PID update interval, the current is distinctly non-zero throughout, by the end recovering maybe some ~20% of its initial value. The approximation here to a “zero current voltage” does not appear to be a very good one.

2.2.3. “Mean zero current” as an approximation to the “zero current voltage”

The above analysis does suggest a different approximation: instead of dropping from the steady state applied voltage to the voltage of the back emf, to what voltage would we have to drop in order to have the mean (i.e: average) current over the PID interval be zero? We compute that here as a function of the steady state velocity Ω .

```
vappFromVelocity[Ω, {}]
```

```
currentFromΩAndDrop = curStepGenericNoExternalTorque /. vapp0 → vappFromVelocity[Ω, {}] // FullSimplify
```

$$\frac{(Bafter R + Ke Kt \eta N^2 + B R \eta N^2) \Omega}{Kt \eta N^2}$$

$$\frac{1}{Bafter R + (Ke Kt + B R) \eta N^2} \left((Bafter + B \eta N^2) \left(\Delta vappConst + \left(Ke + \frac{R (Bafter + B \eta N^2)}{Kt \eta N^2} \right) \Omega \right) - e^{-\frac{t (Bafter L + Jafter R + (B L + J R) \eta N^2)}{2 L (Jafter + J \eta N^2)}} \Delta vappConst (Bafter + B \eta N^2) \right. \\ \left. \cosh \left[\frac{t \sqrt{(-4 L (Jafter + J \eta N^2) (Bafter R + (Ke Kt + B R) \eta N^2) + (Bafter L + Jafter R + (B L + J R) \eta N^2)^2)}}{2 L (Jafter + J \eta N^2)} \right] + \frac{e^{-\frac{t (Bafter L + Jafter R + (B L + J R) \eta N^2)}{2 L (Jafter + J \eta N^2)}} \Delta vappConst}{\Delta vappConst} \right. \\ \left. (Bafter (-Bafter L + Jafter R) + (2 Jafter Ke Kt - 2 B Bafter L + Bafter J R + B Jafter R) \eta N^2 + (2 J Ke Kt - B^2 L + B J R) \eta^2 N^4) \right. \\ \left. \sinh \left[\frac{t \sqrt{(-4 L (Jafter + J \eta N^2) (Bafter R + (Ke Kt + B R) \eta N^2) + (Bafter L + Jafter R + (B L + J R) \eta N^2)^2)}}{2 L (Jafter + J \eta N^2)} \right] \right) / \\ \left(\sqrt{(-4 L (Jafter + J \eta N^2) (Bafter R + (Ke Kt + B R) \eta N^2) + (Bafter L + Jafter R + (B L + J R) \eta N^2)^2)} \right)$$

The value $\Delta vappConst$ in the above is the voltage drop we seek. We next symbolically compute the mean current over the PID interval.

```
meanCurrent = Integrate[currentFromΩAndDrop, {t, 0, pidPeriod}] / pidPeriod
```

$$\frac{1}{pidPeriod (Bafter R + (Ke Kt + B R) \eta N^2)} \left(pidPeriod (Bafter + B \eta N^2) \left(\Delta vappConst + \left(Ke + \frac{R (Bafter + B \eta N^2)}{Kt \eta N^2} \right) \Omega \right) + \right. \\ \left(\Delta vappConst (Bafter (-Bafter L + Jafter R) + (2 Jafter Ke Kt - 2 B Bafter L + Bafter J R + B Jafter R) \eta N^2 + (2 J Ke Kt - B^2 L + B J R) \eta^2 N^4) \right. \\ \left. \left(\sqrt{(-4 L (Jafter + J \eta N^2) (Bafter R + (Ke Kt + B R) \eta N^2) + (Bafter L + Jafter R + (B L + J R) \eta N^2)^2)} + \right. \right. \\ \left. e^{-\frac{pidPeriod (Bafter L + Jafter R + (B L + J R) \eta N^2)}{2 L (Jafter + J \eta N^2)}} \left(-\sqrt{(-4 L (Jafter + J \eta N^2) (Bafter R + (Ke Kt + B R) \eta N^2) + (Bafter L + Jafter R + (B L + J R) \eta N^2)^2)} \right) \right. \\ \left. \cosh \left[\frac{pidPeriod \sqrt{(-4 L (Jafter + J \eta N^2) (Bafter R + (Ke Kt + B R) \eta N^2) + (Bafter L + Jafter R + (B L + J R) \eta N^2)^2)}}{2 L (Jafter + J \eta N^2)} \right] - \right. \\ \left. (Bafter L + Jafter R + (B L + J R) \eta N^2) \sinh \left[\frac{pidPeriod \sqrt{(-4 L (Jafter + J \eta N^2) (Bafter R + (Ke Kt + B R) \eta N^2) + (Bafter L + Jafter R + (B L + J R) \eta N^2)^2)}}{2 L (Jafter + J \eta N^2)} \right] \right) \right) / \\ \left(2 (Bafter R + (Ke Kt + B R) \eta N^2) \sqrt{(-4 L (Jafter + J \eta N^2) (Bafter R + (Ke Kt + B R) \eta N^2) + (Bafter L + Jafter R + (B L + J R) \eta N^2)^2)} - \right. \\ \left. 1 \right. \\ \left. 2 (Bafter R + (Ke Kt + B R) \eta N^2) \right. \\ \Delta vappConst \\ (Bafter + B \eta N^2) \\ \left(Bafter L + Jafter R + (B L + J R) \eta N^2 + e^{-\frac{pidPeriod (Bafter L + Jafter R + (B L + J R) \eta N^2)}{2 L (Jafter + J \eta N^2)}} \left(- (Bafter L + Jafter R + (B L + J R) \eta N^2) \cosh \left[\frac{pidPeriod \sqrt{(-4 L (Jafter + J \eta N^2) (Bafter R + (Ke Kt + B R) \eta N^2) + (Bafter L + Jafter R + (B L + J R) \eta N^2)^2)}}{2 L (Jafter + J \eta N^2)} \right] - \right. \right. \\ \left. \sqrt{(-4 L (Jafter + J \eta N^2) (Bafter R + (Ke Kt + B R) \eta N^2) + (Bafter L + Jafter R + (B L + J R) \eta N^2)^2)} \sinh \left[\frac{pidPeriod \sqrt{(-4 L (Jafter + J \eta N^2) (Bafter R + (Ke Kt + B R) \eta N^2) + (Bafter L + Jafter R + (B L + J R) \eta N^2)^2)}}{2 L (Jafter + J \eta N^2)} \right] \right) \right) \right) \right)$$

We set that equal to zero, and solve for the voltage drop.


```
ΔvappMeanZeroCurrent = ΔvappConst /. uniqueSolve[meanCurrent == 0, ΔvappConst] // FullSimplify // Expand // FullSimplify
```

$$\left(e^{\frac{\text{pidPeriod} (\text{Bafter} L - \text{Jafter} R + (B L + J R) \eta N^2)}{2 L (\text{Jafter} + J \eta N^2)}} \text{pidPeriod} (\text{Bafter} + B \eta N^2) (\text{Bafter} R + (K e K t + B R) \eta N^2)^2 \right. \\ \left. \sqrt{((\text{Bafter} L - \text{Jafter} R)^2 - 2 (2 \text{Jafter} K e K t L + \text{Jafter} R (B L - J R) + \text{Bafter} L (-B L + J R)) \eta N^2 + (B^2 L^2 + J^2 R^2 - 2 J L (2 K e K t + B R)) \eta^2 N^4)} \Omega \right) / \\ \left(K t \eta N^2 \left(e^{\frac{\text{pidPeriod} (\text{Bafter} L - \text{Jafter} R + (B L + J R) \eta N^2)}{2 L (\text{Jafter} + J \eta N^2)}} \sqrt{(-4 L (\text{Jafter} + J \eta N^2) (\text{Bafter} R + (K e K t + B R) \eta N^2) + (\text{Bafter} L + \text{Jafter} R + (B L + J R) \eta N^2)^2)} \right. \right. \\ \left. (\text{Bafter}^2 (L - \text{pidPeriod} R) + \text{Bafter} (-K e K t \text{pidPeriod} + 2 B (L - \text{pidPeriod} R)) \eta N^2 - \right. \\ \left. \eta N^2 (\text{Jafter} K e K t + (J K e K t + B (-B L + K e K t \text{pidPeriod} + B \text{pidPeriod} R)) \eta N^2) \right) - (\text{Bafter}^2 L + (-\text{Jafter} K e K t + 2 B \text{Bafter} L) \eta N^2 + \\ (-J K e K t + B^2 L) \eta^2 N^4) \sqrt{(-4 L (\text{Jafter} + J \eta N^2) (\text{Bafter} R + (K e K t + B R) \eta N^2) + (\text{Bafter} L + \text{Jafter} R + (B L + J R) \eta N^2)^2)} \cosh[\\ (\text{pidPeriod} \sqrt{(-4 L (\text{Jafter} + J \eta N^2) (\text{Bafter} R + (K e K t + B R) \eta N^2) + (\text{Bafter} L + \text{Jafter} R + (B L + J R) \eta N^2)^2}) / (2 L (\text{Jafter} + J \eta N^2)))] + \\ (-\text{Bafter}^3 L^2 + \text{Bafter}^2 L (\text{Jafter} R + (-3 B L + J R) \eta N^2) + \text{Bafter} L \eta N^2 (\text{Jafter} (3 K e K t + 2 B R) + (3 J K e K t - 3 B^2 L + 2 B J R) \eta N^2) + \\ \eta N^2 (\text{Jafter}^2 K e K t R + \text{Jafter} (3 B K e K t L + 2 J K e K t R + B^2 L R) \eta N^2 + (3 B J K e K t L - B^3 L^2 + J^2 K e K t R + B^2 J L R) \eta^2 N^4)) \sinh[\\ (\text{pidPeriod} \sqrt{(-4 L (\text{Jafter} + J \eta N^2) (\text{Bafter} R + (K e K t + B R) \eta N^2) + (\text{Bafter} L + \text{Jafter} R + (B L + J R) \eta N^2)^2}) / (2 L (\text{Jafter} + J \eta N^2)))] \right) \right)$$

This is, ehm, a mouthful. In particular (among many things), it is sensitive to the load on the motor (*Jafter*, *Bafter*), which would significantly complicate its application in the FTC context (teams would need to measure their robots). Nevertheless, let's see this in action for the same example motor and load we looked at previously.

```
dropToMeanZeroCurrent[Ω_, motor_] := Module[{
  v1 = vappFromVelocity[Ω, motor],
  Δv2 = ΔvappMeanZeroCurrent /. motor /. pidPeriod -> .050 /. Ω -> Ω,
  Association[vapp0 -> v1, τapp0 -> 0, ΔvappConst -> Δv2]
}]
dropToMeanZeroCurrent[1, unitlessMotorWithFlywheel] // N
{vapp0 -> 0.0196585, τapp0 -> 0., ΔvappConst -> -0.00220185}
```

```
Clear[fn]
fn[Ω_] :=
curStepGenericNoExternalTorque /. unitlessMotorWithFlywheel /. dropToMeanZeroCurrent[Ω, unitlessMotorWithFlywheel] // N //
FullSimplify
Plot[fn[maxVelExample / 10], {t, 0, .050}, AxesLabel -> {"Seconds", "Amps"}, PlotRange -> All, AxesOrigin -> {0, 0}]
```




At first glance, it would indeed seem to be plausible that we've done the math correctly, i.e., that the mean current over the PID interval is zero (that's just a sanity check: assuming we've set the equations up correctly, the solution must be correct, assuming Mathematica has not encountered a bug). Further, comparing this approximation of the zero current voltage to the previous $vg(t) = vapp(t)$ approximation, it is hard to argue that the mean zero current approximation is not both numerically and conceptually superior.

However, as was noted, its operational difficulties make its deployment a challenge. While perhaps some simplification of this result (e.g., perhaps ignoring the load?) might produce an approximation of intermediate goodness that was deployable, such an investigation has not yet been carried out. Accordingly, for the remainder of the discussion of Scheme A, we will work with the $vg(t) = vapp(t)$ approximation for the zero current voltage.

2.3. Calculating Point Z

Recall the structure of Scheme A:

```
plotScheme[schemeAExample, "Scheme A", "Instantaneous Equivalent Applied DC Volts", 12]
```



We have points S, T, and G in hand; we wish to compute point Z. The principle we will use is to locate Z on the line between T and G. Per the above, we take the ordinate of Z to be the current instantaneous back EMF; our job is to compute the abscissa of Z, which we will denote as `controlProp0` (0% proportional braking occurs at the same location as 0% forward driving). The calculation is pretty straightforward.

```
schemeA = Union[schemeA, {yProp0 → vgFromVelocity[Ω, {}]}]
```

```
{controlProp100 → 0, yForward100 → yMax, yProp0 → Ke Ω, yProp100 → 0, yReverse100 → -yMax}
```

```
schemeAPoints = points /. schemeA /. yMax → batteryVoltage
```

```
{S → {-controlMax, -batteryVoltage}, T → {0, 0}, Z → {controlProp0, Ke Ω}, G → {controlMax, batteryVoltage}}
```

```
schemeAEqn = ((line[T, G] /. schemeAPoints)[controlProp0] == (yProp0 /. schemeA))
```

$$\frac{\text{batteryVoltage controlProp0}}{\text{controlMax}} = \text{Ke } \Omega$$

```
schemeASoln = uniqueSolve[schemeAEqn, controlProp0]
```

```
schemeAResult = schemeAPoints /. schemeASoln
```

```
{controlProp0 →  $\frac{\text{controlMax Ke } \Omega}{\text{batteryVoltage}}$ }
```

```
{S → {-controlMax, -batteryVoltage}, T → {0, 0}, Z → { $\frac{\text{controlMax Ke } \Omega}{\text{batteryVoltage}}$ , Ke Ω}, G → {controlMax, batteryVoltage}}
```


Recall from the modelling paper that K_e is the *before-gearbox* electrical coupling constant. Thus, the abscissa of Z is a multiplication of the motor velocity (in radians / second) by a motor constant and the control range magnitude and a division by the battery voltage.

Let's explore a little how the location of Z varies with velocity Ω . We plot Z at various fractions of the maximum velocity (10%, 50%, and 90%):

```

plotSchemePointsExample[schemeResult_, titlefn_, ylabel_] := Module[{ptsFromVel, plotVel},
  ptsFromVel[ $\Omega$ _] := schemeResult /.
    {controlMax  $\rightarrow$  32767, batteryVoltage  $\rightarrow$  12} /. unitlessMotorWithFlywheel /.  $\Omega \rightarrow \Omega$ ;
  plotVel[ $\Omega$ _] := Module[{title = titlefn[ $\Omega$ ]},
    plotPoints[ptsFromVel[ $\Omega$ ], title, ylabel, ImageSize  $\rightarrow$  300, yOver  $\rightarrow$  1.2]];
  plotVel[maxVelExample * #] & /@ {0.1, 0.5, 0.9} // Row
]
plotSchemePointsExample[schemeAResult, "Scheme A Example,  $\Omega$ =" <> ToString[#] <> " rad/s" &, "Volts"]

```




It is significant to note that as velocity increases, the amount of the control range allocated to the forward driving regime diminishes, ultimately to zero. In many (if not most) FTC motor control situations, the control loop spends the majority of its time in the forward driving regime. There seems to be a tension between these situations.

In addition to the 10%, 50%, and 90% of maximum velocity, we look at the transition point across the whole velocity range. As we require in the development above the transition Z to have a non-negative abscissa, clearly the result should be computed using the absolute value of the velocity, then, if need be, adjusted afterward to correct for the sign of forward vs reverse driving.

```

Clear[plotTransitionExample]
plotTransitionExample[schemeName_, result_, pointName_,  $\Omega$ min_,  $\Omega$ max_] := Module[{expr, cMax = 32767, range},
  expr = abscissa[pointName /. result] /. unitlessMotorWithFlywheel /. {controlMax  $\rightarrow$  cMax, batteryVoltage  $\rightarrow$  12} // N;
  range = {Max[-cMax, expr /.  $\Omega \rightarrow \Omega$ max], Min[cMax, expr /.  $\Omega \rightarrow \Omega$ min]};
  Plot[expr, { $\Omega$ ,  $\Omega$ min,  $\Omega$ max}, AxesOrigin  $\rightarrow$  {0, 0},
    PlotLabel  $\rightarrow$  schemeName <> " Point " <> ToString[pointName] <> " Abscissa vs Velocity",
    AxesLabel  $\rightarrow$  {"rad/s", "control"}, PlotRange  $\rightarrow$  range]
]
plotTransitionExample["SchemeA", schemeAResult, Z, -maxVelExample, maxVelExample]

```



3. Scheme B

3.1. Introduction

We now discuss a second scheme, Scheme B. This scheme is motivated by a few observations.

- In actual practice, the PID loop visits point Z far more often than it visits point T (or S). Thus, to help achieve reproducible results when driving, it is preferable to have any uncertainty or variability in the calculation of the locations of the points to be attributed to point T rather than point Z. That is, all other things being equal, it is preferable to have point Z determined a priori, and point T calculated at runtime.
- As was discussed at some length above, the calculation of an equivalent applied DC voltage that produces instantaneous zero current is an elusive proposition. Moreover the only feasibly-calculable approximation thereto that we have is demonstrably poor.


- In practice in FTC situations, the PID algorithm spends the majority of its time in the forward driving regime. Thus, attenuation of the dynamic range of that regime is of concern so as to avoid precision loss in calculations.
- The point Z is by its definition a point of zero *current*. It thus makes sense to at least explore an instantaneous-*current*-linearization (in contrast to the instantaneous-equivalent-applied-voltage-linearization previously examined).

Scheme B is the exploration of such an instantaneous-current-linearization. From the above observations, we can deduce a few facts.

- The point Z is at the origin: it's ordinate is zero because the y axis is current; it's abscissa is zero because no other *fixed* abscissa is reasonable.
- The point G has an abscissa of controlMax, but it's ordinate will be a function of the battery voltage, the back EMF, and the motor resistance.
- Similarly, point S has an abscissa of -controlMax, and an ordinate which is a function of the battery voltage, the back EMF, and motor resistance.
- The ordinate of point T is a function of the battery voltage, the back EMF, and motor resistance; its abscissa will be an interpolation between Z and S.

The general structure of Scheme B looks something like this:

```
schemeB = {controlProp0 → 0, yProp0 → 0, yReverse100 → -yMax, yForward100 → yMax / 2};
schemeBExample = schemeB ~ Join ~ {controlProp100 → -controlMax / 2, yProp100 → -yMax / 2}
plotScheme[schemeBExample, "Scheme B Example", "Current (Amps)", 3]
```

$$\left\{ \text{controlProp0} \rightarrow 0, \text{yProp0} \rightarrow 0, \text{yReverse100} \rightarrow -y_{\text{Max}}, \text{yForward100} \rightarrow \frac{y_{\text{Max}}}{2}, \text{controlProp100} \rightarrow -\frac{\text{controlMax}}{2}, \text{yProp100} \rightarrow -\frac{y_{\text{Max}}}{2} \right\}$$


Lets explore this in more detail.

3.2. Current as Function of State

We calculate the current in instantaneous steady state (i.e. after the LR transient) as a function of the applied voltage and instantaneous current velocity. Note that as was the case in Schema A, this is only an approximation.

```
ssCurrentFromVoltageVelocity[v_, Ω_] := Module[{eqn, soln},
  eqn = voltageEqn /. {vg[t] → vgFromVelocity[Ω, {}], vapp[t] → v, i'[t] → 0};
  soln = uniqueSolve[eqn, i[t]] // FullSimplify;
  i[t] /. soln
]
ssCurrentFromVoltageVelocity[v, Ω]
```

$$\frac{v - K_e \Omega}{R}$$

3.3. Locating Our Points

With that in hand, we can determine the ordinates of S and G from the relationship of current to the battery voltage and velocity; the ordinate of T is calculated similarly, but with respect to the zero applied volts of 100% state 5. Point Z is at the origin, as was mentioned.

```

schemeB = {
  yForward100 → ssCurrentFromVoltageVelocity[batteryVoltage, Ω],
  yReverse100 → ssCurrentFromVoltageVelocity[-batteryVoltage, Ω],
  yProp100 → ssCurrentFromVoltageVelocity[0, Ω],
  controlProp0 → 0, yProp0 → 0
}

```

$$\left\{ yForward100 \rightarrow \frac{\text{batteryVoltage} - K_e \Omega}{R}, yReverse100 \rightarrow -\frac{\text{batteryVoltage} + K_e \Omega}{R}, yProp100 \rightarrow -\frac{K_e \Omega}{R}, \text{controlProp0} \rightarrow 0, yProp0 \rightarrow 0 \right\}$$

With that we can calculate the abscissa of T by fixing T on a line from S to Z.

```

schemeBPoints = points /. schemeB

```

$$\left\{ S \rightarrow \left\{ -\text{controlMax}, -\frac{\text{batteryVoltage} + K_e \Omega}{R} \right\}, T \rightarrow \left\{ \text{controlProp100}, -\frac{K_e \Omega}{R} \right\}, Z \rightarrow \{0, 0\}, G \rightarrow \left\{ \text{controlMax}, \frac{\text{batteryVoltage} - K_e \Omega}{R} \right\} \right\}$$

```

schemeBEqn = (line[S, Z] /. schemeBPoints)[controlProp100] == (yProp100 /. schemeB)
schemeBSoln = uniqueSolve[schemeBEqn, controlProp100]
schemeBResult = schemeBPoints /. schemeBSoln

```

$$-\frac{\text{batteryVoltage} + K_e \Omega}{R} + \frac{(\text{controlMax} + \text{controlProp100})(\text{batteryVoltage} + K_e \Omega)}{\text{controlMax} R} = -\frac{K_e \Omega}{R}$$


$$\left\{ \text{controlProp100} \rightarrow -\frac{\text{controlMax} K_e \Omega}{\text{batteryVoltage} + K_e \Omega} \right\}$$

$$\left\{ S \rightarrow \left\{ -\text{controlMax}, -\frac{\text{batteryVoltage} + K_e \Omega}{R} \right\}, T \rightarrow \left\{ -\frac{\text{controlMax} K_e \Omega}{\text{batteryVoltage} + K_e \Omega}, -\frac{K_e \Omega}{R} \right\}, Z \rightarrow \{0, 0\}, G \rightarrow \left\{ \text{controlMax}, \frac{\text{batteryVoltage} - K_e \Omega}{R} \right\} \right\}$$

Note that abscissas in this result are only dependent on the instantaneous velocity and battery voltage, the control range, and the motor's pre-gearbox electrical coupling constant: conveniently, the uses of the motor's resistive constant R values have all cancelled out.

3.4. Examples


As we did with Scheme A, we illustrate Scheme B with our working motor example at a few different velocities.



These examples clearly illustrate the most significant tradeoff involved in the use of Scheme B: that we get a knee at the origin (note that we are still *continuous* everywhere). While the knee is relatively shallow at low velocities, it is more pronounced at larger ones. This is not *intrinsically* a problem: architecturally speaking, a PID loop can certainly deal with such a condition; while all other things being equal, we'd prefer not to have the knee, the reality is that all other things are *not* equal, as was discussed at the beginning of this section. From one point of view, this knee is the price we pay for moving the uncertainty and variability of the abscissa of Z that is present in Schema A to an uncertainty and variability of the abscissa of T in Scheme B. Further, the dynamic control range allocated to forward driving remains large, and insensitive to velocity.

These tradeoffs seem worthwhile.

In addition to the 10%, 50%, and 90% of maximum velocity, we look at the transition point across the whole velocity range, as we did with Scheme A. As we require in the development above the transition T to have a non-positive abscissa, clearly the result should be computed using the absolute value of the velocity, then, if need be, adjusted afterward to correct for the sign of forward vs reverse driving.



4. Scheme C

In the introduction we stated the following:

- “S, the point of 100% reverse driving. It is clear that this occurs at the negative extreme of the PID control signal range.”
- G, the point of 100% forward driving. This occurs at the positive extreme of the PID control signal range.”

However, upon further reflection, we can see that these unconditional statements about the position of S and G in the control range can in fact be relaxed. In particular, we ought to ask ourselves the question “is it necessary or worthwhile to *always* provide the PID loop with access to the *full* range of the reverse driving regime?” For example, if we are in Scheme B with some significant velocity, proportional braking is strong, and so maybe we don’t need to map the upper end of reverse driving into the range $[-\text{controlMax}, \text{controlMax}]$ as we have been doing so far. Scheme C explores that possibility.

Recall our result from Scheme B.

schemeBResult


$$\left\{ S \rightarrow \left\{ -\text{controlMax}, -\frac{\text{batteryVoltage} + K_e \Omega}{R} \right\}, T \rightarrow \left\{ -\frac{\text{controlMax} K_e \Omega}{\text{batteryVoltage} + K_e \Omega}, -\frac{K_e \Omega}{R} \right\}, Z \rightarrow \{0, 0\}, G \rightarrow \left\{ \text{controlMax}, \frac{\text{batteryVoltage} - K_e \Omega}{R} \right\} \right\}$$

In Scheme C, we stretch out the points S & T left of the Y axis so that they have the same slope as that of the line from Z to G.


```
computeSchemeC[points_] := Module[{scale, scalePt},
  scale = -(S /. points)[[2]] / (G /. points)[[2]];
  scalePt[pt_] := {pt[[1]] * scale, pt[[2]]};
  {
    S → (S /. points // scalePt),
    T → (T /. points // scalePt),
    Z → (Z /. points),
    G → (G /. points)} // FullSimplify
];
schemeCResult = computeSchemeC[schemeBResult]
```

$$\left\{ S \rightarrow \left\{ \text{controlMax} - \frac{2 \text{batteryVoltage} \text{controlMax}}{\text{batteryVoltage} - K_e \Omega}, -\frac{\text{batteryVoltage} + K_e \Omega}{R} \right\}, \right. \\ \left. T \rightarrow \left\{ -\frac{\text{controlMax} K_e \Omega}{\text{batteryVoltage} + K_e \Omega}, -\frac{K_e \Omega}{R} \right\}, Z \rightarrow \{0, 0\}, G \rightarrow \left\{ \text{controlMax}, \frac{\text{batteryVoltage} - K_e \Omega}{R} \right\} \right\}$$

This produces a linear interpretation of the control signal across the entire control signal range. However, unlike Scheme B, for higher velocities, we rely much more on proportional braking that we do on reverse driving. Which seems to make sense, since at higher velocities proportional braking has more effect than it does at lower velocities.



In addition to the 10%, 50%, and 90% of maximum velocity, we look at the transition point across the whole velocity range as we did with the previous schemes. As we require in the development above the transition T to have a non-positive abscissa, clearly the result should be computed using the absolute value of the velocity, then, if need be, adjusted afterward to correct for the sign of forward vs reverse driving.



5. Scheme D

Consider Scheme B and Scheme C. We can interpret Scheme C as Scheme B scaled in its negative control range such that the slope therein matches the slope in the positive control range. However, there exist intermediate blends between the two extremes, where the slope adjusts from not at all (0%, Scheme B) to fully matching (100%, Scheme C). Scheme D is an exploration of those blends. As such, it is a strict generalization of both Scheme B and Scheme C.

```

Clear[computeSchemeD, schemeDResultFn]
computeSchemeD[points_] := Function[{blend}, computeSchemeD[points, blend]]
computeSchemeD[points_, blend_] := Module[{yS, yG, scale, scalePt},
  yS = -(S /. points)[[2]];
  yG = (G /. points)[[2]];
  scale = line[{0, 1}, {1, yS / yG}][blend];
  scalePt[pt_] := {pt[[1]] * scale, pt[[2]]};
  {
    S → (S /. points // scalePt),
    T → (T /. points // scalePt),
    Z → (Z /. points),
    G → (G /. points)} // FullSimplify
];
schemeDResultFn = computeSchemeD[schemeBResult];
schemeDResultFn[0]
schemeDResultFn[1]
# → abscissa[T /. schemeDResultFn[#]] & /@ {0, 1/2, 1, blend}

```

$$\left\{ S \rightarrow \left\{ -\text{controlMax}, -\frac{\text{batteryVoltage} + \text{Ke } \Omega}{R} \right\}, T \rightarrow \left\{ -\frac{\text{controlMax Ke } \Omega}{\text{batteryVoltage} + \text{Ke } \Omega}, -\frac{\text{Ke } \Omega}{R} \right\}, Z \rightarrow \{0, 0\}, G \rightarrow \left\{ \text{controlMax}, \frac{\text{batteryVoltage} - \text{Ke } \Omega}{R} \right\} \right\}$$

$$\left\{ S \rightarrow \left\{ \text{controlMax} - \frac{2 \text{ batteryVoltage controlMax}}{\text{batteryVoltage} - \text{Ke } \Omega}, -\frac{\text{batteryVoltage} + \text{Ke } \Omega}{R} \right\}, T \rightarrow \left\{ \frac{\text{controlMax Ke } \Omega}{-\text{batteryVoltage} + \text{Ke } \Omega}, -\frac{\text{Ke } \Omega}{R} \right\}, Z \rightarrow \{0, 0\}, G \rightarrow \left\{ \text{controlMax}, \frac{\text{batteryVoltage} - \text{Ke } \Omega}{R} \right\} \right\}$$


$$\left\{ 0 \rightarrow -\frac{\text{controlMax Ke } \Omega}{\text{batteryVoltage} + \text{Ke } \Omega}, \frac{1}{2} \rightarrow -\frac{\text{batteryVoltage controlMax Ke } \Omega}{\text{batteryVoltage}^2 - \text{Ke}^2 \Omega^2}, 1 \rightarrow \frac{\text{controlMax Ke } \Omega}{-\text{batteryVoltage} + \text{Ke } \Omega}, \text{blend} \rightarrow -\frac{\text{controlMax Ke } \Omega (\text{batteryVoltage} + (-1 + 2 \text{blend}) \text{Ke } \Omega)}{(\text{batteryVoltage} - \text{Ke } \Omega) (\text{batteryVoltage} + \text{Ke } \Omega)} \right\}$$


We plot examples of Scheme D for various blending parameters. Take particular note of the scaling of the Y axis across these examples. If Scheme D were to be used, at the present time it is unclear to the author how to best select the blending parameter.

```

plotSchemePointsExample[schemeDResultFn[0], "Scheme D Example blend=0 Ω=" <> ToString[#] <> " rad/s", "Current (Amps)"]
plotSchemePointsExample[schemeDResultFn[1/8], "Scheme D Example blend=1/8 Ω=" <> ToString[#] <> " rad/s", "Current (Amps)"]
plotSchemePointsExample[schemeDResultFn[1/4], "Scheme D Example blend=1/4 Ω=" <> ToString[#] <> " rad/s", "Current (Amps)"]
plotSchemePointsExample[schemeDResultFn[1/2], "Scheme D Example blend=1/2 Ω=" <> ToString[#] <> " rad/s", "Current (Amps)"]
plotSchemePointsExample[schemeDResultFn[3/4], "Scheme D Example blend=3/4 Ω=" <> ToString[#] <> " rad/s", "Current (Amps)"]
plotSchemePointsExample[schemeDResultFn[1], "Scheme D Example blend=1 Ω=" <> ToString[#] <> " rad/s", "Current (Amps)"]

```





6. Implementation

In the above discussion, velocity has been measured in radians / second and battery voltage in volts. However, in the firmware in which the PID loop is implemented, velocity is available in encoder counts per second and battery voltage in mV. Here, we convert our three schemes to those units to aid in implementation.

6.1. Development

Our existing K_e value is in SI quantities (seconds, volts, radians):

K_e /. parameterQuantities

K_e s V / rad

Inside the firmware, velocity is measured in encoder counts per second. Thus, it is convenient to have the K_e constant in units where the angular measure is counts instead of radians, specifically in units of s V / count.

```
Clear[keFwToSi]
keFwToSi[keFw_, factor_] := factor * Quantity[keFw, "Seconds" "Volts" / IndependentUnit["Counts"]] *
  Quantity[countsPerRev, IndependentUnit["Counts"] / "Revolutions"] / Quantity[2 Pi, "Radians" / "Revolutions"]
keFwToSi[
  keFw,
  1 / 1000]


$$\frac{\text{countsPerRev keFw}}{2000 \pi} \text{ s V / rad}$$

```

We similarly convert velocity from counts per second to radians per second.

```
ΩFwToSi[ΩFw_] := Quantity[ΩFw, IndependentUnit["Counts"] / "Seconds"] /
  Quantity[countsPerRev, IndependentUnit["Counts"] / "Revolutions"] * Quantity[2 Pi, "Radians" / "Revolutions"]
ΩFwToSi[
  cps]


$$\frac{2 \text{ cps } \pi}{\text{countsPerRev}} \text{ rad / s}$$

```

Finally, battery voltage in the firmware is available in mV instead of volts.

```
batteryVoltageFwToSi[batFw_] := Quantity[batFw, "mV"]
batteryVoltageFwToSi[12000]

12000 mV
```

We put that all together.

```
fwImplMap = {
  Ω → ΩFwToSi[cps],
  batteryVoltage → batteryVoltageFwToSi[mvBatteryVoltage],
  Ke → keFwToSi[keFw, 1 / 1000],
  R → (R /. parameterQuantities) // siUnits
}


$$\left\{ \Omega \rightarrow \frac{2 \text{ cps } \pi}{\text{countsPerRev}} \text{ rad / s}, \text{ batteryVoltage} \rightarrow \frac{\text{mvBatteryVoltage}}{1000} \text{ kg m}^2 / (\text{s}^3 \text{ A}), \text{ Ke} \rightarrow \frac{\text{countsPerRev keFw}}{2000 \pi} \text{ kg m}^2 / (\text{s}^2 \text{ A rad}), \text{ R} \rightarrow \text{R kg m}^2 / (\text{s}^3 \text{ A}^2) \right\}$$

```

6.2. Apply to Scheme Results

We apply that implementation mapping to our results from our various schemes.

```
schemeAResult /. fwImplMap // FullSimplify // simplifyUnits
Z /. (schemeAResult /. fwImplMap // FullSimplify) // abscissa // Apart // framed


$$\left\{ S \rightarrow \left\{ -\text{controlMax}, -\frac{\text{mvBatteryVoltage}}{1000} \text{ V} \right\}, T \rightarrow \{0, 0\}, Z \rightarrow \left\{ \frac{\text{controlMax cps keFw}}{\text{mvBatteryVoltage}}, \frac{\text{cps keFw}}{1000} \text{ V} \right\}, G \rightarrow \left\{ \text{controlMax}, \frac{\text{mvBatteryVoltage}}{1000} \text{ V} \right\} \right\}$$

```

$$\frac{\text{controlMax cps keFw}}{\text{mvBatteryVoltage}}$$

```
schemeBResult /. fwImplMap // FullSimplify // simplifyUnits
T /. (schemeBResult /. fwImplMap) // abscissa // Together // framed
```

$$\left\{ S \rightarrow \left\{ -\text{controlMax}, -\frac{\text{cps keFw} + \text{mvBatteryVoltage}}{1000 R} A \right\}, \right. \\ \left. T \rightarrow \left\{ -\frac{\text{controlMax cps keFw}}{\text{cps keFw} + \text{mvBatteryVoltage}}, -\frac{\text{cps keFw}}{1000 R} A \right\}, Z \rightarrow \{0, 0\}, G \rightarrow \left\{ \text{controlMax}, \frac{-\text{cps keFw} + \text{mvBatteryVoltage}}{1000 R} A \right\} \right\}$$

$$-\frac{\text{controlMax cps keFw}}{\text{cps keFw} + \text{mvBatteryVoltage}}$$

```
schemeCResult /. fwImplMap // FullSimplify // simplifyUnits
T /. (schemeCResult /. fwImplMap) // abscissa // Together // framed
```

$$\left\{ S \rightarrow \left\{ \frac{\text{controlMax} (\text{cps keFw} + \text{mvBatteryVoltage})}{\text{cps keFw} - \text{mvBatteryVoltage}}, -\frac{\text{cps keFw} + \text{mvBatteryVoltage}}{1000 R} A \right\}, \right. \\ \left. T \rightarrow \left\{ \frac{\text{controlMax cps keFw}}{\text{cps keFw} - \text{mvBatteryVoltage}}, -\frac{\text{cps keFw}}{1000 R} A \right\}, Z \rightarrow \{0, 0\}, G \rightarrow \left\{ \text{controlMax}, \frac{-\text{cps keFw} + \text{mvBatteryVoltage}}{1000 R} A \right\} \right\}$$

$$\frac{\text{controlMax cps keFw}}{\text{cps keFw} - \text{mvBatteryVoltage}}$$

```
schemeDResultFn[blend] /. fwImplMap // FullSimplify // simplifyUnits
T /. (schemeDResultFn[blend] /. fwImplMap) // abscissa // FullSimplify // Factor
Factor /@ (%) // Apart // framed
```

$$\left\{ S \rightarrow \left\{ \text{controlMax} \left(-1 + \frac{2 \text{blend cps keFw}}{\text{cps keFw} - \text{mvBatteryVoltage}} \right), -\frac{\text{cps keFw} + \text{mvBatteryVoltage}}{1000 R} A \right\}, \right. \\ T \rightarrow \left\{ \frac{\text{controlMax cps keFw} ((-1 + 2 \text{blend}) \text{cps keFw} + \text{mvBatteryVoltage})}{\text{cps}^2 \text{keFw}^2 - \text{mvBatteryVoltage}^2}, -\frac{\text{cps keFw}}{1000 R} A \right\}, \\ \left. Z \rightarrow \{0, 0\}, G \rightarrow \left\{ \text{controlMax}, \frac{-\text{cps keFw} + \text{mvBatteryVoltage}}{1000 R} A \right\} \right\}$$

$$\frac{\text{controlMax cps keFw} (-\text{cps keFw} + 2 \text{blend cps keFw} + \text{mvBatteryVoltage})}{(\text{cps keFw} - \text{mvBatteryVoltage}) (\text{cps keFw} + \text{mvBatteryVoltage})}$$

$$\frac{\text{blend controlMax cps keFw}}{\text{cps keFw} - \text{mvBatteryVoltage}} + \frac{(-1 + \text{blend}) \text{controlMax cps keFw}}{\text{cps keFw} + \text{mvBatteryVoltage}}$$

6.3. Conversion of Constants

In order to apply the above, we need to convert our SI Ke constant to the encoder-count based variation.

```
(Ke /. fwImplMap) == (Ke /. parameterQuantities)
uniqueSolve[%, keFw]
keFwExpr = keFw /. %;
```

$$\frac{\text{countsPerRev keFw}}{2000 \pi} \text{ kg m}^2 / (\text{s}^2 \text{A rad}) == \text{Ke s V / rad}$$

$$\left\{ \text{keFw} \rightarrow \frac{2000 \text{Ke} \pi}{\text{countsPerRev}} \right\}$$

What sort of scale does that work out to?

```
keData = {{Ke → 1.066 / 60, countsPerRev → 28}};
keFwExpr /. # & /@ keData
{3.98683}
```

A 6q20 fixed point should be oodles.

7. To Do

- In Scheme B, calculate controlReverse0, dealing with the voltage drop across the catch diode. This doesn't effect the above results, but will slightly alter the rescaling of control signal to duty cycle when in the reverse driving regime.
 - Can we improve on the approximation currently found in ssCurrentFromVoltageVelocity[]?
 - Consider presenting a 'current plot' of Scheme A to facilitate comparison between the schemes.
-

8. Revision History

- 2018.08.07. Initial version.
- 2018.08.13. Significant rewrite.
- 2018.08.14. Fix errors caused by switch from post-gearbox to pre-gearbox Ω . Minor prose corrections. Added Scheme C. Added Implementation section.
- 2018.08.15. Explore behavior of transition point with velocity in more detail.
- 2018.08.16. Use better Implementation math.
- 2018.08.19. Added Scheme D.
- 2018.08.29. Small edits to prose. Change formatting.
- 2018.08.31. Refactor implementation for Scheme D