
Time Constants

Robert Atkinson
26 August 2018

We explore extracting time constants from motor model expressions.

1. Introduction

1.1. Administrative

Before we begin, we load in some previously computed logic (Ref: <https://github.com/rgatkinson/RobotPhysics/blob/master/MotorPhysics-GearsInitialConditions.pdf>)

```
Get[NotebookDirectory[] <> "Utilities.m"]
inputDirectory = FileNameJoin[{NotebookDirectory[], "MotorPhysicsGearsInitialConditions.Output"}] <> $PathnameSeparator;
Get[inputDirectory <> "ParametersUnitsAndAssumptions.m"];
Get[inputDirectory <> "MotorModels.m"];
Get[inputDirectory <> "MotorTimeDomainFunctions.m"];
Get[inputDirectory <> "Misc.m"];
```

```
SetOptions[Plot, LabelStyle -> Directive[Background -> None]];
prettyPrintFontSize = 20;
framed[expr_] := Framed[expr, FrameStyle -> Darker[Green]]
```

2. Finding Time Constants

We define a function that finds time constants from the exponents of exponentials in an expression.

```
Clear[findTimeConstants]
findTimeConstants[expr_] := Module[{exprAnalyze, process, exps, matchQ, tc},
  exprAnalyze = TrigToExp[expr];
  process[x: Exp[Times[factor_?NumericQ, t, rest:_]]] := Module[{}, Sow[-1 / (factor * rest)]];
  process[_] := 0;
  exps = Reap[Scan[process[#, &, exprAnalyze, Infinity]][[2]] // Flatten // Union // FullSimplify;
  exps]
```

3. Motor Model

We explore the time constants in the step responses from our motor model. It turns out that they are all the same.

3.1. Current

```
findTimeConstants[curStepGeneric] // prettyPrint
```

$$\frac{2 L (J \eta N^2 + J a f t e r)}{B L \eta N^2 + J R \eta N^2 + B a f t e r L + J a f t e r R - \sqrt{((B L + J R) \eta N^2 + B a f t e r L + J a f t e r R)^2 - 4 L (J \eta N^2 + J a f t e r) ((K e K t + B R) \eta N^2 + B a f t e r R)}}$$
$$\frac{2 L (J \eta N^2 + J a f t e r)}{B L \eta N^2 + J R \eta N^2 + B a f t e r L + J a f t e r R + \sqrt{((B L + J R) \eta N^2 + B a f t e r L + J a f t e r R)^2 - 4 L (J \eta N^2 + J a f t e r) ((K e K t + B R) \eta N^2 + B a f t e r R)}}$$

3.2. Velocity

```
findTimeConstants[velStepGeneric] // prettyPrint
```

$$\frac{2 L (J \eta N^2 + J a f t e r)}{B L \eta N^2 + J R \eta N^2 + B a f t e r L + J a f t e r R - \sqrt{((B L + J R) \eta N^2 + B a f t e r L + J a f t e r R)^2 - 4 L (J \eta N^2 + J a f t e r) ((K e K t + B R) \eta N^2 + B a f t e r R)}}$$

$$\frac{2 L (J \eta N^2 + J a f t e r)}{B L \eta N^2 + J R \eta N^2 + B a f t e r L + J a f t e r R + \sqrt{((B L + J R) \eta N^2 + B a f t e r L + J a f t e r R)^2 - 4 L (J \eta N^2 + J a f t e r) ((K e K t + B R) \eta N^2 + B a f t e r R)}}$$

3.3. Emf

```
findTimeConstants[emfStepGeneric] // prettyPrint
```

$$\frac{2 L (J \eta N^2 + J a f t e r)}{B L \eta N^2 + J R \eta N^2 + B a f t e r L + J a f t e r R - \sqrt{((B L + J R) \eta N^2 + B a f t e r L + J a f t e r R)^2 - 4 L (J \eta N^2 + J a f t e r) ((K e K t + B R) \eta N^2 + B a f t e r R)}}$$

$$\frac{2 L (J \eta N^2 + J a f t e r)}{B L \eta N^2 + J R \eta N^2 + B a f t e r L + J a f t e r R + \sqrt{((B L + J R) \eta N^2 + B a f t e r L + J a f t e r R)^2 - 4 L (J \eta N^2 + J a f t e r) ((K e K t + B R) \eta N^2 + B a f t e r R)}}$$

4. Digging Deeper

Let's explore those time constants. First, we note that they both do in fact have correct units (seconds).

```
tc = findTimeConstants[curStepGeneric]
tc /. parameterQuantities
```

$$\left\{ \frac{2 L (J a f t e r + J \eta N^2)}{(B a f t e r L + J a f t e r R + B L \eta N^2 + J R \eta N^2 - \sqrt{(-4 L (J a f t e r + J \eta N^2) (B a f t e r R + (K e K t + B R) \eta N^2) + (B a f t e r L + J a f t e r R + (B L + J R) \eta N^2)^2})}, \right.$$

$$\left. \frac{2 L (J a f t e r + J \eta N^2)}{(B a f t e r L + J a f t e r R + B L \eta N^2 + J R \eta N^2 + \sqrt{(-4 L (J a f t e r + J \eta N^2) (B a f t e r R + (K e K t + B R) \eta N^2) + (B a f t e r L + J a f t e r R + (B L + J R) \eta N^2)^2})} \right\}$$

$$\left\{ \frac{2 L (J a f t e r + J \eta N^2)}{(B a f t e r L + J a f t e r R + B L \eta N^2 + J R \eta N^2 - \sqrt{(-4 L (J a f t e r + J \eta N^2) (B a f t e r R + (K e K t + B R) \eta N^2) + (B a f t e r L + J a f t e r R + (B L + J R) \eta N^2)^2})} s, \right.$$

$$\left. \frac{2 L (J a f t e r + J \eta N^2)}{(B a f t e r L + J a f t e r R + B L \eta N^2 + J R \eta N^2 + \sqrt{(-4 L (J a f t e r + J \eta N^2) (B a f t e r R + (K e K t + B R) \eta N^2) + (B a f t e r L + J a f t e r R + (B L + J R) \eta N^2)^2})} s \right\}$$

We rationalize the denominators for easier analysis. This gives us our main result.

```

ClearAll[rationalizeDenominator];
SetAttributes[rationalizeDenominator, Listable];
rationalizeDenominator[expr_] := Module[{num, den, f, scale, x, y},
  num = Numerator[expr];
  den = Denominator[expr];
  f[a_ + b: Power[c_, 1 / 2]] := (a - b);
  f[a_ - b: Power[c_, 1 / 2]] := (a + b);
  f[other_] := 1;
  scale = f[den];
  x = num * scale;
  y = FullSimplify[den * scale];
  x / y
]
rationalizeDenominator /@ tc // prettyPrint // framed

```

$$\frac{B L \eta N^2 + J R \eta N^2 + B a f t e r L + J a f t e r R + \sqrt{((B L + J R) \eta N^2 + B a f t e r L + J a f t e r R)^2 - 4 L (J \eta N^2 + J a f t e r) ((K e K t + B R) \eta N^2 + B a f t e r R)}}{2 ((K e K t + B R) \eta N^2 + B a f t e r R)}$$

$$\frac{B L \eta N^2 + J R \eta N^2 + B a f t e r L + J a f t e r R - \sqrt{((B L + J R) \eta N^2 + B a f t e r L + J a f t e r R)^2 - 4 L (J \eta N^2 + J a f t e r) ((K e K t + B R) \eta N^2 + B a f t e r R)}}{2 ((K e K t + B R) \eta N^2 + B a f t e r R)}$$

To simplify the model to help get some insight, we'll ignore the inductance:

```
noL = Limit[(rationalizeDenominator /@ tc), L -> 0]
```

$$\left\{ \frac{J a f t e r R + J R \eta N^2 + \sqrt{(J a f t e r R + J R \eta N^2)^2}}{2 (B a f t e r R + (K e K t + B R) \eta N^2)}, \frac{J a f t e r R + J R \eta N^2 - \sqrt{(J a f t e r R + J R \eta N^2)^2}}{2 (B a f t e r R + (K e K t + B R) \eta N^2)} \right\}$$

If we examine the numerator of the second of these results, we can see that it is zero. However, the first result simplifies nicely to a electro-mechanical result involving (among other things) the product of resistance and inertia and of resistance and drag (note that this is only an approximation, as we ignored the inductance).

```

(noL // First)
(tcApprox = (noL // First) /. {Power[x_^2, 1/2] -> x} // Simplify) // framed

```

$$\frac{J a f t e r R + J R \eta N^2 + \sqrt{(J a f t e r R + J R \eta N^2)^2}}{2 (B a f t e r R + (K e K t + B R) \eta N^2)}$$

$$\frac{R (J a f t e r + J \eta N^2)}{B a f t e r R + (K e K t + B R) \eta N^2}$$

Let's put some real numbers to this.

```
example = addMotorLoad[motorParameters["AM 60 A"], flywheel[Quantity[5, "kg"], Quantity[10, "cm"]]] // siUnits // clearUnits
```

$$\left\langle \begin{array}{l} R \rightarrow \frac{33}{10}, L \rightarrow \frac{347}{500000}, N \rightarrow 60, \eta \rightarrow \frac{9}{10}, K e \rightarrow \frac{533}{30000}, K t \rightarrow \frac{533}{30000}, \\ B \rightarrow \frac{11}{1080000}, J \rightarrow \frac{347}{108000000000}, J a f t e r \rightarrow \frac{1}{40}, B a f t e r \rightarrow 0, \Delta \tau_{appConst} \rightarrow 0 \end{array} \right\rangle$$

Our full result has two distinct time constants.

```
tc /. example // N
1 / %
```

```
{0.072744, 0.000210854}
```

```
{13.7468, 4742.62}
```

The first of those is the electro-mechanical constant we found in our induction-less approximation, and the second we conclude is an L-R electrical time constant.

```
tcApprox /. example // N
```

```
1 / %
```

```
0.0729347
```

```
13.7109
```

5. Revision History

- 2018.08.26. Initial version.