

Motor Physics, with Gears

Robert Atkinson, Maddy Nguyen, Samantha Hordyk, Cole Welch, and John Fraser
6 August 2018

An exploration of the physics of motors, in both symbolic and numeric form. A basic but sufficiently-complete parameterized model of a motor is presented. The differential equations governing that model are exhibited, and, from those, the (frequency domain) transfer functions of each of the model's variables is developed from first principles. Next, mathematical infrastructure for converting the response of general frequency domain functions input signals is developed, then applied to the transfer functions as a function of the applied-voltage and external-applied-torque inputs. This yields time domain functions for each of the model's variables in terms of the (symbolic) value of the model's variables. With those time domain symbolic functions in hand, the steady-state response of the motor as a function of each of the model's parameters is explored.

1. Administrivia

We begin by loading some handy utilities.


```
Get[NotebookDirectory[] <> "Utilities.m"]
```

```
outputDirectory = FileNameJoin[{NotebookDirectory[], "MotorPhysicsGears.Output"}] <> $PathnameSeparator  
CreateDirectory[outputDirectory] // Quiet;
```

```
C:\ftc\RobotPhysics\MotorPhysicsGears.Output\
```

2. Motor Model

The following figure gives a conceptual overview of a DC motor (ref: <http://www.egr.msu.edu/classes/me451/jchoi/2008/>).



In addition to the the above, we also here incorporate into our model:

- an (optional) N:1 reducing gearbox with associated efficiency
- an autonomous time-dependent external torque

The state of the system can thus be described by the following variables.

- $v_{bat}(t)$, the applied voltage (illustrated as $e_a(t)$ above)
- $i(t)$, the armature current
- $v_g(t)$, the back EMF ('g' for 'generator'; illustrated as $e_b(t)$ above)
- $\theta(t)$, the angular position of the shaft, before the gearbox
- $\omega(t)$, the angular velocity of the shaft, before the gearbox
- $\alpha(t)$, the angular acceleration of the shaft, before the gearbox
- $\tau(t)$, the torque driven by the armature current, before the gearbox

- $\tau_{\text{auto}}(t)$, an autonomous externally applied torque, after the gearbox. Often, this is constant, and caused by gravitational action.

The parameter constants that characterize the electrical aspects of the system are the following

- K_e , the electromotive force constant
- K_t , the motor torque constant
- L , electric inductance
- R , electric resistance

The following parameter constants characterize the mechanical aspects of the system

- J , the moment of inertia of the armature etc before the gearbox
- J_{out} , the moment of inertia of any load or torque attached after the gearbox. May be (net) negative, depending on system; see $\tau_{\text{AfterExpr}}$ below.
- B , motor viscous drag constant
- B_{out} , viscous drag of any load attached after the gearbox
- η , the efficiency of the motor's gear box (our current model does not support direction-dependent gearbox efficiencies).
- N (capital nu), the reduction ratio of the gear box

We collect together basic information about our parameters.

```
parameterNames = {J, Jout, B, Bout, Ke, Kt, R, L, η, N}
parameterAssumptions = {};
parameterAssumptions = Union[parameterAssumptions, # ≥ 0 & /@ {J, B, t}];
(*allow J & B to be zero so we can explore simpler motor models*)
parameterAssumptions = Union[parameterAssumptions, # > 0 & /@ {R, L, Ke, Kt, N, η}];
parameterAssumptions = Union[parameterAssumptions, # ∈ ℝ & /@ {Jout, Bout, constvbat, constτauto}];
parameterAssumptions = Union[parameterAssumptions, #[_] ∈ ℝ & /@ {vbat, i, vg, θ, ω, α, τ, τauto}]

{J, Jout, B, Bout, Ke, Kt, R, L, η, N}
```

```
{Bout ∈ ℝ, constvbat ∈ ℝ, constτauto ∈ ℝ, Jout ∈ ℝ, i[_] ∈ ℝ, vbat[_] ∈ ℝ, vg[_] ∈ ℝ, α[_] ∈ ℝ,
θ[_] ∈ ℝ, τ[_] ∈ ℝ, τauto[_] ∈ ℝ, ω[_] ∈ ℝ, Ke > 0, Kt > 0, L > 0, R > 0, η > 0, N > 0, B ≥ 0, J ≥ 0, t ≥ 0}
```

We set these assumptions as the default system assumptions so they're used automatically by Mathematica

```
$Assumptions = parameterAssumptions

{Bout ∈ ℝ, constvbat ∈ ℝ, constτauto ∈ ℝ, Jout ∈ ℝ, i[_] ∈ ℝ, vbat[_] ∈ ℝ, vg[_] ∈ ℝ, α[_] ∈ ℝ,
θ[_] ∈ ℝ, τ[_] ∈ ℝ, τauto[_] ∈ ℝ, ω[_] ∈ ℝ, Ke > 0, Kt > 0, L > 0, R > 0, η > 0, N > 0, B ≥ 0, J ≥ 0, t ≥ 0}
```

3. Differential Equations

3.1. Differential Equation Development

3.1.1. On Gearboxes and Torques

Our modelled system contains a gearbox with (reducing) gear ratio N and efficiency η . On either side of the gearbox we may have loads, drags, and driving torques, the latter being electrically generated on one side and externally applied on the other. In order to be able to write the torque differential equation for the system as a whole, we need to be able to take these two sets of torques as they are found in their respective *separate* places in the system and rewrite them both as they are *effectively* felt at some *one* point in the system. Commonly, in mechanical engineering this one point is chosen to be just before the gearbox, and the loads / torques after the gearbox are said to be “reflected” to create corresponding equivalent reflected loads and torques at the before-gearbox location. Here, we develop the mathematics of carrying out this reflection.

Just a couple of principles are involved:

- State related to angular position (position, velocity, acceleration, etc.) on the output side is $1/N$ th that of on the input side.
- Torque on the output side is $N\eta$ times that of on the input side.

```
Clear[reflectAngle, reflectTorque]
reflectAngle[expr_, out_, in_, t_] := expr /. {out[t] → in[t] / N, Derivative[n_][out][t] → Derivative[n][in][t] / N}
reflectAngle[expr_] := reflectAngle[expr, out, in, t]
reflectTorque[tauout_] := tauout / (N eta)
```

```
reflectTorque[const tauauto + Jout out''[t] + Bout out'[t]] // Expand
% // reflectAngle
```

$$\frac{\text{const } \tau_{\text{auto}}}{\eta N} + \frac{\text{Bout } \theta_{\text{out}}'[t]}{\eta N} + \frac{\text{Jout } \theta_{\text{out}}''[t]}{\eta N}$$

$$\frac{\text{const } \tau_{\text{auto}}}{\eta N} + \frac{\text{Bout } \theta'[t]}{\eta N^2} + \frac{\text{Jout } \theta''[t]}{\eta N^2}$$

3.1.2. Differential Equation System

With that understanding on how to reflect torques, we can now develop the differential equations that govern the system.

The motor torque is proportional to current.

```
ktEqn = tau[t] == Kt i[t]
```

```
tau[t] == Kt i[t]
```

The back EMF is proportional to velocity.

```
backEmfEqn = vg[t] == Ke theta'[t]
```

```
vg[t] == Ke theta'[t]
```

The after-gearbox angular position is a simple ratio of the before-gearbox position. This also applies to the velocity and acceleration.

```
thetaEqns = Derivative[#][thetaout][t] == Derivative[#][theta][t] / N & /@ Range[0, 2]
```

$$\left\{ \theta_{\text{out}}[t] = \frac{\theta[t]}{N}, \theta_{\text{out}}'[t] = \frac{\theta'[t]}{N}, \theta_{\text{out}}''[t] = \frac{\theta''[t]}{N} \right\}$$

Before the gearbox, the electrically driven motor torque is reduced by viscous drag and the inertia of the motor.

```
tauBeforeExpr = tau[t] + -B theta'[t] - J theta''[t]
```

```
tau[t] - B theta'[t] - J theta''[t]
```

After the gearbox, we have an autonomous, time-dependant externally applied torque, a viscous drag, and an internal load and/or an acceleration-dependent external torque.

```
tauAfterExpr = tauauto[t] - Bout theta'[t] - Jout theta''[t]
```

```
tauauto[t] - Bout theta'[t] - Jout theta''[t]
```

As all the torque must be accounted for, the sum of the before torques and the reflected after torques must be zero.

```
tauEqn = tauBeforeExpr + (tauAfterExpr // reflectTorque // reflectAngle) == 0;
tauEqn = Collect[tauEqn, {theta[t], theta'[t], theta''[t]}]
```

$$\tau[t] + \frac{\tau_{\text{auto}}[t]}{\eta N} + \left(-B - \frac{\text{Bout}}{\eta N^2} \right) \theta'[t] + \left(-J - \frac{\text{Jout}}{\eta N^2} \right) \theta''[t] = 0$$

The voltage drop across the resistance, induction and back EMF must equal the applied voltage (due to one of Kirchhoff's laws).

```
voltageEqn = vbat[t] == R i[t] + L i'[t] + vg[t]
```

```
vbat[t] == R i[t] + vg[t] + L i'[t]
```

We put it all together.

```
diffEqns = {ktEqn, backEmfEqn, voltageEqn, tauEqn} ~Join~ thetaEqns // FullSimplify;
diffEqns // prettyPrint
```

```
Kt i(t) = tau(t)
vg(t) = Ke theta'(t)
R i(t) + vg(t) + L i'(t) = vbat(t)
N (J eta N tau(t) + tauauto(t)) = (B eta N^2 + Bout) theta'(t) + (J eta N^2 + Jout) theta''(t)
theta(t) = N thetaout(t)
theta'(t) = N thetaout'(t)
theta''(t) = N thetaout''(t)
```

3.2. Initial Conditions

Some of the initial conditions we can intuit intellectually:

```
initialConditions = {
  theta[0] == 0,
  i[0] == 0,
  vg[0] == 0,
  vbat[0] == v,
  tauauto[0] == T
}

{theta[0] == 0, i[0] == 0, vg[0] == 0, vbat[0] == v, tauauto[0] == T}
```

Others we can solve for:

```
diffEqFunctionsOfTime = Reap[Scan[If[MatchQ[#, _[t]], Sow[#]] &, diffEqns, Infinity]][[2, 1]] // Union
others = Complement[diffEqFunctionsOfTime /. t -> 0, initialConditions][[All, 1]]]
(eqnsSubsKnownInitial = diffEqns /. t -> 0 /. (initialConditions /. Equal -> Rule) /. UnitStep[tau[0] theta'[0]] -> 1);
(solvedInitialConditions = (uniqueSolve[eqnsSubsKnownInitial, others] /. Rule -> Equal));
(allInitialConditions = solvedInitialConditions ~Join~ initialConditions) // prettyPrint

{i[t], vbat[t], vg[t], theta[t], thetaout[t], tau[t], tauauto[t], i'[t], theta'[t], thetaout'[t], theta''[t], thetaout''[t]}

{thetaout[0], tau[0], i'[0], theta'[0], thetaout'[0], theta''[0], thetaout''[0]}
```

```
thetaout(0) = 0
tau(0) = 0
i'(0) = v/L
theta(0) = 0
thetaout'(0) = 0
theta''(0) = (NT)/(J eta N^2 + Jout)
thetaout''(0) = T/(J eta N^2 + Jout)
theta(0) = 0
i(0) = 0
vg(0) = 0
vbat(0) = v
tauauto(0) = T
```

4. Motor Parameters

4.1. Motor Parameter Units

We develop infrastructure regarding the units in which each of our parameters are specified. This will help us sanity check the structure of our equations, as they must be internally consistent in their use of units (e.g.: no velocities added to masses, etc).

A note on radians as a unit. The Bureau International des Poids et Mesures, the intergovernmental organization that standardizes the SI, officially refers to a radian as “a special name is given to the unit one, in order to facilitate the identification of the quantity involved” (Ref: <https://www.bipm.org/en/publications/si-brochure/section2-2-3.html>). With that as background, we have here chosen to include radians in all places in which they logically belong, as doing so provides valuable insight regarding the quantities being manipulated. Thus, for example, you will find here torque having units of $\text{N m} / \text{radian}$ instead of N m , and the like. As, from the SI perspective, this amounts to multiplying or dividing by the unit one, no semantic change is effected.

Aside: in the opinion of one of the authors (Atkinson), that “a special name given to the unit one” is in any way at all different than a first-class “unit” is a perspective reasonable only if one artificially and myopically limits one’s worldview to contain only the seven units whose dimension is necessarily *experimentally* calibrated as opposed to units whose dimension is mathematically or otherwise calibrated, such as radian, becquerel, motor encoder ticks, etc. And the world is chock-full of the latter. But this is a discussion for another time. See also Jacques E. Romain, “Angle as a Fourth Fundamental Quantity”, Journal of Research of the National Bureau of Standards-B. Mathematics and Mathematical Physics, Vol. 66B, No. 3, July- September 1962.

```

radiansUnits = parseUnit["Radians"];
secondsUnits = parseUnit["Seconds"];
siAngularPositionUnits = radiansUnits;
siAngularVelocityUnits = siAngularPositionUnits / secondsUnits;
siAngularAccelerationUnits = siAngularVelocityUnits / secondsUnits;
siTorqueUnits = parseUnit["N m"] / radiansUnits;
siAngularViscousDragUnits = siTorqueUnits / siAngularVelocityUnits;
siAngularInertialUnits = siTorqueUnits / siAngularAccelerationUnits;

parameterUnits = {
  R → parseUnit["Ohms"],
  L → parseUnit["Henrys"],
  i[t] → parseUnit["Amperes"],
  i'[t] → parseUnit["Amperes"] / secondsUnits,
  i''[t] → parseUnit["Amperes"] / secondsUnits / secondsUnits,
  vbat[t] → parseUnit["Volts"],
  constvbat → parseUnit["Volts"],
  vg[t] → parseUnit["Volts"],

  J → siAngularInertialUnits,
  Jout → siAngularInertialUnits,
  B → siAngularViscousDragUnits,
  Bout → siAngularViscousDragUnits,

  Ke → parseUnit["Seconds Volts / Radians"],
  KeShaft → parseUnit["Seconds Volts / Radians"],

  Kt → parseUnit["Newtons Meters"] / parseUnit["Amperes"] / radiansUnits,
  KtShaft → parseUnit["Newtons Meters"] / parseUnit["Amperes"] / radiansUnits,

  θ[t] → siAngularPositionUnits, θout[t] → siAngularPositionUnits,
  θ'[t] → siAngularVelocityUnits, θout'[t] → siAngularPositionUnits,
  θ''[t] → siAngularAccelerationUnits, θout''[t] → siAngularPositionUnits,
  ω[t] → siAngularVelocityUnits,
  α[t] → siAngularAccelerationUnits,

  τ[t] → siTorqueUnits,
  τauto[t] → siTorqueUnits,
  constrauto → siTorqueUnits,

  N → parseUnit["DimensionlessUnit"],
  η → parseUnit["DimensionlessUnit"]
} // Association
unitsToQuantities[units_] := Module[{rules, makeQuantity},
  rules = Normal[units];
  makeQuantity = Function[{param, unit}, Quantity[param, unit]];
  (#[1] → makeQuantity @@ # & /@ rules) // Association
]
parameterQuantities = unitsToQuantities[parameterUnits]

```

$$\left(\begin{array}{l}
R \rightarrow \text{Ohms}, L \rightarrow \text{Henries}, i[t] \rightarrow \frac{\text{Amperes}}{\text{Seconds}}, i'[t] \rightarrow \frac{\text{Amperes}}{\text{Seconds}^2}, i''[t] \rightarrow \frac{\text{Amperes}}{\text{Seconds}^3}, vbat[t] \rightarrow \text{Volts}, constvbat \rightarrow \text{Volts}, vg[t] \rightarrow \text{Volts}, \\
J \rightarrow \frac{\text{Meters Newtons Seconds}^2}{\text{Radians}^2}, Jout \rightarrow \frac{\text{Meters Newtons Seconds}^2}{\text{Radians}^2}, B \rightarrow \frac{\text{Meters Newtons Seconds}}{\text{Radians}^2}, Bout \rightarrow \frac{\text{Meters Newtons Seconds}}{\text{Radians}^2}, \\
Ke \rightarrow \frac{\text{Seconds Volts}}{\text{Radians}}, KeShaft \rightarrow \frac{\text{Seconds Volts}}{\text{Radians}}, Kt \rightarrow \frac{\text{Meters Newtons}}{\text{Amperes Radians}}, KtShaft \rightarrow \frac{\text{Meters Newtons}}{\text{Amperes Radians}}, \theta[t] \rightarrow \text{Radians}, \\
\thetaout[t] \rightarrow \text{Radians}, \theta'[t] \rightarrow \frac{\text{Radians}}{\text{Seconds}}, \thetaout'[t] \rightarrow \text{Radians}, \theta''[t] \rightarrow \frac{\text{Radians}}{\text{Seconds}^2}, \thetaout''[t] \rightarrow \text{Radians}, \omega[t] \rightarrow \frac{\text{Radians}}{\text{Seconds}}, \alpha[t] \rightarrow \frac{\text{Radians}}{\text{Seconds}^2}, \\
\tau[t] \rightarrow \frac{\text{Meters Newtons}}{\text{Radians}}, \tauauto[t] \rightarrow \frac{\text{Meters Newtons}}{\text{Radians}}, constrauto \rightarrow \frac{\text{Meters Newtons}}{\text{Radians}}, N \rightarrow \text{DimensionlessUnit}, \eta \rightarrow \text{DimensionlessUnit} \end{array} \right)$$

```
(
  {
    R → R Ω, L → L H, i[t] → i[t] A, i'[t] → i'[t] A/s, i''[t] → i''[t] A/s², vbat[t] → vbat[t] V, constvbat → constvbat V,
    vg[t] → vg[t] V, J → J ms⁴N/rad⁴, Jout → Jout ms²N/rad², B → B msN/rad², Bout → Bout msN/rad⁴, Ke → Ke sV/rad,
    KeShaft → KeShaft sV/rad, Kt → Kt mN/(Arad), KtShaft → KtShaft mN/(Arad), θ[t] → θ[t] rad, θout[t] → θout[t] rad,
    θ'[t] → θ'[t] rad/s, θout'[t] → θout'[t] rad, θ''[t] → θ''[t] rad/s², θout''[t] → θout''[t] rad, ω[t] → ω[t] rad/s,
    α[t] → α[t] rad/s², τ[t] → τ[t] mN/rad, τauto[t] → τauto[t] mN/rad, constτauto → constτauto mN/rad, N → N, η → η
  }
)
```

Let's look and see if our units are consistent:

```
diffEqns /. parameterQuantities // ColumnForm
```

```
Kt i[t] mN/rad == τ[t] mN/rad
vg[t] V == Ke θ'[t] V
(R i[t] + vg[t] + L i'[t]) A Ω == vbat[t] V
N (η N τ[t] + τauto[t]) mN/rad == ((Bout + B η N²) θ'[t] + (Jout + J η N²) θ''[t]) mN/rad
θ[t] rad == N θout[t] rad
θ'[t] rad/s == N θout'[t] rad
θ''[t] rad/s² == N θout''[t] rad
```

Huzzah! They check out (if the units were incompatible, Mathematica would've told us).

We save information regarding our parameters for later use.

```
saveDefinitions[outputDirectory <> "ParametersUnitsAndAssumptions.m",
  {parameterUnits, parameterQuantities, parameterAssumptions, radiansUnits, secondsUnits, siAngularPositionUnits,
  siAngularVelocityUnits, siAngularAccelerationUnits, siTorqueUnits, siAngularViscousDragUnits, siAngularInertialUnits}]
```

4.2. On Ke & Kt

A brief digression. It is often remarked that, in compatible units, Ke & Kt have the same magnitude and sign. However, while often true, this relation does not *always* hold.

Recall our differential equations:

```
diffEqns // ColumnForm
```

```
Kt i[t] == τ[t]
vg[t] == Ke θ'[t]
R i[t] + vg[t] + L i'[t] == vbat[t]
N (η N τ[t] + τauto[t]) == (Bout + B η N²) θ'[t] + (Jout + J η N²) θ''[t]
θ[t] == N θout[t]
θ'[t] == N θout'[t]
θ''[t] == N θout''[t]
```

In steady state, the current and speed are constant.

```
(ssEqns = diffEqns /. {i'[t] → 0, θ''[t] → 0}) // ColumnForm
(ssEqns = Eliminate[ssEqns, {vg[t], B, θout[t], θout'[t], τauto[t]}) // ColumnForm
(ssEqns = ssEqns[[1 ;; 2]]) // ColumnForm
```

```
Kt i[t] == τ[t]
vg[t] == Ke θ'[t]
R i[t] + vg[t] == vbat[t]
N (η N τ[t] + τauto[t]) == (Bout + B η N²) θ'[t]
θ[t] == N θout[t]
θ'[t] == N θout'[t]
θ == N θout''[t]
```

```
Kt i[t] == τ[t]
vbat[t] == R i[t] + Ke θ'[t]
```

```
Kt i[t] == τ[t]
vbat[t] == R i[t] + Ke θ'[t]
```

From a power point of view (ref: <https://bit.ly/2Lxka0Z>), we must have (electrical) power in = (mechanical) power out + (electrical + other) power losses:

```
powerEqns = And @@ {
  pwrIn == vbat[t] i[t],
  pwrOut ==  $\tau[t] \theta'[t]$ ,
  pwrLosses ==  $i[t]^2 R + \text{pwrOtherPowerLosses}$ ,
  pwrIn == pwrOut + pwrLosses
};
(ssPowerEqns = Union[ssEqns, powerEqns]) // prettyPrint
```

$$\begin{aligned} \text{pwrIn} &= \text{pwrLosses} + \text{pwrOut} \\ \text{pwrIn} &= i(t) \text{vbat}(t) \\ \text{pwrLosses} &= R i(t)^2 + \text{pwrOtherPowerLosses} \\ \text{pwrOut} &= \tau(t) \theta'(t) \\ Kt i(t) &= \tau(t) \\ \text{vbat}(t) &= R i(t) + K_e \theta'(t) \end{aligned}$$

Let's eliminate a few of the variables that we're not interested in.

```
kBalanceEqns = Eliminate[ssPowerEqns, {i[t],  $\tau[t]$ ,  $\theta'[t]$ , vbat[t]}];
kBalanceEqns // prettyPrint
uniqueSolve[kBalanceEqns[[2]], pwrOtherPowerLosses] /. Rule -> Equal
```

$$\begin{aligned} \text{pwrIn} &= \text{pwrLosses} + \text{pwrOut} \\ Kt \text{pwrOtherPowerLosses} &= (K_e - K_t) \text{pwrOut} \end{aligned}$$

$$\left\{ \text{pwrOtherPowerLosses} == \frac{(K_e - K_t) \text{pwrOut}}{K_t} \right\}$$

We can conclude that if other power losses are zero, then K_e must equal K_t .

4.3. Experimental Motor Data

We have experimental data from several motors (Nguyen, Hordyk, & Fraser, 2017). We load in same.

```
Clear[importMotorData, motorData]
importMotorData[] := importMotorData["Characterized Motors (Gears).xlsx"];
importMotorData[file_] := Module[{},
  Import[NotebookDirectory[] <> file, {"Data", 1}][[2 ;;]]
]
motorData = importMotorData[];
(motorData) // TableForm
```

Name	R	L (uH)	L (H)	K_e (>gear)	K_t (>gear, est)	K_e (<gear)	K_t (<gear, est)	J	b
AM 20 A	2.3	691.	0.000691	0.351	0.351	0.01755	0.01755	9.011×10^{-6}	0.0022
AM 20 B	1.9	684.	0.000684	0.389	0.389	0.01945	0.01945	9.011×10^{-6}	0.0025
AM 20 C	5.1	717.	0.000717	0.385	0.385	0.01925	0.01925	8.931×10^{-6}	0.0028
AM 40 A	2.5	674.	0.000674	0.753	0.753	0.018825	0.018825	0.00002221	0.2269
AM 40 B	3.8	705.	0.000705	0.705	0.705	0.017625	0.017625	0.00001741	0.56
AM 40 C	2.1	716.	0.000716	0.763	0.763	0.019075	0.019075	0.00002471	0.018
AM 60 A	3.3	694.	0.000694	1.066	1.066	0.0177667	0.0177667	0.00001041	0.033
AM 60 B	5.1	696.	0.000696	1.076	1.076	0.0179333	0.0179333	8.421×10^{-6}	0.02
AM 3.7 A	8.9	679.	0.000679	0.099	0.099	0.0267568	0.0267568	0.00002791	0.0001
AM 3.7 B	2.6	797.	0.000797	0.108	0.108	0.0291892	0.0291892	0.00003151	0.0001
AM 3.7 C	8.7	880.	0.00088	0.105	0.105	0.0283784	0.0283784	0.00003091	0.0001
Matrix A	3.8	718.	0.000718	0.34	0.34	0.00643939	0.00643939	9.431×10^{-6}	0.0015
Matrix B	7.8	777.	0.000777	0.363	0.363	0.006875	0.006875	7.761×10^{-6}	0.0019
Matrix C	20.6	658.	0.000658	0.338	0.338	0.00640152	0.00640152	7.231×10^{-6}	0.0018
CoreHex A	3.6	1356.	0.001356	0.822	0.822	0.0226759	0.0226759	0.0007331	0.0112
CoreHex B	11.3	1352.	0.001352	0.858	0.858	0.023669	0.023669	0.0006551	0.008
CoreHex C	5.6	1342.	0.001342	0.711	0.711	0.0196138	0.0196138	0.0004541	0.0078

We build a function to retrieve data from the table. We convert from floating point to rational in order to help delay floating point collapse later on down the line. We include load and external torque parameters at zero level so these can be easily added to later.


```

Clear[motorParameters];
Options[motorParameters] = { Geared → True, Efficiency → "Forward" };
validateOption[motorParameters][Geared, val_] := BooleanQ[val];
validateOption[motorParameters][Efficiency, val_] := MemberQ[{"Forward", "Reverse"}, val];

motorParameters[motorName_, opts: OptionsPattern[]] := motorParameters[motorData, motorName, opts]
motorParameters[motorData_, motorName_, opts: OptionsPattern[]] :=
  validateOptions[] @ Module[{row, paramValues, quantify, qty, assoc},
    row = Select[motorData, #[[1]] == motorName &][[1]];
    paramValues = #[[1]] → toRational[ row[[#[[2]]]] ] & /@ {
      {R, 2}, {L, 4}, {KeShaft, 5}, {KtShaft, 6}, {Ke, 7}, {Kt, 8}, {J, 9}, {B, 10},
      {N, 11}, {η, If[OptionValue[Efficiency] == "Forward", 12, 13]} };
    quantify = Function[{name, value},
      qty = (name /. parameterQuantities);
      name → Quantity[value, QuantityUnit[qty]]
    ];
    assoc = quantify @@ # & /@ paramValues // Association;
    assoc[Jout] = Quantity[0, siAngularInertialUnits];
    assoc[Bout] = Quantity[0, siAngularViscousDragUnits];
    assoc[constauto] = Quantity[0, siTorqueUnits];
    If[! OptionValue[Geared],
      assoc[N] = 1;
      assoc[η] = 1;
      assoc[Ke] = assoc[KeShaft];
      assoc[Kt] = assoc[KtShaft];
    ];
    (* remove the shaft values to avoid confusion *)
    assoc[KeShaft] = .;
    assoc[KtShaft] = .;
    assoc // siUnits // simplifyUnits
  ];

motorParameters["AM 60 A", Geared → False]
motorParameters["AM 60 A", Geared → True, Efficiency → "Reverse"]

```

$$\left\langle \begin{array}{l} R \rightarrow \frac{33}{10} \text{ W/A}^2, L \rightarrow \frac{347}{500\,000} \text{ H}, Ke \rightarrow \frac{533}{500} \text{ kg m}^2 / (\text{s}^2 \text{A rad}), Kt \rightarrow \frac{533}{500} \text{ kg m}^2 / (\text{s}^2 \text{A rad}), J \rightarrow \frac{1041}{100\,000\,000} \text{ kg m}^2 / \text{rad}^2, \\ B \rightarrow \frac{33}{1000} \text{ kg m}^2 / (\text{s rad}^2), N \rightarrow 1, \eta \rightarrow 1, Jout \rightarrow 0 \text{ kg m}^2 / \text{rad}^2, Bout \rightarrow 0 \text{ kg m}^2 / (\text{s rad}^2), constauto \rightarrow 0 \text{ kg m}^2 / (\text{s}^2 \text{rad}) \end{array} \right\rangle$$

$$\left\langle \begin{array}{l} R \rightarrow \frac{33}{10} \text{ W/A}^2, L \rightarrow \frac{347}{500\,000} \text{ H}, Ke \rightarrow \frac{533}{30\,000} \text{ kg m}^2 / (\text{s}^2 \text{A rad}), Kt \rightarrow \frac{533}{30\,000} \text{ kg m}^2 / (\text{s}^2 \text{A rad}), J \rightarrow \frac{1041}{100\,000\,000} \text{ kg m}^2 / \text{rad}^2, \\ B \rightarrow \frac{33}{1000} \text{ kg m}^2 / (\text{s rad}^2), N \rightarrow 60, \eta \rightarrow \frac{4}{5}, Jout \rightarrow 0 \text{ kg m}^2 / \text{rad}^2, Bout \rightarrow 0 \text{ kg m}^2 / (\text{s rad}^2), constauto \rightarrow 0 \text{ kg m}^2 / (\text{s}^2 \text{rad}) \end{array} \right\rangle$$

4.4. Loads on the Motor

motorLoad[] creates a motor load given explicit internal and drag parameters. We allow the 'out' suffix but do not require its use.

```

Clear[motorLoad]
Options[motorLoad] = {
  J → Quantity[0, siAngularInertialUnits], Jout → Quantity[0, siAngularInertialUnits],
  B → Quantity[0, siAngularViscousDragUnits], Bout → Quantity[0, siAngularViscousDragUnits],
  constrauto → Quantity[0, siTorqueUnits]
};
motorLoad[opts : OptionsPattern[]] := Module[{assoc = Association[]},
  assoc[Jout] = OptionValue[J] + OptionValue[Jout];
  assoc[Bout] = OptionValue[B] + OptionValue[Bout];
  assoc[constrauto] = OptionValue[constrauto];
  assoc
]

```

flywheel[] creates a motor load given mass and radius parameters

```

Clear[flywheel]
flywheel[mass_, radius_] := motorLoad[J → mass * radius ^ 2 / 2 / Quantity[1, "Radians^2"]]
flywheel[Quantity[2, "kg"], Quantity[5, "cm"]]

```

```

{Jout → 25 kg cm²/rad², Bout → 0 m s N/rad², constrauto → 0 m N/rad}

```

addMotorLoad[] takes motor parameters and a load and returns new parameters that include the load.

```

addMotorLoad[motor_, load_] := Module[{assoc},
  assoc = Association[motor]; (*explicitly create copy*)
  assoc[Jout] = (Jout /. motor) + (Jout /. load);
  assoc[Bout] = (Bout /. motor) + (Bout /. load);
  assoc[constrauto] = (constrauto /. motor) + (constrauto /. load);
  assoc
]
addMotorLoad[motorParameters["AM 60 A"], flywheel[Quantity[2, "kg"], Quantity[5, "cm"]]]

```

```

{R →  $\frac{33}{10} \text{ W/A}^2$ , L →  $\frac{347}{500000} \text{ H}$ , Ke →  $\frac{533}{30000} \text{ kg m}^2 / (\text{s}^2 \text{ A rad})$ , Kt →  $\frac{533}{30000} \text{ kg m}^2 / (\text{s}^2 \text{ A rad})$ , J →  $\frac{1041}{100000000} \text{ kg m}^2 / \text{rad}^2$ ,
B →  $\frac{33}{1000} \text{ kg m}^2 / (\text{s rad}^2)$ , N → 60,  $\eta \rightarrow \frac{9}{10}$ , Jout → 25 kg cm²/rad², Bout → 0 kg m² / (s rad²), constrauto → 0 m N/rad}

```

4.5. Mass on pulley

Here, we model a mass suspended on a massless “rigid string” from a massless pulley attached to the motor shaft. That string acts at a wrench with a length of the radius of the pulley to deliver a torque to the shaft. rMassOnPulley[] computes that torque.

There is a force pulling down on the mass due to gravity and up due to acceleration in the string that results from the driven angular acceleration of the motor shaft. The sum of these forces is the tension in the string. If the net acceleration is negative (ie: downward), a real (flexible) string would clamp at zero and have the mass in free-fall; we choose not to model that for now. Instead, here, our hypothetical rigid string continues to impart acceleration from the pulley even for negative accelerations. This is a bit odd for a string, but more reasonable if the string were in fact a rigid arm being lifted, or some such.

```

Clear[massOnPulley]
massOnPulley[mass_, radius_] := Module[
  {gravityAcceleration, gravityForce, angularAcceleration,
   tangentialAcceleration, tangentialForce, tension, torque, unit, parts, radians = Quantity["Radians"]},

  (* gravity exerts a constant downward force by acting on the mass *)
  gravityAcceleration = Quantity["StandardAccelerationOfGravity"];
  gravityForce = gravityAcceleration * mass;

  (* angular acceleration, converted to tangential acceleration also acts on the mass to create an upward force *)
  angularAcceleration = Quantity["0"][t, siAngularAccelerationUnits];
  tangentialAcceleration = angularAcceleration * radius / radians;
  tangentialForce = tangentialAcceleration * mass;

  (* the net force (the tension in the string) is the sum of the two forces *)
  tension = gravityForce + tangentialForce;

  (* the tension acts like a wrench with lever arm 'radius' *)
  torque = tension * radius / radians;

  (* we decompose the torque into constant and acceleration-dependent parts for output *)
  unit = QuantityUnit[torque];
  parts = List @@ (QuantityMagnitude[torque] // Apart);
  motorLoad[const:auto → Quantity[parts[[1]], unit], Jout → Quantity[parts[[2]], unit] / angularAcceleration]
]
massOnPulley[Quantity[weightLbs, "Pounds"], Quantity[rInches, "Inches"]]
% // siUnits // clearUnits // N

```

$$\langle \text{Jout} \rightarrow r\text{Inches}^2 \text{ weightLbs } \text{lb in}^2/\text{rad}^2, \text{ Bout} \rightarrow 0 \text{ ms N/rad}^2, \text{ const:auto} \rightarrow \frac{196133 \text{ rInches weightLbs}}{508} \text{ lb in}^2/(\text{s}^2\text{rad}) \rangle$$

$$\langle \text{Jout} \rightarrow 0.00029264 \text{ rInches}^2 \text{ weightLbs}, \text{ Bout} \rightarrow 0., \text{ const:auto} \rightarrow 0.112985 \text{ rInches weightLbs} \rangle$$

4.6. Example Motors

We define a motor and loads that we'll use to illustrate examples. We also define a generic, abstract motor that can help explore things symbolically.

'aMotorClassic' is the configuration we explored in our previous work. Ref: <https://github.com/rgatkinson/RobotPhysics/blob/master/MotorPhysics.pdf>

```

aMotor = motorParameters["AM 60 A"]
aMotorClassic = addMotorLoad[motorParameters["AM 60 A", Geared → False], motorLoad[J → Quantity[1, siAngularInertialUnits]]]
aFlywheel = flywheel[Quantity[1, "kg"], Quantity[5, "cm"]] // siUnits
aPulley = massOnPulley[Quantity[3, "Pounds"], Quantity[2, "Inches"]] // siUnits

```

$$\left\langle \left\{ R \rightarrow \frac{33}{10} \text{W/A}^2, L \rightarrow \frac{347}{500000} \text{H}, K_e \rightarrow \frac{533}{30000} \text{kg m}^2/(\text{s}^2 \text{A rad}), K_t \rightarrow \frac{533}{30000} \text{kg m}^2/(\text{s}^2 \text{A rad}), J \rightarrow \frac{1041}{100000000} \text{kg m}^2/\text{rad}^2, \right. \right.$$

$$\left. B \rightarrow \frac{33}{1000} \text{kg m}^2/(\text{s rad}^2), N \rightarrow 60, \eta \rightarrow \frac{9}{10}, J_{out} \rightarrow 0 \text{kg m}^2/\text{rad}^2, B_{out} \rightarrow 0 \text{kg m}^2/(\text{s rad}^2), \text{constraint} \rightarrow 0 \text{kg m}^2/(\text{s}^2 \text{rad}) \right\} \rangle$$

$$\left\langle \left\{ R \rightarrow \frac{33}{10} \text{W/A}^2, L \rightarrow \frac{347}{500000} \text{H}, K_e \rightarrow \frac{533}{500} \text{kg m}^2/(\text{s}^2 \text{A rad}), K_t \rightarrow \frac{533}{500} \text{kg m}^2/(\text{s}^2 \text{A rad}), J \rightarrow \frac{1041}{100000000} \text{kg m}^2/\text{rad}^2, \right. \right.$$

$$\left. B \rightarrow \frac{33}{1000} \text{kg m}^2/(\text{s rad}^2), N \rightarrow 1, \eta \rightarrow 1, J_{out} \rightarrow 1 \text{kg m}^2/\text{rad}^2, B_{out} \rightarrow 0 \text{kg m}^2/(\text{s rad}^2), \text{constraint} \rightarrow 0 \text{mN/rad} \right\} \rangle$$

$$\left\langle \left\{ J_{out} \rightarrow \frac{1}{800} \text{kg m}^2/\text{rad}^2, B_{out} \rightarrow 0 \text{kg m}^2/(\text{s rad}^2), \text{constraint} \rightarrow 0 \text{kg m}^2/(\text{s}^2 \text{rad}) \right\} \right\rangle$$

$$\left\langle \left\{ J_{out} \rightarrow \frac{2194797400719}{62500000000000} \text{kg m}^2/\text{rad}^2, B_{out} \rightarrow 0 \text{kg m}^2/(\text{s rad}^2), \text{constraint} \rightarrow \frac{3389544870828501}{500000000000000} \text{kg m}^2/(\text{s}^2 \text{rad}) \right\} \right\rangle$$

```

genericMotor = # → (# /. parameterQuantities) & /@ Keys[aMotor]
genericMotorLoad = # → (# /. parameterQuantities) & /@ Keys[aFlywheel]

```

$$\left\{ R \rightarrow R \Omega, L \rightarrow L H, K_e \rightarrow K_e \text{sV/rad}, K_t \rightarrow K_t \text{mN/(A rad)}, J \rightarrow J \text{ms}^2 \text{N/rad}^2, B \rightarrow B \text{mN/rad}^2, \right.$$

$$N \rightarrow N, \eta \rightarrow \eta, J_{out} \rightarrow J_{out} \text{ms}^2 \text{N/rad}^2, B_{out} \rightarrow B_{out} \text{mN/rad}^2, \text{constraint} \rightarrow \text{constraint} \text{mN/rad} \left. \right\}$$

$$\left\{ J_{out} \rightarrow J_{out} \text{ms}^2 \text{N/rad}^2, B_{out} \rightarrow B_{out} \text{mN/rad}^2, \text{constraint} \rightarrow \text{constraint} \text{mN/rad} \right\}$$

5. Laplace Transforms and Transfer Function Models

5.1. Laplace Transforms

Returning to the differential equations, we form the Laplace Transform of each, then solve the for the various transforms. First, we apply the Laplace-Transform[] function to each equation. It automatically makes use of the linearity of the transform, and insinuates itself just around the time dependent parts (ie: the parts dependent on the time variable we told it about, namely t).

```
diffEqnsSimplified = FullSimplify[Eliminate[diffEqns, {η[t]}], parameterAssumptions];
leqns = LaplaceTransform[#, t, s] & /@ diffEqns
leqns // prettyPrint
```

```
{Kt LaplaceTransform[i[t], t, s] == LaplaceTransform[τ[t], t, s],
 LaplaceTransform[vg[t], t, s] == Ke (s LaplaceTransform[θ[t], t, s] - θ[0]),
 R LaplaceTransform[i[t], t, s] + L (-i[0] + s LaplaceTransform[i[t], t, s]) + LaplaceTransform[vg[t], t, s] ==
 LaplaceTransform[vbat[t], t, s], N (η N LaplaceTransform[τ[t], t, s] + LaplaceTransform[τauto[t], t, s]) ==
 (Bout + B η N2) (s LaplaceTransform[θ[t], t, s] - θ[0]) + (Jout + J η N2) (s2 LaplaceTransform[θ[t], t, s] - s θ[0] - θ'[0]),
 LaplaceTransform[θ[t], t, s] == N LaplaceTransform[θout[t], t, s],
 s LaplaceTransform[θ[t], t, s] - θ[0] == N (s LaplaceTransform[θout[t], t, s] - θout[0]),
 s2 LaplaceTransform[θ[t], t, s] - s θ[0] - θ'[0] == N (s2 LaplaceTransform[θout[t], t, s] - s θout[0] - θout'[0])}
```

```
Kt (Li[i(t)](s)) = Li[τ(t)](s)
Li[vg(t)](s) = Ke (s (Li[θ(t)](s)) - θ(0))
R (Li[i(t)](s)) + L (s (Li[i(t)](s)) - i(0)) + Li[vg(t)](s) = Li[vbat(t)](s)
N (η N (Li[τ(t)](s)) + Li[τauto(t)](s)) = (B η N2 + Bout) (s (Li[θ(t)](s)) - θ(0)) + (J η N2 + Jout) ((Li[θ(t)](s)) s2 - θ(0) s - θ'(0))
Li[θ(t)](s) = N (Li[θout(t)](s))
s (Li[θ(t)](s)) - θ(0) = N (s (Li[θout(t)](s)) - θout(0))
(Li[θ(t)](s)) s2 - θ(0) s - θ'(0) = N ((Li[θout(t)](s)) s2 - θout(0) s - θout'(0))
```

Next, we walk those equations, picking up those insinuations and sowing them to the wind, reaping them on the outside, and finally removing duplicates.

```
allXforms = Reap[Scan[(If[MatchQ[#, _LaplaceTransform], Sow[#]] &, leqns, Infinity]][[2, 1]] // Union
```

```
{LaplaceTransform[i[t], t, s], LaplaceTransform[vbat[t], t, s], LaplaceTransform[vg[t], t, s], LaplaceTransform[θ[t], t, s],
 LaplaceTransform[θout[t], t, s], LaplaceTransform[τ[t], t, s], LaplaceTransform[τauto[t], t, s]}
```

The voltage transform is input, as is any autonomous externally applied torque; all the others are outputs. Solve for the outputs. Finally substitute what we know about initial conditions, and simplify as much as we can.

```
voltageXform = LaplaceTransform[vbat[t], t, s];
τautoXform = LaplaceTransform[τauto[t], t, s];
inputXforms = {voltageXform, τautoXform}
inputLabels = {"vbat", "τauto"}
(outputXforms = Complement[allXforms, inputXforms]) // ColumnForm
```

```
{LaplaceTransform[vbat[t], t, s], LaplaceTransform[τauto[t], t, s]}
```

```
{vbat, τauto}
```

```
LaplaceTransform[i[t], t, s]
LaplaceTransform[vg[t], t, s]
LaplaceTransform[θ[t], t, s]
LaplaceTransform[θout[t], t, s]
LaplaceTransform[τ[t], t, s]
```

```
(leqnsToSolve = leqns /. (allInitialConditions /. Equal -> Rule)) // ColumnForm
```

```
Kt LaplaceTransform[i[t], t, s] == LaplaceTransform[τ[t], t, s]
LaplaceTransform[vg[t], t, s] == Ke s LaplaceTransform[θ[t], t, s]
R LaplaceTransform[i[t], t, s] + L s LaplaceTransform[i[t], t, s] + LaplaceTransform[vg[t], t, s] == LaplaceTransform[vbat[t], t, s]
N (η N LaplaceTransform[τ[t], t, s] + LaplaceTransform[τauto[t], t, s]) == s (Bout + B η N2) LaplaceTransform[θ[t], t, s] + s2 (Jout + J η N2) L
LaplaceTransform[θ[t], t, s] == N LaplaceTransform[θout[t], t, s]
s LaplaceTransform[θ[t], t, s] == s N LaplaceTransform[θout[t], t, s]
s2 LaplaceTransform[θ[t], t, s] == s2 N LaplaceTransform[θout[t], t, s]
```

```
(rawSolvedXforms = uniqueSolve[leqnsToSolve, outputXforms]) // ColumnForm

LaplaceTransform[i[t], t, s] → 
$$-\frac{\text{Bout LaplaceTransform[vbat[t], t, s]} - \text{Jout s LaplaceTransform[vbat[t], t, s]} - \text{B } \eta \text{ } \eta^2 \text{ LaplaceTransform[vbat[t], t, s]} - \text{J } \eta \text{ } \eta^2 \text{ LaplaceTransform[vbat[t], t, s]}}{\text{Bout R} + \text{Bout L s} + \text{Jout R s} + \text{Jout L s}^2 + \text{Ke Kt } \eta \text{ } \eta^2 + \text{B R } \eta \text{ } \eta^2 + \text{B L s } \eta \text{ } \eta^2 + \text{J R s } \eta \text{ } \eta^2 + \text{J L s}^2 \eta \text{ } \eta^2}$$

LaplaceTransform[vg[t], t, s] → 
$$\frac{\text{Ke s } ( \text{Kt } \eta \text{ } \eta^2 \text{ LaplaceTransform[vbat[t], t, s]} + (\text{R} + \text{L s}) \text{ N LaplaceTransform[\tauauto[t], t, s]} )}{-\text{Ke Kt s } \eta \text{ } \eta^2 - (\text{R} + \text{L s}) ( \text{s } (\text{Bout} + \text{B } \eta \text{ } \eta^2) + \text{s}^2 (\text{Jout} + \text{J } \eta \text{ } \eta^2) )}$$

LaplaceTransform[\theta[t], t, s] → 
$$\frac{\text{Kt } \eta \text{ } \eta^2 \text{ LaplaceTransform[vbat[t], t, s]} + (\text{R} + \text{L s}) \text{ N LaplaceTransform[\tauauto[t], t, s]}}{-\text{Ke Kt s } \eta \text{ } \eta^2 - (\text{R} + \text{L s}) ( \text{s } (\text{Bout} + \text{B } \eta \text{ } \eta^2) + \text{s}^2 (\text{Jout} + \text{J } \eta \text{ } \eta^2) )}$$

LaplaceTransform[\thetaout[t], t, s] → 
$$-\frac{\text{Kt } \eta \text{ } \eta^2 \text{ LaplaceTransform[vbat[t], t, s]} + (\text{R} + \text{L s}) \text{ N LaplaceTransform[\tauauto[t], t, s]}}{\text{N } ( -\text{Ke Kt s } \eta \text{ } \eta^2 - (\text{R} + \text{L s}) ( \text{s } (\text{Bout} + \text{B } \eta \text{ } \eta^2) + \text{s}^2 (\text{Jout} + \text{J } \eta \text{ } \eta^2) ) )}$$

LaplaceTransform[\tau[t], t, s] → 
$$-\frac{\text{Bout R} + \text{Bout L s} + \text{Jout R s} + \text{Jout L s}^2 + \text{Ke Kt } \eta \text{ } \eta^2 + \text{B R } \eta \text{ } \eta^2 + \text{B L s } \eta \text{ } \eta^2 + \text{J R s } \eta \text{ } \eta^2 + \text{J L s}^2 \eta \text{ } \eta^2}{\text{Bout R} + \text{Bout L s} + \text{Jout R s} + \text{Jout L s}^2 + \text{Ke Kt } \eta \text{ } \eta^2 + \text{B R } \eta \text{ } \eta^2 + \text{B L s } \eta \text{ } \eta^2 + \text{J R s } \eta \text{ } \eta^2 + \text{J L s}^2 \eta \text{ } \eta^2}$$

```

We separate the transforms into applied-voltage- and applied-torque-dependent parts.

```
Clear[separateXforms]
separateXforms[xform_] := Module[{xformVarRules, invertedRules, xformVars, vard},
  xformVarRules = # → Unique["laplaceTransform"] & /@ inputXforms;
  invertedRules = Reverse /@ xformVarRules;
  xformVars = xformVarRules[All, 2];
  vard = xform /. xformVarRules;
  Collect[vard, xformVars] /. invertedRules
]
solvedXforms = #[[1]] → separateXforms[#[[2]]] & /@ rawSolvedXforms;
% // ColumnForm

LaplaceTransform[i[t], t, s] → 
$$-\frac{(-\text{Bout} - \text{Jout s} - \text{B } \eta \text{ } \eta^2 - \text{J } \eta \text{ } \eta^2) \text{ LaplaceTransform[vbat[t], t, s]}}{\text{Bout R} + \text{Bout L s} + \text{Jout R s} + \text{Jout L s}^2 + \text{Ke Kt } \eta \text{ } \eta^2 + \text{B R } \eta \text{ } \eta^2 + \text{B L s } \eta \text{ } \eta^2 + \text{J R s } \eta \text{ } \eta^2 + \text{J L s}^2 \eta \text{ } \eta^2} - \frac{\text{Ke N LaplaceTransform[\tauauto[t], t, s]}}{\text{Bout R} + \text{Bout L s} + \text{Jout R s} + \text{Jout L s}^2 + \text{Ke Kt } \eta \text{ } \eta^2 + \text{B R } \eta \text{ } \eta^2 + \text{B L s } \eta \text{ } \eta^2 + \text{J R s } \eta \text{ } \eta^2 + \text{J L s}^2 \eta \text{ } \eta^2}$$

LaplaceTransform[vg[t], t, s] → 
$$\frac{\text{Ke Kt s } \eta \text{ } \eta^2 \text{ LaplaceTransform[vbat[t], t, s]}}{-\text{Ke Kt s } \eta \text{ } \eta^2 - (\text{R} + \text{L s}) ( \text{s } (\text{Bout} + \text{B } \eta \text{ } \eta^2) + \text{s}^2 (\text{Jout} + \text{J } \eta \text{ } \eta^2) )} - \frac{\text{Ke s } (\text{R} + \text{L s}) \text{ N LaplaceTransform[\tauauto[t], t, s]}}{-\text{Ke Kt s } \eta \text{ } \eta^2 - (\text{R} + \text{L s}) ( \text{s } (\text{Bout} + \text{B } \eta \text{ } \eta^2) + \text{s}^2 (\text{Jout} + \text{J } \eta \text{ } \eta^2) )}$$

LaplaceTransform[\theta[t], t, s] → 
$$\frac{\text{Kt } \eta \text{ } \eta^2 \text{ LaplaceTransform[vbat[t], t, s]}}{-\text{Ke Kt s } \eta \text{ } \eta^2 - (\text{R} + \text{L s}) ( \text{s } (\text{Bout} + \text{B } \eta \text{ } \eta^2) + \text{s}^2 (\text{Jout} + \text{J } \eta \text{ } \eta^2) )} - \frac{(\text{R} + \text{L s}) \text{ N LaplaceTransform[\tauauto[t], t, s]}}{-\text{Ke Kt s } \eta \text{ } \eta^2 - (\text{R} + \text{L s}) ( \text{s } (\text{Bout} + \text{B } \eta \text{ } \eta^2) + \text{s}^2 (\text{Jout} + \text{J } \eta \text{ } \eta^2) )}$$

LaplaceTransform[\thetaout[t], t, s] → 
$$-\frac{\text{Kt } \eta \text{ } \eta^2 \text{ LaplaceTransform[vbat[t], t, s]}}{-\text{Ke Kt s } \eta \text{ } \eta^2 - (\text{R} + \text{L s}) ( \text{s } (\text{Bout} + \text{B } \eta \text{ } \eta^2) + \text{s}^2 (\text{Jout} + \text{J } \eta \text{ } \eta^2) )} - \frac{(\text{R} + \text{L s}) \text{ LaplaceTransform[\tauauto[t], t, s]}}{-\text{Ke Kt s } \eta \text{ } \eta^2 - (\text{R} + \text{L s}) ( \text{s } (\text{Bout} + \text{B } \eta \text{ } \eta^2) + \text{s}^2 (\text{Jout} + \text{J } \eta \text{ } \eta^2) )}$$

LaplaceTransform[\tau[t], t, s] → 
$$-\frac{(-\text{Bout} - \text{Jout s} - \text{B } \eta \text{ } \eta^2 - \text{J } \eta \text{ } \eta^2) \text{ LaplaceTransform[vbat[t], t, s]}}{\text{Bout R} + \text{Bout L s} + \text{Jout R s} + \text{Jout L s}^2 + \text{Ke Kt } \eta \text{ } \eta^2 + \text{B R } \eta \text{ } \eta^2 + \text{B L s } \eta \text{ } \eta^2 + \text{J R s } \eta \text{ } \eta^2 + \text{J L s}^2 \eta \text{ } \eta^2} - \frac{\text{Ke Kt N LaplaceTransform[\tauauto[t], t, s]}}{\text{Bout R} + \text{Bout L s} + \text{Jout R s} + \text{Jout L s}^2 + \text{Ke Kt } \eta \text{ } \eta^2 + \text{B R } \eta \text{ } \eta^2 + \text{B L s } \eta \text{ } \eta^2 + \text{J R s } \eta \text{ } \eta^2 + \text{J L s}^2 \eta \text{ } \eta^2}$$

```

5.2. Transfer Function Models

We use the solved transforms to create TransferFunctionModel[]s for every variable of interest.

```
Clear[divideTerms]
divideTerms[sum_, termDivisors_] := Module[{list},
  list = List @@ sum;
  #[[1]] / #[[2]] & /@ Transpose[{list, termDivisors}]
]
divideTerms[a x + b y, {x, y}]
```

```
{a, b}
```

```
Clear[makeModel]
makeModel[var_] := makeModel[var, ToString[var], 1]
makeModel[var_, outputLabel_, factor_] := TransferFunctionModel[
  factor * Function[{sum}, divideTerms[sum, inputXforms]],
  {LaplaceTransform[var[t], t, s] /. solvedXforms},
  s,
  SystemsModelLabels → {inputLabels, outputLabel}] // FullSimplify // Simplify
```

```

motorPositionModel = makeModel[θ]
motorVelocityModel = makeModel[θ, "ω", s]
motorAccelerationModel = makeModel[θ, "α", s * s]
motorCurrentModel = makeModel[i]
motorEMFModel = makeModel[vg, "vg", 1]
motorTorqueModel = makeModel[τ]

```

$$\left(\begin{array}{c|c} \frac{v_{bat}}{K_t \eta N^2} & \frac{\tau_{auto}}{N (R + L s)} \\ \hline \frac{K_e K_t \eta N^2 s + s (R + L s) (Bout + Jout s + \eta N^2 (B + J s))}{K_e K_t \eta N^2 s + s (R + L s) (Bout + Jout s + \eta N^2 (B + J s))} & \frac{K_e K_t \eta N^2 s + s (R + L s) (Bout + Jout s + \eta N^2 (B + J s))}{K_e K_t \eta N^2 s + s (R + L s) (Bout + Jout s + \eta N^2 (B + J s))} \end{array} \right) \mathcal{T}$$

$$\left(\begin{array}{c|c} \frac{v_{bat}}{K_t \eta N^2} & \frac{\tau_{auto}}{N (R + L s)} \\ \hline \frac{Bout (R + L s) + Jout s (R + L s) + \eta N^2 (K_e K_t + (B + J s) (R + L s))}{Bout (R + L s) + Jout s (R + L s) + \eta N^2 (K_e K_t + (B + J s) (R + L s))} & \frac{Bout (R + L s) + Jout s (R + L s) + \eta N^2 (K_e K_t + (B + J s) (R + L s))}{Bout (R + L s) + Jout s (R + L s) + \eta N^2 (K_e K_t + (B + J s) (R + L s))} \end{array} \right) \mathcal{T}$$

$$\left(\begin{array}{c|c} \frac{v_{bat}}{K_t \eta N^2 s} & \frac{\tau_{auto}}{N s (R + L s)} \\ \hline \frac{Bout (R + L s) + Jout s (R + L s) + \eta N^2 (K_e K_t + (B + J s) (R + L s))}{Bout (R + L s) + Jout s (R + L s) + \eta N^2 (K_e K_t + (B + J s) (R + L s))} & \frac{Bout (R + L s) + Jout s (R + L s) + \eta N^2 (K_e K_t + (B + J s) (R + L s))}{Bout (R + L s) + Jout s (R + L s) + \eta N^2 (K_e K_t + (B + J s) (R + L s))} \end{array} \right) \mathcal{T}$$

$$\left(\begin{array}{c|c} \frac{v_{bat}}{Bout + Jout s + \eta N^2 (B + J s)} & \frac{\tau_{auto}}{K_e N} \\ \hline \frac{K_e K_t \eta N^2 + (R + L s) (Bout + Jout s + \eta N^2 (B + J s))}{K_e K_t \eta N^2 + (R + L s) (Bout + Jout s + \eta N^2 (B + J s))} & \frac{K_e K_t \eta N^2 + (R + L s) (Bout + Jout s + \eta N^2 (B + J s))}{K_e K_t \eta N^2 + (R + L s) (Bout + Jout s + \eta N^2 (B + J s))} \end{array} \right) \mathcal{T}$$

$$\left(\begin{array}{c|c} \frac{v_{bat}}{K_e K_t \eta N^2} & \frac{\tau_{auto}}{K_e N (R + L s)} \\ \hline \frac{Bout (R + L s) + Jout s (R + L s) + \eta N^2 (K_e K_t + (B + J s) (R + L s))}{Bout (R + L s) + Jout s (R + L s) + \eta N^2 (K_e K_t + (B + J s) (R + L s))} & \frac{Bout (R + L s) + Jout s (R + L s) + \eta N^2 (K_e K_t + (B + J s) (R + L s))}{Bout (R + L s) + Jout s (R + L s) + \eta N^2 (K_e K_t + (B + J s) (R + L s))} \end{array} \right) \mathcal{T}$$

$$\left(\begin{array}{c|c} \frac{v_{bat}}{K_t (Bout + Jout s + \eta N^2 (B + J s))} & \frac{\tau_{auto}}{K_e K_t N} \\ \hline \frac{K_e K_t \eta N^2 + (R + L s) (Bout + Jout s + \eta N^2 (B + J s))}{K_e K_t \eta N^2 + (R + L s) (Bout + Jout s + \eta N^2 (B + J s))} & \frac{K_e K_t \eta N^2 + (R + L s) (Bout + Jout s + \eta N^2 (B + J s))}{K_e K_t \eta N^2 + (R + L s) (Bout + Jout s + \eta N^2 (B + J s))} \end{array} \right) \mathcal{T}$$

```

motorPositionOutModel = makeModel[θout]
motorVelocityOutModel = makeModel[θout, "ωout", s]
motorAccelerationOutModel = makeModel[θout, "αout", s * s]

```

$$\left(\begin{array}{c|c} \frac{v_{bat}}{K_t \eta N} & \frac{\tau_{auto}}{R + L s} \\ \hline \frac{K_e K_t \eta N^2 s + s (R + L s) (Bout + Jout s + \eta N^2 (B + J s))}{K_e K_t \eta N^2 s + s (R + L s) (Bout + Jout s + \eta N^2 (B + J s))} & \frac{K_e K_t \eta N^2 s + s (R + L s) (Bout + Jout s + \eta N^2 (B + J s))}{K_e K_t \eta N^2 s + s (R + L s) (Bout + Jout s + \eta N^2 (B + J s))} \end{array} \right) \mathcal{T}$$

$$\left(\begin{array}{c|c} \frac{v_{bat}}{K_t \eta N} & \frac{\tau_{auto}}{R + L s} \\ \hline \frac{Bout (R + L s) + Jout s (R + L s) + \eta N^2 (K_e K_t + (B + J s) (R + L s))}{Bout (R + L s) + Jout s (R + L s) + \eta N^2 (K_e K_t + (B + J s) (R + L s))} & \frac{Bout (R + L s) + Jout s (R + L s) + \eta N^2 (K_e K_t + (B + J s) (R + L s))}{Bout (R + L s) + Jout s (R + L s) + \eta N^2 (K_e K_t + (B + J s) (R + L s))} \end{array} \right) \mathcal{T}$$

$$\left(\begin{array}{c|c} \frac{v_{bat}}{K_t \eta N s} & \frac{\tau_{auto}}{s (R + L s)} \\ \hline \frac{Bout (R + L s) + Jout s (R + L s) + \eta N^2 (K_e K_t + (B + J s) (R + L s))}{Bout (R + L s) + Jout s (R + L s) + \eta N^2 (K_e K_t + (B + J s) (R + L s))} & \frac{Bout (R + L s) + Jout s (R + L s) + \eta N^2 (K_e K_t + (B + J s) (R + L s))}{Bout (R + L s) + Jout s (R + L s) + \eta N^2 (K_e K_t + (B + J s) (R + L s))} \end{array} \right) \mathcal{T}$$

Save the models to a file for later use.

```

saveDefinitions[outputDirectory <> "MotorModels.m",
{diffEqns, motorPositionModel, motorVelocityModel,
motorAccelerationModel, motorCurrentModel, motorEMFModel, motorTorqueModel,
motorPositionOutModel, motorVelocityOutModel, motorAccelerationOutModel}]

```

6. Time Domain Functions

The time domain functions we seek model various states of the motor as a direct function of time. Generally speaking, we may obtain them from the frequency domain by (matrix) multiplying the transfer function by the Laplace transform of the input signal, then taking the inverse Laplace transform of the result. However, if we're starting with the (frequency domain) transfer function and the (time domain) input signal in-hand, we can compute the time

domain function more efficiently using the Laplace transform convolution theorem.

6.1. Time Domain Output Functions

6.1.1. makeTimeDomainFunctionConvolve[]

makeTimeDomainFunctionConvolve[] uses convolution and the theorem that the Laplace transform of a convolution is the product of the Laplace transform of the factors. Ref1: <http://mathfaculty.fullerton.edu/mathews/c2003/laplaceconvolutionmod.html>.

With the convolution approach, we can make explicit use of the fact that in the uses we have we know we want to take the inverse Laplace transform of a product, one of whose factors we already know the time domain inverse for, and thus we can avoid having LaplaceTransform[] do as much heavy lifting as we might ask it to do if we were to transform the input, multiply, then inverse-transform the whole result. This convolution approach also has been shown to be more efficient than the OutputResponse[] builtin function.

```
Clear[makeTimeDomainFunctionConvolve, convolve]
parameterAssumptions

convolve[fExpr_, gExpr_, exprVar_, t_] := Module[{τ(*=Unique["τ"]*)},
  Assuming[Union[{t ≥ 0}, parameterAssumptions], Integrate[(fExpr /. exprVar → τ) * (gExpr /. exprVar → t - τ), {τ, 0, t}]]
]

makeTimeDomainFunctionConvolve[model_, tInputFunctions_] := Module[
  {s(*=Unique["s"]*), τ(*=Unique["τ"]*), modelExpr},
  modelExpr = InverseLaplaceTransform[model[s], s, τ][[1]];
  Function[{t}, Module[{convolutions},
    convolutions = convolve[modelExpr[[#]], (tInputFunctions[[#]])[τ], τ, t] & /@ Range[1, Length[tInputFunctions]];
    {Total[convolutions]} (* put in list to mirror InverseLaplaceTransform[] output structure *)
  ]
]

{Bout ∈ ℝ, constvbat ∈ ℝ, constτauto ∈ ℝ, Jout ∈ ℝ, i[_] ∈ ℝ, vbat[_] ∈ ℝ, vg[_] ∈ ℝ, α[_] ∈ ℝ,
  θ[_] ∈ ℝ, τ[_] ∈ ℝ, τauto[_] ∈ ℝ, ω[_] ∈ ℝ, Ke > 0, Kt > 0, L > 0, R > 0, η > 0, N > 0, B ≥ 0, J ≥ 0, t ≥ 0}
```

6.1.2. makeMotorTimeDomainFunction[]

makeMotorTimeDomainFunction[] takes a transfer function model and returns a function that, when provided with ea and τa time-domain input functions, returns a function of time.

```
Clear[makeMotorTimeDomainFunction]
makeMotorTimeDomainFunction[model_] := makeMotorTimeDomainFunction[model, makeTimeDomainFunctionConvolve]
makeMotorTimeDomainFunction[model_, builder_] := makeMotorTimeDomainFunction[model, builder] = (*cache the result*)
Module[{t(*=Unique["t"]*)},
  Function[{vbat, τauto}, Module[{fn},
    fn = builder[model, {vbat[[#]] &, τauto[[#]] &}];
    fn
  ]]]
```

6.1.3. Time Domain Output Functions for Motor State

We define time domain functions for all the motor state of interest.

```
Clear[motorPosition, motorVelocity, motorAcceleration, motorCurrent, motorEMF, motorTorque]
```

```
motorPosition = makeMotorTimeDomainFunction[motorPositionModel];
motorPositionConstVbat = motorPosition[Vbat &, 0 &];
```

```
motorVelocity = makeMotorTimeDomainFunction[motorVelocityModel];
motorVelocityConstVbat = motorVelocity[Vbat &, 0 &];
```

```
motorAcceleration = makeMotorTimeDomainFunction[motorAccelerationModel];
motorAccelerationConstVbat = motorAcceleration[Vbat &, 0 &];
```



```
motorCurrent = makeMotorTimeDomainFunction[motorCurrentModel];
motorCurrentConstVbat = motorCurrent[Vbat &, 0 &];
```

```
motorEMF = makeMotorTimeDomainFunction[motorEMFModel];
motorEMFConstVbat = motorEMF[Vbat &, 0 &];
```

```
motorTorque = makeMotorTimeDomainFunction[motorTorqueModel];
motorTorqueConstVbat = motorTorque[Vbat &, 0 &];
```

The *output* too.

```
Clear[motorPositionOut, motorVelocityOut, motorAccelerationOut]
```

```
motorPositionOut = makeMotorTimeDomainFunction[motorPositionOutModel];
motorPositionOutConstVbat = motorPositionOut[Vbat &, 0 &];
```

```
motorVelocityOut = makeMotorTimeDomainFunction[motorVelocityOutModel];
motorVelocityOutConstVbat = motorVelocityOut[Vbat &, 0 &];
```

```
motorAccelerationOut = makeMotorTimeDomainFunction[motorAccelerationOutModel];
motorAccelerationOutConstVbat = motorAccelerationOut[Vbat &, 0 &];
```

Save these time domain functions so we can load them in later without having to develop them from scratch.

```
saveDefinitions[outputDirectory <> "MotorTimeDomainFunctions.m",
{
  motorPosition, motorVelocity, motorAcceleration, motorCurrent, motorEMF, motorTorque,
  motorPositionOut, motorVelocityOut, motorAccelerationOut,
  motorPositionConstVbat, motorVelocityConstVbat,
  motorAccelerationConstVbat, motorCurrentConstVbat, motorEMFConstVbat, motorTorqueConstVbat,
  motorPositionOutConstVbat, motorVelocityOutConstVbat, motorAccelerationOutConstVbat,
  makeMotorTimeDomainFunction
}]
```

6.2. Time Domain Input Functions

6.2.1. Keeping track of units

In analogy to `parameterQuantities` above, we define `inputQuantities`, which adds units to our input parameters.

```
parameterQuantities
```

$$\langle \left\{ \begin{array}{l} R \rightarrow R \, \Omega, L \rightarrow L \, H, i[t] \rightarrow i[t] \, A, i'[t] \rightarrow i'[t] \, A/s, i''[t] \rightarrow i''[t] \, A/s^2, vbat[t] \rightarrow vbat[t] \, V, constvbat \rightarrow constvbat \, V, \\ vg[t] \rightarrow vg[t] \, V, J \rightarrow J \, ms^2N/rad^2, Jout \rightarrow Jout \, ms^2N/rad^2, B \rightarrow B \, msN/rad^2, Bout \rightarrow Bout \, msN/rad^2, Ke \rightarrow Ke \, sV/rad, \\ KeShaft \rightarrow KeShaft \, sV/rad, Kt \rightarrow Kt \, mN/(Arad), KtShaft \rightarrow KtShaft \, mN/(Arad), \theta[t] \rightarrow \theta[t] \, rad, \thetaout[t] \rightarrow \thetaout[t] \, rad, \\ \theta'[t] \rightarrow \theta'[t] \, rad/s, \thetaout'[t] \rightarrow \thetaout'[t] \, rad, \theta''[t] \rightarrow \theta''[t] \, rad/s^2, \thetaout''[t] \rightarrow \thetaout''[t] \, rad, \omega[t] \rightarrow \omega[t] \, rad/s, \\ \alpha[t] \rightarrow \alpha[t] \, rad/s^2, \tau[t] \rightarrow \tau[t] \, mN/rad, \tauauto[t] \rightarrow \tauauto[t] \, mN/rad, const\tauauto \rightarrow const\tauauto \, mN/rad, N \rightarrow N, \eta \rightarrow \eta \end{array} \right\} \rangle$$

```
inputQuantities = {
  constvbat \rightarrow Quantity[vbat, "Volts"],
  t \rightarrow Quantity[t, "Seconds"]
}
```

```
{constvbat \rightarrow vbat \, V, t \rightarrow t \, s}
```

6.2.2. Typical Input Examples

In analogy to `aMotor` above, we now define some typical inputs. The first (and only, for now) is a constant 12V input with no externally applied torque.

```
twelveVoltInput = Association[t → Quantity[t, "Seconds"], constvbat → Quantity[12, "Volts"]]
```

$$\langle \{ t \rightarrow t \text{ s}, \text{constvbat} \rightarrow 12 \text{ V} \} \rangle$$


6.3. Examples

We try out our time-domain functions on our example data.


```
Clear[applyTimeFunction]
applyTimeFunction[motorTimeFunction_, motorWithLoad_, input_, t_] :=
  applyTimeFunction[motorTimeFunction, motorWithLoad, input, t] = Module[{generic, withUnits, unitless},
    generic = motorTimeFunction[constvbat &, consttau &][t];
    withUnits = generic /. motorWithLoad /. input // siUnits // FullSimplify;
    unitless = withUnits // clearUnits // N // FullSimplify;
    {withUnits, unitless}]
```

6.3.1. Examples without gears or external torque, using 'classic' load

```
{velUnits, vel} = applyTimeFunction[motorVelocity, aMotorClassic, twelveVoltInput, t];
velUnits // N
Plot[vel, {t, 0, 20}, AxesLabel → {"Seconds", HoldForm[Radians / Second]}, PlotRange → All, AxesOrigin → {0, 0}]
```


$$\left\{ 2.71828^{-0.377374 t \sqrt{\text{kg m} / (\sqrt{\text{kg m}})}} \left(-10.2734 \text{ rad/s} \right) + 2.71828^{-4754.7 t \sqrt{\text{kg m} / (\sqrt{\text{kg m}})}} \left(0.000815385 \text{ rad/s} \right) + 10.2726 \text{ rad/s} \right\}$$


```
{torUnits, tor} = applyTimeFunction[motorTorque, aMotorClassic, twelveVoltInput, t];
torUnits // N
Plot[tor, {t, 0, 20}, AxesLabel → {"Seconds", HoldForm[Netwon Meters / Radian]}, PlotRange → All, AxesOrigin → {0, 0}]
```

$$\left\{ 2.71828^{-4754.7 t \sqrt{\text{kg m} / (\sqrt{\text{kg m}})}} \left(-3.87693 \text{ kg m}^2 / (\text{s}^2 \text{ rad}) \right) + 0.338995 \text{ kg m}^2 / (\text{s}^2 \text{ rad}) + 2.71828^{-0.377374 t \sqrt{\text{kg m} / (\sqrt{\text{kg m}})}} \left(3.53793 \text{ kg m}^2 / (\text{s}^2 \text{ rad}) \right) \right\}$$



```
{curUnits, cur} = applyTimeFunction[motorCurrent, aMotorClassic, twelveVoltInput, t];
curUnits // N
Plot[cur, {t, 0, 20}, AxesLabel → {"Seconds", HoldForm[Amps]}, PlotRange → All, AxesOrigin → {0, 0}]
```

$$\left\{ 2.71828^{-4754.7 t \sqrt{\text{kg m}} / (\sqrt{\text{kg m}})} \left(-3.63689 \text{ A} \right) + 0.318007 \text{ A} + 2.71828^{-0.377374 t \sqrt{\text{kg m}} / (\sqrt{\text{kg m}})} \left(3.31888 \text{ A} \right) \right\}$$




```
{emfUnits, emf} = applyTimeFunction[motorEMF, aMotorClassic, twelveVoltInput, t];
emfUnits // N // simplifyUnits
Plot[{emf, constvbat /. twelveVoltInput}, {t, 0, 20},
  AxesLabel → {"Seconds", HoldForm[Volts]}, PlotRange → All, AxesOrigin → {0, 0}]
```

$$\left\{ 2.71828^{-0.377374 t \sqrt{\text{kg m}} / (\sqrt{\text{kg m}})} \left(-10.9514 \text{ V} \right) + 2.71828^{-4754.7 t \sqrt{\text{kg m}} / (\sqrt{\text{kg m}})} \left(0.000869201 \text{ V} \right) + 10.9506 \text{ V} \right\}$$



```
{posUnits, pos} = applyTimeFunction[motorPosition, aMotorClassic, twelveVoltInput, t];
posUnits // N
Plot[pos, {t, 0, 20}, AxesLabel → {"Seconds", HoldForm[Radians]}, PlotRange → All, AxesOrigin → {0, 0}]
```

$$\left\{ 2.71828^{-4754.7 t \sqrt{\text{kg m}} / (\sqrt{\text{kg m}})} \left(-1.7149 \times 10^{-7} \text{ rad} \right) + \right. \\ \left. 2.71828^{-0.377374 t \sqrt{\text{kg m}} / (\sqrt{\text{kg m}})} \left(27.2234 \text{ rad} \right) + 4.12469 \times 10^{-8} \left(-6.60011 \times 10^8 + 2.49051 \times 10^8 t \right) \text{ rad} \right\}$$




6.3.2. Examples with gear, using flywheel load

```
unitlessMotorWithFlywheel =
  addMotorLoad[motorParameters["AM 60 A"], flywheel[Quantity[3, "kg"], Quantity[8, "cm"]]] // siUnits // clearUnits // N
  twelveVoltInput // siUnits // clearUnits // N
{velUnits2, vel2} =
  applyTimeFunction[motorVelocity, unitlessMotorWithFlywheel, twelveVoltInput // siUnits // clearUnits // N, t];
Plot[vel2, {t, 0, .01}, AxesLabel -> {"Seconds", HoldForm[Radians / Second]}, PlotRange -> All, AxesOrigin -> {0, 0}]

<| R -> 3.3, L -> 0.000694, Ke -> 0.0177667, Kt -> 0.0177667,
  J -> 0.00001041, B -> 0.033, N -> 60.,  $\eta$  -> 0.9, Jout -> 0.0096, Bout -> 0., const $\tau$ auto -> 0. |>
```


```
<| t -> t, constvbat -> 12. |>
```



6.3.3. Examples without gear, using flywheel load

```
unitlessMotorWithFlywheelNoGear =
  addMotorLoad[motorParameters["AM 60 A", Geared -> False], flywheel[Quantity[3, "kg"], Quantity[8, "cm"]]] // siUnits //
  clearUnits // N
{velUnits3, vel3} = applyTimeFunction[motorVelocity, unitlessMotorWithFlywheelNoGear,
  twelveVoltInput // siUnits // clearUnits // N, t];
Plot[vel3, {t, 0, 0.25}, AxesLabel -> {"Seconds", HoldForm[Radians / Second]}, PlotRange -> All, AxesOrigin -> {0, 0}]

<| R -> 3.3, L -> 0.000694, Ke -> 1.066, Kt -> 1.066, J -> 0.00001041, B -> 0.033, N -> 1.,  $\eta$  -> 1., Jout -> 0.0096, Bout -> 0., const $\tau$ auto -> 0. |>
```



7. Steady State

7.1. Calculating Steady State Values : The Final Value Theorem

We explore the steady state behavior, the limit of the time domain functions as time goes to infinity.

One way to calculate said limit is to brute-force calculate the limit in question. While that works, and we can and have previously calculated same (Ref: <https://github.com/rgatkinson/RobotPhysics/blob/master/MotorPhysics.pdf>), it is very slow to do so.

A more efficient way to calculate the steady state behavior is to avail ourselves of the use of the Final Value Theorem (Ref: <https://en.wikipedia.org/wiki/>).

Final_value_theorem). The final value theorem tells us that if the time-domain limit exists, it is equal to the limit of the Laplace transform of the time domain function as $s \rightarrow 0$.

```
Clear[ssFinalValueTheorem]
ssFinalValueTheorem[model_, inputs_] := Module[{s(= Unique[]*), t(= Unique[]*), xformedInputs, expr},
  xformedInputs = LaplaceTransform[#, t, s] &@ inputs;
  expr = (model[s].xformedInputs)[[1]];
  (*Limit is really slow if we include assumptions (!?), so don't*)
  Block[{$Assumptions = {}}, Limit[s expr, s -> 0] // FullSimplify]
]
```

```
(ss = {
  ssPos -> ssFinalValueTheorem[motorPositionModel, {constvbat &, constrauto &}] // Factor,
  ssVel -> ssFinalValueTheorem[motorVelocityModel, {constvbat &, constrauto &}] // Factor,
  ssAcc -> ssFinalValueTheorem[motorAccelerationModel, {constvbat &, constrauto &}] // Factor,
  ssEmf -> ssFinalValueTheorem[motorEMFModel, {constvbat &, constrauto &}] // Factor,
  ssCur -> ssFinalValueTheorem[motorCurrentModel, {constvbat &, constrauto &}] // Factor,
  ssTor -> ssFinalValueTheorem[motorTorqueModel, {constvbat &, constrauto &}] // Factor
}) // prettyPrint
ss /. {N -> 1, η -> 1, constrauto -> 0} // FullSimplify // prettyPrint
```

```
ssPos -> Indeterminate
ssVel -> 
$$\frac{N(\text{constrauto } R + \text{constvbat } Kt \eta N)}{Ke Kt \eta N^2 + B R \eta N^2 + Bout R}$$

ssAcc -> 0
ssEmf -> 
$$\frac{Ke N(\text{constrauto } R + \text{constvbat } Kt \eta N)}{Ke Kt \eta N^2 + B R \eta N^2 + Bout R}$$

ssCur -> 
$$\frac{B \text{constvbat } \eta N^2 - \text{constrauto } Ke N + Bout \text{constvbat}}{Ke Kt \eta N^2 + B R \eta N^2 + Bout R}$$

ssTor -> 
$$\frac{Kt(B \text{constvbat } \eta N^2 - \text{constrauto } Ke N + Bout \text{constvbat})}{Ke Kt \eta N^2 + B R \eta N^2 + Bout R}$$

```

```
ssPos -> Indeterminate
ssVel -> 
$$\frac{\text{constvbat } Kt}{Ke Kt + (B + Bout) R}$$

ssAcc -> 0
ssEmf -> 
$$\frac{\text{constvbat } Ke Kt}{Ke Kt + (B + Bout) R}$$

ssCur -> 
$$\frac{(B + Bout) \text{constvbat}}{Ke Kt + (B + Bout) R}$$

ssTor -> 
$$\frac{(B + Bout) \text{constvbat } Kt}{Ke Kt + (B + Bout) R}$$

```

The “Indeterminate” result regarding position is Mathematica informing us that it has proven to itself that the limit does not exist.

```
saveDefinitions[outputDirectory <> "SteadyState.m", {ss}]
```

7.2. On the Applicability of the Final Value Theorem

Can we always use the Final Value Theorem?

No, the Final Value Theorem only applies under certain conditions. Specifically: that all non-zero poles of the transfer function must have negative real parts, and the transfer function must have at most one pole at the origin (Ref: <http://www-personal.umich.edu/~dsbaero/others/39-FVTrevisited.pdf>). The predicate exhibited in the ConditionalExpression of the brute-force result captures exactly these conditions. When using the Final Value Theorem on any concrete system, we will need to test the conditions in any particular application. Lets illustrate by looking at the velocity model.

```
motorVelocityModel
(velPoles = TransferFunctionPoles[motorVelocityModel]) // MatrixForm
```

$$\begin{pmatrix} \frac{v_{bat}}{Kt \eta N^2} & \frac{\tau_{auto}}{N (R + L s)} \\ \frac{Bout (R + L s) + Jout s (R + L s) + \eta N^2 (Ke Kt + (B + J s) (R + L s))}{Bout (R + L s) + Jout s (R + L s) + \eta N^2 (Ke Kt + (B + J s) (R + L s))} & \frac{Bout (R + L s) + Jout s (R + L s) + \eta N^2 (Ke Kt + (B + J s) (R + L s))}{Bout (R + L s) + Jout s (R + L s) + \eta N^2 (Ke Kt + (B + J s) (R + L s))} \end{pmatrix} \mathcal{T}$$

$$\begin{pmatrix} \frac{-Bout L - Jout R - B L \eta N^2 - J R \eta N^2 - \sqrt{-4 (Jout L + J L \eta N^2) (Bout R + Ke Kt \eta N^2 + B R \eta N^2) + (Bout L + Jout R + B L \eta N^2 + J R \eta N^2)^2}}{2 (Jout L + J L \eta N^2)} & \frac{-Bout L - Jout R - B L \eta N^2 - J R \eta N^2 - \sqrt{-4 (Jout L + J L \eta N^2) (Bout R + Ke Kt \eta N^2 + B R \eta N^2) + (Bout L + Jout R + B L \eta N^2 + J R \eta N^2)^2}}{2 (Jout L + J L \eta N^2)} \\ \frac{-Bout L - Jout R - B L \eta N^2 - J R \eta N^2 + \sqrt{-4 (Jout L + J L \eta N^2) (Bout R + Ke Kt \eta N^2 + B R \eta N^2) + (Bout L + Jout R + B L \eta N^2 + J R \eta N^2)^2}}{2 (Jout L + J L \eta N^2)} & \frac{-Bout L - Jout R - B L \eta N^2 - J R \eta N^2 + \sqrt{-4 (Jout L + J L \eta N^2) (Bout R + Ke Kt \eta N^2 + B R \eta N^2) + (Bout L + Jout R + B L \eta N^2 + J R \eta N^2)^2}}{2 (Jout L + J L \eta N^2)} \end{pmatrix}$$

The poles are the same for both applied-voltage and applied-torque input since the transfer function denominator is the same for both inputs, in both cases the zeros of:

```
den = motorVelocityModel[s][[1, 1]] // Denominator
```

$$Bout (R + L s) + Jout s (R + L s) + (Ke Kt + (B + J s) (R + L s)) \eta N^2$$

In order to get a feel for where these zeros are in practice, let's look at that denominator for an actual example motor. We compute where the poles occur for one of our examples

```
den /. aMotorClassic /. parameterQuantities // siUnits // clearUnits
```

```
Solve[% == 0, s] // N
```

$$\frac{284089}{250000} + \left(\frac{33}{1000} + \frac{1041 s}{100000000} \right) \left(\frac{33}{10} + \frac{347 s}{500000} \right) + \left(\frac{33}{10} + \frac{347 s}{500000} \right) s$$

$$\{ \{s \rightarrow -0.377374\}, \{s \rightarrow -4754.7\} \}$$

This, for this example, the Final Value Theorem applies. As another example, adding a ridiculous additional load moves the poles, but they remain in the LHS of the complex plane.

```
den /. addMotorLoad[aMotorClassic, flywheel[Quantity[1000, "kg"], Quantity[1, "m"]]] /. parameterQuantities // siUnits // clearUnits
```

```
Solve[% == 0, s] // N
```

$$\frac{284089}{250000} + \left(\frac{33}{1000} + \frac{1041 s}{100000000} \right) \left(\frac{33}{10} + \frac{347 s}{500000} \right) + 501 \left(\frac{33}{10} + \frac{347 s}{500000} \right) s$$

$$\{ \{s \rightarrow -0.000753194\}, \{s \rightarrow -4755.04\} \}$$

An analogous analytic development awaits a future revision to this document, as do examples that involve gears or externally applied torque.

8. More Administrivia

```
saveDefinitions[outputDirectory <> "Misc.m",
{importMotorData, motorParameters, motorLoad, flywheel, addMotorLoad, massOnPulley, applyTimeFunction}]
```

9. Revision History

- 2018.07.23 Added acceleration-dependant external torque. Corrected “viscous friction” to “viscous drag”. Clarified usage of units.
- 2018.08.06 Added gearbox support, removed time consuming internal tests. Regularized state variable names. Removed (flawed) “Back EMF vs Applied Voltage” discussion