

# Motor Physics

Robert Atkinson, Maddy Nguyen, Samantha Hordyk, Cole Welch, and John Fraser  
23 July 2018

An exploration of the physics of motors, in both symbolic and numeric form. A basic but sufficiently-complete parameterized model of a motor is presented. The differential equations governing that model are exhibited, and, from those, the (frequency domain) transfer functions of each of the model's variables is developed from first principles. Next, mathematical infrastructure for converting the response of general frequency domain functions input signals is developed, then applied to the transfer functions and their to applied-voltage and applied-torque inputs. This yields time domain functions for each of the model's variables in terms of the (symbolic) value of the model's variables. With those time domain symbolic functions in hand, the steady-state response of the motor as a function of each of the model's parameters is explored.


## Administrivia

We begin by loading some handy utilities.

```
Get[NotebookDirectory[] <> "Utilities.m"]
```

## Motor Model

The following gives a conceptual overview of a DC motor (ref: <http://www.egr.msu.edu/classes/me451/jchoi/2008/>).



In addition to the the above, we allow here for an externally applied torque.

With that considered, the state of the system can be described by the following variables.

- $e_a$ , the applied voltage
- $i_a$ , the armature current
- $e_b$ , the back EMF
- $\theta$ , the angular position
- $\tau$ , the torque driven by the armature current
- $\tau_a$ , the externally applied torque, if any

The parameter constants that characterize the motor are the following

- $J$ , Moment of inertia
- $b$ , Motor viscous drag constant
- $K_e$ , Electromotive force constant
- $K_t$ , Motor torque constant
- $R$ , Electric resistance

### ■ L Electric inductance

Further, the moment of inertia  $J$  and viscous drag constant  $b$  can be decomposed into additive contributions from both the motor and the load; we'll return to that later.

```
motorParameterNames = {J, b, Ke, Kt, R, L}
```

```
{J, b, Ke, Kt, R, L}
```

## Differential Equations

### Differential Equation Development

The system can be modelled as a set of differential equations

```
diffEqns = {}
```

```
{}
```

Drive torque is proportional to current.

```
diffEqns = Union[diffEqns, {τ[t] == Kt i[t]}]
```

```
{τ[t] == Kt i[t]}
```

The back EMF is proportional to velocity.

```
diffEqns = Union[diffEqns, {e[t] == Ke θ'[t]}]
```

```
{e[t] == Ke θ'[t], τ[t] == Kt i[t]}
```

Viscous drag reduces the torque, and what's left accelerates the inertial mass. There is an external torque which is linearly dependant angular acceleration as well as having an entirely autonomous term (wherein in many applications is embodied a constant external-torque due to gravity).

```
motorParameterNames = Union[motorParameterNames, {ατ}]
```

```
fnτa[t_] := ατ θ''[t] + τa[t]
```

```
diffEqns = Union[diffEqns, {τ[t] + fnτa[t] - b θ'[t] == J θ''[t]}]
```

```
{b, J, Ke, Kt, L, R, ατ}
```

```
{e[t] == Ke θ'[t], τ[t] == Kt i[t], τ[t] + τa[t] - b θ'[t] + ατ θ''[t] == J θ''[t]}
```

The voltage drop across the resistive, inductive, and back EMF must equal the applied voltage (due to one of Kirchhoff's laws).

```
diffEqns = Union[diffEqns, {ea[t] == R i[t] + L i'[t] + e[t]}]
```

```
{e[t] == Ke θ'[t], ea[t] == e[t] + R i[t] + L i'[t], τ[t] == Kt i[t], τ[t] + τa[t] - b θ'[t] + ατ θ''[t] == J θ''[t]}
```

```
diffEqns // prettyPrint
```

```
e(t) = Ke θ'(t)
```

```
ea(t) = e(t) + R i(t) + L i'(t)
```

```
τ(t) = Kt i(t)
```

```
τ(t) + τa(t) - b θ'(t) + ατ θ''(t) = J θ''(t)
```

### Initial Conditions

Some of the initial conditions we can intuit intellectually:

```
initialConditions = {
   $\theta[0] == 0,$ 
   $i[0] == 0,$ 
   $e[0] == 0,$ 
   $ea[0] == v,$ 
   $\tau a[0] == T$ 
};
```

Others we can solve for:

```
others = {i'[0],  $\theta'[0]$ ,  $\theta''[0]$ ,  $\tau[0]$ };
(eqnsSubsKnownInitial = diffEqns /. t -> 0 /. (initialConditions /. Equal -> Rule));
(solvedInitialConditions = (uniqueSolve[eqnsSubsKnownInitial, others] /. Rule -> Equal));
(allInitialConditions = solvedInitialConditions ~Join~ initialConditions) // prettyPrint
```

```
 $i'(0) = \frac{v}{L}$ 
 $\theta'(0) = 0$ 
 $\theta''(0) = \frac{T}{J - \alpha \tau}$ 
 $\tau(0) = 0$ 
 $\theta(0) = 0$ 
 $i(0) = 0$ 
 $e(0) = 0$ 
 $ea(0) = v$ 
 $\tau a(0) = T$ 
```

## Motor Parameters

We now take a moment to carefully work out the units of each parameter and variable. Some units are obvious and clear. Let's write those down.

```

siTorqueUnits = parseUnit["N m"] / parseUnit["Radians"];

parameterUnits = {
  R → parseUnit["Ohms"],
  L → parseUnit["Henrys"],
  θ[t] → parseUnit["Radians"],
  θ'[t] → parseUnit["Radians per Second"],
  Ω[t] → parseUnit["Radians per Second"],
  θ''[t] → parseUnit["Radians per Second per Second"],
  α[t] → parseUnit["Radians per Second per Second"],
  i[t] → parseUnit["Amperes"],
  i'[t] → parseUnit["Amperes"] / parseUnit["Second"],
  i''[t] → parseUnit["Amperes"] / parseUnit["Second"] / parseUnit["Second"],
  ea[t] → parseUnit["Volts"],
  e[t] → parseUnit["Volts"],
  τ[t] → siTorqueUnits,
  τa[t] → siTorqueUnits,
  ατ → siTorqueUnits / parseUnit["Radians"] * parseUnit["Seconds"]^2
} // Association

unitsToQuantities[units_] := Module[{rules, makeQuantity},
  rules = Normal[units];
  makeQuantity = Function[{param, unit}, Quantity[param, unit]];
  (#[1] → makeQuantity @@ # & /@ rules) // Association
]
parameterQuantities = unitsToQuantities[parameterUnits]

```

$$\left\langle \begin{array}{l} R \rightarrow \text{Ohms}, L \rightarrow \text{Henrys}, \theta[t] \rightarrow \text{Radians}, \theta'[t] \rightarrow \frac{\text{Radians}}{\text{Seconds}}, \Omega[t] \rightarrow \frac{\text{Radians}}{\text{Seconds}}, \\ \theta''[t] \rightarrow \frac{\text{Radians}}{\text{Seconds}^2}, \alpha[t] \rightarrow \frac{\text{Radians}}{\text{Seconds}^2}, i[t] \rightarrow \text{Amperes}, i'[t] \rightarrow \frac{\text{Amperes}}{\text{Seconds}}, i''[t] \rightarrow \frac{\text{Amperes}}{\text{Seconds}^2}, ea[t] \rightarrow \text{Volts}, \\ e[t] \rightarrow \text{Volts}, \tau[t] \rightarrow \frac{\text{Meters Newtons}}{\text{Radians}}, \tau a[t] \rightarrow \frac{\text{Meters Newtons}}{\text{Radians}}, \alpha\tau \rightarrow \frac{\text{Meters Newtons Seconds}^2}{\text{Radians}^2} \end{array} \right\rangle$$

$$\left\langle \begin{array}{l} R \rightarrow R \Omega, L \rightarrow L H, \theta[t] \rightarrow \theta[t] \text{ rad}, \theta'[t] \rightarrow \theta'[t] \text{ rad/s}, \Omega[t] \rightarrow \Omega[t] \text{ rad/s}, \\ \theta''[t] \rightarrow \theta''[t] \text{ rad/s}^2, \alpha[t] \rightarrow \alpha[t] \text{ rad/s}^2, i[t] \rightarrow i[t] \text{ A}, i'[t] \rightarrow i'[t] \text{ A/s}, i''[t] \rightarrow i''[t] \text{ A/s}^2, \\ ea[t] \rightarrow ea[t] \text{ V}, e[t] \rightarrow e[t] \text{ V}, \tau[t] \rightarrow \tau[t] \text{ mN/rad}, \tau a[t] \rightarrow \tau a[t] \text{ mN/rad}, \alpha\tau \rightarrow \alpha\tau \text{ ms}^2\text{N/rad}^2 \end{array} \right\rangle$$

A note on units. The Bureau International des Poids et Mesures, the intergovernmental organization that standardizes the SI, officially refers to a radian as “a special name is given to the unit one, in order to facilitate the identification of the quantity involved” (Ref: <https://www.bipm.org/en/publication-s/si-brochure/section2-2-3.html>). With that as background, we have here chosen to include radians in all places in which they logically belong, as doing so provides valuable insight regarding the quantities being manipulated. Thus, for example, you will find here torque having units of N m / radian instead of N m, and the like. As, from the SI perspective, this amounts to multiplying or dividing by the unit one, no semantic change is effected.

Aside: that “a special name given to the unit one” is in any way at all different than a first-class “unit” is a perspective reasonable only if one artificially and myopically limits one’s worldview to contain only the seven units whose dimension is necessarily *experimentally* calibrated as opposed to units whose dimension is mathematically or otherwise calibrated, such as radian, becquerel, motor encoder ticks, etc. And the world is chock-full of the latter. But this is a discussion for another time. See also Jacques E. Romain, “Angle as a Fourth Fundamental Quantity”, Journal of Research of the National Bureau of Standards-B. Mathematics and Mathematical Physics, Vol. 66B, No. 3, July- September 1962.

Let’s look at how far our tabulation of parameter units takes us:

```
diffEqns /. parameterQuantities // ColumnForm
```

$$\begin{aligned} e[t] \text{ V} &= K e \left( \theta'[t] \text{ rad/s} \right) \\ ea[t] \text{ V} &= (e[t] + R i[t] + L i'[t]) A \Omega \\ \tau[t] \text{ mN/rad} &= K t \left( i[t] \text{ A} \right) \\ b \left( -\theta'[t] \text{ rad/s} \right) + (\tau[t] + \tau a[t] + \alpha\tau \theta''[t]) \text{ mN/rad} &= J \left( \theta''[t] \text{ rad/s}^2 \right) \end{aligned}$$

We can see that if we write down the units for either b or J, the remaining units will be fully determined. We choose to specify J.

```
siAngularInertialUnits = parseUnit["kg m^2"] / parseUnit["Radians^2"];
```

```
parameterUnits[J] = siAngularInertialUnits;
parameterQuantities = unitsToQuantities[parameterUnits]
diffEqns /. parameterQuantities // ColumnForm
remainingUnits = uniqueSolve[diffEqns[{{1, 3, 4}}], {Ke, Kt, b}]
```

$$\left\{ \begin{array}{l} R \rightarrow R \Omega, L \rightarrow L H, \theta[t] \rightarrow \theta[t] \text{ rad}, \theta'[t] \rightarrow \theta'[t] \text{ rad/s}, \Omega[t] \rightarrow \Omega[t] \text{ rad/s}, \theta''[t] \rightarrow \theta''[t] \text{ rad/s}^2, \\ \alpha[t] \rightarrow \alpha[t] \text{ rad/s}^2, i[t] \rightarrow i[t] \text{ A}, i'[t] \rightarrow i'[t] \text{ A/s}, i''[t] \rightarrow i''[t] \text{ A/s}^2, ea[t] \rightarrow ea[t] \text{ V}, \\ e[t] \rightarrow e[t] \text{ V}, \tau[t] \rightarrow \tau[t] \text{ mN/rad}, \tau a[t] \rightarrow \tau a[t] \text{ mN/rad}, \alpha \tau \rightarrow \alpha \tau \text{ ms}^2 \text{ N/rad}^2, J \rightarrow J \text{ kgm}^2 \text{ /rad}^2 \end{array} \right\}$$

```
e[t] V == Ke ( \theta'[t] rad/s )
ea[t] V == ( e[t] + R i[t] + L i'[t] ) A \Omega
\tau[t] mN/rad == Kt ( i[t] A )
b ( -\theta'[t] rad/s ) + ( \tau[t] + \tau a[t] + \alpha \tau \theta''[t] ) mN/rad == J \theta''[t] kgm^2 / (s^2 rad)
```

$$\left\{ Ke \rightarrow \frac{e[t]}{\theta'[t]}, Kt \rightarrow \frac{\tau[t]}{i[t]}, b \rightarrow \frac{\tau[t] + \tau a[t] - J \theta''[t] + \alpha \tau \theta''[t]}{\theta'[t]} \right\}$$

Let's check how those work.

```
remainingUnits /. parameterQuantities
```

$$\left\{ Ke \rightarrow \frac{e[t]}{\theta'[t]} \text{ sV/rad}, Kt \rightarrow \frac{\tau[t]}{i[t]} \text{ mN/(A rad)}, b \rightarrow \frac{\tau[t] + \tau a[t] - J \theta''[t] + \alpha \tau \theta''[t]}{\theta'[t]} \text{ msN/rad}^2 \right\}$$

Huzzah! They check out (if the units were incompatible, Mathematica would've told us). Let's the remaining units to our kit.

```
(parameterUnits[#] = QuantityUnit[# /. (remainingUnits /. parameterQuantities)]) & /@ remainingUnits[{{All, 1}}]
parameterQuantities = unitsToQuantities[parameterUnits]
siAngularViscousDragUnits = parseUnit["N m s"] / parseUnit["Radians^2"];
```

$$\left\{ \frac{\text{Seconds Volts}}{\text{Radians}}, \frac{\text{Meters Newtons}}{\text{Amperes Radians}}, \frac{\text{Meters Newtons Seconds}}{\text{Radians}^2} \right\}$$

$$\left\{ \begin{array}{l} R \rightarrow R \Omega, L \rightarrow L H, \theta[t] \rightarrow \theta[t] \text{ rad}, \theta'[t] \rightarrow \theta'[t] \text{ rad/s}, \Omega[t] \rightarrow \Omega[t] \text{ rad/s}, \theta''[t] \rightarrow \theta''[t] \text{ rad/s}^2, \alpha[t] \rightarrow \alpha[t] \text{ rad/s}^2, \\ i[t] \rightarrow i[t] \text{ A}, i'[t] \rightarrow i'[t] \text{ A/s}, i''[t] \rightarrow i''[t] \text{ A/s}^2, ea[t] \rightarrow ea[t] \text{ V}, e[t] \rightarrow e[t] \text{ V}, \tau[t] \rightarrow \tau[t] \text{ mN/rad}, \\ \tau a[t] \rightarrow \tau a[t] \text{ mN/rad}, \alpha \tau \rightarrow \alpha \tau \text{ ms}^2 \text{ N/rad}^2, J \rightarrow J \text{ kgm}^2 \text{ /rad}^2, Ke \rightarrow Ke \text{ sV/rad}, Kt \rightarrow Kt \text{ mN/(A rad)}, b \rightarrow b \text{ msN/rad}^2 \end{array} \right\}$$

## Motor Parameters Redux

### On Ke & Kt

It is often remarked that, in compatible units, Ke & Kt have the same magnitude and sign. However, while often true, this relation does not always hold. Recall our differential equations:

```
diffEqns // ColumnForm
```

```
e[t] == Ke \theta'[t]
ea[t] == e[t] + R i[t] + L i'[t]
\tau[t] == Kt i[t]
\tau[t] + \tau a[t] - b \theta'[t] + \alpha \tau \theta''[t] == J \theta''[t]
```

In steady state, the current and speed are constant.

```
(ssEqns = diffEqns /. {i'[t] → 0, θ'[t] → 0}) // ColumnForm
(ssEqns = Eliminate[ssEqns, {e[t], b}]) // ColumnForm
```

```
e[t] == Ke θ'[t]
ea[t] == e[t] + R i[t]
τ[t] == Kt i[t]
τ[t] + τa[t] - b θ'[t] == 0
```

```
ea[t] == R i[t] + Ke θ'[t]
τ[t] == Kt i[t]
```

From a power point of view (ref: <https://bit.ly/2Lxka0Z>), we must have (electrical) power in = (mechanical) power out + (electrical + other) power losses:

```
powerEqns = And @@ {
  pwrIn == ea[t] i[t],
  pwrOut == τ[t] θ'[t],
  pwrLosses == i[t]^2 R + pwrOtherPowerLosses,
  pwrIn == pwrOut + pwrLosses
};
(ssPowerEqns = ssEqns ~Join~ powerEqns) // prettyPrint
```

```
ea(t) = R i(t) + Ke θ'(t)
τ(t) = Kt i(t)
pwrIn = ea(t) i(t)
pwrOut = τ(t) θ'(t)
pwrLosses = R i(t)^2 + pwrOtherPowerLosses
pwrIn = pwrLosses + pwrOut
```

Let's eliminate a few of the variables that we're not interested in.

```
kBalanceEqns = Eliminate[ssPowerEqns, {i[t], τ[t], θ'[t], ea[t]}];
kBalanceEqns // prettyPrint
uniqueSolve[kBalanceEqns[[2]], pwrOtherPowerLosses] /. Rule → Equal
```

```
pwrIn = pwrLosses + pwrOut
Kt pwrOtherPowerLosses == (Ke - Kt) pwrOut
```

```
{pwrOtherPowerLosses == (Ke - Kt) pwrOut / Kt}
```

We can conclude that if other power losses are zero, then Ke must equal Kt.

## Experimental Motor Data

We have experimental data from several motors (Nguyen, Hordyk, & Fraser, 2017). We load in same.

```
(motorData = Import[NotebookDirectory[] <> "Characterized Motors.xlsx", {"Data", 1}]) // TableForm
```

Name	J	b	K	R	L (uH)	L (H)	J
AM 20 A	9.0110-6	0.0022	0.351	2.3	691.	0.000691	9.011 × 10 <sup>-6</sup>
AM 20 B	9.0110-6	0.0025	0.389	1.9	684.	0.000684	9.011 × 10 <sup>-6</sup>
AM 20 C	8.9310-6	0.0028	0.385	5.1	717.	0.000717	8.931 × 10 <sup>-6</sup>
AM 40 A	2.2210-5	0.2269	0.753	2.5	674.	0.000674	0.0002221
AM 40 B	1.7410-5	0.56	0.705	3.8	705.	0.000705	0.0001741
AM 40 C	2.4710-5	0.018	0.763	2.1	716.	0.000716	0.0002471
AM 60 A	1.0410-5	0.033	1.066	3.3	694.	0.000694	0.0001041
AM 60 B	8.4210-6	0.02	1.076	5.1	696.	0.000696	8.421 × 10 <sup>-6</sup>
AM 3.7 A	2.7910-5	0.00014	0.099	8.9	679.	0.000679	0.0002791
AM 3.7 B	3.1510-5	0.000176	0.108	2.6	797.	0.000797	0.0003151
AM 3.7 C	3.0910-5	0.00017	0.105	8.7	880.	0.00088	0.0003091
Matrix A	9.4310-6	0.00151	0.34	3.8	718.	0.000718	9.431 × 10 <sup>-6</sup>
Matrix B	7.7610-6	0.00191	0.363	7.8	777.	0.000777	7.761 × 10 <sup>-6</sup>
Matrix C	7.2310-6	0.00186	0.338	20.6	658.	0.000658	7.231 × 10 <sup>-6</sup>
CoreHex A	7.3310-4	0.0112	0.822	3.6	1356.	0.001356	0.0007331
CoreHex B	6.5510-4	0.008	0.858	11.3	1352.	0.001352	0.0006551
CoreHex C	4.5410-4	0.0078	0.711	5.6	1342.	0.001342	0.0004541

We build a function to retrieve data from the table. We convert from floating point to rational in order to help delay floating point collapse later on down the line. We allow for optional load inertia and viscous drag.

```
siAngularInertialUnits = parseUnit["kg m^2"] / parseUnit["Radians^2"];
siAngularViscousDragUnits = parseUnit["N m s"] / parseUnit["Radians^2"];
siTorqueUnits = parseUnit["N m"] / parseUnit["Radians"];

Clear[motorParameters]
Options[motorParameters] = {
  JLoad → Quantity[0, siAngularInertialUnits],
  bLoad → Quantity[0, siAngularViscousDragUnits]
};
motorParameters[motorName_, opts : OptionsPattern[]] := Module[{row, paramValues, quantify, assoc},
  row = Select[motorData, #[[1]] == motorName &][[1]];
  paramValues = #[[1]] → toRational[row[[#][2]]] & /@ {{J, 8}, {b, 3}, {Ke, 4}, {Kt, 4}, {R, 5}, {L, 7}};
  quantify = Function[{name, value},
    name → ((name /. parameterQuantities) /. name → value)
  ];
  assoc = quantify @@ # & /@ paramValues // Association;
  assoc[J] = assoc[J] + OptionValue[JLoad];
  assoc[b] = assoc[b] + OptionValue[bLoad];
  assoc
];

motorParameters["AM 60 A", JLoad → Quantity[1, siAngularInertialUnits]]
```

$$\left\langle J \rightarrow \frac{100001041}{100000000} \text{ kg m}^2/\text{rad}^2, b \rightarrow \frac{33}{1000} \text{ m s N/rad}^2, Ke \rightarrow \frac{533}{500} \text{ s V/rad}, Kt \rightarrow \frac{533}{500} \text{ m N/(A rad)}, R \rightarrow \frac{33}{10} \Omega, L \rightarrow \frac{347}{500000} \text{ H} \right\rangle$$

## Working Examples

### Example motors

We define a motor with a load that we'll use to illustrate examples. We also define a generic, abstract motor that can help explore things symbolically

```
siAngularInertialUnits
aMotor = motorParameters["AM 60 A", JLoad → Quantity[1, siAngularInertialUnits]]
```

$$\frac{\text{Kilograms Meters}^2}{\text{Radians}^2}$$

$$\left\langle J \rightarrow \frac{100001041}{100000000} \text{ kg m}^2/\text{rad}^2, b \rightarrow \frac{33}{1000} \text{ m s N/rad}^2, Ke \rightarrow \frac{533}{500} \text{ s V/rad}, Kt \rightarrow \frac{533}{500} \text{ m N/(A rad)}, R \rightarrow \frac{33}{10} \Omega, L \rightarrow \frac{347}{500000} \text{ H} \right\rangle$$

```
genericMotor = # → (# /. parameterQuantities) & /@ motorParameterNames
```

$$\left\{ b \rightarrow b \text{ m s N/rad}^2, J \rightarrow J \text{ kg m}^2/\text{rad}^2, Ke \rightarrow Ke \text{ s V/rad}, Kt \rightarrow Kt \text{ m N/(A rad)}, L \rightarrow L \text{ H}, R \rightarrow R \Omega, \alpha \tau \rightarrow \alpha \tau \text{ m s}^2 \text{ N/rad}^2 \right\}$$

## Laplace Transforms and Transfer Function Models

### Laplace Transforms

Returning to the differential equations, we form the Laplace Transform of each, then solve the for the various transforms. First, we apply the Laplace-Transform[] function to each equation. It automatically makes use of the linearity of the transform, and insinuates itself just around the time dependent parts (ie: the parts dependent on the time variable we told it about, namely  $t$ ).

```
leqns = LaplaceTransform[#, t, s] &/@ diffEqns
leqns // prettyPrint
```

```
{LaplaceTransform[e[t], t, s] == Ke (s LaplaceTransform[theta[t], t, s] - theta[0]), LaplaceTransform[ea[t], t, s] ==
  LaplaceTransform[e[t], t, s] + R LaplaceTransform[i[t], t, s] + L (-i[0] + s LaplaceTransform[i[t], t, s]),
  LaplaceTransform[tau[t], t, s] == Kt LaplaceTransform[i[t], t, s],
  LaplaceTransform[tau[t], t, s] + LaplaceTransform[taua[t], t, s] - b (s LaplaceTransform[theta[t], t, s] - theta[0]) +
  alpha (s^2 LaplaceTransform[theta[t], t, s] - s theta[0] - theta'[0]) == J (s^2 LaplaceTransform[theta[t], t, s] - s theta[0] - theta'[0])}
```

```
Li[theta(t)](s) = Ke (s (Li[theta(t)](s)) - theta(0))
Li[ea(t)](s) = Li[e(t)](s) + R (Li[i(t)](s)) + L (s (Li[i(t)](s)) - i(0))
Li[tau(t)](s) = Kt (Li[i(t)](s))
Li[tau(t)](s) + Li[ta(t)](s) - b (s (Li[theta(t)](s)) - theta(0)) + alpha ((Li[theta(t)](s)) s^2 - theta(0) s - theta'(0)) = J ((Li[theta(t)](s)) s^2 - theta(0) s - theta'(0))
```

Next, we walk those equations, picking up those insinuations and sowing them to the wind, reaping them on the outside, and finally removing duplicates

```
allXforms = Reap[Scan[If[MatchQ[#, _LaplaceTransform], Sow[#]] &, leqns, Infinity]][[2, 1]] // Union
```

```
{LaplaceTransform[e[t], t, s], LaplaceTransform[ea[t], t, s], LaplaceTransform[i[t], t, s],
  LaplaceTransform[theta[t], t, s], LaplaceTransform[tau[t], t, s], LaplaceTransform[taua[t], t, s]}
```

The voltage transform is input; all the others are outputs. Solve for the outputs. Finally substitute what we know about initial conditions, and simplify as much as we can.

```
voltageXform = LaplaceTransform[ea[t], t, s];
tinXform = LaplaceTransform[taua[t], t, s]
inputXforms = {voltageXform, tinXform};
outputXforms = Complement[allXforms, inputXforms]
rawSolvedXforms = uniqueSolve[leqns, outputXforms]
rawSolvedXforms = rawSolvedXforms /. (allInitialConditions /. Equal -> Rule) // FullSimplify
```

```
LaplaceTransform[taua[t], t, s]
```

```
{LaplaceTransform[e[t], t, s], LaplaceTransform[i[t], t, s], LaplaceTransform[theta[t], t, s], LaplaceTransform[tau[t], t, s]}
```

```
{LaplaceTransform[e[t], t, s] ->
  -(((-Ke Kt Li[0] - Ke Kt LaplaceTransform[ea[t], t, s] - Ke R LaplaceTransform[taua[t], t, s] - Ke L s LaplaceTransform[taua[t], t, s] -
    J Ke R theta'[0] - J Ke L s theta'[0] + Ke R alpha theta'[0] + Ke L s alpha theta'[0]) / (Ke Kt + b R + b L s + J R s + J L s^2 - R s alpha - L s^2 alpha)),
  LaplaceTransform[i[t], t, s] -> -((-b Li[0] - J L s i[0] + L s alpha i[0] - b LaplaceTransform[ea[t], t, s] - J s
    LaplaceTransform[ea[t], t, s] + s alpha LaplaceTransform[ea[t], t, s] + Ke LaplaceTransform[taua[t], t, s] + J Ke theta'[0] - Ke alpha theta'[0]) /
    (Ke Kt + b R + b L s + J R s + J L s^2 - R s alpha - L s^2 alpha)), LaplaceTransform[theta[t], t, s] ->
  -((-Kt Li[0] - Kt LaplaceTransform[ea[t], t, s] - R LaplaceTransform[taua[t], t, s] - L s LaplaceTransform[taua[t], t, s] -
    Ke Kt theta[0] - b R theta[0] - b L s theta[0] - J R s theta[0] - J L s^2 theta[0] + R s alpha theta[0] + L s^2 alpha theta[0] - J R theta'[0] - J L s theta'[0] + R alpha theta'[0] + L s alpha theta'[0]) /
    (s (Ke Kt + b R + b L s + J R s + J L s^2 - R s alpha - L s^2 alpha))), LaplaceTransform[tau[t], t, s] ->
  -((-b Kt Li[0] - J Kt L s i[0] + Kt L s alpha i[0] - b Kt LaplaceTransform[ea[t], t, s] - J Kt s LaplaceTransform[ea[t], t, s] +
    Kt s alpha LaplaceTransform[ea[t], t, s] + Ke Kt LaplaceTransform[taua[t], t, s] +
    J Ke Kt theta'[0] - Ke Kt alpha theta'[0]) / (Ke Kt + b R + b L s + J R s + J L s^2 - R s alpha - L s^2 alpha))}
```

```
{LaplaceTransform[e[t], t, s] -> (Ke (Kt LaplaceTransform[ea[t], t, s] + (R + L s) LaplaceTransform[taua[t], t, s])) /
  (Ke Kt + (R + L s) (b + s (J - alpha))), LaplaceTransform[i[t], t, s] ->
  ((b + s (J - alpha)) LaplaceTransform[ea[t], t, s] - Ke LaplaceTransform[taua[t], t, s]) / (Ke Kt + (R + L s) (b + s (J - alpha))),
  LaplaceTransform[theta[t], t, s] -> (Kt LaplaceTransform[ea[t], t, s] + (R + L s) LaplaceTransform[taua[t], t, s]) /
  (s (Ke Kt + (R + L s) (b + s (J - alpha)))), LaplaceTransform[tau[t], t, s] ->
  (Kt ((b + s (J - alpha)) LaplaceTransform[ea[t], t, s] - Ke LaplaceTransform[taua[t], t, s])) / (Ke Kt + (R + L s) (b + s (J - alpha)))}
```

We separate the transforms into applied-voltage- and applied-torque-dependent parts.



```

Clear[separate, decompose, inputLabels]
separate[xform_] := Module[{xformVarRules, invertedRules, xformVars, vard},
  xformVarRules = # -> Unique["laplaceTransform"] & /@ inputXforms;
  invertedRules = Reverse /@ xformVarRules;
  xformVars = xformVarRules[All, 2];
  vard = xform /. xformVarRules;
  Collect[vard, xformVars] /. invertedRules
]
decompose[sum_] := {sum[[1]] / voltageXform, sum[[2]] / tauXform}
inputLabels = {"ea", "ta"}

solvedXforms = #[[1]] -> separate[#[[2]]] & /@ rawSolvedXforms;
% // ColumnForm

```

```
{ea, ta}
```

$$\begin{aligned}
 \text{LaplaceTransform}[e[t], t, s] &\rightarrow \frac{K e K t \text{LaplaceTransform}[ea[t], t, s]}{K e K t + (R + L s) (b + s (J - \alpha \tau))} + \frac{K e (R + L s) \text{LaplaceTransform}[ta[t], t, s]}{K e K t + (R + L s) (b + s (J - \alpha \tau))} \\
 \text{LaplaceTransform}[i[t], t, s] &\rightarrow \frac{(b + s (J - \alpha \tau)) \text{LaplaceTransform}[ea[t], t, s]}{K e K t + (R + L s) (b + s (J - \alpha \tau))} - \frac{K e \text{LaplaceTransform}[ta[t], t, s]}{K e K t + (R + L s) (b + s (J - \alpha \tau))} \\
 \text{LaplaceTransform}[\theta[t], t, s] &\rightarrow \frac{K t \text{LaplaceTransform}[ea[t], t, s]}{s (K e K t + (R + L s) (b + s (J - \alpha \tau)))} + \frac{(R + L s) \text{LaplaceTransform}[ta[t], t, s]}{s (K e K t + (R + L s) (b + s (J - \alpha \tau)))} \\
 \text{LaplaceTransform}[\tau[t], t, s] &\rightarrow \frac{K t (b + s (J - \alpha \tau)) \text{LaplaceTransform}[ea[t], t, s]}{K e K t + (R + L s) (b + s (J - \alpha \tau))} - \frac{K e K t \text{LaplaceTransform}[ta[t], t, s]}{K e K t + (R + L s) (b + s (J - \alpha \tau))}
 \end{aligned}$$

## Transfer Function Models

We use the solved transforms to create TransferFunctionModel[]s for every variable of interest.

```

Clear[makeModel]
makeModel[var_] := makeModel[var, ToString[var], 1]
makeModel[var_, outputLabel_, factor_] := TransferFunctionModel[
  factor * Function[{sum}, decompose[sum]]
  [LaplaceTransform[var[t], t, s] /. solvedXforms],
  s,
  SystemsModelLabels -> {inputLabels, outputLabel}]

```

```

motorPositionModel = makeModel[θ]
motorVelocityModel = makeModel[θ, "α", s]
motorAccelerationModel = makeModel[θ, "α", s * s]
motorCurrentModel = makeModel[i]
motorEMFModel = makeModel[e]
motorTorqueModel = makeModel[τ]

```

$$\left( \begin{array}{c|c} \frac{ea}{Kt} & \frac{\tau a}{R+Ls} \\ \hline \frac{s(KeKt + (R+Ls)(b + (J-\alpha\tau)s))}{Kt} & \frac{s(KeKt + (R+Ls)(b + (J-\alpha\tau)s))}{R+Ls} \end{array} \right) \mathcal{T}$$

$$\left( \begin{array}{c|c} \frac{ea}{Kt} & \frac{\tau a}{R+Ls} \\ \hline \frac{KeKt + (R+Ls)(b + (J-\alpha\tau)s)}{Kt} & \frac{KeKt + (R+Ls)(b + (J-\alpha\tau)s)}{R+Ls} \end{array} \right) \mathcal{T}$$

$$\left( \begin{array}{c|c} \frac{ea}{Kts} & \frac{\tau a}{s(R+Ls)} \\ \hline \frac{KeKt + (R+Ls)(b + (J-\alpha\tau)s)}{Kts} & \frac{KeKt + (R+Ls)(b + (J-\alpha\tau)s)}{s(R+Ls)} \end{array} \right) \mathcal{T}$$

$$\left( \begin{array}{c|c} \frac{ea}{b + (J-\alpha\tau)s} & \frac{\tau a}{Ke} \\ \hline \frac{KeKt + (R+Ls)(b + (J-\alpha\tau)s)}{b + (J-\alpha\tau)s} & \frac{KeKt + (R+Ls)(b + (J-\alpha\tau)s)}{Ke} \end{array} \right) \mathcal{T}$$

$$\left( \begin{array}{c|c} \frac{ea}{KeKt} & \frac{\tau a}{Ke(R+Ls)} \\ \hline \frac{KeKt + (R+Ls)(b + (J-\alpha\tau)s)}{KeKt} & \frac{KeKt + (R+Ls)(b + (J-\alpha\tau)s)}{Ke(R+Ls)} \end{array} \right) \mathcal{T}$$

$$\left( \begin{array}{c|c} \frac{ea}{Kt(b + (J-\alpha\tau)s)} & \frac{\tau a}{KeKt} \\ \hline \frac{KeKt + (R+Ls)(b + (J-\alpha\tau)s)}{Kt(b + (J-\alpha\tau)s)} & \frac{KeKt + (R+Ls)(b + (J-\alpha\tau)s)}{KeKt} \end{array} \right) \mathcal{T}$$

## Time Domain Functions

The time domain functions we seek model various states of the motor as a direct function of time. Generally speaking, we obtain them from the frequency domain by (matrix) multiplying the transfer function by the Laplace transform of the input signal, then taking the inverse Laplace transform of the result. However, if we're starting with the (frequency domain) transfer function and the (time domain) input signal in-hand, we can compute the time domain function more efficiently using the Laplace transform convolution theorem.

### Time Domain Output Functions

#### makeTimeDomainFunctionLaplace[]

makeTimeDomainFunctionLaplace[] computes the time domain function using the approach of obtaining everything in frequency domain then converting to time domain as a final step.

```

Clear[transformInputFunctions, makeTimeDomainFunctionLaplace]
transformInputFunctions[tInputFunctions_List, t_, s_] := LaplaceTransform[#, [t], t, s] & /@ tInputFunctions
makeTimeDomainFunctionLaplace[model_, tInputFunctions_] := Module[
  {s = Unique["s"], t = Unique["t"], sInputs, sOutput, tOutput},
  sInputs = transformInputFunctions[tInputFunctions, t, s];
  sOutput = model[s] . sInputs;
  tOutput = InverseLaplaceTransform[sOutput, s, t];
  tOutput /. t -> # &
]

```

#### makeTimeDomainFunctionConvolve[]

makeTimeDomainFunctionConvolve[] uses convolution and the theorem that the Laplace transform of a convolution is the product of the Laplace transform of the factors. Ref1: <http://mathfaculty.fullerton.edu/mathews/c2003/laplaceconvolutionmod.html>.

With the convolution approach, we can make explicit use of the fact that in the uses we have we know we want to take the inverse Laplace transform of a product, one of whose factors we already know the time domain inverse for, and thus we can avoid having LaplaceTransform[] do as much heavy lifting as we ask it to do in the other approach.

```
Clear[makeTimeDomainFunctionConvolve, convolve]
convolve[fExpr_, gExpr_, exprVar_, t_] := Module[{τ = Unique["τ"]},
  Assuming[t ≥ 0, Integrate[(fExpr /. exprVar → τ) * (gExpr /. exprVar → t - τ), {τ, 0, t}]]
]

makeTimeDomainFunctionConvolve[model_, tInputFunctions_] := Module[
  {s = Unique["s"], τ = Unique["τ"], modelExpr},
  modelExpr = InverseLaplaceTransform[model[s], s, τ][[1]];
  Function[{t}, Module[{convolutions},
    convolutions = convolve[modelExpr[[#]], (tInputFunctions[[#]])[τ], τ, t] &@ Range[1, Length[tInputFunctions]];
    {Total[convolutions]} (* put in list to mirror InverseLaplaceTransform[] output structure *)
  ]
]
]
```

### Small Tests

A few tests comparing both approaches.

```
testModel = TransferFunctionModel[
  {LaplaceTransform[1, t, s], LaplaceTransform[(Sin[t] + 7 Exp[-t]), t, s]}, s, SystemsModelLabels → {inputLabels, "out"}]

{{makeTimeDomainFunctionConvolve @@ #}[t], (makeTimeDomainFunctionLaplace @@ #)[t]} &@ {
  {TransferFunctionModel[1 / (s^2 + 1), s], {2 Cos[#] &} (*Example 12.29 in Ref1*)},
  {testModel, {v &, T &}}
} // ColumnForm
```

$$\left( \begin{array}{c|cc} & ea & \tau a \\ \hline out & \frac{1}{s} & \frac{8 + s + 7 s^2}{s (1 + s) (1 + s^2)} \end{array} \right) \mathcal{T}$$

```
{{t Sin[t]}, {t Sin[t]}}
{{t v + T (8 - 7 e^-t - Cos[t])}, {t v + 8 T - 7 e^-t T - T Cos[t]}}
```

### makeMotorTimeDomainFunction[]

makeMotorTimeDomainFunction[] takes a transfer function model and returns a function that, when provided with ea and τa time-domain input functions, returns a function of time.

```
Clear[makeMotorTimeDomainFunction]
makeMotorTimeDomainFunction[model_] := makeMotorTimeDomainFunction[model, makeTimeDomainFunctionConvolve]
makeMotorTimeDomainFunction[model_, builder_] := Module[{ea = Unique["ea"], τa = Unique["τa"], t = Unique["t"], expr},
  expr = builder[model, {(1 &, *)ea[#] &, τa[#] &}][t];
  Function[{eaActual, τaActual}, Module[{exprT},
    exprT = expr /. {ea → eaActual, τa → τaActual};
    Function[{tActual}, exprT /. t → tActual]]
  ]]
```

### Larger Tests

We compare makeTimeDomainFunctionLaplace[] and makeTimeDomainFunctionConvolve[] on substantially-sized transfer functions. We see that, symbolically, for these tests they produce identical results, as they should. However, the makeTimeDomainFunctionConvolve[] approach is significantly faster, as therein we are asking InverseLaplaceTransform[] to significantly less heavy lifting. Moreover, and related to that point, the convolution approach is able to successfully compute time domain functions for some transfer functions and inputs that stymie makeTimeDomainFunctionLaplace[].

```
exampleConvolveE = makeMotorTimeDomainFunction[motorVelocityModel, makeTimeDomainFunctionConvolve][v &, 0 &][t] // Timing
exampleConvolveI = makeMotorTimeDomainFunction[motorVelocityModel, makeTimeDomainFunctionConvolve][0 &, 1 &][t] // Timing
```

$$\left\{ 10.0156, \left\{ \text{Kt} \left( \frac{1}{\text{Ke Kt} + \text{b R}} \right. \right. \right. \\ \left. \left( \sqrt{\left( \text{b}^2 \text{L}^2 + \text{J}^2 \text{R}^2 + 2 \text{b L R} \alpha \tau + \alpha \tau \left( 4 \text{Ke Kt L} + \text{R}^2 \alpha \tau \right) - 2 \text{J} \left( 2 \text{Ke Kt L} + \text{R} \left( \text{b L} + \text{R} \alpha \tau \right) \right) \right)} + 2 \text{e}^{-\frac{\text{t} \left( \text{b L} + \text{J R} - \text{R} \alpha \tau + \sqrt{\text{b}^2 \text{L}^2 + \text{J}^2 \text{R}^2 + 4 \text{Ke Kt L} \alpha \tau + 2 \text{b L R} \alpha \tau + \text{R}^2 \alpha \tau^2 - 2 \text{J} \left( 2 \text{Ke Kt L} + \text{b L R} + \text{R}^2 \alpha \tau \right)} \right)}{2 \text{L} \left( \text{J} - \alpha \tau \right)}} \right. \\ \left. \text{L} \left( \text{J} - \alpha \tau \right) \left( 1 / \left( \text{b L} + \text{J R} - \text{R} \alpha \tau + \sqrt{\left( \text{b}^2 \text{L}^2 + \text{J}^2 \text{R}^2 + 4 \text{Ke Kt L} \alpha \tau + 2 \text{b L R} \alpha \tau + \text{R}^2 \alpha \tau^2 - 2 \text{J} \left( 2 \text{Ke Kt L} + \text{b L R} + \text{R}^2 \alpha \tau \right) \right)} \right) + \right. \\ \left. \left. \left. \text{e}^{\frac{\text{t} \left( \text{b}^2 \text{L}^2 + \text{J}^2 \text{R}^2 + 4 \text{Ke Kt L} \alpha \tau + 2 \text{b L R} \alpha \tau + \text{R}^2 \alpha \tau^2 - 2 \text{J} \left( 2 \text{Ke Kt L} + \text{b L R} + \text{R}^2 \alpha \tau \right) \right)}{\text{L} \left( \text{J} - \alpha \tau \right)}} \right/ \left( -\text{b L} - \text{J R} + \text{R} \alpha \tau + \sqrt{\left( \text{b}^2 \text{L}^2 + \text{J}^2 \text{R}^2 + 4 \text{Ke Kt L} \alpha \tau + 2 \text{b L R} \alpha \tau + \right. \right. \right. \\ \left. \left. \left. \text{R}^2 \alpha \tau^2 - 2 \text{J} \left( 2 \text{Ke Kt L} + \text{b L R} + \text{R}^2 \alpha \tau \right) \right) \right) \right) \vee \left/ \left( \sqrt{\left( \text{b L} + \text{R} \left( \text{J} - \alpha \tau \right) \right)^2 - 4 \text{L} \left( \text{Ke Kt} + \text{b R} \right) \left( \text{J} - \alpha \tau \right)} \right) \right\} \right\}$$

$$\left\{ 14.0469, \left\{ \frac{1}{\text{Ke Kt} + \text{b R}} 2 \text{ R } (\text{J} - \alpha \tau) \sqrt{(\text{b}^2 \text{ L}^2 + \text{J}^2 \text{ R}^2 + 2 \text{ b L R } \alpha \tau + \alpha \tau (4 \text{ Ke Kt L} + \text{R}^2 \alpha \tau) - 2 \text{ J } (2 \text{ Ke Kt L} + \text{R} (\text{b L} + \text{R } \alpha \tau)))} \right. \right. \\ \left. \left. - \frac{\tau (\text{b L} + \text{J R R } \alpha \tau + \sqrt{\text{b}^2 \text{ L}^2 + \text{J}^2 \text{ R}^2 + 4 \text{ Ke Kt L } \alpha \tau + 2 \text{ b L R } \alpha \tau + \text{R}^2 \alpha \tau^2 - 2 \text{ J } (2 \text{ Ke Kt L} + \text{b L R} + \text{R}^2 \alpha \tau)})}{2 \text{ L } (\text{J} - \alpha \tau)} \right. \right. \\ \left. \left. \text{L } (\text{J} - \alpha \tau) \left( \left( e^{\frac{\tau (\sqrt{\text{b}^2 \text{ L}^2 + \text{J}^2 \text{ R}^2 + 2 \text{ b L R } \alpha \tau + 4 \text{ Ke Kt L} + \text{R}^2 \alpha \tau} - 2 \text{ J } (2 \text{ Ke Kt L} + \text{R} (\text{b L} + \text{R } \alpha \tau)))}{\text{L } (\text{J} - \alpha \tau)}} \right. \right. \right. \right. \\ \left. \left. \left( -\text{b L} + \text{J R} - \text{R } \alpha \tau + \sqrt{(\text{b}^2 \text{ L}^2 + \text{J}^2 \text{ R}^2 + 4 \text{ Ke Kt L } \alpha \tau + 2 \text{ b L R } \alpha \tau + \text{R}^2 \alpha \tau^2 - 2 \text{ J } (2 \text{ Ke Kt L} + \text{b L R} + \text{R}^2 \alpha \tau)))} \right) \right) \right) \right) \Bigg/ \\ \left( -\text{b L} - \text{J R} + \text{R } \alpha \tau + \sqrt{(\text{b}^2 \text{ L}^2 + \text{J}^2 \text{ R}^2 + 4 \text{ Ke Kt L } \alpha \tau + 2 \text{ b L R } \alpha \tau + \text{R}^2 \alpha \tau^2 - 2 \text{ J } (2 \text{ Ke Kt L} + \text{b L R} + \text{R}^2 \alpha \tau)))} \right) - \\ \left( \text{b L} - \text{J R} + \text{R } \alpha \tau + \sqrt{(\text{b}^2 \text{ L}^2 + \text{J}^2 \text{ R}^2 + 4 \text{ Ke Kt L } \alpha \tau + 2 \text{ b L R } \alpha \tau + \text{R}^2 \alpha \tau^2 - 2 \text{ J } (2 \text{ Ke Kt L} + \text{b L R} + \text{R}^2 \alpha \tau)))} \right) \Bigg/ \\ \left( \text{b L} + \text{J R} - \text{R } \alpha \tau + \sqrt{(\text{b}^2 \text{ L}^2 + \text{J}^2 \text{ R}^2 + 4 \text{ Ke Kt L } \alpha \tau + 2 \text{ b L R } \alpha \tau + \text{R}^2 \alpha \tau^2 - 2 \text{ J } (2 \text{ Ke Kt L} + \text{b L R} + \text{R}^2 \alpha \tau)))} \right) \Bigg) \Bigg) \Bigg/ \\ \left( 2 \sqrt{(\text{b L} + \text{R } (\text{J} - \alpha \tau))^2 - 4 \text{ L } (\text{Ke Kt} + \text{b R}) (\text{J} - \alpha \tau) (\text{J} - \alpha \tau)} \right) \Bigg\} \Bigg\}$$

```
exampleLaplaceE = makeMotorTimeDomainFunction[motorVelocityModel, makeTimeDomainFunctionLaplace][v &, 0 &][t] // Timing
exampleLaplaceT = makeMotorTimeDomainFunction[motorVelocityModel, makeTimeDomainFunctionLaplace][0 &, 1 &][t] // Timing
```

$$\left\{ 39.3438, \left\{ Kt \left( \frac{1}{Ke Kt + b R} - b e^{\left( t \left( -\frac{b}{2 (J \alpha \tau)} - \frac{J R}{2 L (J \alpha \tau)} - \frac{R \alpha \tau}{2 L (J \alpha \tau)} - \sqrt{\frac{-4 (Ke Kt + b R) (J L - L \alpha \tau) + (b L + J R - R \alpha \tau)^2}{2 L (J \alpha \tau)}} \right)} \right) L + b e^{\left( t \left( -\frac{b L}{2 (J L - L \alpha \tau)} - \frac{J R}{2 (J L - L \alpha \tau)} - \frac{R \alpha \tau}{2 (J L - L \alpha \tau)} + \sqrt{\frac{-4 (Ke Kt + b R) (J L - L \alpha \tau) + (b L + J R - R \alpha \tau)^2}{2 (J L - L \alpha \tau)}} \right)} \right) L - \right. \right. \\ e^{\left( t \left( -\frac{b}{2 (J \alpha \tau)} - \frac{J R}{2 L (J \alpha \tau)} + \frac{R \alpha \tau}{2 L (J \alpha \tau)} - \sqrt{\frac{-4 (Ke Kt + b R) (J L - L \alpha \tau) + (b L + J R - R \alpha \tau)^2}{2 L (J \alpha \tau)}} \right)} \right) J R + e^{\left( t \left( -\frac{b L}{2 (J L - L \alpha \tau)} - \frac{J R}{2 (J L - L \alpha \tau)} + \frac{R \alpha \tau}{2 (J L - L \alpha \tau)} + \sqrt{\frac{-4 (Ke Kt + b R) (J L - L \alpha \tau) + (b L + J R - R \alpha \tau)^2}{2 (J L - L \alpha \tau)}} \right)} \right) J R + \\ e^{\left( t \left( -\frac{b}{2 (J \alpha \tau)} - \frac{J R}{2 L (J \alpha \tau)} + \frac{R \alpha \tau}{2 L (J \alpha \tau)} - \sqrt{\frac{-4 (Ke Kt + b R) (J L - L \alpha \tau) + (b L + J R - R \alpha \tau)^2}{2 L (J \alpha \tau)}} \right)} \right) R \alpha \tau - e^{\left( t \left( -\frac{b L}{2 (J L - L \alpha \tau)} - \frac{J R}{2 (J L - L \alpha \tau)} + \frac{R \alpha \tau}{2 (J L - L \alpha \tau)} + \sqrt{\frac{-4 (Ke Kt + b R) (J L - L \alpha \tau) + (b L + J R - R \alpha \tau)^2}{2 (J L - L \alpha \tau)}} \right)} \right) R \alpha \tau + \\ e^{\left( t \left( -\frac{b}{2 (J \alpha \tau)} - \frac{J R}{2 L (J \alpha \tau)} + \frac{R \alpha \tau}{2 L (J \alpha \tau)} - \sqrt{\frac{-4 (Ke Kt + b R) (J L - L \alpha \tau) + (b L + J R - R \alpha \tau)^2}{2 L (J \alpha \tau)}} \right)} \right) \sqrt{-4 (Ke Kt + b R) (J L - L \alpha \tau) + (b L + J R - R \alpha \tau)^2} + \\ e^{\left( t \left( -\frac{b L}{2 (J L - L \alpha \tau)} - \frac{J R}{2 (J L - L \alpha \tau)} + \frac{R \alpha \tau}{2 (J L - L \alpha \tau)} + \sqrt{\frac{-4 (Ke Kt + b R) (J L - L \alpha \tau) + (b L + J R - R \alpha \tau)^2}{2 (J L - L \alpha \tau)}} \right)} \right) \sqrt{-4 (Ke Kt + b R) (J L - L \alpha \tau) + (b L + J R - R \alpha \tau)^2} \left. \right) \Bigg/ \\ \left( 2 (Ke Kt + b R) \sqrt{-4 (Ke Kt + b R) (J L - L \alpha \tau) + (b L + J R - R \alpha \tau)^2} \right) \Bigg\} \Bigg\} \Bigg\}$$

$$\left\{ 39.6406, \right. \\ \left\{ \frac{R}{Ke Kt + b R} - \frac{1}{2} e^{\left( -\frac{b}{2(3-\alpha\tau)} - \frac{3R}{2L(3-\alpha\tau)} + \frac{R\alpha\tau}{2L(3-\alpha\tau)} - \frac{\sqrt{-4(Ke Kt + b R)(3L - L\alpha\tau) + (bL + 3R - R\alpha\tau)^2}}{2L(3-\alpha\tau)} \right)} Ke Kt L - 2 e^{\left( -\frac{bL}{2(3L - L\alpha\tau)} - \frac{3R}{2(3L - L\alpha\tau)} + \frac{R\alpha\tau}{2(3L - L\alpha\tau)} + \frac{\sqrt{-4(Ke Kt + b R)(3L - L\alpha\tau) + (bL + 3R - R\alpha\tau)^2}}{2(3L - L\alpha\tau)} \right)} Ke Kt L + \right. \\ b e^{\left( -\frac{b}{2(3-\alpha\tau)} - \frac{3R}{2L(3-\alpha\tau)} + \frac{R\alpha\tau}{2L(3-\alpha\tau)} - \frac{\sqrt{-4(Ke Kt + b R)(3L - L\alpha\tau) + (bL + 3R - R\alpha\tau)^2}}{2L(3-\alpha\tau)} \right)} LR - b e^{\left( -\frac{bL}{2(3L - L\alpha\tau)} - \frac{3R}{2(3L - L\alpha\tau)} + \frac{R\alpha\tau}{2(3L - L\alpha\tau)} + \frac{\sqrt{-4(Ke Kt + b R)(3L - L\alpha\tau) + (bL + 3R - R\alpha\tau)^2}}{2(3L - L\alpha\tau)} \right)} LR - \\ e^{\left( -\frac{b}{2(3-\alpha\tau)} - \frac{3R}{2L(3-\alpha\tau)} + \frac{R\alpha\tau}{2L(3-\alpha\tau)} - \frac{\sqrt{-4(Ke Kt + b R)(3L - L\alpha\tau) + (bL + 3R - R\alpha\tau)^2}}{2L(3-\alpha\tau)} \right)} JR^2 + e^{\left( -\frac{bL}{2(3L - L\alpha\tau)} - \frac{3R}{2(3L - L\alpha\tau)} + \frac{R\alpha\tau}{2(3L - L\alpha\tau)} + \frac{\sqrt{-4(Ke Kt + b R)(3L - L\alpha\tau) + (bL + 3R - R\alpha\tau)^2}}{2(3L - L\alpha\tau)} \right)} JR^2 + \\ e^{\left( -\frac{b}{2(3-\alpha\tau)} - \frac{3R}{2L(3-\alpha\tau)} + \frac{R\alpha\tau}{2L(3-\alpha\tau)} - \frac{\sqrt{-4(Ke Kt + b R)(3L - L\alpha\tau) + (bL + 3R - R\alpha\tau)^2}}{2L(3-\alpha\tau)} \right)} R^2 \alpha\tau - e^{\left( -\frac{bL}{2(3L - L\alpha\tau)} - \frac{3R}{2(3L - L\alpha\tau)} + \frac{R\alpha\tau}{2(3L - L\alpha\tau)} + \frac{\sqrt{-4(Ke Kt + b R)(3L - L\alpha\tau) + (bL + 3R - R\alpha\tau)^2}}{2(3L - L\alpha\tau)} \right)} R^2 \alpha\tau + \\ e^{\left( -\frac{b}{2(3-\alpha\tau)} - \frac{3R}{2L(3-\alpha\tau)} + \frac{R\alpha\tau}{2L(3-\alpha\tau)} - \frac{\sqrt{-4(Ke Kt + b R)(3L - L\alpha\tau) + (bL + 3R - R\alpha\tau)^2}}{2L(3-\alpha\tau)} \right)} R \sqrt{-4(Ke Kt + b R)(3L - L\alpha\tau) + (bL + 3R - R\alpha\tau)^2} + \\ e^{\left( -\frac{bL}{2(3L - L\alpha\tau)} - \frac{3R}{2(3L - L\alpha\tau)} + \frac{R\alpha\tau}{2(3L - L\alpha\tau)} + \frac{\sqrt{-4(Ke Kt + b R)(3L - L\alpha\tau) + (bL + 3R - R\alpha\tau)^2}}{2(3L - L\alpha\tau)} \right)} R \sqrt{-4(Ke Kt + b R)(3L - L\alpha\tau) + (bL + 3R - R\alpha\tau)^2} \Bigg\} \\ \left. \left( 2(Ke Kt + b R) \sqrt{-4(Ke Kt + b R)(3L - L\alpha\tau) + (bL + 3R - R\alpha\tau)^2} \right) \right\} \Bigg\}$$

At first glance, the two approaches look like they might be producing quite different results. However, a little bit of work verifies that the applied-voltage functions are the same.

```
exampleConvolveE[[2, 1]] // Factor // Simplify
exampleLaplaceE[[2, 1]] // Factor // Simplify
%% / %
```

$$\begin{aligned}
& - \left( \left( e^{-\frac{t \left( b L + J R + \sqrt{(b L + R (J - \alpha T))^2 - 4 L (K e K t + b R) (J - \alpha T)} - R \alpha T \right)}{2 L (J - \alpha T)}} K t \left( -b L + b e^{\frac{t \sqrt{(b L + R (J - \alpha T))^2 - 4 L (K e K t + b R) (J - \alpha T)}}{L (J - \alpha T)}} L - J R + e^{\frac{t \sqrt{(b L + R (J - \alpha T))^2 - 4 L (K e K t + b R) (J - \alpha T)}}{L (J - \alpha T)}} J R + \right. \right. \\
& \left. \left. \sqrt{(b L + R (J - \alpha T))^2 - 4 L (K e K t + b R) (J - \alpha T)} + e^{\frac{t \sqrt{(b L + R (J - \alpha T))^2 - 4 L (K e K t + b R) (J - \alpha T)}}{L (J - \alpha T)}} \sqrt{(b L + R (J - \alpha T))^2 - 4 L (K e K t + b R) (J - \alpha T)} - \right. \right. \\
& \left. \left. 2 e^{\frac{t \left( b L + J R + \sqrt{(b L + R (J - \alpha T))^2 - 4 L (K e K t + b R) (J - \alpha T)} - R \alpha T \right)}{2 L (J - \alpha T)}} \sqrt{(b L + R (J - \alpha T))^2 - 4 L (K e K t + b R) (J - \alpha T)} + R \alpha T - e^{\frac{t \sqrt{(b L + R (J - \alpha T))^2 - 4 L (K e K t + b R) (J - \alpha T)}}{L (J - \alpha T)}} R \alpha T \right) \sqrt{\right.} \\
& \left. \left( 2 (K e K t + b R) \sqrt{(b L + R (J - \alpha T))^2 - 4 L (K e K t + b R) (J - \alpha T)} \right) \right)
\end{aligned}$$

$$\begin{aligned}
& - \left( \left( e^{-\frac{t \left( b L + J R + \sqrt{(b L + R (J - \alpha T))^2 - 4 L (K e K t + b R) (J - \alpha T)} - R \alpha T \right)}{2 L (J - \alpha T)}} K t \left( -b L + b e^{\frac{t \sqrt{(b L + R (J - \alpha T))^2 - 4 L (K e K t + b R) (J - \alpha T)}}{L (J - \alpha T)}} L - J R + e^{\frac{t \sqrt{(b L + R (J - \alpha T))^2 - 4 L (K e K t + b R) (J - \alpha T)}}{L (J - \alpha T)}} J R + \right. \right. \\
& \left. \left. \sqrt{(b L + R (J - \alpha T))^2 - 4 L (K e K t + b R) (J - \alpha T)} + e^{\frac{t \sqrt{(b L + R (J - \alpha T))^2 - 4 L (K e K t + b R) (J - \alpha T)}}{L (J - \alpha T)}} \sqrt{(b L + R (J - \alpha T))^2 - 4 L (K e K t + b R) (J - \alpha T)} - \right. \right. \\
& \left. \left. 2 e^{\frac{t \left( b L + J R + \sqrt{(b L + R (J - \alpha T))^2 - 4 L (K e K t + b R) (J - \alpha T)} - R \alpha T \right)}{2 L (J - \alpha T)}} \sqrt{(b L + R (J - \alpha T))^2 - 4 L (K e K t + b R) (J - \alpha T)} + R \alpha T - e^{\frac{t \sqrt{(b L + R (J - \alpha T))^2 - 4 L (K e K t + b R) (J - \alpha T)}}{L (J - \alpha T)}} R \alpha T \right) \sqrt{\right.} \\
& \left. \left( 2 (K e K t + b R) \sqrt{(b L + R (J - \alpha T))^2 - 4 L (K e K t + b R) (J - \alpha T)} \right) \right)
\end{aligned}$$

1

That the applied-torque functions are equivalent is a little harder to verify, but we do (finally) manage it with a bit more work than we needed for the applied-voltage functions.

```
convolvedFull = exampleConvolveτ[[2, 1]] // Expand // FullSimplify // Expand // FullSimplify
laplaceFull = exampleLaplaceτ[[2, 1]] // Expand // FullSimplify // Expand // FullSimplify
convolvedFull / laplaceFull
```

$$\frac{1}{2 (K e K t + b R)} \left( 2 R - e^{-\frac{t (b L + J R + \sqrt{(b L + R (J - \alpha \tau))^2 - 4 L (K e K t + b R) (J - \alpha \tau)})}{2 L (J - \alpha \tau)}} R - e^{-\frac{t (-b L - J R + \sqrt{(b L + R (J - \alpha \tau))^2 - 4 L (K e K t + b R) (J - \alpha \tau)})}{2 L (J - \alpha \tau)}} R + \right. \\ \left. e^{-\frac{t (b L + J R + \sqrt{(b L + R (J - \alpha \tau))^2 - 4 L (K e K t + b R) (J - \alpha \tau)})}{2 L (J - \alpha \tau)}} \left( -1 + e^{\frac{t \sqrt{(b L + R (J - \alpha \tau))^2 - 4 L (K e K t + b R) (J - \alpha \tau)}}{L (J - \alpha \tau)}} \right) (2 K e K t L + R (b L + R (-J + \alpha \tau))) \right) / \\ \left( \sqrt{(b L + R (J - \alpha \tau))^2 - 4 L (K e K t + b R) (J - \alpha \tau)} \right)$$

$$\frac{1}{2 (K e K t + b R)} \left( 2 R - e^{-\frac{t (b L + J R + \sqrt{(b L + R (J - \alpha \tau))^2 - 4 L (K e K t + b R) (J - \alpha \tau)})}{2 L (J - \alpha \tau)}} R - e^{-\frac{t (-b L - J R + \sqrt{(b L + R (J - \alpha \tau))^2 - 4 L (K e K t + b R) (J - \alpha \tau)})}{2 L (J - \alpha \tau)}} R + \right. \\ \left. e^{-\frac{t (b L + J R + \sqrt{(b L + R (J - \alpha \tau))^2 - 4 L (K e K t + b R) (J - \alpha \tau)})}{2 L (J - \alpha \tau)}} \left( -1 + e^{\frac{t \sqrt{(b L + R (J - \alpha \tau))^2 - 4 L (K e K t + b R) (J - \alpha \tau)}}{L (J - \alpha \tau)}} \right) (2 K e K t L + R (b L + R (-J + \alpha \tau))) \right) / \\ \left( \sqrt{(b L + R (J - \alpha \tau))^2 - 4 L (K e K t + b R) (J - \alpha \tau)} \right)$$

1

Huzzah. Going forward, we will use the convolution approach.

### Time Domain Output Functions for Motor State

We define time domain functions for all the motor state of interest.

```
Clear[motorPosition, motorVelocity, motorAcceleration, motorCurrent, motorEMF, motorTorque]
```

```
motorPosition = makeMotorTimeDomainFunction[motorPositionModel];
```

```
motorVelocity = makeMotorTimeDomainFunction[motorVelocityModel];
```

```
motorAcceleration = makeMotorTimeDomainFunction[motorAccelerationModel];
```

```
motorCurrent = makeMotorTimeDomainFunction[motorCurrentModel];
```

```
motorEMF = makeMotorTimeDomainFunction[motorEMFModel];
```

```
motorTorque = makeMotorTimeDomainFunction[motorTorqueModel];
```

### Time Domain Input Functions

#### Keeping track of units

In analogy to parameterQuantities above, we define inputQuantities, which adds units to our input parameters.

## parameterQuantities

$$\left\langle \begin{aligned} &R \rightarrow R \Omega, L \rightarrow L H, \theta[t] \rightarrow \theta[t] \text{ rad}, \theta'[t] \rightarrow \theta'[t] \text{ rad/s}, \Omega[t] \rightarrow \Omega[t] \text{ rad/s}, \theta''[t] \rightarrow \theta''[t] \text{ rad/s}^2, \alpha[t] \rightarrow \alpha[t] \text{ rad/s}^2, \\ &i[t] \rightarrow i[t] \text{ A}, i'[t] \rightarrow i'[t] \text{ A/s}, i''[t] \rightarrow i''[t] \text{ A/s}^2, ea[t] \rightarrow ea[t] \text{ V}, e[t] \rightarrow e[t] \text{ V}, \tau[t] \rightarrow \tau[t] \text{ mN/rad}, \\ &\tau a[t] \rightarrow \tau a[t] \text{ mN/rad}, \alpha \tau \rightarrow \alpha \tau \text{ ms}^2 \text{ N/rad}^2, J \rightarrow J \text{ kg m}^2 \text{ /rad}^2, Ke \rightarrow Ke \text{ sV/rad}, Kt \rightarrow Kt \text{ mN/(A rad)}, b \rightarrow b \text{ msN/rad}^2 \end{aligned} \right\rangle$$

```
inputQuantities = {
  ea → Quantity[ea, "Volts"],
  τa → Quantity[τa, siTorqueUnits],
  cτ → Quantity[τa, siTorqueUnits],
  t → Quantity[t, "Seconds"]
}
```

```
{ea → ea V, τa → τa mN/rad, cτ → τa mN/rad, t → t s}
```

## Mass on pulley

Here, we model a mass suspended on a massless “rigid string” from a massless pulley attached to the motor shaft. That string acts at a wrench with a length of the radius of the pulley to deliver a torque to the shaft. `τMassOnPulley[]` computes that torque.

There is a force pulling down on the mass due to gravity and up due to acceleration in the string that results from the driven angular acceleration of the shaft. The sum of these forces is the tension in the string. If the net acceleration is negative (ie: downward), a real (flexible) string would clamp at zero and have the mass in free-fall; we choose not to model that for now. Instead, here, our hypothetical rigid string continues to impart acceleration from the pulley even for negative accelerations. This is a bit odd for a string, but more reasonable if the string were in fact a rigid arm being lifted, or some such.

```
Clear[τMassOnPulley]
τMassOnPulley[mass_, radius_][t_] := Module[
  {gravityAcceleration, gravityForce, angularAcceleration,
   tangentialAcceleration, tangentialForce, tension, torque, unit, parts, radians = Quantity["Radians"]},

  (* gravity exerts a constant force *)
  gravityAcceleration = Quantity["StandardAccelerationOfGravity"];
  gravityForce = gravityAcceleration * mass;

  (* angular acceleration is resisted by mass, creates tangential force resisting acceleration *)
  angularAcceleration = θ''[t] /. parameterQuantities;
  tangentialAcceleration = angularAcceleration * radius / radians;
  tangentialForce = tangentialAcceleration * mass;

  tension = gravityForce + tangentialForce;

  (*tension acts like a wrench with lever arm 'radius'*)
  torque = tension * radius / radians;
  unit = QuantityUnit[torque];
  parts = List @@ (QuantityMagnitude[torque] // Apart);
  {cτ → Quantity[parts[[1]], unit], ατ → Quantity[parts[[2]], unit] / angularAcceleration}
]
τMassOnPulley[Quantity[weightLbs, "Pounds"], Quantity[rInches, "Inches"]][t]
% // siUnits // clearUnits // N
```

```
{cτ →  $\frac{196133 \text{ rInches weightLbs}}{508} \text{ lb in}^2 / (\text{s}^2 \text{ rad})$ , ατ →  $\text{rInches}^2 \text{ weightLbs lb in}^2 / \text{rad}^2$ }
```

```
{cτ → 0.112985 rInches weightLbs, ατ → 0.00029264 rInches2 weightLbs}
```

## Typical Input Examples

In analogy to `aMotor` above, we now define some typical inputs. The first has no externally applied torque.



```
anInput = Module[{},
{
  t → Quantity[t, "Seconds"],
  ea → Quantity[12, "Volts"],
  τa → Quantity[0, siTorqueUnits],
  cτ → Quantity[0, siTorqueUnits],
  ατ → Quantity[0, (ατ /. parameterUnits)]
}] // Association

{t → t s, ea → 12 V, τa → 0 mN/rad, cτ → 0 mN/rad, ατ → 0 ms2N/rad2 |}
```

A second has external torque from a mass on a pulley.

```
τMassOnPulley[Quantity[3, "Pounds"], Quantity[2, "Inches"]][t] // N
τMassOnPulley[Quantity[3, "Pounds"], Quantity[2, "Inches"]][t] // siUnits // N
anInputPulley = {t → Quantity[t, "Seconds"], ea → Quantity[12, "Volts"]} ~Join~
Module[{pulley = τMassOnPulley[Quantity[3, "Pounds"], Quantity[2, "Inches"]]},
  Print[pulley[t]];
  pulley = pulley[t] // siUnits;
  {cτ → (cτ /. pulley), τa → (cτ /. pulley), ατ → (ατ /. pulley)}] // Association
```

```
{cτ → 2316.53 lbin2/(s2rad), ατ → 12. lbin2/rad2}
```

```
{cτ → 0.677909 kg m2/(s2rad), ατ → 0.00351168 kg m2/rad2}
```

```
{cτ →  $\frac{588399}{254}$  lbin2/(s2rad), ατ → 12 lbin2/rad2}
```

```
{t → t s, ea → 12 V, cτ →  $\frac{3389544870828501}{500000000000000}$  kg m2/(s2rad),
τa →  $\frac{3389544870828501}{500000000000000}$  kg m2/(s2rad), ατ →  $\frac{2194797400719}{625000000000000}$  kg m2/rad2 |}
```

A third leaves quantities with symbolic parameters

```
genericInputτ = Module[{},
{
  t → Quantity[t, "Seconds"],
  ea → Quantity[ea, "Volts"],
  τa → Quantity[cτ, siTorqueUnits],
  cτ → Quantity[cτ, siTorqueUnits],
  ατ → Quantity[ατ, (ατ /. parameterUnits)]
}] // Association

{t → t s, ea → ea V, τa → cτ mN/rad, cτ → cτ mN/rad, ατ → ατ ms2N/rad2 |}
```

## Examples

We try out our time-domain functions on our example data. Notice how the units come out correctly: they flow automatically from beginning to end

```
anInput // N
anInputPulley // N

{t → t s, ea → 12. V, τa → 0. mN/rad, cτ → 0. mN/rad, ατ → 0. ms2N/rad2 |}
```

```
{t → t s, ea → 12. V, cτ → 0.677909 kg m2/(s2rad), τa → 0.677909 kg m2/(s2rad), ατ → 0.00351168 kg m2/rad2 |}
```

```
velUnits = motorVelocity[ea &, ta &][t] /. aMotor /. anInput // N // siUnits // FullSimplify
```

$$\left\{ e^{-0.377374 t} \left( -10.2734 \text{ rad/s} \right) + e^{-4754.7 t} \left( 0.000815385 \text{ rad/s} \right) + 10.2726 \text{ rad/s} \right\}$$


```
vel = velUnits // clearUnits
```

```
velτ = motorVelocity[ea &, ta &][t] /. aMotor /. anInputPulley // N // siUnits // clearUnits // FullSimplify
```

```
Plot[{vel, velτ}, {t, 0, 10}]
```

$$\left\{ 10.2726 + 0.000815385 e^{-4754.7 t} - 10.2734 e^{-0.377374 t} \right\}$$

$$\left\{ 12.0691 + 0.00081827 e^{-4754.7 t} - 12.0699 e^{-0.378704 t} \right\}$$



Good: the model with the (positive) external torque achieves greater velocity, as it should.

```
torUnits = motorTorque[ea &, ta &][t] /. aMotor /. anInput // N // siUnits // FullSimplify
```

```
torτ = motorTorque[ea &, ta &][t] /. aMotor /. anInputPulley // N // siUnits // clearUnits // FullSimplify
```


```
tor = torUnits // clearUnits
```

```
Plot[{tor, torτ}, {t, 0, 10}]
```

$$\left\{ e^{-4754.7 t} \left( -3.87693 \text{ kg m}^2 / (\text{s}^2 \text{ rad}) \right) + 0.338995 \text{ kg m}^2 / (\text{s}^2 \text{ rad}) + e^{-0.377374 t} \left( 3.53793 \text{ kg m}^2 / (\text{s}^2 \text{ rad}) \right) \right\}$$

$$\left\{ -0.279629 - 3.87698 e^{-4754.7 t} + 4.15661 e^{-0.378704 t} \right\}$$


$$\left\{ 0.338995 - 3.87693 e^{-4754.7 t} + 3.53793 e^{-0.377374 t} \right\}$$



```
posUnits = motorPosition[ea &, ta &][t] /. aMotor /. anInput // N // siUnits // FullSimplify
pos = posUnits // clearUnits
Plot[{pos}, {t, 0, 10}]
```

$$\left\{ e^{-4754.7 t} \left( -1.7149 \times 10^{-7} \text{ rad} \right) + e^{-0.377374 t} \left( 27.2234 \text{ rad} \right) + (-27.2234 + 10.2726 t) \text{ rad} \right\}$$


$$\{-27.2234 - 1.7149 \times 10^{-7} e^{-4754.7 t} + 27.2234 e^{-0.377374 t} + 10.2726 t\}$$



```
curUnits = motorCurrent[ea &, ta &][t] /. aMotor /. anInput // N // siUnits // FullSimplify
cur = curUnits // clearUnits
Plot[{cur}, {t, 0, 10}]
```

$$\left\{ e^{-4754.7 t} \left( -3.63689 \text{ A} \right) + 0.318007 \text{ A} + e^{-0.377374 t} \left( 3.31888 \text{ A} \right) \right\}$$


$$\{0.318007 - 3.63689 e^{-4754.7 t} + 3.31888 e^{-0.377374 t}\}$$



```
emfUnits = motorEMF[ea &, ta &][t] /. aMotor /. anInput // N // siUnits // FullSimplify
emf = emfUnits // clearUnits
Plot[{emf}, {t, 0, 10}]
```

$$\left\{ e^{-0.377374 t} \left( -10.9514 \text{ kg m}^2 / (\text{s}^3 \text{ A}) \right) + e^{-4754.7 t} \left( 0.000869201 \text{ kg m}^2 / (\text{s}^3 \text{ A}) \right) + 10.9506 \text{ kg m}^2 / (\text{s}^3 \text{ A}) \right\}$$

$$\{10.9506 + 0.000869201 e^{-4754.7 t} - 10.9514 e^{-0.377374 t}\}$$



## Steady State

### Calculating Steady State Values

#### Brute Force vs The Final Value Theorem

We explore the steady state behavior, the limit of the time domain functions as time goes to infinity.

One way to calculate said limit is to brute-force calculate the limit in question. Let's explore that with velocity.

```
parameterAssumptions = (# >= 0 & /@ Complement[Keys[aMotor ~Join~ anInput], {tau, alpha}]) ~Join~ {tau ∈ ℝ, alpha ∈ ℝ}
genericVelExpr =
  motorVelocity[ea &, tau &][t] /. genericMotor /. genericInputtau // siUnits // clearUnits // clearUnits // FullSimplify
genericVelLimit = Limit[genericVelExpr, t → Infinity, Assumptions → parameterAssumptions]
```

```
{b >= 0, ctau >= 0, ea >= 0, J >= 0, Ke >= 0, Kt >= 0, L >= 0, R >= 0, tau >= 0, tau ∈ ℝ, alpha ∈ ℝ}
```

$$\left\{ \frac{1}{2 (Ke Kt + b R)} \left( 2 (ea Kt + ctau R) + \frac{1}{2 e^{-\frac{1}{2} t \left( \frac{R}{L} + \frac{b}{J - \alpha \tau} \right)}} \left( -(ea Kt + ctau R) \cosh \left[ \frac{1}{2 J L - 2 L \alpha \tau} t \sqrt{(-4 J Ke Kt L + b^2 L^2 - 2 b J L R + J^2 R^2 + 4 Ke Kt L \alpha \tau + 2 b L R \alpha \tau - 2 J R^2 \alpha \tau + R^2 \alpha \tau^2)} \right] + \left( 2 ctau Ke Kt L + b L (-ea Kt + ctau R) + ea Kt R (-J + \alpha \tau) + ctau R^2 (-J + \alpha \tau) \right) \sinh \left[ \frac{1}{2 J L - 2 L \alpha \tau} t \sqrt{(-4 J Ke Kt L + b^2 L^2 - 2 b J L R + J^2 R^2 + 4 Ke Kt L \alpha \tau + 2 b L R \alpha \tau - 2 J R^2 \alpha \tau + R^2 \alpha \tau^2)} \right] \right) \right) / \left( \sqrt{(b^2 L^2 + J^2 R^2 + 2 b L R \alpha \tau + \alpha \tau (4 Ke Kt L + R^2 \alpha \tau) - 2 J (2 Ke Kt L + R (b L + R \alpha \tau)))} \right) \right\}$$

```
{ConditionalExpression[
  ea Kt + ctau R
  Ke Kt + b R
  , (sqrt((-4 J Ke Kt L + b^2 L^2 - 2 b J L R + J^2 R^2 + 4 Ke Kt L alpha tau + 2 b L R alpha tau - 2 J R^2 alpha tau + R^2 alpha tau^2)) |
  1 / (sqrt((b^2 L^2 + J^2 R^2 + 2 b L R alpha tau + alpha tau (4 Ke Kt L + R^2 alpha tau) - 2 J (2 Ke Kt L + b L R + R^2 alpha tau)))) ∈ ℝ &&
  L (J - alpha tau) (b L + J R - R alpha tau) > 0 && L (J - alpha tau) sqrt((-4 J Ke Kt L + b^2 L^2 - 2 b J L R + J^2 R^2 + 4 Ke Kt L alpha tau + 2 b L R alpha tau - 2 J R^2 alpha tau + R^2 alpha tau^2)) > 0 &&
  L (J - alpha tau) (b L + J R - R alpha tau - sqrt((-4 J Ke Kt L + b^2 L^2 - 2 b J L R + J^2 R^2 + 4 Ke Kt L alpha tau + 2 b L R alpha tau - 2 J R^2 alpha tau + R^2 alpha tau^2))) > 0]}]
```

A more efficient way to calculate the steady state behavior is to avail ourselves of the use of the Final Value Theorem (Ref: [https://en.wikipedia.org/wiki/Final\\_value\\_theorem](https://en.wikipedia.org/wiki/Final_value_theorem)). The final value theorem tells us that if the time-domain limit exists, it is equal to the limit of the Laplace transform of the time domain function as  $s \rightarrow 0$ .

```
Clear[ssFinalValueTheorem]
ssFinalValueTheorem[model_, inputs_] := Module[{s = Unique[], t = Unique[], expr},
  expr = model[s] . (LaplaceTransform[#, t, s] & /@ inputs);
  Limit[s expr, s → 0][[1]]
]
```

```
(ss = {
  ssPos → ssFinalValueTheorem[motorPositionModel, {ea &, cτ &}] // Factor,
  ssVel → ssFinalValueTheorem[motorVelocityModel, {ea &, cτ &}] // Factor,
  ssAcc → ssFinalValueTheorem[motorAccelerationModel, {ea &, cτ &}] // Factor,
  ssEmf → ssFinalValueTheorem[motorEMFModel, {ea &, cτ &}] // Factor,
  ssCur → ssFinalValueTheorem[motorCurrentModel, {ea &, cτ &}] // Factor,
  ssTor → ssFinalValueTheorem[motorTorqueModel, {ea &, cτ &}] // Factor
}) // prettyPrint
```

$$\begin{aligned} \text{ssPos} &\rightarrow \text{Indeterminate} \\ \text{ssVel} &\rightarrow \frac{ea \, Kt + c\tau \, R}{Ke \, Kt + b \, R} \\ \text{ssAcc} &\rightarrow 0 \\ \text{ssEmf} &\rightarrow \frac{Ke (ea \, Kt + c\tau \, R)}{Ke \, Kt + b \, R} \\ \text{ssCur} &\rightarrow \frac{b \, ea - c\tau \, Ke}{Ke \, Kt + b \, R} \\ \text{ssTor} &\rightarrow \frac{(b \, ea - c\tau \, Ke) \, Kt}{Ke \, Kt + b \, R} \end{aligned}$$

We observe that, for velocity, at least, the Final Value Theorem approach gives the same value as does the brute force approach. Huzzah! Moreover, computing the Final Value Theorem values is much more efficient.

### On the Applicability of the Final Value Theorem

Can we always use the Final Value Theorem?

No, the Final Value Theorem only applies under certain conditions. Specifically: that all non-zero poles of the transfer function must have negative real parts, and the transfer function must have at most one pole at the origin (Ref: <http://www-personal.umich.edu/~dsbaero/others/39-FVTrevisited.pdf>). The predicate exhibited in the ConditionalExpression of the brute-force result captures exactly these conditions. When using the Final Value Theorem on any concrete system, we will need to test the conditions in any particular application. Lets illustrate by looking at the velocity model.

```
motorVelocityModel
(velPoles = TransferFunctionPoles[motorVelocityModel]) // MatrixForm
```

$$\left( \begin{array}{c|c} \frac{ea}{Kt} & \frac{\tau a}{R + L \, s} \\ \hline \Omega & \frac{Ke \, Kt + (R + L \, s) \, (b + (J - \alpha \tau) \, s)}{Ke \, Kt + (R + L \, s) \, (b + (J - \alpha \tau) \, s)} \end{array} \right) \mathcal{T}$$

$$\left( \begin{array}{c} \frac{-b \, L - J \, R + R \, \alpha \tau - \sqrt{-4 \, (Ke \, Kt + b \, R) \, (J \, L - L \, \alpha \tau) + (b \, L + J \, R - R \, \alpha \tau)^2}}{2 \, (J \, L - L \, \alpha \tau)} \\ \frac{-b \, L - J \, R + R \, \alpha \tau + \sqrt{-4 \, (Ke \, Kt + b \, R) \, (J \, L - L \, \alpha \tau) + (b \, L + J \, R - R \, \alpha \tau)^2}}{2 \, (J \, L - L \, \alpha \tau)} \end{array} \right) \left( \begin{array}{c} \frac{-b \, L - J \, R + R \, \alpha \tau - \sqrt{-4 \, (Ke \, Kt + b \, R) \, (J \, L - L \, \alpha \tau) + (b \, L + J \, R - R \, \alpha \tau)^2}}{2 \, (J \, L - L \, \alpha \tau)} \\ \frac{-b \, L - J \, R + R \, \alpha \tau + \sqrt{-4 \, (Ke \, Kt + b \, R) \, (J \, L - L \, \alpha \tau) + (b \, L + J \, R - R \, \alpha \tau)^2}}{2 \, (J \, L - L \, \alpha \tau)} \end{array} \right)$$

The poles are the same for both applied-voltage and applied-torque input since the transfer function denominator is the same for both inputs, in both cases being the zeros of:

```
den = motorVelocityModel[s][[1, 2]] // Denominator
```

$$Ke \, Kt + (R + L \, s) \, (b + s \, (J - \alpha \tau))$$

In order to get a feel for where these zeros are in practice, let's look at that denominator for an actual example motor. We compute where the poles occur as a function of  $\alpha \tau$ .

```
polesατ = den /. aMotor /. parameterQuantities // siUnits // clearUnits
Solve[polesατ == 0, s] // N
```

$$\frac{284 \, 089}{250 \, 000} + \left( \frac{33}{10} + \frac{347 \, s}{500 \, 000} \right) \left( \frac{33}{1000} + s \left( \frac{100 \, 001 \, 041}{100 \, 000 \, 000} - \alpha \tau \right) \right)$$


$$\left\{ \left\{ s \rightarrow \left( 14.4092 \left( 1.65003 \times 10^{10} - 1.65 \times 10^{10} \alpha \tau - 1. \sqrt{2.72173 \times 10^{20} - 5.44423 \times 10^{20} \alpha \tau + 2.7225 \times 10^{20} \alpha \tau^2} \right) \right) / \left( -1.00001 \times 10^8 + 1. \times 10^8 \alpha \tau \right) \right\}, \right. \\ \left. \left\{ s \rightarrow \left( 14.4092 \left( 1.65003 \times 10^{10} - 1.65 \times 10^{10} \alpha \tau + \sqrt{2.72173 \times 10^{20} - 5.44423 \times 10^{20} \alpha \tau + 2.7225 \times 10^{20} \alpha \tau^2} \right) \right) / \left( -1.00001 \times 10^8 + 1. \times 10^8 \alpha \tau \right) \right\} \right\}$$

By exploring the system a bit, we can see that for certain values of  $\alpha \tau$  (in this one example, apparently any  $\alpha \tau$  bigger than one), poles exist in the right hand plane, and thus the Final Value Theorem does not apply:

```

Manipulate[
  Column[{polesατ /. {ατ → var},
    poles → (s /. Solve[Evaluate[polesατ /. {ατ → var}] == 0, s] // ReIm),
    Row[
      {Plot3D[polesατ /. {s → x + I y, ατ → var} // Abs, {x, -range, range},
        {y, -range, range}, ImageSize → Medium, AxesLabel → {"Re", "Im"}, PlotLabel → "Abs"},
      Plot3D[polesατ /. {s → x + I y, ατ → var} // Arg, {x, -range, range}, {y, -range, range},
        ImageSize → Medium, AxesLabel → {"Re", "Im"}, PlotLabel → "Arg"}
    ]}],
  {{var, 1.1, "ατ"}, -10, 10}, {{range, 10}, 0, 10000}]

```



Using the “hard-way” limit previously calculated, we can verify the observation the steady state velocity is unbounded for such  $\alpha\tau$ .

```

genericVellimit /. aMotor /. parameterQuantities // siUnits // clearUnits // N
% /. ατ → # & /@ {0, 1/2, 1, 2}

{ConditionalExpression[0.803048 (3.3 cτ + 1.066 ea),
  (√(10.8869 - 21.7769 ατ + 10.89 ατ²) | 1 / (√(10.8902 + 0.000151153 ατ - 2.00002 (0.00165284 + 10.89 ατ) + ατ (0.00315452 + 10.89 ατ))) ) ∈
  ℝ && 0.000694 (3.30006 - 3.3 ατ) (1.00001 - 1. ατ) > 0. && 0.000694 (1.00001 - 1. ατ) √(10.8869 - 21.7769 ατ + 10.89 ατ²) > 0. &&
  0.000694 (1.00001 - 1. ατ) (3.30006 - 3.3 ατ - 1. √(10.8869 - 21.7769 ατ + 10.89 ατ²)) > 0. ]}

{{0.803048 (3.3 cτ + 1.066 ea)}, {0.803048 (3.3 cτ + 1.066 ea)}, {Undefined}, {Undefined}}

```

An analogous analytic development awaits a future revision to this document.

```

Collect[den // Expand, s]

Ke Kt + b R + s² (J L - L ατ) + s (b L + J R - R ατ)

```

## Back EMF vs Applied Voltage

We want to know the steady-state velocity at which the back EMF balances the input voltage. Thus, we need EMF in terms of speed. We have EMF from voltage, and speed from voltage. So we need to invert the latter, then compose.

```
Clear[ssAppliedVoltageFromVelocity]
ssAppliedVoltageFromVelocity[velocity_] := Module[{eqn, velSym = Unique["vel"]},
  eqn = velSym == (ssVel /. ss);
  ea /. uniqueSolve[eqn, ea][[1]] /. velSym → velocity
]
ssAppliedVoltageFromVelocity[Ω]
```

$$\frac{-c\tau R + K_e K_t \Omega + b R \Omega}{K_t}$$

```
Clear[ssEmfFromAppliedVoltage, ssEmfFromVelocity]
ssEmfFromAppliedVoltage[voltage_] := (ssEmf /. ss) /. ea → voltage
ssEmfFromVelocity[velocity_] := ssEmfFromAppliedVoltage[ssAppliedVoltageFromVelocity[velocity]]
ssEmfFromVelocity[Ω]
% // FullSimplify
```

$$\frac{K_e (K_e K_t \Omega + b R \Omega)}{K_e K_t + b R}$$

$$K_e \Omega$$

Well, that's a result now, isn't it? Of course, this is expected, given the underlying differential equations:

```
diffEqns[[1]]
```

$$e[t] == K_e \theta'[t]$$

However, that we arrive at the same result the long-way around boosts confidence that the intervening math is in fact correct.

So, at what velocity does the applied voltage balance the back EMF?

```
ssEmfFromVelocity[Ω] == ssAppliedVoltageFromVelocity[Ω] // FullSimplify
uniqueSolve[%, Ω]
emfThresholdVelocity = %[[1]]
```

$$\frac{R (c\tau - b \Omega)}{K_t} == 0$$

$$\left\{ \Omega \rightarrow \frac{c\tau}{b} \right\}$$

$$\Omega \rightarrow \frac{c\tau}{b}$$

That's interesting: the equalizing velocity is dependent on the motor (through its viscous drag parameter), the external system (through the externally applied torque), but *not* the externally applied voltage. Also: if there is no external torque, then the threshold velocity is zero, which is as it should be, since in that situation the back EMF can never match the externally applied voltage due to losses except when all is stopped.

A quick check that the units are correct:

```
emfThresholdVelocity /. parameterQuantities /. inputQuantities
```

$$\Omega \rightarrow \frac{\tau a}{b} \text{ rad/s}$$

What does this look like for an actual motor with our example weight attached?

```
(Ω /. emfThresholdVelocity) /. aMotor /. anInputPulley // siUnits
UnitConvert[%, "rpm"] // N
UnitConvert[%, "Revolutions / Second"] * Quantity[1120, IndependentUnit["ticks"] / "Revolutions"]
```

$$\frac{102713480934197}{5000000000000} \text{ rad/s}$$

196.168 rev/min

3661.81 ticks /s

## Revision History

- 2018.07.23 Added acceleration-dependant external torque. Corrected “viscous friction” to “viscous drag”. Clarified usage of units.