
On Proportional Braking in H Bridges

Robert Atkinson
13 August 2018

Most H bridge designs use one of two common switching modalities, which are usually termed 'lock anti-phase' and 'sign-magnitude' respectively (Ref: <http://www.modularcircuits.com/blog/articles/h-bridge-secrets>). However, the STMicroelectronics VN5050E H bridge, used in FTC motor controllers for nearly a decade, employs neither of these modalities, instead using what can be called an 'asynchronous sign-magnitude' switching modality. In addition to its driving modes, this modality admits a mode wherein dynamic-braking is applied in proportional to the PWM duty cycle, which we term 'proportional braking'. Here, we explore why using proportional braking might be advantageous, and the details of doing so.

1. Introduction


1.1. Administrative

Before we begin, we load in some previously computed logic.

```
Get[NotebookDirectory[] <> "Utilities.m"]
inputDirectory = FileNameJoin[{NotebookDirectory[], "MotorPhysicsGearsInitialConditions.Output"}] <> $PathnameSeparator;
Get[inputDirectory <> "ParametersUnitsAndAssumptions.m"];
Get[inputDirectory <> "MotorModels.m"];
Get[inputDirectory <> "MotorTimeDomainFunctions.m"];
Get[inputDirectory <> "Misc.m"];
```

1.2. H Bridge Architecture

In order to appreciate the design task explored in this document, it is important to have a comfortable, detailed working understanding of the operation of H bridges. From a simplified perspective, an H Bridge can abstractly be thought of as a circuit involving a motor, a voltage source, and four switches (Ref: https://en.wikipedia.org/wiki/H_bridge):




In this document, we need to be able to discuss with some precision the state of these switches. As there are four switches, there are 16 possible states. We will number those states by assigning weights to each switch, and summing over the weights of the switches which are closed in any particular state. We assign weights as follows: $S1 \rightarrow 8$, $S2 \rightarrow 4$, $S3 \rightarrow 2$, and $S4 \rightarrow 1$. With that nomenclature in hand, the motor's response in each of the 16 states is as follows:

State #	S1	S2	S3	S4	Instantaneous Velocity	
					CW	CCW
9	1	0	0	1	Drive CW	Drive CW
6	0	1	1	0	Drive CCW	Drive CCW
0	0	0	0	0	Coast	Coast
8	1	0	0	0	Coast	Brake to VCC
4	0	1	0	0	Brake to GND	Coast
2	0	0	1	0	Brake to VCC	Coast
1	0	0	0	1	Coast	Brake to GND
5	0	1	0	1	Brake to GND	Brake to GND
10	1	0	1	0	Brake to VCC	Brake to VCC
12	0	0	0	0	Short Circuit	Short Circuit
3	0	0	0	0	Short Circuit	Short Circuit
7	0	0	0	0	Short Circuit	Short Circuit
11	0	0	0	0	Short Circuit	Short Circuit
13	0	0	0	0	Short Circuit	Short Circuit
14	0	0	0	0	Short Circuit	Short Circuit
15	0	0	0	0	Short Circuit	Short Circuit

Note that in a design where an H bridge is driven by pulse width modulation, we can talk with precision about the operation of the bridge by naming the state used in the “on” part of the PWM cycle, the state used in the “off” part of the cycle, and the duty-cycle percentage of the “on” state as a fraction of the PWM framing period (which is contextually herein taken as fixed). Thus, for example, we might talk about an H bridge using a 6/2 mode being driven at a 75% duty cycle, or a bridge in 5/0 mode with a 30% duty cycle, etc.

States 9 and 6 are the driving states. In state 9, the motor turns one direction, which we have arbitrarily labeled as CW (or, more precisely, it attempts to increase its angular velocity in the CW direction). In state 6, the motor turns CCW. In state 0 it coasts, while in states 5 and 10 rheostatic dynamic braking is employed (Ref: https://en.wikipedia.org/wiki/Dynamic_braking), either through the ground rail or the power rail. Seven of the states are short circuits, and should be avoided.

States 8, 4, 2, and 1 deserve special mention. Their behavior is dependent not only on the state of the switches, but also on the sign of the instantaneous angular velocity. This is due to consequences of the fact that a more complete picture of an H bridge must include the so-called ‘catch diodes’ whose presence is necessary to avoid inductive voltage spikes during switching transitions (Ref: <http://modularcircuits.tantosonline.com/blog/articles/h-bridge-secrets/h-bridges-the-basics/>).



Let us consider state 8, where S1 (aka Q1) is closed, and the remaining switches are open. If we are driving CW, and, say, switching between state 9 and state 8 over a PWM frame (which we will denote as the 9/8 mode), then as switch S4 is opened, the inductive motor current briefly discharges across D1 (in FTC motors, ‘briefly’ is about 1ms). Because the shaft is turning CW, the polarity of the motor’s back EMF is such that it cannot drive current through that same route. All is well: we drive forward.

However, consider the case where we had been driving in the CCW direction, using, say, the 6/2 mode, so that the motor shaft is currently rotating CCW. Then, were we to switch to 9/8 mode, we find that the polarity of the motor’s back EMF is the other way around; it can now also drive current through the diode, and we find ourselves in a dynamic braking situation.

States 4, 2, and 1 have analogous behavior.

1.3. Forward Driving vs Proportional Braking vs Reverse Driving

The STMicroelectronics VNH5050E H bridges used in FTC use 9/8 and 6/2 driving modes. Thus, the behaviors of states 8 & 2 mentioned above are pertinent.

Suppose we are driving CW, cruising along, forward driving using mode 9/8 (the 6/2 case is analogous), having developed some non-zero velocity in the shaft. Then, suppose further that our PID loop discerns that we're going a little too fast. At first, it will ask us to take our foot off the gas a little, which we do by reducing the 9/8 duty cycle. But if after that we're *still* going too fast, the PID loop will ask us to take stronger measures. What the current (August 2018) firmware does in response to that request is to start driving in the other direction (i.e. CCW, using mode 6/2). This we'll call reverse driving (reverse in the sense that it's opposite to the direction of the shaft rotation). Reverse driving brings in to play the 'brake to VCC' behavior of state 2.

The problem is that it's *impossible to decelerate gently drive using reverse driving*. This is because even 6/2 reverse driving with only the minimum possible 0% duty cycle is effectively (except for a diode voltage drop) a fully dynamic braking situation; as we increase the duty cycle percentage, this braking is only augmented by additional motor torque actively driving backwards. The transition from 0% forward driving to 0% reverse driving is thus a very abrupt one.

With asynchronous sign-magnitude H bridges such as the VNH5050E there is a third mode we can avail ourselves of to help ease this transition: mode 5/0, which we call proportional braking. In state 5, S2 and S4 are closed, causing dynamic braking; in state 0, all switches are open. A 0% duty cycle proportional braking thus is a (zero-torque) open circuit where we coast, just as we do in 0% 9/8 forward driving. At the other extreme, 100% proportional braking is full dynamic braking, just as 0% reverse driving is (again, ignoring the diode drop for the moment). In between, the balance of braking vs coasting is controlled by the PWM duty cycle. Thus, proportional braking provides a transitional regime between forward driving and reverse driving that we can exploit to respond smoothly to the PID loop's request for increasing amounts of deceleration.

The design challenge is to select the points in the control signal coming from the PID loop where we transition from one regime to the other. There are four points of relevance, which are as follows.


- S, the point of 100% reverse driving. It is clear that this occurs at the negative extreme of the PID control signal range.
- G, the point of 100% forward driving. This occurs at the positive extreme of the PID control signal range.
- Z, the point of 0% proportional braking and 0% forward driving.
- T, the point of 0% reverse driving and 100% proportional braking (we again ignore the diode for now).

As a general scheme, these four points are structurally related relative to each the order illustrated below. Forward driving is shown in green, proportional braking in blue, and reverse driving in red.

```
Clear[line, points, pointNames, plotPoints];
line[{x1_, y1_}, {x2_, y2_}] := Function[{x}, y1 + (y2 - y1) / (x2 - x1) * (x - x1)]
points = {S -> {-controlMax, yReverse100},
  T -> {controlProp100, yProp100}, Z -> {controlProp0, yProp0}, G -> {controlMax, yForward100}};
pointNames = {S, T, Z, G};
plotPoints[pointsN_, title_, yMax_, opts: OptionsPattern[]] := Module[
  {listPlot, range, reverse, braking, forward, yMax},
  yMax = pointsN[All, 2, 2] // Abs // Max;
  range[x_, p1_, p2_] := {x, (p1 /. pointsN)[[1]], (p2 /. pointsN)[[1]]};
  listPlot = ListPlot[{S -> "S", T -> "T", Z -> "Z", G -> "G"} /. pointsN,
    PlotRange -> {-yMax * 1.1, yMax * 1.1}, PlotStyle -> Black, AxesLabel -> {"control", yMax}, PlotLabel -> title, opts];
  reverse = Plot[Evaluate[line[S, T] /. pointsN][x], Evaluate[range[x, S, T]], PlotStyle -> Red, opts];
  braking = Plot[Evaluate[line[T, Z] /. pointsN][x], Evaluate[range[x, T, Z]], PlotStyle -> Blue, opts];
  forward = Plot[Evaluate[line[Z, G] /. pointsN][x], Evaluate[range[x, Z, G]], PlotStyle -> Darker[Green], opts];
  Show[listPlot, reverse, braking, forward]
]
```

```
examplePoints = {S -> {-32767, -100}, T -> {-16384, -70}, Z -> {8192, 30}, G -> {32767, 92}}
plotPoints[examplePoints, "Relative Structure of Points", "y"]

{S -> {-32767, -100}, T -> {-16384, -70}, Z -> {8192, 30}, G -> {32767, 92}}
```



Each time we run the PID control loop (currently every 50ms), we calculate the coordinates for the four points. From those, together with the control output from the PID loop, the location of control value on the curve is determined: its driving regime (forward, braking, reverse), and the relative fractional location within that regime. This latter fractional location is then used to proportionally scale the PWM value from a 0% to 100% duty cycle. The H bridge operating mode corresponding to the driving regime and the calculated duty cycle are then passed to the H bridge.

Our job here in this present document is to figure out how to locate the coordinates of the four points.

1.4. Roadmap

In what follows, the location of the abscissas of S & G and either Z or T in the PID control range will be determined a priori based on external considerations, and the location of the remaining point then determined from the other three using arguments of linearity. We will examine two different schemes, Scheme A and Scheme B.

2. Scheme A


2.1. Introduction: Points F, R, and T

The signature characteristic of the first scheme we examine is that it attempts to map in a linear manner the control signal coming from the PID loop to an “instantaneous equivalent applied DC voltage”. That is, for any H bridge switching mode and duty cycle, this scheme attempts to determine the applied voltage that would produce the same motor response in a dual system that lacked the H bridge but instead had a DC voltage source directly connected to the motor.

For some switching modes and duty cycle combinations, this equivalence is clear.

- At G (100% forward driving; e.g. 100% state 9), the motor is connected between battery and ground 100% of the time. The equivalent applied voltage is thus clearly the battery voltage.
- Similarly, at S (100% reverse driving), the equivalent applied voltage is the negative battery voltage
- At T (100% proportional braking, ie: 100% state 5), the motor leads are shorted together; the equivalent applied voltage is thus zero volts. The same is also true (ignoring the diode) for 0% reverse driving (e.g.: 100% state 2). Further, given the equivalent voltages of G and S just discussed, if the overall scheme is to map the control signal to equivalent voltage *linearly*, then it follows that T must have a control value of zero.

With these in hand, and recognizing that Z must lie between T and F, we can visualize Scheme A as having the following overall shape. The particular value of Z shown is for illustrative purposes only.



2.2. Point Z

The situation at Z is a bit trickier. Here (0% proportional braking/100% state 0, and 0% forward driving, 100% state 8) we have an open circuit, either literally, or nearly so (because of the diode). How can we map this open circuit into a closed circuit equivalence with some applied voltage?

The approach we examine is to recognize that in an open circuit, the current is zero (and thus the torque is zero as well, but we'll focus on current). We then ask ourselves, given the instantaneous state of the system (ie: some instantaneous velocity Ω), in the H-bridgeless-dual is there a DC voltage we could apply that would yield zero current? The answer is, messily, "sort of".

2.2.1. $v_g(t) = v_{app}(t)$ as an approximation to the "zero current voltage"

Recall (see <https://github.com/rgatkinson/RobotPhysics/blob/master/MotorPhysicsGearsInitialConditions.pdf>) the differential equation in our model that links applied DC voltage $v_{app}(t)$, the back EMF voltage $v_g(t)$, and the current $i(t)$ is the following:

```
(voltageEqn = diffEqns[{3}]) // prettyPrint
```

$$L \dot{i}(t) + R i(t) + v_g(t) = v_{app}(t)$$

An approximate argument goes something like this. If the current is zero, then the second term on the left disappears. Further, while initially there's a transient, the current soon stabilizes, and thus the first term will also disappear. That leaves us with the following:

```
voltageEqn /. {i[t] -> 0, i'[t] -> 0} // prettyPrint
```

$$v_g(t) = v_{app}(t)$$

That is, if the applied voltage matches the generated EMF, there should be no current. However intuitive that equivalence might be (or not), it is clearly only an approximation, for (at least) two reasons:

- The current isn't instantaneously zero, but instead must adjust through a transient
- After the voltage is dropped from the voltage that produced steady state velocity Ω to that of the back EMF, the velocity necessarily declines, which produces less back EMF. Thus, whatever voltage balance there was to achieve zero necessarily degrades.

The important question, then, is "how good of an approximation is it?" Let us examine that in some detail.

2.2.2. Analyzing $v_g(t) = v_{app}(t)$

It turns out that in the modelling work referenced above, we have a complete, closed form, analytic result for the current as a function of time in a motor with gearbox as the voltage steps at time zero from steady state applied voltage v_{app0} to some subsequent applied voltage $v_{app0} + \Delta v_{appConst}$. This result is also parameterized on the motor model parameters (R, L, etc), the motor load after the gearbox(if any), and an optional externally applied torque (if any). We illustrate that result here in its entirety in order that the reader may have some understanding of its complexity and completeness; however, in future we will refrain from directly illustrating such complex functions, as they are very difficult to appreciate in detail.

curStepGeneric

$$\left(\frac{2 K_e L \Delta v_{appConst} N (J a f t e r + J \eta N^2)}{\left(\left(-1 + e^{-\frac{t \left(B a f t e r L + J a f t e r R + B L \eta N^2 + J R \eta N^2 - \sqrt{-4 L (J a f t e r + J \eta N^2) (B a f t e r R + (K_e K_t + B R) \eta N^2) + (B a f t e r L + J a f t e r R + (B L + J R) \eta N^2)^2}}}{2 L (J a f t e r + J \eta N^2)} \right) \right) / (B a f t e r L + J a f t e r R + B L \eta N^2 + J R \eta N^2 - \sqrt{-4 L (J a f t e r + J \eta N^2) (B a f t e r R + (K_e K_t + B R) \eta N^2) + (B a f t e r L + J a f t e r R + (B L + J R) \eta N^2)^2}) \right) - \left(-1 + e^{-\frac{t \left(B a f t e r L + J a f t e r R + B L \eta N^2 + J R \eta N^2 - \sqrt{-4 L (J a f t e r + J \eta N^2) (B a f t e r R + (K_e K_t + B R) \eta N^2) + (B a f t e r L + J a f t e r R + (B L + J R) \eta N^2)^2}}}{2 L (J a f t e r + J \eta N^2)} \right) \right) / (B a f t e r L + J a f t e r R + B L \eta N^2 + J R \eta N^2 + \sqrt{-4 L (J a f t e r + J \eta N^2) (B a f t e r R + (K_e K_t + B R) \eta N^2) + (B a f t e r L + J a f t e r R + (B L + J R) \eta N^2)^2}) \right) \right) / \left(\sqrt{-4 L (J a f t e r + J \eta N^2) (B a f t e r R + (K_e K_t + B R) \eta N^2) + (B a f t e r L + J a f t e r R + (B L + J R) \eta N^2)^2} \right) + \left(\sqrt{-4 L (J a f t e r + J \eta N^2) (B a f t e r R + (K_e K_t + B R) \eta N^2) + (B a f t e r L + J a f t e r R + (B L + J R) \eta N^2)^2} \right) + \left(\Delta v_{appConst} \left(- \left(\left(-1 + e^{-\frac{t \left(B a f t e r L + J a f t e r R + B L \eta N^2 + J R \eta N^2 - \sqrt{-4 L (J a f t e r + J \eta N^2) (B a f t e r R + (K_e K_t + B R) \eta N^2) + (B a f t e r L + J a f t e r R + (B L + J R) \eta N^2)^2}}}{2 L (J a f t e r + J \eta N^2)} \right) \right) \right) L^2 (J a f t e r + J \eta N^2) / (B a f t e r L + \right.$$

[illegible]

$$\frac{B L \eta N^2 + J R \eta N^2 + \sqrt{\left(-4 L \left(J a f t e r + J \eta N^2\right) \left(B a f t e r R + \left(K e K t + B R\right) \eta N^2\right) + \left(B a f t e r L + J a f t e r R + \left(B L + J R\right) \eta N^2\right)^2\right)}}{\left(2 L \sqrt{\left(-4 \left(J a f t e r L + J L \eta N^2\right) \left(B a f t e r R + K e K t \eta N^2 + B R \eta N^2\right) + \left(B a f t e r L + J a f t e r R + B L \eta N^2 + J R \eta N^2\right)^2\right)} + B a f t e r v a p p \theta + N \left(B v a p p \theta \eta N - K e \tau a p p \theta\right)\right)} + \frac{B a f t e r R + \left(K e K t + B R\right) \eta N^2}{\left(2 L \sqrt{\left(-4 \left(J a f t e r L + J L \eta N^2\right) \left(B a f t e r R + K e K t \eta N^2 + B R \eta N^2\right) + \left(B a f t e r L + J a f t e r R + B L \eta N^2 + J R \eta N^2\right)^2\right)} + B a f t e r v a p p \theta + N \left(B v a p p \theta \eta N - K e \tau a p p \theta\right)\right)}$$

If we forget about the externally-applied torque, the result does become simpler, but is still significantly complex:

```
curStepGenericNoExternalTorque = curStepGeneric /. {tauapp0 -> 0, deltaappConst -> 0} // FullSimplify // Expand // FullSimplify
```

$$\frac{1}{B a f t e r R + \left(K e K t + B R\right) \eta N^2} \left(\left(v a p p \theta + \Delta v a p p C o n s t \right) \left(B a f t e r + B \eta N^2 \right) - e^{\frac{t \left(B a f t e r L - J a f t e r R + \left(B L + J R \right) \eta N^2 \right)}{2 L \left(J a f t e r + J \eta N^2 \right)}} \Delta v a p p C o n s t \left(B a f t e r + B \eta N^2 \right) \right. \\ \left. C o s h \left[\left(t \sqrt{\left(-4 L \left(J a f t e r + J \eta N^2\right) \left(B a f t e r R + \left(K e K t + B R\right) \eta N^2\right) + \left(B a f t e r L + J a f t e r R + \left(B L + J R \right) \eta N^2\right)^2\right)} \right) / \left(2 L \left(J a f t e r + J \eta N^2 \right) \right) \right] + \right. \\ \left. e^{\frac{t \left(B a f t e r L - J a f t e r R + \left(B L + J R \right) \eta N^2 \right)}{2 L \left(J a f t e r + J \eta N^2 \right)}} \Delta v a p p C o n s t \right. \\ \left. \left(B a f t e r \left(-B a f t e r L + J a f t e r R \right) + \left(2 J a f t e r K e K t - 2 B B a f t e r L + B a f t e r J R + B J a f t e r R \right) \eta N^2 + \left(2 J K e K t - B^2 L + B J R \right) \eta^2 N^4 \right) \right. \\ \left. S i n h \left[\left(t \sqrt{\left(-4 L \left(J a f t e r + J \eta N^2\right) \left(B a f t e r R + \left(K e K t + B R\right) \eta N^2\right) + \left(B a f t e r L + J a f t e r R + \left(B L + J R \right) \eta N^2\right)^2\right)} \right) / \left(2 L \left(J a f t e r + J \eta N^2 \right) \right) \right] \right) / \\ \left(\sqrt{\left(-4 L \left(J a f t e r + J \eta N^2\right) \left(B a f t e r R + \left(K e K t + B R\right) \eta N^2\right) + \left(B a f t e r L + J a f t e r R + \left(B L + J R \right) \eta N^2\right)^2\right)} \right)$$

Because of this complexity, in order to explore the effect on current of the drop of applied voltage to the back EMF we choose (initially, at least) to do so in the context of a concrete motor and particular loads. Here, we use an Andy Mark NeveRest 60 motor with either a 10kg or a 1kg flywheel attached:

```
aMotor = motorParameters["AM 60 A"];
aFlywheel = flywheel[Quantity[10, "kg"], Quantity[10, "cm"]] // siUnits;
aMotorWithFlywheel = addMotorLoad[aMotor, aFlywheel] // siUnits;
unitlessMotorWithFlywheel = aMotorWithFlywheel // siUnits // clearUnits;
aMotorWithFlywheel
```

$$\left\{ R \rightarrow \frac{33}{10} \text{ kg m}^2 / (\text{s}^2 \text{ A}^2), L \rightarrow \frac{347}{500000} \text{ kg m}^2 / (\text{s}^2 \text{ A}^2), N \rightarrow 60, \eta \rightarrow \frac{9}{10}, K e \rightarrow \frac{533}{30000} \text{ kg m}^2 / (\text{s}^2 \text{ A rad}), \right. \\ K t \rightarrow \frac{533}{30000} \text{ kg m}^2 / (\text{s}^2 \text{ A rad}), B \rightarrow \frac{11}{1080000} \text{ kg m}^2 / (\text{s rad}^2), J \rightarrow \frac{347}{10800000000} \text{ kg m}^2 / \text{rad}^2, \\ \left. J a f t e r \rightarrow \frac{1}{20} \text{ kg m}^2 / \text{rad}^2, B a f t e r \rightarrow 0 \text{ kg m}^2 / (\text{s rad}^2), \Delta \tau a p p C o n s t \rightarrow 0 \text{ kg m}^2 / (\text{s}^2 \text{ rad}) \right\}$$

We want to explore the current as a function of the steady state velocity Ω . To do that, we first construct a function that gives us the applied voltage that must be in effect to achieve the steady state velocity Ω .

```
vappFromVelocity[Omega_, motor_] := Module[{eqn},
  eqn = {theta'[ss0] /. (initialConditions /. Equal -> Rule) /. tauapp0 -> 0 /. motor} == Omega;
  vapp0 /. (uniqueSolve[eqn, vapp0] // First)]
vappFromVelocity[Omega, unitlessMotorWithFlywheel]
```

$$\frac{314339 \Omega}{266500}$$

We also construct a function that gives us the back EMF as a function of Ω .

```
vgFromVelocity[Omega_, motor_] := Module[{v1, eqn},
  v1 = vappFromVelocity[Omega, motor];
  {vg[ss0] /. (initialConditions /. Equal -> Rule) /. tauapp0 -> 0 /. motor} /. vapp0 -> v1]
vgFromVelocity[Omega, unitlessMotorWithFlywheel]
```

$$\frac{533 \Omega}{500}$$



This should, of course, come out as $K_e \Omega$, and, gladly, it does (remembering we're dealing here with post-gearbox K_e and Ω). With these in hand, we can define a function that pairs up the two voltages in question as an input to our general current result.

```
dropToVg[Ω_, motor_] := Module[{v1 = vappFromVelocity[Ω, motor], v2 = vgFromVelocity[Ω, motor]},
  Association[vapp0 → v1, ΔvappConst → v2 - v1]
]
dropToVg[2, unitlessMotorWithFlywheel]
```

```
{vapp0 →  $\frac{314\,339}{133\,250}$ , ΔvappConst →  $-\frac{121}{533}$ }
```

We put that all together, illustrating the result for an steady state velocity of $\Omega=1$ radians / second. We plot for 50ms, as that is the PID update interval in the FTC expansion hub firmware.

```
Clear[fn]
fn[Ω_] :=
  curStepGenericNoExternalTorque /. unitlessMotorWithFlywheel /. dropToVg[Ω, unitlessMotorWithFlywheel] // N // FullSimplify
Plot[fn[1], {t, 0, .050}, AxesLabel → {"Seconds", HoldForm[Amps]}, PlotRange → All, AxesOrigin → {0, 0}]
Plot[fn[1], {t, 0, .050}, AxesLabel → {"Seconds", HoldForm[Amps]}, PlotRange → {0, 0.005}, AxesOrigin → {0, 0}]
```



(Note: the shape of these plots is characteristic; the vertical axis scales linearly with initial velocity:)

```
fn[n Ω] / fn[Ω] // FullSimplify
```

```
n
```

It is worthy of note that the current never does, in fact, reach zero. Further, over the course of the PID update interval, the current is distinctly non-zero throughout, by the end recovering maybe some ~20% of its initial value. The approximation here to a “zero current voltage” does not appear to be a very good one.

2.2.3. “Mean zero current” as an approximation to the “zero current voltage”

The above analysis does suggest a different approximation: instead of dropping from the steady state applied voltage to the voltage of the back emf, to what voltage would we have to drop in order to have the mean (i.e: average) current over the PID interval be zero? We compute that here as a function of the steady state velocity Ω .


```
vappFromVelocity[Ω, {}]
currentFromΩAndDrop = curStepGenericNoExternalTorque /. vapp0 → vappFromVelocity[Ω, {}] // FullSimplify
```

$$\frac{(Bafter R + Ke Kt \eta N^2 + B R \eta N^2) \Omega}{Kt \eta N}$$

$$\frac{1}{Bafter R + (Ke Kt + B R) \eta N^2} \left((Bafter + B \eta N^2) \left(\Delta vappConst + \frac{Bafter R \Omega}{Kt \eta N} + \left(Ke + \frac{B R}{Kt} \right) N \Omega \right) - e^{-\frac{t (Bafter L + Jafter R + (B L + J R) \eta N^2)}{2 L (Jafter + J \eta N^2)}} \Delta vappConst (Bafter + B \eta N^2) \right. \\ \left. \cosh \left[\left(t \sqrt{-4 L (Jafter + J \eta N^2) (Bafter R + (Ke Kt + B R) \eta N^2) + (Bafter L + Jafter R + (B L + J R) \eta N^2)^2} \right) / (2 L (Jafter + J \eta N^2)) \right] + \right. \\ \left. e^{-\frac{t (Bafter L + Jafter R + (B L + J R) \eta N^2)}{2 L (Jafter + J \eta N^2)}} \Delta vappConst \right. \\ \left. (Bafter (-Bafter L + Jafter R) + (2 Jafter Ke Kt - 2 B Bafter L + Bafter J R + B Jafter R) \eta N^2 + (2 J Ke Kt - B^2 L + B J R) \eta^2 N^4) \right. \\ \left. \sinh \left[\left(t \sqrt{-4 L (Jafter + J \eta N^2) (Bafter R + (Ke Kt + B R) \eta N^2) + (Bafter L + Jafter R + (B L + J R) \eta N^2)^2} \right) / (2 L (Jafter + J \eta N^2)) \right] \right) / \\ \left(\sqrt{-4 L (Jafter + J \eta N^2) (Bafter R + (Ke Kt + B R) \eta N^2) + (Bafter L + Jafter R + (B L + J R) \eta N^2)^2} \right)$$

The value $\Delta vappConst$ in the above is the voltage drop we seek. We next symbolically compute the mean current over the PID interval.

```
meanCurrent = Integrate[currentFromΩAndDrop, {t, 0, pidPeriod}] / pidPeriod
```

$$\left(pidPeriod (Bafter + B \eta N^2) \left(\Delta vappConst + \frac{Bafter R \Omega}{Kt \eta N} + \left(Ke + \frac{B R}{Kt} \right) N \Omega \right) + \right. \\ \left(\Delta vappConst (Bafter (-Bafter L + Jafter R) + (2 Jafter Ke Kt - 2 B Bafter L + Bafter J R + B Jafter R) \eta N^2 + (2 J Ke Kt - B^2 L + B J R) \eta^2 N^4) \right. \\ \left. \left(\sqrt{-4 L (Jafter + J \eta N^2) (Bafter R + (Ke Kt + B R) \eta N^2) + (Bafter L + Jafter R + (B L + J R) \eta N^2)^2} + \right. \right. \\ \left. e^{-\frac{pidPeriod (Bafter L + Jafter R + (B L + J R) \eta N^2)}{2 L (Jafter + J \eta N^2)}} \left(-\sqrt{-4 L (Jafter + J \eta N^2) (Bafter R + (Ke Kt + B R) \eta N^2) + (Bafter L + Jafter R + (B L + J R) \eta N^2)^2} \right) \right. \\ \left. \cosh \left[\left(pidPeriod \sqrt{-4 L (Jafter + J \eta N^2) (Bafter R + (Ke Kt + B R) \eta N^2) + (Bafter L + Jafter R + (B L + J R) \eta N^2)^2} \right) / \right. \right. \\ \left. \left. (2 L (Jafter + J \eta N^2)) \right] - (Bafter L + Jafter R + (B L + J R) \eta N^2) \sinh \left[\left(pidPeriod \sqrt{-4 L (Jafter + J \eta N^2) \right. \right. \right. \\ \left. \left. (Bafter R + (Ke Kt + B R) \eta N^2) + (Bafter L + Jafter R + (B L + J R) \eta N^2)^2 \right) / (2 L (Jafter + J \eta N^2)) \right] \right) \right) / \\ \left(2 (Bafter R + (Ke Kt + B R) \eta N^2) \sqrt{-4 L (Jafter + J \eta N^2) (Bafter R + (Ke Kt + B R) \eta N^2) + (Bafter L + Jafter R + (B L + J R) \eta N^2)^2} \right) - \\ \frac{1}{2 (Bafter R + (Ke Kt + B R) \eta N^2)} \\ \Delta vappConst \\ (Bafter + B \eta N^2) \\ \left(Bafter L + Jafter R + (B L + J R) \eta N^2 + e^{-\frac{pidPeriod (Bafter L + Jafter R + (B L + J R) \eta N^2)}{2 L (Jafter + J \eta N^2)}} \left(- (Bafter L + Jafter R + (B L + J R) \eta N^2) \right. \right. \\ \left. \cosh \left[\left(pidPeriod \sqrt{-4 L (Jafter + J \eta N^2) (Bafter R + (Ke Kt + B R) \eta N^2) + (Bafter L + Jafter R + (B L + J R) \eta N^2)^2} \right) / \right. \right. \\ \left. \left. (2 L (Jafter + J \eta N^2)) \right] - \sqrt{-4 L (Jafter + J \eta N^2) (Bafter R + (Ke Kt + B R) \eta N^2) + (Bafter L + Jafter R + (B L + J R) \eta N^2)^2} \right) \\ \left. \sinh \left[\left(pidPeriod \sqrt{-4 L (Jafter + J \eta N^2) (Bafter R + (Ke Kt + B R) \eta N^2) + (Bafter L + Jafter R + (B L + J R) \eta N^2)^2} \right) / \right. \right. \\ \left. \left. (2 L (Jafter + J \eta N^2)) \right] \right) \right) / (pidPeriod (Bafter R + (Ke Kt + B R) \eta N^2))$$

We set that equal to zero, and solve for the voltage drop.

```
 $\Delta v_{app} \text{MeanZeroCurrent} = \Delta v_{app} \text{Const} /. \text{uniqueSolve}[\text{meanCurrent} == 0, \Delta v_{app} \text{Const}] // \text{FullSimplify} // \text{Expand} // \text{FullSimplify}$ 
```


$$\left(e^{\frac{\text{pidPeriod} (\text{Bafter} L - \text{Jafter} R) (B L + J R) \eta N^2}{2 L (\text{Jafter} + J \eta N^2)}} \text{pidPeriod} (\text{Bafter} + B \eta N^2) (\text{Bafter} R + (K_e K_t + B R) \eta N^2)^2 \right. \\ \left. \sqrt{((\text{Bafter} L - \text{Jafter} R)^2 - 2 (2 \text{Jafter} K_e K_t L + \text{Jafter} R (B L - J R) + \text{Bafter} L (-B L + J R)) \eta N^2 + (B^2 L^2 + J^2 R^2 - 2 J L (2 K_e K_t + B R)) \eta^2 N^4) \Omega} \right) / \\ \left(K_t \eta N \left(e^{\frac{\text{pidPeriod} (\text{Bafter} L - \text{Jafter} R) (B L + J R) \eta N^2}{2 L (\text{Jafter} + J \eta N^2)}} \sqrt{(-4 L (\text{Jafter} + J \eta N^2) (\text{Bafter} R + (K_e K_t + B R) \eta N^2) + (\text{Bafter} L + \text{Jafter} R + (B L + J R) \eta N^2)^2)} \right. \right. \\ \left. (\text{Bafter}^2 (L - \text{pidPeriod} R) + \text{Bafter} (-K_e K_t \text{pidPeriod} + 2 B (L - \text{pidPeriod} R)) \eta N^2 - \right. \\ \left. \eta N^2 (\text{Jafter} K_e K_t + (J K_e K_t + B (-B L + K_e K_t \text{pidPeriod} + B \text{pidPeriod} R)) \eta N^2) - (\text{Bafter}^2 L + (-\text{Jafter} K_e K_t + 2 B \text{Bafter} L) \eta N^2 + \right. \\ \left. (-J K_e K_t + B^2 L) \eta^2 N^4) \sqrt{(-4 L (\text{Jafter} + J \eta N^2) (\text{Bafter} R + (K_e K_t + B R) \eta N^2) + (\text{Bafter} L + \text{Jafter} R + (B L + J R) \eta N^2)^2)} \cosh[\right. \\ \left. (\text{pidPeriod} \sqrt{(-4 L (\text{Jafter} + J \eta N^2) (\text{Bafter} R + (K_e K_t + B R) \eta N^2) + (\text{Bafter} L + \text{Jafter} R + (B L + J R) \eta N^2)^2)}) / (2 L (\text{Jafter} + J \eta N^2)) \right] + \\ \left. (-\text{Bafter}^3 L^2 + \text{Bafter}^2 L (\text{Jafter} R + (-3 B L + J R) \eta N^2) + \text{Bafter} L \eta N^2 (\text{Jafter} (3 K_e K_t + 2 B R) + (3 J K_e K_t - 3 B^2 L + 2 B J R) \eta N^2) + \right. \\ \left. \eta N^2 (\text{Jafter}^2 K_e K_t R + \text{Jafter} (3 B K_e K_t L + 2 J K_e K_t R + B^2 L R) \eta N^2 + (3 B J K_e K_t L - B^3 L^2 + J^2 K_e K_t R + B^2 J L R) \eta^2 N^4) \right) \sinh[\\ \left. (\text{pidPeriod} \sqrt{(-4 L (\text{Jafter} + J \eta N^2) (\text{Bafter} R + (K_e K_t + B R) \eta N^2) + (\text{Bafter} L + \text{Jafter} R + (B L + J R) \eta N^2)^2)}) / (2 L (\text{Jafter} + J \eta N^2)) \right] \right) \right)$$

This is, ehm, a mouthful. In particular (among many things), it is sensitive to the load on the motor (*Jafter*, *Bafter*), which would significantly complicate its application in the FTC context (teams would need to measure their robots). Nevertheless, let's see this in action for the same example motor and load we looked at previously.

```
dropToMeanZeroCurrent[Ω_, motor_] := Module[{
  v1 = vappFromVelocity[Ω, motor],
  Δv2 = ΔvappMeanZeroCurrent /. motor /. pidPeriod → .050 /. Ω → Ω,
  Association[vapp0 → v1, τapp0 → 0, ΔvappConst → Δv2]
}]
dropToMeanZeroCurrent[1, unitlessMotorWithFlywheel] // N

<| vapp0 → 1.17951, τapp0 → 0., ΔvappConst → -0.132111 |>
```

```
Clear[fn]
fn[Ω_] :=
  curStepGenericNoExternalTorque /. unitlessMotorWithFlywheel /. dropToMeanZeroCurrent[Ω, unitlessMotorWithFlywheel] // N //
  FullSimplify
Plot[fn[1], {t, 0, .050}, AxesLabel → {"Seconds", HoldForm[Amps]}, PlotRange → All, AxesOrigin → {0, 0}]
```




At first glance, it would indeed seem to be plausible that we've done the math correctly, i.e., that the mean current over the PID interval is zero (that's just a sanity check: assuming we've set the equations up correctly, the solution must be correct, assuming Mathematica has not encountered a bug). Further, comparing this approximation of the zero current voltage to the previous $vg(t) = v_{app}(t)$ approximation, it is hard to argue that the mean zero current approximation is not both numerically and conceptually superior.

However, as was noted, its operational difficulties make its deployment a challenge. While perhaps some simplification of this result (e.g., perhaps ignoring the load?) might produce an approximation of intermediate goodness that was deployable, such an investigation has not yet been carried out. Accordingly, for the remainder of the discussion of Scheme A, we will work with the $vg(t) = v_{app}(t)$ approximation for the zero current voltage.

2.3. Calculating Point Z

Recall the structure of Scheme A:

```
plotScheme[schemeAExample, "Scheme A", "Instantaneous Equivalent Applied DC Volts", 12]
```



We have points R, T, and F in hand; we wish to compute point Z. The principle we will use is to locate Z on the line between T and F. Per the above, we take the ordinate of Z to be the current instantaneous back EMF; our job is to compute the abscissa of Z, which we will denote as controlProp0 (0% proportional braking occurs at the same location as 0% forward driving). The calculation is pretty straightforward.

```
schemeA = Union[schemeA, {yProp0 → vgFromVelocity[Ω, {}]}]
```

```
{controlProp100 → 0, yForward100 → yMax, yProp0 → Ke N Ω, yProp100 → 0, yReverse100 → -yMax}
```

```
schemeAPoints = points /. schemeA /. yMax → batteryVoltage
```

```
{S → {-controlMax, -batteryVoltage}, T → {0, 0}, Z → {controlProp0, Ke N Ω}, G → {controlMax, batteryVoltage}}
```

```
eqn = (line[T, G] /. schemeAPoints)[controlProp0] == (yProp0 /. schemeA)
```

$$\frac{\text{batteryVoltage controlProp0}}{\text{controlMax}} = \text{Ke N } \Omega$$

```
schemeASoln = uniqueSolve[eqn, controlProp0]
```

$$\left\{ \text{controlProp0} \rightarrow \frac{\text{controlMax Ke N } \Omega}{\text{batteryVoltage}} \right\}$$

Recall from the modelling paper that Ke is the *before*-gearbox electrical coupling constant; Ke N is simply the corresponding after-gearbox value. Thus, the abscissa of Z is a multiplication of the motor velocity (in radians / second) by a constant and a division by the battery voltage.

Let's explore a little how the location of Z varies with velocity Ω . The maximum velocity of the motor and load example we've been working with is about 10.2 radians / second when on a 12 volt battery:

```
ssSimple = ss /. {vapp0 → 0, tapp0 → 0, ΔtappConst → 0};
maxVelExample = ssVelAfter /. ssSimple /. unitlessMotorWithFlywheel /. ΔvappConst → 12 // N
```


```
10.1737
```

We explore how Z moves at various fractions of that maximum velocity (10%, 50%, and 90%):

```

schemeAPointsFromVelocityExample[Ω_] := schemeAPoints /. schemeASoln /.
{controlMax → 32767, batteryVoltage → 12} /. unitlessMotorWithFlywheel /. Ω → ΩΩ
plotZFromVel[Ω_] := Module[{title = "Scheme A Example, Ω=" <> ToString[Ω] <> " rad/s"},
  plotPoints[schemeAPointsFromVelocityExample[Ω], title, "Volts", ImageSize → 300]]
plotZFromVel[maxVelExample * #] & /@ {0.1, 0.5, 0.9} // Row

```



It is significant to note that as velocity increases, the amount of the control range allocated to the forward driving regime diminishes, ultimately to zero. In many (if not most) FTC motor control situations, the control loop spends the majority of its time in the forward driving regime. There seems to be a tension between these situations.

3. Scheme B

3.1. Introduction

We now discuss a second scheme, scheme B. This scheme is motivated by a few observations.


- In actual practice, the PID loop visits point Z far more often than it visits point T (or S). Thus, to help achieve reproducible results when driving, it is preferable to have any uncertainty or variability in the calculation of the locations of the points to be attributed to point T rather than point Z. That is, all other things being equal, it is preferable to have point Z determined a priori, and point T calculated at runtime.
- As was discussed at some length above, the calculation of an equivalent applied DC voltage that produces instantaneous zero current is an elusive proposition. Moreover the only feasibly-calculable approximation thereto that we have is demonstrably poor.
- In practice in FTC situations, the PID algorithm spends the majority of its time in the forward driving regime. Thus, attenuation of the dynamic range of that regime is of concern so as to avoid precision loss in calculations.
- The point Z is by its definition a point of zero *current*. It thus makes sense to at least explore an instantaneous-*current*-linearization (in contrast to the instantaneous-equivalent-applied-voltage-linearization previously examined).

Scheme B is the exploration of such an instantaneous-current-linearization. From these observations, we can deduce a few facts.

- The point Z is at the origin: its ordinate is zero, because the y axis is torque; its abscissa is zero because no other *fixed* abscissa is reasonable.
- The point G has an abscissa of controlMax, but its ordinate will be a function of the battery voltage, the back EMF, and the motor resistance.
- Similarly, point S has an abscissa of -controlMax, and an ordinate which is a function of the battery voltage, the back EMF, and motor resistance.
- The ordinate of point T is a function of the battery voltage, the back EMF, and motor resistance; its abscissa will be an interpolation between Z and S.

The general structure of the scheme B looks something like this:

```
schemeB = {controlProp0 → 0, yProp0 → 0, yReverse100 → -yMax, yForward100 → yMax / 2};
schemeBExample = schemeB ~Join~ {controlProp100 → -controlMax / 2, yProp100 → -yMax / 2}
plotScheme[schemeBExample, "Scheme B Example", "Current (Amps)", 3]
```

$$\left\{ \text{controlProp0} \rightarrow 0, \text{yProp0} \rightarrow 0, \text{yReverse100} \rightarrow -\frac{y_{\text{Max}}}{2}, \text{yForward100} \rightarrow \frac{y_{\text{Max}}}{2}, \text{controlProp100} \rightarrow -\frac{\text{controlMax}}{2}, \text{yProp100} \rightarrow -\frac{y_{\text{Max}}}{2} \right\}$$


Lets explore this in more detail.

3.2. Current as Function of State

We calculate the current in steady state (i.e. after the LR transient) as a function of the applied voltage and instantaneous current velocity. Note that as was the case in Schema A, this is only an approximation.

```
ssCurrentFromVoltageVelocity[v_, Ω_] := Module[{eqn, soln},
  eqn = voltageEqn /. {vg[t] → vgFromVelocity[Ω, {}], vapp[t] → v, i'[t] → 0};
  soln = uniqueSolve[eqn, i[t]] // FullSimplify;
  i[t] /. soln
]
```

```
ssCurrentFromVoltageVelocity[v, Ω]
```

$$\frac{v - K_e N \Omega}{R}$$

3.3. Locating Our Points

With that in hand, we can determine the ordinates of S and G from the relationship of current to the battery voltage and velocity; the ordinate of T is calculated similarly, but with respect to the zero applied volts of 100% state 5. Point Z is at the origin, as was mentioned.

```
schemeB = {
  yForward100 → ssCurrentFromVoltageVelocity[batteryVoltage, Ω],
  yReverse100 → ssCurrentFromVoltageVelocity[-batteryVoltage, Ω],
  yProp100 → ssCurrentFromVoltageVelocity[0, Ω],
  controlProp0 → 0, yProp0 → 0
}
```

$$\left\{ \text{yForward100} \rightarrow \frac{\text{batteryVoltage} - K_e N \Omega}{R}, \text{yReverse100} \rightarrow -\frac{\text{batteryVoltage} + K_e N \Omega}{R}, \text{yProp100} \rightarrow -\frac{K_e N \Omega}{R}, \text{controlProp0} \rightarrow 0, \text{yProp0} \rightarrow 0 \right\}$$

With that we can calculate the abscissa of T by fixing T on a line from S to Z.

```
schemeBPoints = points /. schemeB
```

$$\left\{ S \rightarrow \left\{ -\text{controlMax}, -\frac{\text{batteryVoltage} + K_e N \Omega}{R} \right\}, T \rightarrow \left\{ \text{controlProp100}, -\frac{K_e N \Omega}{R} \right\}, Z \rightarrow \{0, 0\}, G \rightarrow \left\{ \text{controlMax}, \frac{\text{batteryVoltage} - K_e N \Omega}{R} \right\} \right\}$$

```
eqn = (line[S, Z] /. schemeBPoints)[controlProp100] == yProp100 /. schemeB
schemeBSoln = uniqueSolve[eqn, controlProp100]
```

$$-\frac{\text{batteryVoltage} + K_e N \Omega}{R} + \frac{(\text{controlMax} + \text{controlProp100}) (\text{batteryVoltage} + K_e N \Omega)}{\text{controlMax} R} = -\frac{K_e N \Omega}{R}$$


$$\left\{ \text{controlProp100} \rightarrow -\frac{\text{controlMax} K_e N \Omega}{\text{batteryVoltage} + K_e N \Omega} \right\}$$

Note that this result is only dependent on the instantaneous velocity and battery voltage, the control range, and the motor's post-gearbox electrical coupling constant: conveniently, the uses of the motor's resistive constant R values have all cancelled out.

3.4. Examples

As we did with Scheme A, we illustrate Scheme B with our working motor example at a few different velocities.

```
schemeBPointsFromVelocityExample[Ω_] := schemeBPoints /. schemeBSoln /.
  {controlMax → 32767, batteryVoltage → 12} /. unitlessMotorWithFlywheel /. Ω → Ω
plotZFromVelB[Ω_] := Module[{title = "Scheme B Example, Ω=" <> ToString[Ω] <> " rad/s"},
  plotPoints[schemeBPointsFromVelocityExample[Ω], title, "Current (Amps)", ImageSize → 300]]
plotZFromVelB[maxVelExample * #] & /@ {0.1, 0.5, 0.9} // Row
```



These examples clearly illustrate the most significant tradeoff involved in the use of Scheme B: that we get a knee at the origin (note that we are still *continuous* everywhere). While the knee is relatively shallow at low velocities, it is more pronounced at larger ones. This is not *intrinsically* a problem: architecturally speaking, a PID loop can certainly deal with such a condition; while all other things being equal, we'd prefer not to have the knee, the reality is that all other things are *not* equal, as was discussed at the beginning of this section. From one point of view, this knee is the price we pay for moving the uncertainty and variability of the abscissa of Z that is present in Schema A to an uncertainty and variability of the abscissa of T in Scheme B. Further, the dynamic control range allocated to forward driving remains large, and insensitive to velocity.

These tradeoffs seem worthwhile.

4. To Do

- Calculate `controlReverse0`, dealing with the voltage drop across the catch diode. This doesn't effect the above results, but will slightly alter the rescaling of control signal to duty cycle when in the reverse driving regime.
- Can we improve on the approximation currently found in `ssCurrentFromVoltageVelocity[]`?

5. Revision History

- 2018.08.07 Initial version.
- 2018.08.13. Significant rewrite.