

Motor Physics, with Gears & Initial Conditions

Robert Atkinson, Maddy Nguyen, Samantha Hordyk, Cole Welch, and John Fraser
11 August 2018

An exploration of the physics of motors, in both symbolic and numeric form. A basic but sufficiently-complete parameterized model of a motor is presented. The differential equations governing that model are exhibited, and, from those, the (frequency domain) transfer functions of each of the model's variables is developed from first principles. Next, mathematical infrastructure for converting the response of general frequency domain functions input signals is developed, then applied to the transfer functions as a function of initial conditions, and the applied-voltage and external-applied-torque inputs. This yields time domain functions for each of the model's variables in terms of the (symbolic) value of the model's variables. With those time domain symbolic functions in hand, the steady-state response of the motor as a function of each of the model's parameters is explored.

1. Administrivia

We begin by loading some handy utilities.


```
Get[NotebookDirectory[] <> "Utilities.m"]
```

```
outputDirectory = FileNameJoin[{NotebookDirectory[], "MotorPhysicsGearsInitialConditions.Output"}] <> $PathnameSeparator  
CreateDirectory[outputDirectory] // Quiet;
```

```
C:\ftc\RobotPhysics\MotorPhysicsGearsInitialConditions.Output\
```

2. Motor Model

The following figure gives a conceptual overview of a DC motor (ref: <http://www.egr.msu.edu/classes/me451/jchoi/2008/>).



In addition to the the above, we also here incorporate into our model:

- an (optional) N:1 reducing gearbox with associated efficiency
- an autonomous time-dependent external torque

The state of the system can thus be described by the following variables.

- $v_{app}(t)$, the applied voltage (illustrated as $e_a(t)$ above)
- $i(t)$, the armature current
- $v_g(t)$, the back EMF ('g' for 'generator'; illustrated as $e_b(t)$ above)
- $\theta(t)$, the angular position of the shaft, before the gearbox
- $\omega(t)$, the angular velocity of the shaft, before the gearbox
- $\alpha(t)$, the angular acceleration of the shaft, before the gearbox
- $\tau(t)$, the torque driven by the armature current, before the gearbox

- $\tau_{app}(t)$, an autonomous externally applied torque, after the gearbox. Often, this is constant, and caused by gravitational action.

The parameter constants that characterize the electrical aspects of the system are the following

- K_e , the electromotive force constant
- K_t , the motor torque constant
- L , electric inductance
- R , electric resistance

The following parameter constants characterize the mechanical aspects of the system

- J , the moment of inertia of the armature etc before the gearbox
- J_{after} , the moment of inertia of any load or torque attached after the gearbox. May be (net) negative, depending on system; see $\tau_{AfterExpr}$ below.
- B , motor viscous drag constant
- B_{after} , viscous drag of any load attached after the gearbox
- η , the efficiency of the motor's gear box (our current model does not support direction-dependent gearbox efficiencies).
- N (capital nu), the reduction ratio of the gear box

We collect together basic information about our parameters.

```
parameterNames = {J, Jafter, B, Bafter, Ke, Kt, R, L, η, N}
parameterAssumptions = {};
parameterAssumptions = Union[parameterAssumptions, # ≥ 0 & /@ {J, B, t}];
(*allow J & B to be zero so we can explore simpler motor models*)
parameterAssumptions = Union[parameterAssumptions, # > 0 & /@ {R, L, Ke, Kt, N, η}];
parameterAssumptions = Union[parameterAssumptions, # ∈ ℝ & /@ {Jafter, Bafter, , vapp0, τapp0, ΔvappConst, ΔτappConst}];
parameterAssumptions = Union[parameterAssumptions, #[_] ∈ ℝ & /@ {Δvapp, i, vg, θ, ω, α, τ, Δτapp}]
```

```
{J, Jafter, B, Bafter, Ke, Kt, R, L, η, N}
```

```
{Bafter ∈ ℝ, Jafter ∈ ℝ, Null ∈ ℝ, vapp0 ∈ ℝ, ΔvappConst ∈ ℝ, ΔτappConst ∈ ℝ, τapp0 ∈ ℝ, i[_] ∈ ℝ, vg[_] ∈ ℝ, α[_] ∈ ℝ,
Δvapp[_] ∈ ℝ, Δτapp[_] ∈ ℝ, θ[_] ∈ ℝ, τ[_] ∈ ℝ, ω[_] ∈ ℝ, Ke > 0, Kt > 0, L > 0, R > 0, η > 0, N > 0, B ≥ 0, J ≥ 0, t ≥ 0}
```

3. Differential Equations

3.1. Differential Equation Development

3.1.1. On Gearboxes and Torques

Our modelled system contains a gearbox with (reducing) gear ratio N and efficiency η . On either side of the gearbox we may have loads, drags, and driving torques, the latter being electrically generated on one side and externally applied on the other. In order to be able to write the torque differential equation for the system as a whole, we need to be able to take these two sets of torques as they are found in their respective *separate* places in the system and rewrite them both as they are *effectively* felt at some *one* point in the system. Commonly, in mechanical engineering this one point is chosen to be just before the gearbox, and the loads / torques after the gearbox are said to be “reflected” to create corresponding equivalent reflected loads and torques at the before-gearbox location. Here, we develop the mathematics of carrying out this reflection.

Just a couple of principles are involved:

- State related to angular position (position, velocity, acceleration, etc.) on the output side is $1/N$ th that of on the input side.
- Torque on the output side is $N\eta$ times that of on the input side.

```
Clear[reflectAngle, reflectTorque]
reflectAngle[expr_, out_, in_, t_] := expr /. {out[t] → in[t] / N, Derivative[n_][out][t] → Derivative[n][in][t] / N}
reflectAngle[expr_] := reflectAngle[expr, θafter, θ, t]
reflectTorque[τout_] := τout / (N η)
```

```
reflectTorque[ΔτappConst + Jafter θafter''[t] + Bafter θafter'[t]] // Expand
% // reflectAngle
```

$$\frac{\Delta \tau_{app} \text{Const}}{\eta N} + \frac{B_{after} \theta_{after}'[t]}{\eta N} + \frac{J_{after} \theta_{after}''[t]}{\eta N}$$

$$\frac{\Delta \tau_{app} \text{Const}}{\eta N} + \frac{B_{after} \theta'[t]}{\eta N^2} + \frac{J_{after} \theta''[t]}{\eta N^2}$$

3.1.2. Differential Equation System

With that understanding on how to reflect torques, we can now develop the differential equations that govern the system.

The motor torque is proportional to current.

```
ktEqn = τ[t] == Kt i[t]
```

```
τ[t] == Kt i[t]
```

The back EMF is proportional to velocity.

```
backEmfEqn = vg[t] == Ke θ'[t]
```

```
vg[t] == Ke θ'[t]
```

The after-gearbox angular position is a simple ratio of the before-gearbox position. This also applies to the velocity and acceleration.

```
θafterEqns = Derivative[#][θafter][t] == Derivative[#][θ][t] / N & /@ Range[0, 2]
```

$$\left\{ \theta_{after}[t] = \frac{\theta[t]}{N}, \theta_{after}'[t] = \frac{\theta'[t]}{N}, \theta_{after}''[t] = \frac{\theta''[t]}{N} \right\}$$

Before the gearbox, the electrically driven motor torque is reduced by viscous drag and the inertia of the motor.

```
τBeforeExpr = τ[t] + -B θ'[t] - J θ''[t]
```

```
τ[t] - B θ'[t] - J θ''[t]
```

After the gearbox, we have an autonomous, time-dependant externally applied torque, a viscous drag, and an internal load and/or an acceleration-dependent external torque.

```
τAfterExpr = τapp[t] - Bafter θafter'[t] - Jafter θafter''[t]
```

```
τapp[t] - Bafter θafter'[t] - Jafter θafter''[t]
```

As all the torque must be accounted for, the sum of the before torques and the reflected after torques must be zero.

```
τEqn = τBeforeExpr + (τAfterExpr // reflectTorque // reflectAngle) == 0;
τEqn = Collect[τEqn, {θ[t], θ'[t], θ''[t]}]
```

$$\tau[t] + \frac{\tau_{app}[t]}{\eta N} + \left(-B - \frac{B_{after}}{\eta N^2}\right) \theta'[t] + \left(-J - \frac{J_{after}}{\eta N^2}\right) \theta''[t] = 0$$

The voltage drop across the resistance, induction and back EMF must equal the applied voltage (due to one of Kirchhoff's laws).

```
voltageEqn = vapp[t] == R i[t] + L i'[t] + vg[t]
```

```
vapp[t] == R i[t] + vg[t] + L i'[t]
```

We put it all together.

```
diffEqns = {ktEqn, backEmfEqn, voltageEqn, tauEqn} ~Join~ thetaAfterEqns // FullSimplify;
diffEqns // prettyPrint
```

```
Kt i(t) = tau(t)
vg(t) = Ke theta'(t)
R i(t) + vg(t) + L i'(t) = vapp(t)

$$\frac{\eta \tau(t) N^2 + \text{rapp}(t) N - (B \eta N^2 + \text{Bafter}) \theta(t) - (J \eta N^2 + \text{Jafter}) \theta'(t)}{\eta N} = 0$$

thetaAfter(t) =  $\frac{\theta(t)}{N}$ 
thetaAfter'(t) =  $\frac{\theta'(t)}{N}$ 
thetaAfter''(t) =  $\frac{\theta''(t)}{N}$ 
```

3.2. Initial Conditions

3.2.1. Steady state before t==0

We will assume that initially, the motor is in steady state at some velocity because of the application of some particular applied voltage and external torque. In steady state, the derivative of current and velocity are zero.

```
someInitialConditions = {
  vapp[ss0] == vapp0,
  tauapp[ss0] == tauapp0,
  theta[ss0] == 0, (* arbitrary *)
  theta''[ss0] == 0, (* steady state condition *)
  i'[ss0] == 0 (* steady state condition *)
}

{vapp[ss0] == vapp0, tauapp[ss0] == tauapp0, theta[ss0] == 0, theta''[ss0] == 0, i'[ss0] == 0}
```

From those, we can calculate the values of the remainder of the state.

```
diffEqFunctionsOfTime = Reap[Scan[If[MatchQ[#, _[t]], Sow[#]] &, diffEqns, Infinity]][[2, 1]] // Union
others = Complement[diffEqFunctionsOfTime /. t -> ss0, someInitialConditions][[All, 1]];
eqnsSubsSs0 = diffEqns /. t -> ss0 /. (someInitialConditions /. Equal -> Rule);
(solvedSs0 = (uniqueSolve[eqnsSubsSs0, others] /. Rule -> Equal));
(initialConditions = solvedSs0 ~Join~ someInitialConditions) // prettyPrint

{i[t], vapp[t], vg[t], theta[t], thetaAfter[t], tau[t], tauapp[t], i'[t], theta'[t], thetaAfter'[t], theta''[t], thetaAfter''[t]}
```

```
i(ss0) ==  $-\frac{B \text{vapp0} \eta N^2 + \text{Ke} \text{rapp0} N - \text{Bafter} \text{vapp0}}{\text{Ke} Kt \eta N^2 + B R \eta N^2 + \text{Bafter} R}$ 
vg(ss0) ==  $\frac{\text{Ke} N (Kt \text{vapp0} \eta N + R \text{rapp0})}{\text{Ke} Kt \eta N^2 + B R \eta N^2 + \text{Bafter} R}$ 
thetaAfter(ss0) == 0
tau(ss0) ==  $\frac{Kt (B \text{vapp0} \eta N^2 - \text{Ke} \text{rapp0} N + \text{Bafter} \text{vapp0})}{\text{Ke} Kt \eta N^2 + B R \eta N^2 + \text{Bafter} R}$ 
theta'(ss0) ==  $\frac{N (Kt \text{vapp0} \eta N + R \text{rapp0})}{\text{Ke} Kt \eta N^2 + B R \eta N^2 + \text{Bafter} R}$ 
thetaAfter'(ss0) ==  $-\frac{Kt \text{vapp0} \eta N - R \text{rapp0}}{\text{Ke} Kt \eta N^2 + B R \eta N^2 + \text{Bafter} R}$ 
thetaAfter''(ss0) == 0
vapp(ss0) == vapp0
tauapp(ss0) == tauapp0
theta(ss0) == 0
theta'(ss0) == 0
i'(ss0) == 0
```

3.2.2. Initial conditions at t==0

Recall our differential equations.

```
diffEqns // prettyPrint

Kt i(t) = τ(t)
vg(t) = Ke θ'(t)
R i(t) + vg(t) + L i'(t) = vapp(t)

$$\frac{\eta \tau(t) N^2 + \tau_{app}(t) N - (B \eta N^2 + B_{after}) \theta(t) - (J \eta N^2 + J_{after}) \theta'(t)}{\eta N} = 0$$


$$\theta_{after}(t) = \frac{\theta(t)}{N}$$


$$\theta'_{after}(t) = \frac{\theta'(t)}{N}$$


$$\theta''_{after}(t) = \frac{\theta''(t)}{N}$$

```

In order to be able to use Laplace transforms, we need to adjust these so that the input values are zero at t==0. To do so, we substitute in a bias for both inputs.

```
(ΔdiffEqns = diffEqns /. {vapp[t] → vapp0 + Δvapp[t], τapp[t] → τapp0 + Δτapp[t]}) // prettyPrint

Kt i(t) = τ(t)
vg(t) = Ke θ'(t)
R i(t) + vg(t) + L i'(t) = vapp0 + Δvapp(t)

$$\frac{\eta \tau(t) N^2 + (\tau_{app0} + \Delta\tau_{app}(t)) N - (B \eta N^2 + B_{after}) \theta(t) - (J \eta N^2 + J_{after}) \theta'(t)}{\eta N} = 0$$


$$\theta_{after}(t) = \frac{\theta(t)}{N}$$


$$\theta'_{after}(t) = \frac{\theta'(t)}{N}$$


$$\theta''_{after}(t) = \frac{\theta''(t)}{N}$$

```

We now set up the initial conditions applicable for t >= 0.

```
initialConditionsRules = initialConditions /. Equal → Rule;
ΔsomeInitialConditions = {
  Δvapp[0] == ΔvappConst,
  Δτapp[0] == ΔτappConst,
  θ[0] == 0,
  i[0] == i[ss0] /. initialConditionsRules /. ss0 → 0,
  θ'[0] == θ'[ss0] /. initialConditionsRules /. ss0 → 0
}

{Δvapp[0] == ΔvappConst, Δτapp[0] == ΔτappConst, θ[0] == 0,
 i[0] == -
$$\frac{B_{after} v_{app0} - B v_{app0} \eta N^2 + Ke N \tau_{app0}}{B_{after} R + Ke Kt \eta N^2 + B R \eta N^2}, \theta'[0] == \frac{N (Kt v_{app0} \eta N + R \tau_{app0})}{B_{after} R + Ke Kt \eta N^2 + B R \eta N^2}}$$

```

```

ΔDiffEqFunctionsOfTime = Reap[Scan[If[MatchQ[#, _[t]], Sow[#]] &, ΔdiffEqns, Infinity]][[2, 1]] // Union
Δothers = Complement[ΔDiffEqFunctionsOfTime /. t → 0, ΔsomeInitialConditions[All, 1]]]
ΔeqnsSubsKnownInitial = ΔdiffEqns /. t → 0 /. (ΔsomeInitialConditions /. Equal → Rule);
ΔsolvedInitialConditions = (uniqueSolve[ΔeqnsSubsKnownInitial, Δothers] /. Rule → Equal);
(ΔinitialConditions = ΔsolvedInitialConditions ~Join~ ΔsomeInitialConditions) // prettyPrint

```

```
{i[t], vg[t], Δvapp[t], Δτapp[t], θ[t], θafter[t], τ[t], i'[t], θ'[t], θafter'[t], θ''[t], θafter''[t]}
```

```
{vg[0], θafter[0], τ[0], i'[0], θafter'[0], θ''[0], θafter''[0]}
```

$$vg(0) = \frac{Ke N (Kt vapp0 \eta N + R \tauapp0)}{Ke Kt \eta N^2 + B R \eta N^2 + Bafter R}$$

$$\thetaafter(0) = 0$$

$$\tau(0) = -\frac{B Kt vapp0 \eta N^2 + Ke Kt \tauapp0 N - Bafter Kt vapp0}{Ke Kt \eta N^2 + B R \eta N^2 + Bafter R}$$

$$i'(0) = \frac{\Delta vappConst}{L}$$

$$\thetaafter'(0) = -\frac{Kt vapp0 \eta N - R \tauapp0}{Ke Kt \eta N^2 + B R \eta N^2 + Bafter R}$$

$$\theta''(0) = \frac{\Delta \tauappConst N}{J \eta N^2 + Jafter}$$

$$\thetaafter''(0) = \frac{\Delta \tauappConst}{J \eta N^2 + Jafter}$$

$$\Delta vapp(0) = \Delta vappConst$$

$$\Delta \tauapp(0) = \Delta \tauappConst$$

$$\theta(0) = 0$$

$$i(0) = -\frac{B vapp0 \eta N^2 + Ke \tauapp0 N - Bafter vapp0}{Ke Kt \eta N^2 + B R \eta N^2 + Bafter R}$$

$$\theta'(0) = \frac{N (Kt vapp0 \eta N + R \tauapp0)}{Ke Kt \eta N^2 + B R \eta N^2 + Bafter R}$$

4. Motor Parameters

4.1. Motor Parameter Units

We develop infrastructure regarding the units in which each of our parameters are specified. This will help us sanity check the structure of our equations, as they must be internally consistent in their use of units (e.g.: no velocities added to masses, etc).

A note on radians as a unit. The Bureau International des Poids et Mesures, the intergovernmental organization that standardizes the SI, officially refers to a radian as “a special name is given to the unit one, in order to facilitate the identification of the quantity involved” (Ref: <https://www.bipm.org/en/publications/si-brochure/section2-2-3.html>). With that as background, we have here chosen to include radians in all places in which they logically belong, as doing so provides valuable insight regarding the quantities being manipulated. Thus, for example, you will find here torque having units of N m / radian instead of N m, and the like. As, from the SI perspective, this amounts to multiplying or dividing by the unit one, no semantic change is effected.

Aside: in the opinion of one of the authors (Atkinson), that “a special name given to the unit one” is in any way at all different than a first-class “unit” is a perspective reasonable only if one artificially and myopically limits one’s worldview to contain only the seven units whose dimension is necessarily *experimentally* calibrated as opposed to units whose dimension is mathematically or otherwise calibrated, such as radian, becquerel, motor encoder ticks, etc. And the world is chock-full of the latter. But this is a discussion for another time. See also Jacques E. Romain, “Angle as a Fourth Fundamental Quantity”, Journal of Research of the National Bureau of Standards-B. Mathematics and Mathematical Physics, Vol. 66B, No. 3, July- September 1962.

```

radiansUnits = parseUnit["Radians"];
secondsUnits = parseUnit["Seconds"];
siAngularPositionUnits = radiansUnits;
siAngularVelocityUnits = siAngularPositionUnits / secondsUnits;
siAngularAccelerationUnits = siAngularVelocityUnits / secondsUnits;
siTorqueUnits = parseUnit["N m"] / radiansUnits;
siAngularViscousDragUnits = siTorqueUnits / siAngularVelocityUnits;
siAngularInertialUnits = siTorqueUnits / siAngularAccelerationUnits;

parameterUnits = {
  R → parseUnit["Ohms"],
  L → parseUnit["Henrys"],
  i[t] → parseUnit["Amperes"],
  i'[t] → parseUnit["Amperes"] / secondsUnits,
  i''[t] → parseUnit["Amperes"] / secondsUnits / secondsUnits,
  Δvapp[t] → parseUnit["Volts"],
  ΔvappConst → parseUnit["Volts"],
  vg[t] → parseUnit["Volts"],
  vappθ → parseUnit["Volts"],
  vappθ → parseUnit["Volts"],

  J → siAngularInertialUnits,
  Jafter → siAngularInertialUnits,
  B → siAngularViscousDragUnits,
  Bafter → siAngularViscousDragUnits,

  Ke → parseUnit["Seconds Volts / Radians"],
  Kt → parseUnit["Newtons Meters"] / parseUnit["Amperes"] / radiansUnits,

  θ[t] → siAngularPositionUnits, θafter[t] → siAngularPositionUnits,
  θ'[t] → siAngularVelocityUnits, θafter'[t] → siAngularPositionUnits,
  θ''[t] → siAngularAccelerationUnits, θafter''[t] → siAngularPositionUnits,
  ω[t] → siAngularVelocityUnits,
  α[t] → siAngularAccelerationUnits,

  τ[t] → siTorqueUnits,
  Δτapp[t] → siTorqueUnits,
  ΔτappConst → siTorqueUnits,
  τappθ → siTorqueUnits,
  τappθ → siTorqueUnits,

  N → parseUnit["DimensionlessUnit"],
  η → parseUnit["DimensionlessUnit"]
} // Association
unitsToQuantities[units_] := Module[{rules, makeQuantity},
  rules = Normal[units];
  makeQuantity = Function[{param, unit}, Quantity[param, unit]];
  ({#[1] → makeQuantity @@ # & /@ rules} // Association)
]
parameterQuantities = unitsToQuantities[parameterUnits]

```

$$\left(\begin{array}{l}
 R \rightarrow \text{Ohms}, L \rightarrow \text{Henrys}, i[t] \rightarrow \frac{\text{Amperes}}{\text{Seconds}}, i'[t] \rightarrow \frac{\text{Amperes}}{\text{Seconds}^2}, i''[t] \rightarrow \frac{\text{Amperes}}{\text{Seconds}^2}, \Delta v_{app}[t] \rightarrow \text{Volts}, \Delta v_{app} \text{Const} \rightarrow \text{Volts}, \\
 v_g[t] \rightarrow \text{Volts}, v_{app\theta} \rightarrow \text{Volts}, J \rightarrow \frac{\text{Meters Newtons Seconds}^2}{\text{Radians}^2}, J_{after} \rightarrow \frac{\text{Meters Newtons Seconds}^2}{\text{Radians}^2}, B \rightarrow \frac{\text{Meters Newtons Seconds}}{\text{Radians}^2}, \\
 B_{after} \rightarrow \frac{\text{Meters Newtons Seconds}}{\text{Radians}^2}, K_e \rightarrow \frac{\text{Seconds Volts}}{\text{Radians}}, K_t \rightarrow \frac{\text{Meters Newtons}}{\text{Amperes Radians}}, \theta[t] \rightarrow \text{Radians}, \theta_{after}[t] \rightarrow \text{Radians}, \theta'[t] \rightarrow \frac{\text{Radians}}{\text{Seconds}}, \\
 \theta_{after}'[t] \rightarrow \text{Radians}, \theta''[t] \rightarrow \frac{\text{Radians}}{\text{Seconds}^2}, \theta_{after}''[t] \rightarrow \text{Radians}, \omega[t] \rightarrow \frac{\text{Radians}}{\text{Seconds}}, \alpha[t] \rightarrow \frac{\text{Radians}}{\text{Seconds}^2}, \tau[t] \rightarrow \frac{\text{Meters Newtons}}{\text{Radians}}, \\
 \Delta \tau_{app}[t] \rightarrow \frac{\text{Meters Newtons}}{\text{Radians}}, \Delta \tau_{app} \text{Const} \rightarrow \frac{\text{Meters Newtons}}{\text{Radians}}, \tau_{app\theta} \rightarrow \frac{\text{Meters Newtons}}{\text{Radians}}, N \rightarrow \text{DimensionlessUnit}, \eta \rightarrow \text{DimensionlessUnit} \end{array} \right)$$

```

⟨ | R → R Ω , L → L H , i[t] → i[t] A , i'[t] → i'[t] A/s , i''[t] → i''[t] A/s² , Δvapp[t] → Δvapp[t] V , ΔvappConst → ΔvappConst V ,
vg[t] → vg[t] V , vapp0 → vapp0 V , J → J ms⁴N/rad⁴ , Jafter → Jafter ms⁴N/rad⁴ , B → B msN/rad² , Bafter → Bafter msN/rad² ,
Ke → Ke sV/rad , Kt → Kt mN/(Arad) , θ[t] → θ[t] rad , θafter[t] → θafter[t] rad , θ'[t] → θ'[t] rad/s ,
θafter'[t] → θafter'[t] rad , θ''[t] → θ''[t] rad/s² , θafter''[t] → θafter''[t] rad , ω[t] → ω[t] rad/s , α[t] → α[t] rad/s² ,
τ[t] → τ[t] mN/rad , Δτapp[t] → Δτapp[t] mN/rad , ΔτappConst → ΔτappConst mN/rad , τapp0 → τapp0 mN/rad , N → N , η → η | ⟩

```

Let's look and see if our units are consistent:

```
ΔdiffEqns /. parameterQuantities // ColumnForm
```

```

Kt i[t] mN/rad == τ[t] mN/rad
vg[t] V == Ke θ'[t] V
(R i[t] + vg[t] + L i'[t]) A Ω == (vapp0 + Δvapp[t]) V
N (τapp0 + Δτapp[t]) + η N² τ[t] - (Bafter + B η N²) θ'[t] - (Jafter + J η N²) θ''[t] mN/rad == 0
θafter[t] rad ==  $\frac{\theta[t]}{N}$  rad
θafter'[t] rad ==  $\frac{\theta'[t]}{N}$  rad/s
θafter''[t] rad ==  $\frac{\theta''[t]}{N}$  rad/s²

```

Huzzah! They check out (if the units were incompatible, Mathematica would've told us). How about our initial conditions?


```
initialConditions /. parameterQuantities // siUnits // simplifyUnits // ColumnForm
ΔinitialConditions /. parameterQuantities // siUnits // simplifyUnits // ColumnForm
```

$$\begin{aligned} i[ss0] &= -\frac{-B \text{after} \text{vapp}0 - B \text{vapp}0 \eta \text{N}^2 + K_e \text{N} \tau_{\text{app}0}}{B \text{after} R + K_e K_t \eta \text{N}^2 + B R \eta \text{N}^2} \text{A} \\ \text{vg}[ss0] &= \frac{K_e \text{N} (K_t \text{vapp}0 \eta \text{N} + R \tau_{\text{app}0})}{B \text{after} R + K_e K_t \eta \text{N}^2 + B R \eta \text{N}^2} \text{V} \\ \theta_{\text{after}}[ss0] &= 0 \\ \tau[ss0] &= \frac{K_t (B \text{after} \text{vapp}0 + B \text{vapp}0 \eta \text{N}^2 - K_e \text{N} \tau_{\text{app}0})}{B \text{after} R + K_e K_t \eta \text{N}^2 + B R \eta \text{N}^2} \text{kg m}^2 / (\text{s}^2 \text{rad}) \\ \theta'[ss0] &= \frac{\text{N} (K_t \text{vapp}0 \eta \text{N} + R \tau_{\text{app}0})}{B \text{after} R + K_e K_t \eta \text{N}^2 + B R \eta \text{N}^2} \text{rad/s} \\ \theta_{\text{after}}'[ss0] &= -\frac{-K_t \text{vapp}0 \eta \text{N} - R \tau_{\text{app}0}}{B \text{after} R + K_e K_t \eta \text{N}^2 + B R \eta \text{N}^2} \text{rad/s} \\ \theta_{\text{after}}''[ss0] &= 0 \\ \text{vapp}[ss0] &= \text{vapp}0 \text{V} \\ \tau_{\text{app}}[ss0] &= \tau_{\text{app}0} \text{kg m}^2 / (\text{s}^2 \text{rad}) \\ \theta[ss0] &= 0 \\ \theta''[ss0] &= 0 \\ i'[ss0] &= 0 \end{aligned}$$

$$\begin{aligned} \text{vg}[0] &= \frac{K_e \text{N} (K_t \text{vapp}0 \eta \text{N} + R \tau_{\text{app}0})}{B \text{after} R + K_e K_t \eta \text{N}^2 + B R \eta \text{N}^2} \text{V} \\ \theta_{\text{after}}[0] &= 0 \\ \tau[0] &= -\frac{-B \text{after} K_t \text{vapp}0 - B K_t \text{vapp}0 \eta \text{N}^2 + K_e K_t \text{N} \tau_{\text{app}0}}{B \text{after} R + K_e K_t \eta \text{N}^2 + B R \eta \text{N}^2} \text{kg m}^2 / (\text{s}^2 \text{rad}) \\ i'[0] &= \frac{\Delta \text{vappConst}}{L} \text{A/s} \\ \theta_{\text{after}}'[0] &= -\frac{-K_t \text{vapp}0 \eta \text{N} - R \tau_{\text{app}0}}{B \text{after} R + K_e K_t \eta \text{N}^2 + B R \eta \text{N}^2} \text{rad/s} \\ \theta''[0] &= \frac{\Delta \tau_{\text{appConst}} \text{N}}{J \text{after} + J \eta \text{N}^2} \text{rad/s}^2 \\ \theta_{\text{after}}''[0] &= \frac{\Delta \tau_{\text{appConst}}}{J \text{after} + J \eta \text{N}^2} \text{rad/s}^2 \\ \Delta \text{vapp}[0] &= \Delta \text{vappConst} \text{V} \\ \Delta \tau_{\text{app}}[0] &= \Delta \tau_{\text{appConst}} \text{kg m}^2 / (\text{s}^2 \text{rad}) \\ \theta[0] &= 0 \\ i[0] &= -\frac{-B \text{after} \text{vapp}0 - B \text{vapp}0 \eta \text{N}^2 + K_e \text{N} \tau_{\text{app}0}}{B \text{after} R + K_e K_t \eta \text{N}^2 + B R \eta \text{N}^2} \text{A} \\ \theta'[0] &= \frac{\text{N} (K_t \text{vapp}0 \eta \text{N} + R \tau_{\text{app}0})}{B \text{after} R + K_e K_t \eta \text{N}^2 + B R \eta \text{N}^2} \text{rad/s} \end{aligned}$$

Good. We save information regarding our parameters for later use.

```
saveDefinitions[outputDirectory <> "ParametersUnitsAndAssumptions.m",
{parameterUnits, parameterQuantities, parameterAssumptions, radiansUnits, secondsUnits, siAngularPositionUnits,
siAngularVelocityUnits, siAngularAccelerationUnits, siTorqueUnits, siAngularViscousDragUnits, siAngularInertialUnits}]
```

4.2. On Ke & Kt

A brief digression. It is often remarked that, in compatible units, Ke & Kt have the same magnitude and sign. However, while often true, this relation does not *always* hold.

Recall our differential equations:

```
diffEqns // ColumnForm
```

```
Kt i[t] == τ[t]
vg[t] == Ke θ'[t]
R i[t] + vg[t] + L i'[t] == vapp[t]

$$\frac{\eta N^2 \tau[t] + N \tau \text{app}[t] - (\text{Bafter} \cdot B \eta N^2) \theta'[t] - (\text{Jafter} \cdot J \eta N^2) \theta''[t]}{\eta N} == 0$$


$$\theta \text{after}[t] == \frac{\theta[t]}{N}$$


$$\theta \text{after}'[t] == \frac{\theta'[t]}{N}$$


$$\theta \text{after}''[t] == \frac{\theta''[t]}{N}$$

```

In steady state, the current and speed are constant.

```
(ssEqns = diffEqns /. {i'[t] → 0, θ''[t] → 0}) // ColumnForm
(ssEqns = Eliminate[ssEqns, {vg[t], B, θafter[t], θafter'[t], θafter''[t], τapp[t]}]) // ColumnForm
(ssEqns = ssEqns[[1 ;; 2]]) // ColumnForm
```

```
Kt i[t] == τ[t]
vg[t] == Ke θ'[t]
R i[t] + vg[t] == vapp[t]

$$\frac{\eta N^2 \tau[t] + N \tau \text{app}[t] - (\text{Bafter} \cdot B \eta N^2) \theta'[t]}{\eta N} == 0$$


$$\theta \text{after}[t] == \frac{\theta[t]}{N}$$


$$\theta \text{after}'[t] == \frac{\theta'[t]}{N}$$


$$\theta \text{after}''[t] == 0$$

```

```
Kt i[t] == τ[t]
vapp[t] == R i[t] + Ke θ'[t]
η ≠ 0
N ≠ 0
```

```
Kt i[t] == τ[t]
vapp[t] == R i[t] + Ke θ'[t]
```

From a power point of view (ref: <https://bit.ly/2Lxka0Z>), we must have (electrical) power in = (mechanical) power out + (electrical + other) power losses:

```
powerEqns = And @@ {
  pwrIn == vapp[t] i[t],
  pwrOut == τ[t] θ'[t],
  pwrLosses == i[t]^2 R + pwrOtherPowerLosses,
  pwrIn == pwrOut + pwrLosses
};
(ssPowerEqns = Union[ssEqns, powerEqns]) // prettyPrint
```

```
pwrIn = pwrLosses + pwrOut
pwrIn = i(t) vapp(t)
pwrLosses = R i(t)^2 + pwrOtherPowerLosses
pwrOut = τ(t) θ'(t)
Kt i(t) = τ(t)
vapp(t) = R i(t) + Ke θ'(t)
```

Let's eliminate a few of the variables that we're not interested in.

```
kBalanceEqns = Eliminate[ssPowerEqns, {i[t], τ[t], θ'[t], vapp[t]}];
kBalanceEqns // prettyPrint
uniqueSolve[kBalanceEqns[[2]], pwrOtherPowerLosses] /. Rule → Equal
```

```
pwrIn = pwrLosses + pwrOut
Kt pwrOtherPowerLosses == (Ke - Kt) pwrOut
```

```
{pwrOtherPowerLosses == 
$$\frac{(Ke - Kt) pwrOut}{Kt}}$$
}
```

We can conclude that if other power losses are zero, then Ke must equal Kt.

4.3. Experimental Motor Data

We have experimental data from several motors (Nguyen, Hordyk, & Fraser, 2017). We load in same.

```
Clear[importMotorData, motorData]
importMotorData[] := importMotorData["Characterized Motors (Gears).xlsx"];
importMotorData[file_] := Module[{},
  Import[NotebookDirectory[] <> file, {"Data", 1}][[2 ;;]]
]
motorData = importMotorData[];
(motorData) // TableForm
```

Name	R	L (uH)	L (H)	Ke (>gear)	Kt (>gear, est)	Ke (<gear)	Kt (<gear, est)	J	b
AM 20 A	2.3	691.	0.000691	0.351	0.351	0.01755	0.01755	9.011×10^{-6}	0.0022
AM 20 B	1.9	684.	0.000684	0.389	0.389	0.01945	0.01945	9.011×10^{-6}	0.0025
AM 20 C	5.1	717.	0.000717	0.385	0.385	0.01925	0.01925	8.931×10^{-6}	0.0028
AM 40 A	2.5	674.	0.000674	0.753	0.753	0.018825	0.018825	0.00002221	0.2269
AM 40 B	3.8	705.	0.000705	0.705	0.705	0.017625	0.017625	0.00001741	0.56
AM 40 C	2.1	716.	0.000716	0.763	0.763	0.019075	0.019075	0.00002471	0.018
AM 60 A	3.3	694.	0.000694	1.066	1.066	0.0177667	0.0177667	0.00001041	0.033
AM 60 B	5.1	696.	0.000696	1.076	1.076	0.0179333	0.0179333	8.421×10^{-6}	0.02
AM 3.7 A	8.9	679.	0.000679	0.099	0.099	0.0267568	0.0267568	0.00002791	0.0001
AM 3.7 B	2.6	797.	0.000797	0.108	0.108	0.0291892	0.0291892	0.00003151	0.0001
AM 3.7 C	8.7	880.	0.00088	0.105	0.105	0.0283784	0.0283784	0.00003091	0.0001
Matrix A	3.8	718.	0.000718	0.34	0.34	0.00643939	0.00643939	9.431×10^{-6}	0.0015
Matrix B	7.8	777.	0.000777	0.363	0.363	0.006875	0.006875	7.761×10^{-6}	0.0019
Matrix C	20.6	658.	0.000658	0.338	0.338	0.00640152	0.00640152	7.231×10^{-6}	0.0018
CoreHex A	3.6	1356.	0.001356	0.822	0.822	0.0226759	0.0226759	0.0007331	0.0112
CoreHex B	11.3	1352.	0.001352	0.858	0.858	0.023669	0.023669	0.0006551	0.008
CoreHex C	5.6	1342.	0.001342	0.711	0.711	0.0196138	0.0196138	0.0004541	0.0078

We build a function to retrieve data from the table. We convert from floating point to rational in order to help delay floating point collapse later on down the line. We include load and external torque parameters at zero level so these can be easily added to later.

```

Clear[motorParameters];
Options[motorParameters] = { WithGearbox → True, Efficiency → "Forward" };
validateOption[motorParameters][WithGearbox, val_] := BooleanQ[val];
validateOption[motorParameters][Efficiency, val_] := MemberQ[{"Forward", "Reverse"}, val];

motorParameters[motorName_, opts: OptionsPattern[]] := motorParameters[motorData, motorName, opts]
motorParameters[motorData_, motorName_, opts: OptionsPattern[]] :=
  validateOptions[] @ Module[{row, paramValues, keQuants, ktQuants, quantify, qty, assoc},
    row = Select[motorData, #[[1]] == motorName &][[1]];
    paramValues = #[[1]] → toRational[ row[[#[[2]]]] ] & /@ {
      {R, 2}, {L, 4}, {KeWithGearbox, 5}, {KtWithGearbox, 6}, {KeNoGearbox, 7}, {KtNoGearbox, 8}, {J, 9}, {B, 10},
      {N, 11}, {η, If[OptionValue[Efficiency] == "Forward", 12, 13]} };
    keQuants = # → Quantity[#, (Ke /. parameterQuantities) // QuantityUnit] & /@ {KeWithGearbox, KeNoGearbox};
    ktQuants = # → Quantity[#, (Kt /. parameterQuantities) // QuantityUnit] & /@ {KtWithGearbox, KtNoGearbox};
    quantify = Function[{name, value},
      qty = (name /. parameterQuantities /. keQuants /. ktQuants);
      name → Quantity[value, QuantityUnit[qty]]
    ];
    assoc = quantify @@ # & /@ paramValues // Association;

    (* add in defaults that we can later override *)
    assoc[Jafter] = Quantity[0, siAngularInertialUnits];
    assoc[Bafter] = Quantity[0, siAngularViscousDragUnits];
    assoc[ΔτappConst] = Quantity[0, siTorqueUnits];
    (*assoc[ΔvappConst] = Quantity[0, "Volts"];*)

    If[OptionValue[WithGearbox],
      assoc[Ke] = assoc[KeNoGearbox];
      assoc[Kt] = assoc[KtNoGearbox];
      ,
      assoc[N] = 1;
      assoc[η] = 1;
      assoc[Ke] = assoc[KeWithGearbox];
      assoc[Kt] = assoc[KtWithGearbox];
    ];

    (* remove the shaft values to avoid confusion *)
    assoc[KeWithGearbox] = .;
    assoc[KtWithGearbox] = .;
    assoc[KeNoGearbox] = .;
    assoc[KtNoGearbox] = .;
    assoc // siUnits // simplifyUnits
  ];

motorParameters["AM 60 A", WithGearbox → False]
motorParameters["AM 60 A", WithGearbox → True, Efficiency → "Reverse"]

```

$$\left\langle \left| R \rightarrow \frac{33}{10} \text{ W/A}^2, L \rightarrow \frac{347}{500\,000} \text{ H}, J \rightarrow \frac{1041}{100\,000\,000} \text{ kg m}^2/\text{rad}^2, B \rightarrow \frac{33}{1000} \text{ kg m}^2/(\text{s rad}^2), N \rightarrow 1, \eta \rightarrow 1, \text{Jafter} \rightarrow 0 \text{ kg m}^2/\text{rad}^2, \right. \right. \\
 \left. \left. \text{Bafter} \rightarrow 0 \text{ kg m}^2/(\text{s rad}^2), \Delta\tau_{\text{appConst}} \rightarrow 0 \text{ kg m}^2/(\text{s}^2 \text{ rad}), Ke \rightarrow \frac{533}{500} \text{ kg m}^2/(\text{s}^2 \text{ A rad}), Kt \rightarrow \frac{533}{500} \text{ kg m}^2/(\text{s}^2 \text{ A rad}) \right| \right\rangle$$

$$\left\langle \left| R \rightarrow \frac{33}{10} \text{ W/A}^2, L \rightarrow \frac{347}{500\,000} \text{ H}, J \rightarrow \frac{1041}{100\,000\,000} \text{ kg m}^2/\text{rad}^2, B \rightarrow \frac{33}{1000} \text{ kg m}^2/(\text{s rad}^2), N \rightarrow 60, \eta \rightarrow \frac{4}{5}, \text{Jafter} \rightarrow 0 \text{ kg m}^2/\text{rad}^2, \right. \right. \\
 \left. \left. \text{Bafter} \rightarrow 0 \text{ kg m}^2/(\text{s rad}^2), \Delta\tau_{\text{appConst}} \rightarrow 0 \text{ kg m}^4/(\text{s}^4 \text{ rad}), Ke \rightarrow \frac{533}{30\,000} \text{ kg m}^2/(\text{s}^2 \text{ A rad}), Kt \rightarrow \frac{533}{30\,000} \text{ kg m}^2/(\text{s}^2 \text{ A rad}) \right| \right\rangle$$

4.4. Loads on the Motor

motorLoad[] creates a motor load given explicit internal and drag parameters. We allow the 'after' suffix but do not require its use.

```
Clear[motorLoad]
Options[motorLoad] = {
  J → Quantity[0, siAngularInertialUnits], Jafter → Quantity[0, siAngularInertialUnits],
  B → Quantity[0, siAngularViscousDragUnits], Bafter → Quantity[0, siAngularViscousDragUnits],
  ΔτappConst → Quantity[0, siTorqueUnits]
};
motorLoad[opts : OptionsPattern[]] := Module[{assoc = Association[]},
  assoc[Jafter] = OptionValue[J] + OptionValue[Jafter];
  assoc[Bafter] = OptionValue[B] + OptionValue[Bafter];
  assoc[ΔτappConst] = OptionValue[ΔτappConst];
  assoc
]
```

flywheel[] creates a motor load given mass and radius parameters

```
Clear[flywheel]
flywheel[mass_, radius_] := motorLoad[J → mass * radius ^ 2 / 2 / Quantity[1, "Radians^2"]]
flywheel[Quantity[2, "kg"], Quantity[5, "cm"]]
```

$$\left\langle \left| \begin{array}{l} \text{Jafter} \rightarrow 25 \text{ kg cm}^2/\text{rad}^2, \text{ Bafter} \rightarrow 0 \text{ m s N}/\text{rad}^2, \Delta\tau_{\text{appConst}} \rightarrow 0 \text{ m N}/\text{rad} \end{array} \right| \right\rangle$$

addMotorLoad[] takes motor parameters and a load and returns new parameters that include the load.

```
addMotorLoad[motor_, load_] := Module[{assoc},
  assoc = Association[motor]; (*explicitly create copy*)
  assoc[Jafter] = (Jafter /. motor) + (Jafter /. load);
  assoc[Bafter] = (Bafter /. motor) + (Bafter /. load);
  assoc[ΔτappConst] = (ΔτappConst /. motor) + (ΔτappConst /. load);
  assoc
]
addMotorLoad[motorParameters["AM 60 A"], flywheel[Quantity[2, "kg"], Quantity[5, "cm"]]]
```

$$\left\langle \left| \begin{array}{l} R \rightarrow \frac{33}{10} \text{ W/A}^2, L \rightarrow \frac{347}{500000} \text{ H}, J \rightarrow \frac{1041}{100000000} \text{ kg m}^2/\text{rad}^2, B \rightarrow \frac{33}{1000} \text{ kg m}^2/(\text{s rad}^2), N \rightarrow 60, \eta \rightarrow \frac{9}{10}, \text{Jafter} \rightarrow 25 \text{ kg cm}^2/\text{rad}^2, \\ \text{Bafter} \rightarrow 0 \text{ kg m}^2/(\text{s rad}^2), \Delta\tau_{\text{appConst}} \rightarrow 0 \text{ m N}/\text{rad}, K_e \rightarrow \frac{533}{30000} \text{ kg m}^2/(\text{s}^2 \text{ A rad}), K_t \rightarrow \frac{533}{30000} \text{ kg m}^2/(\text{s}^2 \text{ A rad}) \end{array} \right| \right\rangle$$

4.5. Mass on pulley

Here, we model a mass suspended on a massless “rigid string” from a massless pulley attached to the motor shaft. That string acts at a wrench with a length of the radius of the pulley to deliver a torque to the shaft. τMassOnPulley[] computes that torque.

There is a force pulling down on the mass due to gravity and up due to acceleration in the string that results from the driven angular acceleration of the motor shaft. The sum of these forces is the tension in the string. If the net acceleration is negative (ie: downward), a real (flexible) string would clamp at zero and have the mass in free-fall; we choose not to model that for now. Instead, here, our hypothetical rigid string continues to impart acceleration from the pulley even for negative accelerations. This is a bit odd for a string, but more reasonable if the string were in fact a rigid arm being lifted, or some such.

```

Clear[massOnPulley]
massOnPulley[mass_, radius_] := Module[
  {gravityAcceleration, gravityForce, angularAcceleration,
    tangentialAcceleration, tangentialForce, tension, torque, unit, parts, radians = Quantity["Radians"]},

  (* gravity exerts a constant downward force by acting on the mass *)
  gravityAcceleration = Quantity["StandardAccelerationOfGravity"];
  gravityForce = gravityAcceleration * mass;

  (* angular acceleration, converted to tangential acceleration also acts on the mass to create an upward force *)
  angularAcceleration = Quantity["0", "t", "siAngularAccelerationUnits"];
  tangentialAcceleration = angularAcceleration * radius / radians;
  tangentialForce = tangentialAcceleration * mass;

  (* the net force (the tension in the string) is the sum of the two forces *)
  tension = gravityForce + tangentialForce;

  (* the tension acts like a wrench with lever arm 'radius' *)
  torque = tension * radius / radians;

  (* we decompose the torque into constant and acceleration-dependent parts for output *)
  unit = QuantityUnit[torque];
  parts = List @@ (QuantityMagnitude[torque] // Apart);
  motorLoad[ΔτappConst → Quantity[parts[[1]], unit], Jafter → Quantity[parts[[2]], unit] / angularAcceleration]
]
massOnPulley[Quantity[weightLbs, "Pounds"], Quantity[rInches, "Inches"]]
% // siUnits // clearUnits // N

```

$$\left\langle \left| \text{Jafter} \rightarrow r\text{Inches}^2 \text{ weightLbs lb in}^2/\text{rad}^2, \text{ Bafter} \rightarrow 0 \text{ mN/rad}^2, \Delta\tau_{\text{appConst}} \rightarrow \frac{196.133 \text{ rInches weightLbs}}{508} \text{ lb in}^2/(\text{s}^2\text{rad}) \right| \right\rangle$$

$$\left\langle \left| \text{Jafter} \rightarrow 0.00029264 \text{ rInches}^2 \text{ weightLbs}, \text{ Bafter} \rightarrow 0., \Delta\tau_{\text{appConst}} \rightarrow 0.112985 \text{ rInches weightLbs} \right| \right\rangle$$

4.6. Example Motors

We define a motor and loads that we'll use to illustrate examples. We also define a generic, abstract motor that can help explore things symbolically.

'aMotorClassic' is the configuration we explored in our previous work. Ref: <https://github.com/rgatkinson/RobotPhysics/blob/master/MotorPhysics.pdf>

```

aMotorClassic =
  addMotorLoad[motorParameters["AM 60 A", WithGearbox → False], motorLoad[J → Quantity[1, "siAngularInertialUnits"]]]
aFlywheel = flywheel[Quantity[1, "kg"], Quantity[5, "cm"]] // siUnits
aPulley = massOnPulley[Quantity[3, "Pounds"], Quantity[2, "Inches"]] // siUnits

```

$$\left\langle \left| R \rightarrow \frac{33}{10} \text{ W/A}^2, L \rightarrow \frac{347}{500000} \text{ H}, J \rightarrow \frac{1041}{100000000} \text{ kg m}^2/\text{rad}^2, B \rightarrow \frac{33}{1000} \text{ kg m}^2/(\text{s rad}^2), N \rightarrow 1, \eta \rightarrow 1, \text{Jafter} \rightarrow 1 \text{ kg m}^2/\text{rad}^2, \right. \right. \\ \left. \left. \text{Bafter} \rightarrow 0 \text{ kg m}^2/(\text{s rad}^2), \Delta\tau_{\text{appConst}} \rightarrow 0 \text{ mN/rad}, K_e \rightarrow \frac{533}{500} \text{ kg m}^2/(\text{s}^2 \text{A rad}), K_t \rightarrow \frac{533}{500} \text{ kg m}^2/(\text{s}^2 \text{A rad}) \right| \right\rangle$$

$$\left\langle \left| \text{Jafter} \rightarrow \frac{1}{800} \text{ kg m}^2/\text{rad}^2, \text{ Bafter} \rightarrow 0 \text{ kg m}^2/(\text{s rad}^2), \Delta\tau_{\text{appConst}} \rightarrow 0 \text{ kg m}^2/(\text{s}^2 \text{rad}) \right| \right\rangle$$

$$\left\langle \left| \text{Jafter} \rightarrow \frac{2194797400719}{625000000000000} \text{ kg m}^2/\text{rad}^2, \text{ Bafter} \rightarrow 0 \text{ kg m}^2/(\text{s rad}^2), \Delta\tau_{\text{appConst}} \rightarrow \frac{3389544870828501}{500000000000000} \text{ kg m}^2/(\text{s}^2 \text{rad}) \right| \right\rangle$$

5. Laplace Transforms and Transfer Function Models

5.1. Laplace Transforms

Returning to the differential equations, we form the Laplace Transform of each, then solve the for the various transforms. First, we apply the LaplaceTransform[] function to each equation. It automatically makes use of the linearity and differential properties of the transform, and insinuates itself just around the time dependent parts (ie: the parts dependent on the time variable we told it about, namely t).

```
leqns = LaplaceTransform[#, t, s] &/@ ΔdiffEqns;
leqns // prettyPrint
```

$$\begin{aligned} Kt(\mathcal{L}_t[i(t)](s)) &= \mathcal{L}_t[\tau(t)](s) \\ \mathcal{L}_t[v_g(t)](s) &= Ke(s(\mathcal{L}_t[\theta(t)](s)) - \theta(0)) \\ R(\mathcal{L}_t[i(t)](s)) + L(s(\mathcal{L}_t[i(t)](s)) - i(0)) + \mathcal{L}_t[v_g(t)](s) &= \frac{v_{app0}}{s} + \mathcal{L}_t[\Delta v_{app}(t)](s) \\ \frac{\eta(\mathcal{L}_t[\tau(t)](s))N^2 + \frac{r_{app0}N}{s} + \mathcal{L}_t[\Delta r_{app}(t)](s))N - (B\eta N^2 + Bafter)(s(\mathcal{L}_t[\theta(t)](s)) - \theta(0)) - (J\eta N^2 + Jafter)(\mathcal{L}_t[\theta(t)](s))s^2 - \theta(0)s - \theta'(0)}{\eta N} &= 0 \\ \mathcal{L}_t[\theta_{after}(t)](s) &= \frac{\mathcal{L}_t[\theta(t)](s)}{N} \\ s(\mathcal{L}_t[\theta_{after}(t)](s)) - \theta_{after}(0) &= \frac{s(\mathcal{L}_t[\theta(t)](s)) - \theta(0)}{N} \\ (\mathcal{L}_t[\theta_{after}(t)](s))s^2 - \theta_{after}(0)s - \theta_{after}'(0) &= \frac{(\mathcal{L}_t[\theta(t)](s))s^2 - \theta(0)s - \theta'(0)}{N} \end{aligned}$$

Next, we walk those equations, picking up those insinuations and sowing them to the wind, reaping them on the outside, and finally removing duplicates.

```
allXforms = Reap[Scan[(If[MatchQ[#, _LaplaceTransform], Sow[#]] &, leqns, Infinity]][[2, 1]] // Union

{LaplaceTransform[i[t], t, s], LaplaceTransform[v_g[t], t, s], LaplaceTransform[Δv_app[t], t, s], LaplaceTransform[Δr_app[t], t, s],
LaplaceTransform[θ[t], t, s], LaplaceTransform[θ_after[t], t, s], LaplaceTransform[τ[t], t, s]}
```

The voltage transform is input, as is any autonomous externally applied torque; all the others are outputs. Solve for the outputs. Finally substitute what we know about initial conditions, and simplify as much as we can.

```
Δv_appXform = LaplaceTransform[Δv_app[t], t, s];
Δr_appXform = LaplaceTransform[Δr_app[t], t, s];
inputXforms = {Δv_appXform, Δr_appXform}
inputLabels = {"1", "v_app", "r_app"}
(outputXforms = Complement[allXforms, inputXforms]) // ColumnForm

{LaplaceTransform[Δv_app[t], t, s], LaplaceTransform[Δr_app[t], t, s]}
```

```
{1, v_app, r_app}
```

```
LaplaceTransform[i[t], t, s]
LaplaceTransform[v_g[t], t, s]
LaplaceTransform[θ[t], t, s]
LaplaceTransform[θ_after[t], t, s]
LaplaceTransform[τ[t], t, s]
```

```
(leqnsToSolve = leqns /. (ΔinitialConditions /. Equal → Rule)) // ColumnForm
```

$$\begin{aligned} Kt \text{ LaplaceTransform}[i[t], t, s] &= \text{LaplaceTransform}[\tau[t], t, s] \\ \text{LaplaceTransform}[v_g[t], t, s] &= Ke s \text{ LaplaceTransform}[\theta[t], t, s] \\ R \text{ LaplaceTransform}[i[t], t, s] + L \left(\frac{-Bafter v_{app0} - B v_{app0} \eta N^2 + Ke N \tau_{app0}}{Bafter R - Ke Kt \eta N^2 + B R \eta N^2} + s \text{ LaplaceTransform}[i[t], t, s] \right) + \text{LaplaceTransform}[v_g[t], t, s] &= \frac{v_{app0}}{s} + La \\ \frac{\eta \tau_{app0}}{s} + \eta \text{ LaplaceTransform}[\Delta r_{app}[t], t, s] - s (Bafter + B \eta N^2) \text{ LaplaceTransform}[\theta[t], t, s] - (Jafter + J \eta N^2) \left(-\frac{\eta (Kt v_{app0} \eta N + R \tau_{app0})}{Bafter R - Ke Kt \eta N^2 + B R \eta N^2} + s^2 \text{ LaplaceTransform}[\theta[t], t, s] \right) + \eta N^2 \text{ LaplaceTransform}[\tau[t], t, s] &= \\ \text{LaplaceTransform}[\theta_{after}[t], t, s] &= \frac{\text{LaplaceTransform}[\theta[t], t, s]}{N} \\ s \text{ LaplaceTransform}[\theta_{after}[t], t, s] &= \frac{s \text{ LaplaceTransform}[\theta[t], t, s]}{N} \\ -Kt v_{app0} \eta N - R \tau_{app0} + s^2 \text{ LaplaceTransform}[\theta_{after}[t], t, s] &= \frac{-\eta (Kt v_{app0} \eta N + R \tau_{app0}) + s^2 \text{ LaplaceTransform}[\theta[t], t, s]}{Bafter R - Ke Kt \eta N^2 + B R \eta N^2} \end{aligned}$$

```
(rawSolvedXforms = uniqueSolve[leqnsToSolve, outputXforms] // Simplify) // ColumnForm
```

$$\begin{aligned} \text{LaplaceTransform}[i[t], t, s] &\rightarrow \frac{\text{Bafter vapp0} + N (B \text{ vapp0} \eta N - \text{Ke} \tau \text{app0})}{s (\text{Bafter } R + (\text{Ke } Kt + B R) \eta N^2)} + \frac{(\text{Bafter} + \text{Jafter } s + (B + J s) \eta N^2) \text{LaplaceTransform}[\Delta \text{vapp}[t], t, s]}{\text{Bafter } (R + L s) + \text{Jafter } s (R + L s) + (\text{Ke } Kt + (B + J s) (R + L s)) \eta N^2} - \frac{\text{Ke } N \text{LaplaceTransform}[\Delta \tau \text{app}[t], t, s]}{\text{Bafter } (R + L s) + \text{Jafter } s (R + L s) + (\text{Ke } Kt + (B + J s) (R + L s)) \eta N^2} \\ \text{LaplaceTransform}[vg[t], t, s] &\rightarrow \text{Ke } N \left(\frac{Kt \text{ vapp0} \eta N + R \tau \text{app0}}{s (\text{Bafter } R + (\text{Ke } Kt + B R) \eta N^2)} + \frac{Kt \eta N \text{LaplaceTransform}[\Delta \text{vapp}[t], t, s]}{\text{Bafter } (R + L s) + \text{Jafter } s (R + L s) + (\text{Ke } Kt + (B + J s) (R + L s)) \eta N^2} + \frac{(R + L s) \text{LaplaceTransform}[\Delta \tau \text{app}[t], t, s]}{\text{Bafter } (R + L s) + \text{Jafter } s (R + L s) + (\text{Ke } Kt + (B + J s) (R + L s)) \eta N^2} \right) \\ \text{LaplaceTransform}[\theta[t], t, s] &\rightarrow \frac{N \left(\frac{Kt \text{ vapp0} \eta N + R \tau \text{app0}}{\text{Bafter } R + (\text{Ke } Kt + B R) \eta N^2} + \frac{Kt s \eta N \text{LaplaceTransform}[\Delta \text{vapp}[t], t, s]}{\text{Bafter } (R + L s) + \text{Jafter } s (R + L s) + (\text{Ke } Kt + (B + J s) (R + L s)) \eta N^2} + \frac{s (R + L s) \text{LaplaceTransform}[\Delta \tau \text{app}[t], t, s]}{\text{Bafter } (R + L s) + \text{Jafter } s (R + L s) + (\text{Ke } Kt + (B + J s) (R + L s)) \eta N^2} \right)}{s^2} \\ \text{LaplaceTransform}[\theta \text{after}[t], t, s] &\rightarrow \frac{\frac{Kt \text{ vapp0} \eta N + R \tau \text{app0}}{\text{Bafter } R + (\text{Ke } Kt + B R) \eta N^2} + \frac{Kt s \eta N \text{LaplaceTransform}[\Delta \text{vapp}[t], t, s]}{\text{Bafter } (R + L s) + \text{Jafter } s (R + L s) + (\text{Ke } Kt + (B + J s) (R + L s)) \eta N^2} + \frac{s (R + L s) \text{LaplaceTransform}[\Delta \tau \text{app}[t], t, s]}{\text{Bafter } (R + L s) + \text{Jafter } s (R + L s) + (\text{Ke } Kt + (B + J s) (R + L s)) \eta N^2}}{s^2} \\ \text{LaplaceTransform}[\tau[t], t, s] &\rightarrow \text{Kt} \left(\frac{\text{Bafter vapp0} + N (B \text{ vapp0} \eta N - \text{Ke} \tau \text{app0})}{s (\text{Bafter } R + (\text{Ke } Kt + B R) \eta N^2)} + \frac{(\text{Bafter} + \text{Jafter } s + (B + J s) \eta N^2) \text{LaplaceTransform}[\Delta \text{vapp}[t], t, s]}{\text{Bafter } (R + L s) + \text{Jafter } s (R + L s) + (\text{Ke } Kt + (B + J s) (R + L s)) \eta N^2} - \frac{\text{Ke } N \text{LaplaceTransform}[\Delta \tau \text{app}[t], t, s]}{\text{Bafter } (R + L s) + \text{Jafter } s (R + L s) + (\text{Ke } Kt + (B + J s) (R + L s)) \eta N^2} \right) \end{aligned}$$

We separate the transforms into applied-voltage- and applied-torque-dependent parts.

```
Clear[separateXforms]
separateXforms[xform_] := Module[{xformVarRules, invertedRules, xformVars, vard, rawCollected, coeffs, const, terms},
  xformVarRules = # -> Unique["laplaceTransform"] & /@ inputXforms;
  invertedRules = Reverse /@ xformVarRules;
  xformVars = xformVarRules[All, 2];
  vard = xform /. xformVarRules;
  rawCollected = Collect[vard, xformVars] /. invertedRules;
  coeffs = Coefficient[rawCollected, #] & /@ inputXforms;
  const = Together[rawCollected /. {# -> 0 & /@ inputXforms}];
  (* we multiply const by s to make the units come out right (?) *)
  const = {const[[1]] * s // Simplify};
  terms = const ~Join~ (#[[1]] * #[[2]] & /@ Transpose[{coeffs, inputXforms}]);
  Plus @@ terms
]
solvedXforms = #[[1]] -> separateXforms[#[[2]]] & /@ rawSolvedXforms;
% // ColumnForm
```

$$\begin{aligned} \text{LaplaceTransform}[i[t], t, s] &\rightarrow \frac{\text{Bafter vapp0} + N (B \text{ vapp0} \eta N - \text{Ke} \tau \text{app0})}{\text{Bafter } R + (\text{Ke } Kt + B R) \eta N^2} + \frac{(\text{Bafter} + \text{Jafter } s + (B + J s) \eta N^2) \text{LaplaceTransform}[\Delta \text{vapp}[t], t, s]}{\text{Bafter } (R + L s) + \text{Jafter } s (R + L s) + (\text{Ke } Kt + (B + J s) (R + L s)) \eta N^2} - \frac{\text{Ke } N \text{LaplaceTransform}[\Delta \tau \text{app}[t], t, s]}{\text{Bafter } (R + L s) + \text{Jafter } s (R + L s) + (\text{Ke } Kt + (B + J s) (R + L s)) \eta N^2} \\ \text{LaplaceTransform}[vg[t], t, s] &\rightarrow \frac{\text{Ke } N (Kt \text{ vapp0} \eta N + R \tau \text{app0})}{\text{Bafter } R + (\text{Ke } Kt + B R) \eta N^2} + \frac{\text{Ke } Kt \eta N^2 \text{LaplaceTransform}[\Delta \text{vapp}[t], t, s]}{\text{Bafter } (R + L s) + \text{Jafter } s (R + L s) + (\text{Ke } Kt + (B + J s) (R + L s)) \eta N^2} + \frac{\text{Ke } (R + L s) N \text{LaplaceTransform}[\Delta \tau \text{app}[t], t, s]}{\text{Bafter } (R + L s) + \text{Jafter } s (R + L s) + (\text{Ke } Kt + (B + J s) (R + L s)) \eta N^2} \\ \text{LaplaceTransform}[\theta[t], t, s] &\rightarrow \frac{N (Kt \text{ vapp0} \eta N + R \tau \text{app0})}{s (\text{Bafter } R + (\text{Ke } Kt + B R) \eta N^2)} + \frac{Kt \eta N^2 \text{LaplaceTransform}[\Delta \text{vapp}[t], t, s]}{s (\text{Bafter } (R + L s) + \text{Jafter } s (R + L s) + (\text{Ke } Kt + (B + J s) (R + L s)) \eta N^2)} + \frac{(R + L s) N \text{LaplaceTransform}[\Delta \tau \text{app}[t], t, s]}{s (\text{Bafter } (R + L s) + \text{Jafter } s (R + L s) + (\text{Ke } Kt + (B + J s) (R + L s)) \eta N^2)} \\ \text{LaplaceTransform}[\theta \text{after}[t], t, s] &\rightarrow \frac{Kt \text{ vapp0} \eta N + R \tau \text{app0}}{s (\text{Bafter } R + (\text{Ke } Kt + B R) \eta N^2)} + \frac{Kt \eta N \text{LaplaceTransform}[\Delta \text{vapp}[t], t, s]}{s (\text{Bafter } (R + L s) + \text{Jafter } s (R + L s) + (\text{Ke } Kt + (B + J s) (R + L s)) \eta N^2)} + \frac{s (R + L s) \text{LaplaceTransform}[\Delta \tau \text{app}[t], t, s]}{s (\text{Bafter } (R + L s) + \text{Jafter } s (R + L s) + (\text{Ke } Kt + (B + J s) (R + L s)) \eta N^2)} \\ \text{LaplaceTransform}[\tau[t], t, s] &\rightarrow \frac{Kt (\text{Bafter vapp0} + N (B \text{ vapp0} \eta N - \text{Ke} \tau \text{app0}))}{\text{Bafter } R + (\text{Ke } Kt + B R) \eta N^2} + \frac{Kt (\text{Bafter} + \text{Jafter } s + (B + J s) \eta N^2) \text{LaplaceTransform}[\Delta \text{vapp}[t], t, s]}{\text{Bafter } (R + L s) + \text{Jafter } s (R + L s) + (\text{Ke } Kt + (B + J s) (R + L s)) \eta N^2} - \frac{\text{Ke } Kt N \text{LaplaceTransform}[\Delta \tau \text{app}[t], t, s]}{\text{Bafter } (R + L s) + \text{Jafter } s (R + L s) + (\text{Ke } Kt + (B + J s) (R + L s)) \eta N^2} \end{aligned}$$

5.2. Transfer Function Models

We use the solved transforms to create TransferFunctionModel[]s for every variable of interest.

```
Clear[divideTerms]
divideTerms[sum_, termDivisors_] := Module[{list},
  list = List @@ sum;
  #[[1]] / #[[2]] & /@ Transpose[{list, termDivisors}]
]
divideTerms[a x + b y, {x, y}]
```

```
{a, b}
```

```
Clear[makeModel]
makeModel[var_] := makeModel[var, ToString[var], 1]
makeModel[var_, outputLabel_, factor_] := TransferFunctionModel[
  factor * Function[{sum}, divideTerms[sum, {1} ~Join~ inputXforms]]
  [LaplaceTransform[var[t], t, s] /. solvedXforms],
  s,
  SystemsModelLabels -> {inputLabels, outputLabel}] // FullSimplify // Simplify
```



```

motorPositionModel = makeModel[ $\theta$ ]
motorVelocityModel = makeModel[ $\theta$ , " $\omega$ ", s]
motorAccelerationModel = makeModel[ $\theta$ , " $\alpha$ ", s * s]
motorCurrentModel = makeModel[i]
motorEMFModel = makeModel[vg, "vg", 1]
motorTorqueModel = makeModel[ $\tau$ ]

```

$$\left(\begin{array}{c} \theta \\ \omega \\ \alpha \end{array} \right) = \left(\begin{array}{c} \frac{1}{N (Kt v_{app} \theta \eta N + R \tau_{app} \theta)} \frac{v_{app}}{Kt \eta N^2} \frac{\tau_{app}}{N (R + L s)} \\ \frac{1}{Bafter R + (Ke Kt + B R) \eta N^2} s \left(Ke Kt \eta N^2 + (R + L s) (Bafter + Jafter s + \eta N^2 (B + J s)) \right) s \left(Ke Kt \eta N^2 + (R + L s) (Bafter + Jafter s + \eta N^2 (B + J s)) \right) \end{array} \right)$$

$$\left(\begin{array}{c} \omega \\ \alpha \end{array} \right) = \left(\begin{array}{c} \frac{1}{Bafter R + (Ke Kt + B R) \eta N^2} \frac{v_{app}}{Kt \eta N^2} \frac{\tau_{app}}{N (R + L s)} \\ \frac{1}{Bafter R + (Ke Kt + B R) \eta N^2} \frac{v_{app}}{Kt \eta N^2} \frac{\tau_{app}}{N (R + L s)} \end{array} \right)$$

$$\left(\begin{array}{c} \alpha \\ i \end{array} \right) = \left(\begin{array}{c} \frac{1}{Bafter R + (Ke Kt + B R) \eta N^2} \frac{v_{app}}{Kt \eta N^2} \frac{\tau_{app}}{N s (R + L s)} \\ \frac{1}{Bafter R + (Ke Kt + B R) \eta N^2} \frac{v_{app}}{Kt \eta N^2} \frac{\tau_{app}}{N s (R + L s)} \end{array} \right)$$

$$\left(\begin{array}{c} i \\ v_g \end{array} \right) = \left(\begin{array}{c} \frac{1}{Bafter v_{app} \theta + N (B v_{app} \theta \eta N - Ke \tau_{app} \theta)} \frac{v_{app}}{Bafter + Jafter s + \eta N^2 (B + J s)} \frac{\tau_{app}}{Ke N} \\ \frac{1}{Bafter R + (Ke Kt + B R) \eta N^2} \frac{v_{app}}{Ke Kt \eta N^2 + (R + L s) (Bafter + Jafter s + \eta N^2 (B + J s))} \frac{\tau_{app}}{Ke Kt \eta N^2 + (R + L s) (Bafter + Jafter s + \eta N^2 (B + J s))} \end{array} \right)$$

$$\left(\begin{array}{c} v_g \\ \tau \end{array} \right) = \left(\begin{array}{c} \frac{1}{Ke N (Kt v_{app} \theta \eta N + R \tau_{app} \theta)} \frac{v_{app}}{Ke Kt \eta N^2} \frac{\tau_{app}}{Ke N (R + L s)} \\ \frac{1}{Bafter R + (Ke Kt + B R) \eta N^2} \frac{v_{app}}{Ke Kt \eta N^2 + (R + L s) (Bafter + Jafter s + \eta N^2 (B + J s))} \frac{\tau_{app}}{Ke Kt \eta N^2 + (R + L s) (Bafter + Jafter s + \eta N^2 (B + J s))} \end{array} \right)$$

$$\left(\begin{array}{c} \tau \end{array} \right) = \left(\begin{array}{c} \frac{1}{Kt (Bafter v_{app} \theta + N (B v_{app} \theta \eta N - Ke \tau_{app} \theta))} \frac{v_{app}}{Kt (Bafter + Jafter s + \eta N^2 (B + J s))} \frac{\tau_{app}}{Ke Kt N} \\ \frac{1}{Bafter R + (Ke Kt + B R) \eta N^2} \frac{v_{app}}{Ke Kt \eta N^2 + (R + L s) (Bafter + Jafter s + \eta N^2 (B + J s))} \frac{\tau_{app}}{Ke Kt \eta N^2 + (R + L s) (Bafter + Jafter s + \eta N^2 (B + J s))} \end{array} \right)$$

```

motorPositionOutModel = makeModel[ $\theta_{after}$ ]
motorVelocityOutModel = makeModel[ $\theta_{after}$ , " $\omega_{out}$ ", s]
motorAccelerationOutModel = makeModel[ $\theta_{after}$ , " $\alpha_{out}$ ", s * s]

```

$$\left(\begin{array}{c} \theta_{after} \\ \omega_{out} \\ \alpha_{out} \end{array} \right) = \left(\begin{array}{c} \frac{1}{Kt v_{app} \theta \eta N + R \tau_{app} \theta} \frac{v_{app}}{Kt \eta N} \frac{\tau_{app}}{R + L s} \\ \frac{1}{Bafter R + (Ke Kt + B R) \eta N^2} s \left(Ke Kt \eta N^2 + (R + L s) (Bafter + Jafter s + \eta N^2 (B + J s)) \right) s \left(Ke Kt \eta N^2 + (R + L s) (Bafter + Jafter s + \eta N^2 (B + J s)) \right) \end{array} \right)$$

$$\left(\begin{array}{c} \omega_{out} \\ \alpha_{out} \end{array} \right) = \left(\begin{array}{c} \frac{1}{Bafter R + (Ke Kt + B R) \eta N^2} \frac{v_{app}}{Kt \eta N} \frac{\tau_{app}}{R + L s} \\ \frac{1}{Bafter R + (Ke Kt + B R) \eta N^2} \frac{v_{app}}{Kt \eta N} \frac{\tau_{app}}{R + L s} \end{array} \right)$$

$$\left(\begin{array}{c} \alpha_{out} \end{array} \right) = \left(\begin{array}{c} \frac{1}{Bafter R + (Ke Kt + B R) \eta N^2} \frac{v_{app}}{Kt \eta N} \frac{\tau_{app}}{s (R + L s)} \\ \frac{1}{Bafter R + (Ke Kt + B R) \eta N^2} \frac{v_{app}}{Kt \eta N} \frac{\tau_{app}}{s (R + L s)} \end{array} \right)$$

Save the models to a file for later use.

```

saveDefinitions[outputDirectory <> "MotorModels.m",
{diffEqns,  $\Delta$ diffEqns, initialConditions,  $\Delta$ initialConditions, motorPositionModel,
motorVelocityModel, motorAccelerationModel, motorCurrentModel, motorEMFModel, motorTorqueModel,
motorPositionOutModel, motorVelocityOutModel, motorAccelerationOutModel}]

```

5.3. Units Verification

```
motorVelocityModel[s][[1]]
modelTest = % /. parameterQuantities /. s -> Quantity[s, "per second"];
inputs = {1, Quantity[vapp, "Volts"], Quantity[tauapp, siTorqueUnits]}
modelTest.inputs
```

$$\left\{ \frac{N (Kt vapp \theta \eta N + R \tau app \theta)}{Bafter R + (Ke Kt + B R) \eta N^2}, \frac{Kt \eta N^2}{Ke Kt \eta N^2 + (R + L s) (Bafter + Jafter s + (B + J s) \eta N^2)}, \frac{(R + L s) N}{Ke Kt \eta N^2 + (R + L s) (Bafter + Jafter s + (B + J s) \eta N^2)} \right\}$$

$$\{1, vapp V, \tau app mN/rad\}$$

$$\left(\frac{Kt vapp \eta N^2}{Ke Kt \eta N^2 + (R + L s) (Bafter + Jafter s + (B + J s) \eta N^2)} + \frac{(R + L s) N \tau app}{Ke Kt \eta N^2 + (R + L s) (Bafter + Jafter s + (B + J s) \eta N^2)} + \frac{N (Kt vapp \theta \eta N + R \tau app \theta)}{Bafter R + (Ke Kt + B R) \eta N^2} \right) rad/s$$

6. Time Domain Functions

The time domain functions we seek model various states of the motor as a direct function of time. Generally speaking, we may obtain them from the frequency domain by (matrix) multiplying the transfer function by the Laplace transform of the input signal, then taking the inverse Laplace transform of the result. However, if we're starting with the (frequency domain) transfer function and the (time domain) input signal in-hand, we can compute the time domain function more efficiently using the Laplace transform convolution theorem.

6.1. Time Domain Output Functions

6.1.1. makeTimeDomainFunctionConvolve[]

makeTimeDomainFunctionConvolve[] uses convolution and the theorem that the Laplace transform of a convolution is the product of the Laplace transform of the factors. Ref1: <http://mathfaculty.fullerton.edu/mathews/c2003/laplaceconvolutionmod.html>.

With the convolution approach, we can make explicit use of the fact that in the uses we have we know we want to take the inverse Laplace transform of a product, one of whose factors we already know the time domain inverse for, and thus we can avoid having LaplaceTransform[] do as much heavy lifting as we might ask it to do if we were to transform the input, multiply, then inverse-transform the whole result. This convolution approach also has been shown to be more efficient than the OutputResponse[] builtin function.

```
Clear[makeTimeDomainFunctionConvolve, convolve]
parameterAssumptions

convolve[fExpr_, gExpr_, exprVar_, t_] := Module[{tau = Unique["tau"]},
  Assuming[Union[{t >= 0}, parameterAssumptions], Integrate[(fExpr /. exprVar -> tau) * (gExpr /. exprVar -> t - tau), {tau, 0, t}]]
]

makeTimeDomainFunctionConvolve[model_, tInputFunctions_] := Module[
  {s = Unique["s"], tau = Unique["tau"], modelExpr},
  modelExpr = InverseLaplaceTransform[model[s], s, tau][[1]];
  Function[{t}, Module[{convolutions},
    convolutions = convolve[modelExpr[[#]], (tInputFunctions[[#]])[tau, tau, t] & /@ Range[1, Length[tInputFunctions]]];
    {Total[convolutions]} (* put in list to mirror InverseLaplaceTransform[] output structure *)
  ]
]

{Bafter ∈ ℝ, Jafter ∈ ℝ, Null ∈ ℝ, vapp0 ∈ ℝ, ΔvappConst ∈ ℝ, ΔtauappConst ∈ ℝ, tauapp0 ∈ ℝ, i[_] ∈ ℝ, vg[_] ∈ ℝ, α[_] ∈ ℝ,
  Δvapp[_] ∈ ℝ, Δtauapp[_] ∈ ℝ, θ[_] ∈ ℝ, τ[_] ∈ ℝ, ω[_] ∈ ℝ, Ke > 0, Kt > 0, L > 0, R > 0, η > 0, N > 0, B ≥ 0, J ≥ 0, t ≥ 0}
```

6.1.2. makeMotorTimeDomainFunction[]

makeMotorTimeDomainFunction[] takes a transfer function model and returns a function that, when provided with ea and tau time-domain input functions, returns a function of time.

```

Clear[makeMotorTimeDomainFunction]
makeMotorTimeDomainFunction[model_] := makeMotorTimeDomainFunction[model, makeTimeDomainFunctionConvolve]
makeMotorTimeDomainFunction[model_, builder_] := makeMotorTimeDomainFunction[model, builder] = (*cache the result*)
Module[{t(*=Unique["t"]*)},
  Function[{vappfn, tappfn}, Module[{fn},
    fn = builder[model, {1 &, vappfn[#] &, tappfn[#] &}];
    fn
  ]
]]

```

6.1.3. Time Domain Output Functions for Motor State

We define time domain functions for all the motor state of interest.

```
Clear[motorPosition, motorVelocity, motorAcceleration, motorCurrent, motorEMF, motorTorque]
```

```
motorPosition = makeMotorTimeDomainFunction[motorPositionModel];
```

```
motorVelocity = makeMotorTimeDomainFunction[motorVelocityModel];
```

```
motorAcceleration = makeMotorTimeDomainFunction[motorAccelerationModel];
```

```
motorCurrent = makeMotorTimeDomainFunction[motorCurrentModel];
```

```
motorEMF = makeMotorTimeDomainFunction[motorEMFModel];
```

```
motorTorque = makeMotorTimeDomainFunction[motorTorqueModel];
```

The *output* too.

```
Clear[motorPositionOut, motorVelocityOut, motorAccelerationOut]
```

```
motorPositionOut = makeMotorTimeDomainFunction[motorPositionOutModel];
```

```
motorVelocityOut = makeMotorTimeDomainFunction[motorVelocityOutModel];
```

```
motorAccelerationOut = makeMotorTimeDomainFunction[motorAccelerationOutModel];
```

Save these time domain functions so we can load them in later without having to develop them from scratch.

```

saveDefinitions[outputDirectory <> "MotorTimeDomainFunctions.m",
{
  motorPosition, motorVelocity, motorAcceleration, motorCurrent, motorEMF, motorTorque,
  motorPositionOut, motorVelocityOut, motorAccelerationOut,
  makeMotorTimeDomainFunction, makeTimeDomainFunctionConvolve
}]

```

6.2. Time Domain Input Functions

6.2.1. Typical Input Examples

In analogy to aMotor above, we now define some typical inputs. The first (and only, for now) is a constant 12V input with no externally applied torque.

```

applyTwelveVolts = Association[t → Quantity[t, "Seconds"],
  vapp0 → Quantity[0, "Volts"], tapp0 → Quantity[0, siTorqueUnits], ΔvappConst → Quantity[12, "Volts"]
<| t → t s, vapp0 → 0 V, tapp0 → 0 mN/rad, ΔvappConst → 12 V |>

```

```
steadyTwelveVolts = Association[t → Quantity[t, "Seconds"],
  vapp0 → Quantity[12, "Volts"], τapp0 → Quantity[0, siTorqueUnits], ΔvappConst → Quantity[0, "Volts"]]
```

```
{t → t s, vapp0 → 12 V, τapp0 → 0 mN/rad, ΔvappConst → 0 V}
```

```
brakeFromTwelveVolts = Association[t → Quantity[t, "Seconds"],
  vapp0 → Quantity[12, "Volts"], τapp0 → Quantity[0, siTorqueUnits], ΔvappConst → Quantity[-12, "Volts"]]
```

```
{t → t s, vapp0 → 12 V, τapp0 → 0 mN/rad, ΔvappConst → -12 V}
```

6.3. Examples


We try out our time-domain functions on our example data.

```
Clear[applyTimeFunction, applyTimeFunctionGeneric]
applyTimeFunctionGeneric[motorTimeFunction_, t_] := applyTimeFunctionGeneric[motorTimeFunction, t] = Module[{},
  motorTimeFunction[ΔvappConst &, ΔτappConst &][t]]

applyTimeFunction[motorTimeFunction_, motorWithLoad_, input_, t_] :=
  applyTimeFunction[motorTimeFunction, motorWithLoad, input, t] = Module[{generic, withUnits, unitless},
    generic = applyTimeFunctionGeneric[motorTimeFunction, t];
    withUnits = generic /. {input ~Join~ motorWithLoad} // siUnits // FullSimplify;
    unitless = withUnits // clearUnits // N // FullSimplify;
    {withUnits, unitless}]
```


6.3.1. Examples without gears or external torque, using 'classic' load

```
{velUnits, vel} = applyTimeFunction[motorVelocity, aMotorClassic, applyTwelveVolts, t];
velUnits // N
Plot[vel, {t, 0, 20}, AxesLabel → {"Seconds", HoldForm[Radians / Second]}, PlotRange → All, AxesOrigin → {0, 0}]
```

$$\left\{ \left\{ 2.71828^{-0.377374 t \sqrt{\text{kg m} / (\sqrt{\text{kg m}})}} \left(-10.2734 \text{ rad/s} \right) + 2.71828^{-4754.7 t \sqrt{\text{kg m} / (\sqrt{\text{kg m}})}} \left(0.000815385 \text{ rad/s} \right) + 10.2726 \text{ rad/s} \right\} \right\}$$



```
{steadyVelUnits, steadyVel} = applyTimeFunction[motorVelocity, aMotorClassic, steadyTwelveVolts, t];
steadyVelUnits // N
Plot[steadyVel, {t, 0, 20}, AxesLabel → {"Seconds", HoldForm[Radians / Second]}, PlotRange → All, AxesOrigin → {0, 0}]
```

$$\left\{ \left\{ 2.71828^{-4754.7 t \sqrt{\text{kg m} / (\sqrt{\text{kg m}})}} + 2.71828^{-0.377374 t \sqrt{\text{kg m} / (\sqrt{\text{kg m}})}} \right\} \left(0. \text{ rad/s} \right) + 10.2726 \text{ rad/s} \right\}$$




```
{brakeVelUnits, brakeVel} = applyTimeFunction[motorVelocity, aMotorClassic, brakeFromTwelveVolts, t];
brakeVelUnits // N
Plot[brakeVel, {t, 0, 20}, AxesLabel → {"Seconds", HoldForm[Radians / Second]}, PlotRange → All, AxesOrigin → {0, 0}]
```

$$\left\{ \left\{ 2.71828^{-4754.7 t \sqrt{\text{kg m} / (\sqrt{\text{kg m}})}} \left(-0.000815385 \text{ rad/s} \right) + \left(1. + 2.71828^{-0.377374 t \sqrt{\text{kg m} / (\sqrt{\text{kg m}})}} \right) \left(0. \text{ rad/s} \right) + 2.71828^{-0.377374 t \sqrt{\text{kg m} / (\sqrt{\text{kg m}})}} \left(10.2734 \text{ rad/s} \right) \right\} \right\}$$




```
{torUnits, tor} = applyTimeFunction[motorTorque, aMotorClassic, applyTwelveVolts, t];
torUnits // N
Plot[tor, {t, 0, 20}, AxesLabel → {"Seconds", HoldForm[Netwon Meters / Radian]}, PlotRange → All, AxesOrigin → {0, 0}]
```

$$\left\{ \left\{ 2.71828^{-4754.7 t \sqrt{\text{kg m} / (\sqrt{\text{kg m}})}} \left(-3.87693 \text{ kg m}^2 / (\text{s}^2 \text{ rad}) \right) + 0.338995 \text{ kg m}^2 / (\text{s}^2 \text{ rad}) + 2.71828^{-0.377374 t \sqrt{\text{kg m} / (\sqrt{\text{kg m}})}} \left(3.53793 \text{ kg m}^2 / (\text{s}^2 \text{ rad}) \right) \right\} \right\}$$




```
{curUnits, cur} = applyTimeFunction[motorCurrent, aMotorClassic, applyTwelveVolts, t];
curUnits // N
Plot[cur, {t, 0, 20}, AxesLabel → {"Seconds", HoldForm[Amps]}, PlotRange → All, AxesOrigin → {0, 0}]
```

$$\left\{ \left\{ 2.71828^{-4754.7 t \sqrt{\text{kg m}/(\sqrt{\text{kg m}})}} \left(-3.63689 \text{ A} \right) + 0.318007 \text{ A} + 2.71828^{-0.377374 t \sqrt{\text{kg m}/(\sqrt{\text{kg m}})}} \left(3.31888 \text{ A} \right) \right\} \right\}$$




```
{emfUnits, emf} = applyTimeFunction[motorEMF, aMotorClassic, applyTwelveVolts, t];
emfUnits // N // simplifyUnits
Plot[{emf, ΔvappConst /. applyTwelveVolts}, {t, 0, 20},
  AxesLabel → {"Seconds", HoldForm[Volts]}, PlotRange → All, AxesOrigin → {0, 0}]
```

$$\left\{ \left\{ 2.71828^{-0.377374 t \sqrt{\text{kg m}/(\sqrt{\text{kg m}})}} \left(-10.9514 \text{ V} \right) + 2.71828^{-4754.7 t \sqrt{\text{kg m}/(\sqrt{\text{kg m}})}} \left(0.000869201 \text{ V} \right) + 10.9506 \text{ V} \right\} \right\}$$



```
{posUnits, pos} = applyTimeFunction[motorPosition, aMotorClassic, applyTwelveVolts, t];
posUnits // N
Plot[pos, {t, 0, 20}, AxesLabel → {"Seconds", HoldForm[Radians]}, PlotRange → All, AxesOrigin → {0, 0}]
```

$$\left\{ \left\{ 2.71828^{-4754.7 t \sqrt{\text{kg m}/(\sqrt{\text{kg m}})}} \left(-1.7149 \times 10^{-7} \text{ rad} \right) + \right. \right. \\ \left. \left. 2.71828^{-0.377374 t \sqrt{\text{kg m}/(\sqrt{\text{kg m}})}} \left(27.2234 \text{ rad} \right) + 4.12469 \times 10^{-8} \left(-6.60011 \times 10^8 + 2.49051 \times 10^8 t \right) \text{ rad} \right\} \right\}$$




6.3.2. Examples with gear, using flywheel load

```
unitlessMotorWithFlywheel =
  addMotorLoad[motorParameters["AM 60 A"], flywheel[Quantity[3, "kg"], Quantity[8, "cm"]]] // siUnits // clearUnits // N
  applyTwelveVolts // siUnits // clearUnits // N
  {velUnits2, vel2} =
    applyTimeFunction[motorVelocity, unitlessMotorWithFlywheel, applyTwelveVolts // siUnits // clearUnits // N, t];
  Plot[vel2, {t, 0, .01}, AxesLabel -> {"Seconds", HoldForm[Radians / Second]}, PlotRange -> All, AxesOrigin -> {0, 0}]

<| R -> 3.3, L -> 0.000694, J -> 0.00001041, B -> 0.033, N -> 60.,  $\eta$  -> 0.9,
  Jafter -> 0.0096, Bafter -> 0.,  $\Delta\tau_{appConst}$  -> 0., Ke -> 0.0177667, Kt -> 0.0177667 |>
```


```
<| t -> t, vapp0 -> 0.,  $\tau_{app0}$  -> 0.,  $\Delta v_{appConst}$  -> 12. |>
```



6.3.3. Examples without gear, using flywheel load

```
unitlessMotorWithFlywheelNoGear =
  addMotorLoad[motorParameters["AM 60 A", WithGearbox -> False], flywheel[Quantity[3, "kg"], Quantity[8, "cm"]]] // siUnits //
  clearUnits // N
  {velUnits3, vel3} = applyTimeFunction[motorVelocity, unitlessMotorWithFlywheelNoGear,
    applyTwelveVolts // siUnits // clearUnits // N, t];
  Plot[vel3, {t, 0, 0.25}, AxesLabel -> {"Seconds", HoldForm[Radians / Second]}, PlotRange -> All, AxesOrigin -> {0, 0}]

<| R -> 3.3, L -> 0.000694, J -> 0.00001041, B -> 0.033, N -> 1.,
   $\eta$  -> 1., Jafter -> 0.0096, Bafter -> 0.,  $\Delta\tau_{appConst}$  -> 0., Ke -> 1.066, Kt -> 1.066 |>
```



7. Steady State

7.1. Calculating Steady State Values : The Final Value Theorem

We explore the steady state behavior, the limit of the time domain functions as time goes to infinity.

One way to calculate said limit is to brute-force calculate the limit in question. While that works, and we can and have previously calculated same (Ref: <https://github.com/rgatkinson/RobotPhysics/blob/master/MotorPhysics.pdf>), it is very slow to do so.

A more efficient way to calculate the steady state behavior is to avail ourselves of the use of the Final Value Theorem (Ref: <https://en.wikipedia.org/wiki/>

Final_value_theorem). The final value theorem tells us that if the time-domain limit exists, it is equal to the limit of the Laplace transform of the time domain function as $s \rightarrow 0$.

```
Clear[ssFinalValueTheorem]
ssFinalValueTheorem[model_, inputs_] := Module[{s(Unique[ ]*), t(Unique[ ]*), xformedInputs, expr},
  xformedInputs = LaplaceTransform[#, t, s] &@ inputs;
  expr = (model[s].xformedInputs)[[1]];
  (*Limit is really slow if we include assumptions (!?), so don't*)
  Block[{$Assumptions = {}}, Limit[s expr, s -> 0] // FullSimplify]
]

(ss = {
  ssPos -> ssFinalValueTheorem[motorPositionModel, {1 &, ΔvappConst &, ΔτappConst &}] // Factor,
  ssVel -> ssFinalValueTheorem[motorVelocityModel, {1 &, ΔvappConst &, ΔτappConst &}] // Factor,
  ssAcc -> ssFinalValueTheorem[motorAccelerationModel, {1 &, ΔvappConst &, ΔτappConst &}] // Factor,
  ssEmf -> ssFinalValueTheorem[motorEMFModel, {1 &, ΔvappConst &, ΔτappConst &}] // Factor,
  ssCur -> ssFinalValueTheorem[motorCurrentModel, {1 &, ΔvappConst &, ΔτappConst &}] // Factor,
  ssTor -> ssFinalValueTheorem[motorTorqueModel, {1 &, ΔvappConst &, ΔτappConst &}] // Factor
} // FullSimplify) // prettyPrint

simpleSs =
ss /. {N -> 1, η -> 1, Bafter -> 0, ΔτappConst -> 0, τapp0 -> 0, vapp0 -> 0, ΔvappConst -> vbat} // FullSimplify // prettyPrint

ssPos -> Indeterminate
ssVel -> 
$$\frac{N(K_t(v_{app0} + \Delta v_{app} \text{Const}) \eta N + R(\Delta \tau_{app} \text{Const} + \tau_{app0}))}{(K_e K_t + B R) \eta N^2 + B_{after} R}$$

ssAcc -> 0
ssEmf -> 
$$\frac{K_e N(K_t(v_{app0} + \Delta v_{app} \text{Const}) \eta N + R(\Delta \tau_{app} \text{Const} + \tau_{app0}))}{(K_e K_t + B R) \eta N^2 + B_{after} R}$$

ssCur -> 
$$\frac{B_{after}(v_{app0} + \Delta v_{app} \text{Const}) + N(B(v_{app0} + \Delta v_{app} \text{Const}) \eta N - K_e(\Delta \tau_{app} \text{Const} + \tau_{app0}))}{(K_e K_t + B R) \eta N^2 + B_{after} R}$$

ssTor -> 
$$\frac{K_t(B_{after}(v_{app0} + \Delta v_{app} \text{Const}) + N(B(v_{app0} + \Delta v_{app} \text{Const}) \eta N - K_e(\Delta \tau_{app} \text{Const} + \tau_{app0})))}{(K_e K_t + B R) \eta N^2 + B_{after} R}$$


ssPos -> Indeterminate
ssVel -> 
$$\frac{K_t v_{bat}}{K_e K_t + B R}$$

ssAcc -> 0
ssEmf -> 
$$\frac{K_e K_t v_{bat}}{K_e K_t + B R}$$

ssCur -> 
$$\frac{B v_{bat}}{K_e K_t + B R}$$

ssTor -> 
$$\frac{B K_t v_{bat}}{K_e K_t + B R}$$

```

The “Indeterminate” result regarding position is Mathematica informing us that it has proven to itself that the limit does not exist.

```
saveDefinitions[outputDirectory <> "SteadyState.m", {ss, simpleSs}]
```

7.2. On the Applicability of the Final Value Theorem

Can we always use the Final Value Theorem?

No, the Final Value Theorem only applies under certain conditions. Specifically: that all non-zero poles of the transfer function must have negative real parts, and the transfer function must have at most one pole at the origin (Ref: <http://www-personal.umich.edu/~dsbaero/others/39-FVTrevisited.pdf>). The predicate exhibited in the ConditionalExpression of the brute-force result captures exactly these conditions. When using the Final Value Theorem on any concrete system, we will need to test the conditions in any particular application. Lets illustrate by looking at the velocity model.


```
motorVelocityModel
```

```
(velPoles = TransferFunctionPoles[motorVelocityModel] // Flatten // Union)
```

$$\omega = \frac{1}{N (K_t v_{app0} \eta N + R \tau_{app0})} \frac{v_{app}}{K_t \eta N^2} \frac{\tau_{app}}{N (R + L s)}$$

$$\frac{B a f t e r R + (K_e K_t + B R) \eta N^2}{K_e K_t \eta N^2 + (R + L s) (B a f t e r + J a f t e r s + \eta N^2 (B + J s))} \frac{K_e K_t \eta N^2 + (R + L s) (B a f t e r + J a f t e r s + \eta N^2 (B + J s))}{N (R + L s)}$$

$$\left\{ \frac{-B a f t e r L - J a f t e r R - B L \eta N^2 - J R \eta N^2 - \sqrt{-4 (J a f t e r L + J L \eta N^2) (B a f t e r R + K_e K_t \eta N^2 + B R \eta N^2) + (B a f t e r L + J a f t e r R + B L \eta N^2 + J R \eta N^2)^2}}{2 (J a f t e r L + J L \eta N^2)}, \right.$$

$$\left. \frac{-B a f t e r L - J a f t e r R - B L \eta N^2 - J R \eta N^2 + \sqrt{-4 (J a f t e r L + J L \eta N^2) (B a f t e r R + K_e K_t \eta N^2 + B R \eta N^2) + (B a f t e r L + J a f t e r R + B L \eta N^2 + J R \eta N^2)^2}}{2 (J a f t e r L + J L \eta N^2)} \right\}$$

In order to get a feel for where these zeros are in practice, let's look at them for an actual example motor. We compute where the poles occur for one of our examples;

```
velPoles /. aMotorClassic // siUnits // clearUnits // N
```

```
{-4754.7, -0.377374}
```

This, for this example, the Final Value Theorem applies (at least for velocity). As another example, adding a ridiculous additional load moves the poles, but they remain in the LHS of the complex plane.

```
velPoles /. addMotorLoad[aMotorClassic, flywheel[Quantity[1000, "kg"], Quantity[1, "m"]]] // siUnits // clearUnits // N
```

```
{-4755.04, -0.000753194}
```

An analogous analytic development awaits a future revision to this document, as do examples that involve gears or externally applied torque.

8. More Administrivia

```
saveDefinitions[outputDirectory <> "Misc.m",
{importMotorData, motorParameters, motorLoad, flywheel, addMotorLoad, massOnPulley, applyTimeFunction}]
```

9. Revision History

- 2018.07.23 Added acceleration-dependant external torque. Corrected “viscous friction” to “viscous drag”. Clarified usage of units.
- 2018.08.06 Added gearbox support, removed time consuming internal tests. Regularized state variable names. Removed (flawed) “Back EMF vs Applied Voltage” discussion
- 2018.08.06 Renamed variables. Remove setting \$Assumptions (that can slow things down)
- 2018.08.11 Added initial conditions support