

Motor Physics, with Gears & Initial Conditions

Robert Atkinson, Maddy Nguyen, Samantha Hordyk, Cole Welch, and John Fraser
27 August 2018

An exploration of the physics of motors, in both symbolic and numeric form. A basic but sufficiently-complete parameterized model of a motor is presented. The differential equations governing that model are exhibited, and, from those, the (frequency domain) transfer functions of each of the model's variables is developed from first principles. Next, mathematical infrastructure for converting the response of general frequency domain functions input signals is developed, then applied to the transfer functions as a function of initial conditions, and the applied-voltage and external-applied-torque inputs. This yields time domain functions for each of the model's variables in terms of the (symbolic) value of the model's variables. With those time domain symbolic functions in hand, the steady-state response of the motor as a function of each of the model's parameters is explored.

1. Administrivia

We begin by loading some handy utilities.

```
Get[NotebookDirectory[] <> "Utilities.m"]
```


```
outputDirectory = FileNameJoin[{NotebookDirectory[], "MotorPhysicsGearsInitialConditions.Output"}] <> $PathnameSeparator  
CreateDirectory[outputDirectory] // Quiet;
```

```
C:\ftc\RobotPhysics\MotorPhysicsGearsInitialConditions.Output\
```

```
SetOptions[Plot, LabelStyle -> Directive[Background -> None]];
```

2. Motor Model

The following figure gives a conceptual overview of a DC motor (ref: <http://www.egr.msu.edu/classes/me451/jchoi/2008/>).



In addition to the the above, we also here incorporate into our model:

- an (optional) N:1 reducing gearbox with associated efficiency
- an autonomous time-dependent external torque

The state of the system can thus be described by the following variables.

- $v_{app}(t)$, the applied voltage (illustrated as $e_a(t)$ above)
- $i(t)$, the armature current
- $v_g(t)$, the back EMF ('g' for 'generator'; illustrated as $e_b(t)$ above)
- $\theta(t)$, the angular position of the shaft, before the gearbox
- $\omega(t)$, the angular velocity of the shaft, before the gearbox

- $\alpha(t)$, the angular acceleration of the shaft, before the gearbox
- $\tau(t)$, the torque driven by the armature current, before the gearbox
- $\tau_{app}(t)$, an autonomous externally applied torque, after the gearbox. Often, this is constant, and caused by gravitational action.

The parameter constants that characterize the electrical aspects of the system are the following

- K_e , the electromotive force constant
- K_t , the motor torque constant
- L , electric inductance
- R , electric resistance

The following parameter constants characterize the mechanical aspects of the system

- J , the moment of inertia of the armature etc before the gearbox
- J_{after} , the moment of inertia of any load or torque attached after the gearbox. May be (net) negative, depending on system; see $\tau_{AfterExpr}$ below.
- B , motor viscous drag constant
- B_{after} , viscous drag of any load attached after the gearbox
- η , the efficiency of the motor's gear box (our current model does not support direction-dependent gearbox efficiencies).
- N (capital nu), the reduction ratio of the gear box

We collect together basic information about our parameters.

```
parameterNames = {J, Jafter, B, Bafter, Ke, Kt, R, L, η, N}
parameterAssumptions = {};
parameterAssumptions = Union[parameterAssumptions, # ≥ 0 & /@ {J, B, t}];
(*allow J & B to be zero so we can explore simpler motor models*)
parameterAssumptions = Union[parameterAssumptions, # > 0 & /@ {R, L, Ke, Kt, N, η}];
parameterAssumptions = Union[parameterAssumptions, # ∈ ℝ & /@ {Jafter, Bafter, vapp0, τapp0, ΔvappConst, ΔτappConst}];
parameterAssumptions = Union[parameterAssumptions, #[_] ∈ ℝ & /@ {Δvapp, i, vg, θ, ω, α, τ, τafter, Δτapp}]

{J, Jafter, B, Bafter, Ke, Kt, R, L, η, N}

{Bafter ∈ ℝ, Jafter ∈ ℝ, vapp0 ∈ ℝ, ΔvappConst ∈ ℝ, ΔτappConst ∈ ℝ, τapp0 ∈ ℝ, i[_] ∈ ℝ, vg[_] ∈ ℝ, α[_] ∈ ℝ, Δvapp[_] ∈ ℝ,
Δτapp[_] ∈ ℝ, θ[_] ∈ ℝ, τ[_] ∈ ℝ, τafter[_] ∈ ℝ, ω[_] ∈ ℝ, Ke > 0, Kt > 0, L > 0, R > 0, η > 0, N > 0, B ≥ 0, J ≥ 0, t ≥ 0}
```

3. Differential Equations

3.1. Differential Equation Development

3.1.1. On Gearboxes and Torques

Our modelled system contains a gearbox with (reducing) gear ratio N and efficiency η . On either side of the gearbox we may have loads, drags, and driving torques, the latter being electrically generated on one side and externally applied on the other. In order to be able to write the torque differential equation for the system as a whole, we need to be able to take these two sets of torques as they are found in their respective *separate* places in the system and rewrite them both as they are *effectively* felt at some *one* point in the system. Commonly, in mechanical engineering this one point is chosen to be just before the gearbox, and the loads / torques after the gearbox are said to be “reflected” to create corresponding equivalent reflected loads and torques at the before-gearbox location. Here, we develop the mathematics of carrying out this reflection.

Just a couple of principles are involved:

- State related to angular position (position, velocity, acceleration, etc.) on the output side is $1/N$ th that of on the input side.
- Torque on the output side is $N\eta$ times that of on the input side.

```

Clear[reflectAngle, reflectTorque, reflectAngleCoefficient, reflectDrag, reflectInertia]
reflectAngle[expr_, out_, in_, t_] := expr /. {out[t] → in[t] / N, Derivative[n_][out][t] → Derivative[n][in][t] / N}
reflectAngle[expr_] := reflectAngle[expr, θafter, θ, t]
reflectTorque[τout_] := τout / (N η)
reflectAngleCoefficient[coeff_] := coeff / N
reflectDrag[b_] := b // reflectAngleCoefficient // reflectTorque
reflectInertia[j_] := j // reflectAngleCoefficient // reflectTorque

```

```

reflectTorque[ΔτappConst + Bafter θafter'[t] + Jafter θafter''[t]] // reflectAngle // Expand
{reflectDrag[Bafter], reflectInertia[Jafter]}

```

$$\frac{\Delta \tau_{app} \text{Const}}{\eta N} + \frac{B_{after} \theta' [t]}{\eta N^2} + \frac{J_{after} \theta'' [t]}{\eta N^2}$$

$$\left\{ \frac{B_{after}}{\eta N^2}, \frac{J_{after}}{\eta N^2} \right\}$$

3.1.2. Differential Equation System

With that understanding on how to reflect torques, we can now develop the differential equations that govern the system.

The motor torque is proportional to current.

$$k_t \text{Eqn} = \tau[t] == K_t i[t]$$

$$\tau[t] == K_t i[t]$$

The back EMF is proportional to velocity.

$$\text{backEmfEqn} = v_g[t] == K_e \theta' [t]$$

$$v_g[t] == K_e \theta' [t]$$

The after-gearbox angular position is a simple ratio of the before-gearbox position. This also applies to the velocity and acceleration.

$$\theta_{after} \text{Eqns} = \text{Derivative}[\#][\theta_{after}][t] == \text{Derivative}[\#][\theta][t] / N \ \& \ /@ \ \text{Range}[0, 2]$$

$$\left\{ \theta_{after}[t] = \frac{\theta[t]}{N}, \theta_{after}'[t] = \frac{\theta'[t]}{N}, \theta_{after}''[t] = \frac{\theta''[t]}{N} \right\}$$

Before the gearbox, the electrically driven motor torque is reduced by viscous drag and the inertia of the motor.

$$\tau_{BeforeExpr} = \tau[t] + -B \theta' [t] - J \theta'' [t]$$

$$\tau[t] - B \theta' [t] - J \theta'' [t]$$

After the gearbox, we have an autonomous, time-dependant externally applied torque, a viscous drag, and an internal load and/or an acceleration-dependent external torque.

$$\tau_{AfterExpr} = \tau_{app}[t] - B_{after} \theta_{after}' [t] - J_{after} \theta_{after}'' [t]$$

$$\tau_{app}[t] - B_{after} \theta_{after}' [t] - J_{after} \theta_{after}'' [t]$$

As all the torque must be accounted for, the sum of the before torques and the reflected after torques must be zero.

$$\tau \text{Eqn} = \tau_{BeforeExpr} + (\tau_{AfterExpr} // \text{reflectTorque} // \text{reflectAngle}) == 0;$$

$$\tau \text{Eqn} = \text{Collect}[\tau \text{Eqn}, \{\theta[t], \theta'[t], \theta''[t]\}];$$

$$\tau \text{Eqns} = \{\tau \text{Eqn}, \text{reflectTorque}[\tau_{after}[t]] == \tau[t]\}$$

$$\left\{ \tau[t] + \frac{\tau_{app}[t]}{\eta N} + \left(-B - \frac{B_{after}}{\eta N^2} \right) \theta'[t] + \left(-J - \frac{J_{after}}{\eta N^2} \right) \theta''[t] == 0, \frac{\tau_{after}[t]}{\eta N} == \tau[t] \right\}$$

The voltage drop across the resistance, induction and back EMF must equal the applied voltage (due to one of Kirchhoff's laws).

```
voltageEqn = vapp[t] == R i[t] + L i'[t] + vg[t]
vapp[t] == R i[t] + vg[t] + L i'[t]
```

We put it all together.

```
diffEqns = {ktEqn, backEmfEqn, voltageEqn} ~Join~ τEqns ~Join~ θafterEqns // FullSimplify;
diffEqns // prettyPrint
```

$$K_t i(t) = \tau(t)$$

$$v_g(t) = K_e \theta'(t)$$

$$R i(t) + v_g(t) + L i'(t) = v_{app}(t)$$

$$\frac{\eta \tau(t) N^2 + \tau_{app}(t) N - (B \eta N^2 + B_{after}) \theta'(t) - (J \eta N^2 + J_{after}) \theta''(t)}{\eta N} = 0$$

$$\tau(t) = \frac{\tau_{after}(t)}{\eta N}$$

$$\theta_{after}(t) = \frac{\theta(t)}{N}$$

$$\theta'_{after}(t) = \frac{\theta'(t)}{N}$$

$$\theta''_{after}(t) = \frac{\theta''(t)}{N}$$

3.2. Initial Conditions

3.2.1. Steady state before $t=0$

We will assume that initially, the motor is in steady state at some velocity because of the application of some particular applied voltage and external torque. In steady state, the derivative of current and velocity are zero.

```
someInitialConditions = {
  vapp[ss0] == vapp0,
  τapp[ss0] == τapp0,
  θ[ss0] == 0, (* arbitrary *)
  θ''[ss0] == 0, (* steady state condition *)
  i'[ss0] == 0 (* steady state condition *)
}

{vapp[ss0] == vapp0, τapp[ss0] == τapp0, θ[ss0] == 0, θ''[ss0] == 0, i'[ss0] == 0}
```

From those, we can calculate the values of the remainder of the state.

```
diffEqFunctionsOfTime = Reap[Scan[If[MatchQ[#, _[t]], Sow[#]] &, diffEqns, Infinity]][[2, 1]] // Union
others = Complement[diffEqFunctionsOfTime /. t -> ss0, someInitialConditions[[All, 1]]];
eqnsSubsSs0 = diffEqns /. t -> ss0 /. (someInitialConditions /. Equal -> Rule);
(solvedSs0 = (uniqueSolve[eqnsSubsSs0, others] /. Rule -> Equal));
(initialConditions = solvedSs0 ~Join~ someInitialConditions) // prettyPrint

{i[t], vapp[t], vg[t],  $\theta$ [t],  $\theta$ after[t],  $\tau$ [t],  $\tau$ after[t],  $\tau$ app[t], i'[t],  $\theta'$ [t],  $\theta$ after'[t],  $\theta''$ [t],  $\theta$ after''[t]}
```

```
i(ss0) = - $\frac{B \text{vapp0 } \eta N^2 + K_e \text{rapp0 } N - B \text{after vapp0}}{K_e K_t \eta N^2 + B R \eta N^2 + B \text{after } R}$ 
vg(ss0) = - $\frac{K_e K_t \text{vapp0 } \eta N^2 - K_e R \text{rapp0 } N}{K_e K_t \eta N^2 + B R \eta N^2 + B \text{after } R}$ 
 $\theta$ after(ss0) = 0
 $\tau$ (ss0) =  $\frac{K_t (B \text{vapp0 } \eta N^2 - K_e \text{rapp0 } N + B \text{after vapp0})}{K_e K_t \eta N^2 + B R \eta N^2 + B \text{after } R}$ 
 $\tau$ after(ss0) =  $\frac{K_t \eta N (B \text{vapp0 } \eta N^2 - K_e \text{rapp0 } N + B \text{after vapp0})}{K_e K_t \eta N^2 + B R \eta N^2 + B \text{after } R}$ 
 $\theta'$ (ss0) = - $\frac{K_t \text{vapp0 } \eta N^2 - R \text{rapp0 } N}{K_e K_t \eta N^2 + B R \eta N^2 + B \text{after } R}$ 
 $\theta$ after'(ss0) = - $\frac{K_t \text{vapp0 } \eta N - R \text{rapp0}}{K_e K_t \eta N^2 + B R \eta N^2 + B \text{after } R}$ 
 $\theta$ after''(ss0) = 0
vapp(ss0) = vapp0
rapp(ss0) = rapp0
 $\theta$ (ss0) = 0
 $\theta'$ (ss0) = 0
i'(ss0) = 0
```

3.2.2. Initial conditions at t==0

Recall our differential equations.

```
diffEqns // prettyPrint
```

```
Kt i(t) =  $\tau$ (t)
vg(t) =  $K_e \theta'(t)$ 
R i(t) + vg(t) + L i'(t) = vapp(t)
 $\frac{\eta \tau(t) N^2 + \text{rapp}(t) N - (B \eta N^2 + B \text{after}) \theta'(t) - (J \eta N^2 + J \text{after}) \theta''(t)}{\eta N} = 0$ 
 $\tau(t) = \frac{\tau \text{after}(t)}{\eta N}$ 
 $\theta \text{after}(t) = \frac{\theta(t)}{N}$ 
 $\theta \text{after}'(t) = \frac{\theta'(t)}{N}$ 
 $\theta \text{after}''(t) = \frac{\theta''(t)}{N}$ 
```

In order to be able to use Laplace transforms, we need to adjust these so that the input values are zero at t==0. To do so, we substitute in a bias for both inputs.

```
( $\Delta$ diffEqns = diffEqns /. {vapp[t] -> vapp0 +  $\Delta$ vapp[t],  $\tau$ app[t] ->  $\tau$ app0 +  $\Delta$  $\tau$ app[t]}) // prettyPrint
```

```
Kt i(t) =  $\tau$ (t)
vg(t) =  $K_e \theta'(t)$ 
R i(t) + vg(t) + L i'(t) = vapp0 +  $\Delta$ vapp(t)
 $\frac{\eta \tau(t) N^2 + (\text{rapp0} + \Delta \text{rapp}(t)) N - (B \eta N^2 + B \text{after}) \theta'(t) - (J \eta N^2 + J \text{after}) \theta''(t)}{\eta N} = 0$ 
 $\tau(t) = \frac{\tau \text{after}(t)}{\eta N}$ 
 $\theta \text{after}(t) = \frac{\theta(t)}{N}$ 
 $\theta \text{after}'(t) = \frac{\theta'(t)}{N}$ 
 $\theta \text{after}''(t) = \frac{\theta''(t)}{N}$ 
```

We now set up the initial conditions applicable for t >= 0.

```

initialConditionsRules = initialConditions /. Equal → Rule;
ΔsomeInitialConditions = {
  Δvapp[0] == ΔvappConst,
  Δτapp[0] == ΔτappConst,
  θ[0] == 0,
  i[0] == i[ss0] /. initialConditionsRules /. ss0 → 0,
  θ'[0] == θ'[ss0] /. initialConditionsRules /. ss0 → 0
}

```

```

{Δvapp[0] == ΔvappConst, Δτapp[0] == ΔτappConst, θ[0] == 0,
  i[0] ==  $-\frac{\text{Bafter vapp0} - \text{B vapp0 } \eta \text{ N}^2 + \text{Ke N } \tau \text{ app0}}{\text{Bafter R} + \text{Ke Kt } \eta \text{ N}^2 + \text{B R } \eta \text{ N}^2}$ , θ'[0] ==  $-\frac{-\text{Kt vapp0 } \eta \text{ N}^2 - \text{R N } \tau \text{ app0}}{\text{Bafter R} + \text{Ke Kt } \eta \text{ N}^2 + \text{B R } \eta \text{ N}^2}$ }}

```

```

ΔDiffEqFunctionsOfTime = Reap[Scan[If[MatchQ[#, _[t]], Sow[#]] &, ΔdiffEqns, Infinity]][[2, 1]] // Union
Δothers = Complement[ΔDiffEqFunctionsOfTime /. t → 0, ΔsomeInitialConditions[All, 1]]
ΔeqnsSubsKnownInitial = ΔdiffEqns /. t → 0 /. (ΔsomeInitialConditions /. Equal → Rule);
ΔsolvedInitialConditions = (uniqueSolve[ΔeqnsSubsKnownInitial, Δothers] /. Rule → Equal);
(ΔinitialConditions = ΔsolvedInitialConditions ~Join~ ΔsomeInitialConditions) // prettyPrint

```

```

{i[t], vg[t], Δvapp[t], Δτapp[t], θ[t], θafter[t], τ[t], τafter[t], i'[t], θ'[t], θafter'[t], θ''[t], θafter''[t]}

```

```

{vg[0], θafter[0], τ[0], τafter[0], i'[0], θafter'[0], θ''[0], θafter''[0]}

```

```

vg(0) =  $\frac{\text{Ke}(\text{Kt vapp0 } \eta \text{ N}^2 + \text{R rapp0 N})}{\text{Ke Kt } \eta \text{ N}^2 + \text{B R } \eta \text{ N}^2 + \text{Bafter R}}$ 
θafter(0) = 0
τ(0) =  $-\frac{\text{B Kt vapp0 } \eta \text{ N}^2 + \text{Ke Kt rapp0 N} - \text{Bafter Kt vapp0}}{\text{Ke Kt } \eta \text{ N}^2 + \text{B R } \eta \text{ N}^2 + \text{Bafter R}}$ 
τafter(0) =  $\frac{\eta \text{ N}(\text{B Kt vapp0 } \eta \text{ N}^2 - \text{Ke Kt rapp0 N} + \text{Bafter Kt vapp0})}{\text{Ke Kt } \eta \text{ N}^2 + \text{B R } \eta \text{ N}^2 + \text{Bafter R}}$ 
i'(0) =  $\frac{\Delta \text{vappConst}}{L}$ 
θafter'(0) =  $-\frac{-\text{Kt vapp0 } \eta \text{ N} - \text{R rapp0}}{\text{Ke Kt } \eta \text{ N}^2 + \text{B R } \eta \text{ N}^2 + \text{Bafter R}}$ 
θ''(0) =  $\frac{\Delta \text{rappConst N}}{J \eta \text{ N}^2 + \text{Jafter}}$ 
θafter''(0) =  $\frac{\Delta \text{rappConst}}{J \eta \text{ N}^2 + \text{Jafter}}$ 
Δvapp(0) = ΔvappConst
Δτapp(0) = ΔτappConst
θ(0) = 0
i(0) =  $-\frac{\text{B vapp0 } \eta \text{ N}^2 + \text{Ke rapp0 N} - \text{Bafter vapp0}}{\text{Ke Kt } \eta \text{ N}^2 + \text{B R } \eta \text{ N}^2 + \text{Bafter R}}$ 
θ'(0) =  $-\frac{-\text{Kt vapp0 } \eta \text{ N}^2 - \text{R rapp0 N}}{\text{Ke Kt } \eta \text{ N}^2 + \text{B R } \eta \text{ N}^2 + \text{Bafter R}}$ 

```

4. Motor Parameters

4.1. Motor Parameter Units

We develop infrastructure regarding the units in which each of our parameters are specified. This will help us sanity check the structure of our equations, as they must be internally consistent in their use of units (e.g.: no velocities added to masses, etc).

A note on radians as a unit. The Bureau International des Poids et Mesures, the intergovernmental organization that standardizes the SI, officially refers to a radian as “a special name is given to the unit one, in order to facilitate the identification of the quantity involved” (Ref: <https://www.bipm.org/en/publications/si-brochure/section2-2-3.html>). With that as background, we have here chosen to include radians in all places in which they logically belong, as doing so provides valuable insight regarding the quantities being manipulated. Thus, for example, you will find here torque having units of N m / radian instead of N m, and the like. As, from the SI perspective, this amounts to multiplying or dividing by the unit one, no semantic change is effected.

Aside: in the opinion of one of the authors (Atkinson), that “a special name given to the unit one” is in any way at all different than a first-class “unit” is a perspective reasonable only if one artificially and myopically limits one’s worldview to contain only the seven units whose dimension is necessarily *experimentally* calibrated as opposed to units whose dimension is mathematically or otherwise calibrated, such as radian, becquerel, motor encoder ticks, etc. And the world is chock-full of the latter. But this is a discussion for another time. See also Jacques E. Romain, “Angle as a Fourth Fundamental

Quantity", Journal of Research of the National Bureau of Standards-B. Mathematics and Mathematical Physics, Vol. 66B, No. 3, July- September 1962.

```

radiansUnits = parseUnit["Radians"];
secondsUnits = parseUnit["Seconds"];
siAngularPositionUnits = radiansUnits;
siAngularVelocityUnits = siAngularPositionUnits / secondsUnits;
siAngularAccelerationUnits = siAngularVelocityUnits / secondsUnits;
siTorqueUnits = parseUnit["N m"] / radiansUnits;
siAngularViscousDragUnits = siTorqueUnits / siAngularVelocityUnits;
siAngularInertialUnits = siTorqueUnits / siAngularAccelerationUnits;

parameterUnits = {
  R → parseUnit["Ohms"],
  L → parseUnit["Henrys"],
  i[t] → parseUnit["Amperes"],
  i'[t] → parseUnit["Amperes"] / secondsUnits,
  i''[t] → parseUnit["Amperes"] / secondsUnits / secondsUnits,
  Δvapp[t] → parseUnit["Volts"],
  ΔvappConst → parseUnit["Volts"],
  vg[t] → parseUnit["Volts"],
  vapp0 → parseUnit["Volts"],
  vapp0 → parseUnit["Volts"],

  J → siAngularInertialUnits,
  Jafter → siAngularInertialUnits,
  B → siAngularViscousDragUnits,
  Bafter → siAngularViscousDragUnits,

  Ke → parseUnit["Seconds Volts / Radians"],
  Kt → parseUnit["Newtons Meters"] / parseUnit["Amperes"] / radiansUnits,

  θ[t] → siAngularPositionUnits, θafter[t] → siAngularPositionUnits,
  θ'[t] → siAngularVelocityUnits, θafter'[t] → siAngularPositionUnits,
  θ''[t] → siAngularAccelerationUnits, θafter''[t] → siAngularPositionUnits,
  ω[t] → siAngularVelocityUnits,
  α[t] → siAngularAccelerationUnits,

  τ[t] → siTorqueUnits,
  τafter[t] → siTorqueUnits,
  Δτapp[t] → siTorqueUnits,
  ΔτappConst → siTorqueUnits,
  τapp0 → siTorqueUnits,
  τapp0 → siTorqueUnits,

  N → parseUnit["DimensionlessUnit"],
  η → parseUnit["DimensionlessUnit"]
} // Association;
unitsToQuantities[units_] := Module[{rules, makeQuantity},
  rules = Normal[units];
  makeQuantity = Function[{param, unit}, Quantity[param, unit]];
  (#[1] → makeQuantity @@ # & /@ rules) // Association
]
parameterQuantities = unitsToQuantities[parameterUnits]

⟨ { R → R Ω, L → L H, i[t] → i[t] A, i'[t] → i'[t] A/s, i''[t] → i''[t] A/s², Δvapp[t] → Δvapp[t] V,
  ΔvappConst → ΔvappConst V, vg[t] → vg[t] V, vapp0 → vapp0 V, J → J ms⁴N/rad⁴, Jafter → Jafter ms⁴N/rad⁴,
  B → B msN/rad², Bafter → Bafter msN/rad², Ke → Ke sV/rad, Kt → Kt mN/(Arad), θ[t] → θ[t] rad,
  θafter[t] → θafter[t] rad, θ'[t] → θ'[t] rad/s, θafter'[t] → θafter'[t] rad, θ''[t] → θ''[t] rad/s²,
  θafter''[t] → θafter''[t] rad, ω[t] → ω[t] rad/s, α[t] → α[t] rad/s², τ[t] → τ[t] mN/rad, τafter[t] → τafter[t] mN/rad,
  Δτapp[t] → Δτapp[t] mN/rad, ΔτappConst → ΔτappConst mN/rad, τapp0 → τapp0 mN/rad, N → N, η → η } ⟩

```

Let's look and see if our units are consistent:

```
ΔdiffEqns /. parameterQuantities // ColumnForm
```

$$Kt i[t] \text{ mN/rad} == \tau[t] \text{ mN/rad}$$

$$vg[t] \text{ V} == Ke \theta'[t] \text{ V}$$

$$(R i[t] + vg[t] + L i'[t]) A \Omega == (vapp\theta + \Delta vapp[t]) \text{ V}$$

$$\frac{N (\tau app\theta + \Delta \tau app[t]) + \eta N^2 \tau[t] - (B after + B \eta N^2) \theta'[t] - (J after + J \eta N^2) \theta''[t]}{\eta N} \text{ mN/rad} == 0$$

$$\tau[t] \text{ mN/rad} == \frac{\tau after[t]}{\eta N} \text{ mN/rad}$$

$$\theta after[t] \text{ rad} == \frac{\theta[t]}{N} \text{ rad}$$

$$\theta after'[t] \text{ rad} == \frac{\theta'[t]}{N} \text{ rad/s}$$

$$\theta after''[t] \text{ rad} == \frac{\theta''[t]}{N} \text{ rad/s}^2$$

Huzzah! They check out (if the units were incompatible, Mathematica would've told us). How about our initial conditions?


```
initialConditions /. parameterQuantities // siUnits // simplifyUnits // ColumnForm
ΔinitialConditions /. parameterQuantities // siUnits // simplifyUnits // ColumnForm
```

```
i[ss0] == - (Bafter vapp0 - B vapp0 η N^2 + Ke N τapp0) / (Bafter R + Ke Kt η N^2 + B R η N^2) A
vg[ss0] == - (Ke Kt vapp0 η N^2 - Ke R N τapp0) / (Bafter R + Ke Kt η N^2 + B R η N^2) V
ϕafter[ss0] == 0
τ[ss0] == (Kt (Bafter vapp0 - B vapp0 η N^2 - Ke N τapp0)) / (Bafter R + Ke Kt η N^2 + B R η N^2) kg m^2 / (s^2 rad)
τafter[ss0] == (Kt η N (Bafter vapp0 - B vapp0 η N^2 - Ke N τapp0)) / (Bafter R + Ke Kt η N^2 + B R η N^2) kg m^2 / (s^2 rad)
Θ'[ss0] == - (Kt vapp0 η N^2 - R N τapp0) / (Bafter R + Ke Kt η N^2 + B R η N^2) rad / s
ϕafter'[ss0] == - (Kt vapp0 η N - R τapp0) / (Bafter R + Ke Kt η N^2 + B R η N^2) rad / s
ϕafter''[ss0] == 0
vapp[ss0] == vapp0 V
τapp[ss0] == τapp0 kg m^2 / (s^2 rad)
Θ[ss0] == 0
Θ''[ss0] == 0
i'[ss0] == 0
```

```
vg[0] == (Ke (Kt vapp0 η N^2 + R N τapp0)) / (Bafter R + Ke Kt η N^2 + B R η N^2) V
ϕafter[0] == 0
τ[0] == - (Bafter Kt vapp0 - B Kt vapp0 η N^2 + Ke Kt N τapp0) / (Bafter R + Ke Kt η N^2 + B R η N^2) kg m^2 / (s^2 rad)
τafter[0] == (η N (Bafter Kt vapp0 - B Kt vapp0 η N^2 - Ke Kt N τapp0)) / (Bafter R + Ke Kt η N^2 + B R η N^2) kg m^2 / (s^2 rad)
i'[0] == (ΔvappConst) / L A / s
ϕafter'[0] == - (Kt vapp0 η N - R τapp0) / (Bafter R + Ke Kt η N^2 + B R η N^2) rad / s
Θ''[0] == (ΔτappConst N) / (Jafter + J η N^2) rad / s^2
ϕafter''[0] == (ΔτappConst) / (Jafter + J η N^2) rad / s^2
Δvapp[0] == ΔvappConst V
Δτapp[0] == ΔτappConst kg m^2 / (s^2 rad)
Θ[0] == 0
i[0] == - (Bafter vapp0 - B vapp0 η N^2 + Ke N τapp0) / (Bafter R + Ke Kt η N^2 + B R η N^2) A
Θ'[0] == - (Kt vapp0 η N^2 - R N τapp0) / (Bafter R + Ke Kt η N^2 + B R η N^2) rad / s
```

Good. We save information regarding our parameters for later use.

```
saveDefinitions[outputDirectory <> "ParametersUnitsAndAssumptions.m",
{parameterUnits, parameterQuantities, parameterAssumptions, radiansUnits, secondsUnits, siAngularPositionUnits,
siAngularVelocityUnits, siAngularAccelerationUnits, siTorqueUnits, siAngularViscousDragUnits, siAngularInertialUnits}]
```

4.2. On Ke & Kt

A brief digression. It is often remarked that, in compatible units, Ke & Kt have the same magnitude and sign. However, while often true, this relation does not *always* hold.

Recall our differential equations:

```
diffEqns // ColumnForm
```

```
Kt i[t] == τ[t]
vg[t] == Ke θ'[t]
R i[t] + vg[t] + L i'[t] == vapp[t]

$$\frac{\eta N^2 \tau[t] + N \tauapp[t] - (Bafter + B \eta N^2) \theta'[t] - (Jafter + J \eta N^2) \theta''[t]}{\eta N} == 0$$

τ[t] ==  $\frac{\tauafter[t]}{\eta N}$ 
θafter[t] ==  $\frac{\theta[t]}{N}$ 
θafter'[t] ==  $\frac{\theta'[t]}{N}$ 
θafter''[t] ==  $\frac{\theta''[t]}{N}$ 
```

In steady state, the current and speed are constant.

```
(ssEqns = diffEqns /. {i'[t] → 0, θ''[t] → 0}) // ColumnForm
(ssEqns = Eliminate[ssEqns, {vg[t], B, θafter[t], θafter'[t], θafter''[t], τapp[t]}]) // ColumnForm
(ssEqns = ssEqns[[1 ;; 2]]) // ColumnForm
```

```
Kt i[t] == τ[t]
vg[t] == Ke θ'[t]
R i[t] + vg[t] == vapp[t]

$$\frac{\eta N^2 \tau[t] + N \tauapp[t] - (Bafter + B \eta N^2) \theta'[t]}{\eta N} == 0$$

τ[t] ==  $\frac{\tauafter[t]}{\eta N}$ 
θafter[t] ==  $\frac{\theta[t]}{N}$ 
θafter'[t] ==  $\frac{\theta'[t]}{N}$ 
θafter''[t] == 0
```

```
Kt i[t] == τ[t]
vapp[t] == R i[t] + Ke θ'[t]
τafter[t] == η N τ[t]
η ≠ 0
N ≠ 0
```

```
Kt i[t] == τ[t]
vapp[t] == R i[t] + Ke θ'[t]
```

From a power point of view (ref: <https://bit.ly/2Lxka0Z>), we must have (electrical) power in = (mechanical) power out + (electrical + other) power losses:

```
powerEqns = And @@ {
  pwrIn == vapp[t] i[t],
  pwrOut == τ[t] θ'[t],
  pwrLosses == i[t]^2 R + pwrOtherPowerLosses,
  pwrIn == pwrOut + pwrLosses
};
(ssPowerEqns = Union[ssEqns, powerEqns]) // prettyPrint
```

```
pwrIn = pwrLosses + pwrOut
pwrIn = i(t) vapp(t)
pwrLosses = R i(t)^2 + pwrOtherPowerLosses
pwrOut = τ(t) θ'(t)
Kt i(t) = τ(t)
vapp(t) = R i(t) + Ke θ'(t)
```

Let's eliminate a few of the variables that we're not interested in.

```
kBalanceEqns = Eliminate[ssPowerEqns, {i[t], τ[t], θ'[t], vapp[t]}];
kBalanceEqns // prettyPrint
uniqueSolve[kBalanceEqns[[2]], pwrOtherPowerLosses] /. Rule → Equal
```

```
pwrIn = pwrLosses + pwrOut
Kt pwrOtherPowerLosses == (Ke - Kt) pwrOut
```

```
{pwrOtherPowerLosses ==  $\frac{(Ke - Kt) pwrOut}{Kt}$ }
```

We can conclude that if other power losses are zero, then K_e must equal K_t .

4.3. Experimental Motor Data

We have experimental data from several motors (Nguyen, Hordyk, & Fraser, 2017). We load in same. Note that the motor parameters herein have all been measured *after* the attached gearbox.

```
Clear[importMotorData, motorData]
importMotorData[] := importMotorData["Characterized Motors (Gears)v2.xlsx"];
importMotorData[file_] := Module[{}],
  Import[NotebookDirectory[] <> file, {"Data", 1}][[2 ;;]]
]
motorData = importMotorData[];
(motorData) // TableForm
```

Name	Model	R	L (uH)	L (H)	K_e	K_t (est)	J	B	Ticks/Rev	N	η_f
AM 20 A	am-3102	2.3	691.	0.000691	0.351	0.351	9.011×10^{-6}	0.0022	560.	20.	0.9
AM 20 B		1.9	684.	0.000684	0.389	0.389	9.011×10^{-6}	0.0025	560.	20.	0.9
AM 20 C		5.1	717.	0.000717	0.385	0.385	8.931×10^{-6}	0.0028	560.	20.	0.9
AM 40 A	am-2964a	2.5	674.	0.000674	0.753	0.753	0.0002221	0.2269	1120.	40.	0.9
AM 40 B		3.8	705.	0.000705	0.705	0.705	0.0001741	0.56	1120.	40.	0.9
AM 40 C		2.1	716.	0.000716	0.763	0.763	0.0002471	0.018	1120.	40.	0.9
AM 60 A	am-3103	3.3	694.	0.000694	1.066	1.066	0.0001041	0.033	1680.	60.	0.9
AM 60 B		5.1	696.	0.000696	1.076	1.076	8.421×10^{-6}	0.02	1680.	60.	0.9
AM 3.7 A	am-3461	8.9	679.	0.000679	0.099	0.099	0.0002791	0.00014	44.4	3.7	0.9
AM 3.7 B		2.6	797.	0.000797	0.108	0.108	0.0003151	0.000176	44.4	3.7	0.9
AM 3.7 C		8.7	880.	0.00088	0.105	0.105	0.0003091	0.00017	44.4	3.7	0.9
Matrix A	14-0011	3.8	718.	0.000718	0.34	0.34	9.431×10^{-6}	0.00151	1478.4	52.8	0.9
Matrix B		7.8	777.	0.000777	0.363	0.363	7.761×10^{-6}	0.00191	1478.4	52.8	0.9
Matrix C	REV-41-1300	20.6	658.	0.000658	0.338	0.338	7.231×10^{-6}	0.00186	1478.4	52.8	0.9
CoreHex A		3.6	1356.	0.001356	0.822	0.822	0.0007331	0.0112	288.	72.	0.9
CoreHex B		11.3	1352.	0.001352	0.858	0.858	0.0006551	0.008	288.	72.	0.9
CoreHex C		5.6	1342.	0.001342	0.711	0.711	0.0004541	0.0078	288.	72.	0.9
Tetrix	W39530								1440.	52.	

We build a function to retrieve data from the table. We convert from floating point to rational in order to help delay floating point collapse later on down the line. We include load and external torque parameters at zero level so these can be easily added to later.

```

Clear[motorParameters];
Options[motorParameters] = {WithGearbox -> True, Efficiency -> "Forward"};
validateOption[motorParameters][WithGearbox, val_] := BooleanQ[val];
validateOption[motorParameters][Efficiency, val_] := MemberQ[{"Forward", "Reverse"}, val];

motorParameters[motorName_, opts: OptionsPattern[]] := motorParameters[motorData, motorName, opts]
motorParameters[motorData_, motorName_, opts: OptionsPattern[]] :=
  validateOptions[] @ Module[{row, paramValues, quantify, qty, quantifiedRow, result, gearbox},
    row = Select[motorData, #[[1]] == motorName &][[1]];
    paramValues = #[[1]] -> toRational[ row[[#[[2]]]] ] & /@ {
      {R, 3}, {L, 5}, {Ke, 6}, {Kt, 7}, {J, 8}, {B, 9},
      {N, 11}, {η, If[OptionValue[Efficiency] == "Forward", 12, 13]};
    quantify = Function[{name, value},
      qty = (name /. parameterQuantities);
      name -> Quantity[value, QuantityUnit[qty]]
    ];
    quantifiedRow = quantify @@ # & /@ paramValues // Association;

    (* An empty container that we'll add to *)
    result = Association[];
    result[R] = quantifiedRow[R];
    result[L] = quantifiedRow[L];

    (* Figure out the gearing *)
    If[OptionValue[WithGearbox],
      result[N] = quantifiedRow[N];
      result[η] = quantifiedRow[η];
    ,
      result[N] = 1;
      result[η] = 1;
    ];
    (* WithGearbox being false means what it says: the load is directly on the armature. *)
    (* We still have to reflect the experimental parameters, either way. *)
    gearbox = {N -> quantifiedRow[N], η -> quantifiedRow[η]};

    (* In the result, we need to reflect measured values back before the gearbox *)
    result[Ke] = reflectAngleCoefficient[quantifiedRow[Ke]] /. gearbox;
    result[Kt] = reflectAngleCoefficient[quantifiedRow[Kt]] /. gearbox;
    result[B] = reflectDrag[quantifiedRow[B]] /. gearbox;
    result[J] = reflectInertia[quantifiedRow[J]] /. gearbox;

    (* add in defaults that we can later add thereto *)
    result[Jafter] = Quantity[0, siAngularInertialUnits];
    result[Bafter] = Quantity[0, siAngularViscousDragUnits];
    result[ΔτappConst] = Quantity[0, siTorqueUnits];

    result // siUnits // simplifyUnits
  ];

motorParameters["AM 60 A", WithGearbox -> False]
motorParameters["AM 60 A", WithGearbox -> True, Efficiency -> "Reverse"]

```

$$\left\langle \begin{array}{l} R \rightarrow \frac{33}{10} \text{ W/A}^2, L \rightarrow \frac{347}{500\,000} \text{ H}, N \rightarrow 1, \eta \rightarrow 1, Ke \rightarrow \frac{533}{30\,000} \text{ kg m}^2 / (\text{s}^2 \text{A rad}), Kt \rightarrow \frac{533}{30\,000} \text{ kg m}^2 / (\text{s}^2 \text{A rad}), B \rightarrow \frac{11}{1080\,000} \text{ kg m}^2 / (\text{s rad}^2), \\ J \rightarrow \frac{347}{108\,000\,000\,000} \text{ kg m}^2 / \text{rad}^2, J_{\text{after}} \rightarrow 0 \text{ kg m}^2 / \text{rad}^2, B_{\text{after}} \rightarrow 0 \text{ kg m}^2 / (\text{s rad}^2), \Delta\tau_{\text{appConst}} \rightarrow 0 \text{ kg m}^2 / (\text{s}^2 \text{rad}) \end{array} \right\rangle$$

$$\left\langle \begin{array}{l} R \rightarrow \frac{33}{10} \text{ W/A}^2, L \rightarrow \frac{347}{500\,000} \text{ H}, N \rightarrow 60, \eta \rightarrow \frac{4}{5}, K_e \rightarrow \frac{533}{30\,000} \text{ kg m}^2 / (\text{s}^2 \text{A rad}), K_t \rightarrow \frac{533}{30\,000} \text{ kg m}^2 / (\text{s}^2 \text{A rad}), B \rightarrow \frac{11}{960\,000} \text{ kg m}^2 / (\text{s rad}^2), \\ J \rightarrow \frac{347}{96\,000\,000\,000} \text{ kg m}^2 / \text{rad}^2, J_{\text{after}} \rightarrow 0 \text{ kg m}^2 / \text{rad}^2, B_{\text{after}} \rightarrow 0 \text{ kg m}^2 / (\text{s rad}^2), \Delta\tau_{\text{appConst}} \rightarrow 0 \text{ kg m}^2 / (\text{s}^2 \text{rad}) \end{array} \right\rangle$$

4.4. Loads on the Motor

motorLoad[] creates a motor load given explicit internal and drag parameters. We allow the 'after' suffix but do not require its use.

```
Clear[motorLoad]
Options[motorLoad] = {
  J → Quantity[0, siAngularInertialUnits], Jafter → Quantity[0, siAngularInertialUnits],
  B → Quantity[0, siAngularViscousDragUnits], Bafter → Quantity[0, siAngularViscousDragUnits],
  ΔτappConst → Quantity[0, siTorqueUnits]
};
motorLoad[opts : OptionsPattern[]] := Module[{assoc = Association[]},
  assoc[Jafter] = OptionValue[J] + OptionValue[Jafter];
  assoc[Bafter] = OptionValue[B] + OptionValue[Bafter];
  assoc[ΔτappConst] = OptionValue[ΔτappConst];
  assoc
]
```

flywheel[] creates a motor load given mass and radius parameters

```
Clear[flywheel]
flywheel[mass_, radius_] := motorLoad[J → mass * radius ^ 2 / 2 / Quantity[1, "Radians^2"]
flywheel[Quantity[2, "kg"], Quantity[5, "cm"]]
```

$$\left\langle J_{\text{after}} \rightarrow 25 \text{ kg cm}^2 / \text{rad}^2, B_{\text{after}} \rightarrow 0 \text{ ms N} / \text{rad}^2, \Delta\tau_{\text{appConst}} \rightarrow 0 \text{ mN} / \text{rad} \right\rangle$$

addMotorLoad[] takes motor parameters and a load and returns new parameters that include the load.

```
addMotorLoad[motor_, load_] := Module[{assoc},
  assoc = Association[motor]; (*explicitly create copy*)
  assoc[Jafter] = (Jafter /. motor) + (Jafter /. load);
  assoc[Bafter] = (Bafter /. motor) + (Bafter /. load);
  assoc[ΔτappConst] = (ΔτappConst /. motor) + (ΔτappConst /. load);
  assoc
]
addMotorLoad[motorParameters["AM 60 A"], flywheel[Quantity[2, "kg"], Quantity[5, "cm"]]]
```

$$\left\langle \begin{array}{l} R \rightarrow \frac{33}{10} \text{ W/A}^2, L \rightarrow \frac{347}{500\,000} \text{ H}, N \rightarrow 60, \eta \rightarrow \frac{9}{10}, K_e \rightarrow \frac{533}{30\,000} \text{ kg m}^2 / (\text{s}^2 \text{A rad}), K_t \rightarrow \frac{533}{30\,000} \text{ kg m}^2 / (\text{s}^2 \text{A rad}), B \rightarrow \frac{11}{1080\,000} \text{ kg m}^2 / (\text{s rad}^2), \\ J \rightarrow \frac{347}{108\,000\,000\,000} \text{ kg m}^2 / \text{rad}^2, J_{\text{after}} \rightarrow 25 \text{ kg cm}^2 / \text{rad}^2, B_{\text{after}} \rightarrow 0 \text{ kg m}^2 / (\text{s rad}^2), \Delta\tau_{\text{appConst}} \rightarrow 0 \text{ mN} / \text{rad} \end{array} \right\rangle$$

gearboxlessEquivalent[] takes a motor with load and gearbox and produces the equivalent motor and load in which there is no gearbox.

```

gearboxlessEquivalent[motor_] := Module[{rules, assoc},
  rules = {N → (N /. motor), η → (η /. motor)};
  assoc = Association[motor]; (*explicitly create copy*)
  assoc[J] = assoc[J] + (reflectInertia[assoc[Jafter]] /. rules);
  assoc[B] = assoc[B] + (reflectDrag[assoc[Bafter]] /. rules);
  assoc[ΔτappConst] = (reflectTorque[assoc[ΔτappConst]] /. rules);
  assoc[Jafter] = 0;
  assoc[Bafter] = 0;
  assoc[N] = 1;
  assoc[η] = 1;
  assoc
]
addMotorLoad[motorParameters["AM 60 A"], flywheel[Quantity[2, "kg"], Quantity[5, "cm"]]] // gearboxlessEquivalent

```

$$\left\langle \begin{array}{l} R \rightarrow \frac{33}{10} \text{ W/A}^2, L \rightarrow \frac{347}{500\,000} \text{ H}, N \rightarrow 1, \eta \rightarrow 1, K_e \rightarrow \frac{533}{30\,000} \text{ kg m}^2/(\text{s}^2 \text{ A rad}), K_t \rightarrow \frac{533}{30\,000} \text{ kg m}^2/(\text{s}^2 \text{ A rad}), \\ B \rightarrow \frac{11}{1\,080\,000} \text{ kg m}^2/(\text{s rad}^2), J \rightarrow \frac{251\,041}{32\,400\,000} \text{ kg cm}^2/\text{rad}^2, J_{\text{after}} \rightarrow 0, B_{\text{after}} \rightarrow 0, \Delta\tau_{\text{appConst}} \rightarrow 0 \text{ mN/rad} \end{array} \right\rangle$$

4.5. Mass on pulley

Here, we model a mass suspended on a massless “rigid string” from a massless pulley attached to the motor shaft. That string acts at a wrench with a length of the radius of the pulley to deliver a torque to the shaft. `τMassOnPulley[]` computes that torque.

There is a force pulling down on the mass due to gravity and up due to acceleration in the string that results from the driven angular acceleration of the motor shaft. The sum of these forces is the tension in the string. If the net acceleration is negative (ie: downward), a real (flexible) string would clamp at zero and have the mass in free-fall; we choose not to model that for now. Instead, here, our hypothetical rigid string continues to impart acceleration from the pulley even for negative accelerations. This is a bit odd for a string, but more reasonable if the string were in fact a rigid arm being lifted, or some such.

```

Clear[massOnPulley]
massOnPulley[mass_, radius_] := Module[
  {gravityAcceleration, gravityForce, angularAcceleration,
   tangentialAcceleration, tangentialForce, tension, torque, unit, parts, radians = Quantity["Radians"]},

  (* gravity exerts a constant downward force by acting on the mass *)
  gravityAcceleration = Quantity["StandardAccelerationOfGravity"];
  gravityForce = gravityAcceleration * mass;

  (* angular acceleration, converted to tangential acceleration also acts on the mass to create an upward force *)
  angularAcceleration = Quantity["0.1", "AngularAccelerationUnits"];
  tangentialAcceleration = angularAcceleration * radius / radians;
  tangentialForce = tangentialAcceleration * mass;

  (* the net force (the tension in the string) is the sum of the two forces *)
  tension = gravityForce + tangentialForce;

  (* the tension acts like a wrench with lever arm 'radius' *)
  torque = tension * radius / radians;

  (* we decompose the torque into constant and acceleration-dependent parts for output *)
  unit = QuantityUnit[torque];
  parts = List @@ (QuantityMagnitude[torque] // Apart);
  motorLoad[DeltaAppConst -> Quantity[parts[[1]], unit], Jafter -> Quantity[parts[[2]], unit] / angularAcceleration]
]
massOnPulley[Quantity[weightLbs, "Pounds"], Quantity[rInches, "Inches"]]
% // siUnits // clearUnits // N

<| Jafter -> rInches^2 weightLbs lb in^2 / rad^2, Bafter -> 0 ms N / rad^2, DeltaAppConst ->  $\frac{196.133 \text{ rInches weightLbs}}{508} \text{ lb in}^2 / (\text{s}^2 \text{ rad})$  |>

<| Jafter -> 0.00029264 rInches^2 weightLbs, Bafter -> 0., DeltaAppConst -> 0.112985 rInches weightLbs |>

```

4.6. Example Motors

We define motors and loads that we'll use to illustrate examples.

```

aFlywheel = flywheel[Quantity[10, "kg"], Quantity[10, "cm"]] // siUnits
aPulley = massOnPulley[Quantity[3, "Pounds"], Quantity[2, "Inches"]] // siUnits

```

```

<| Jafter ->  $\frac{1}{20} \text{ kg m}^2 / \text{rad}^2$ , Bafter -> 0 kg m^2 / (s rad^2), DeltaAppConst -> 0 kg m^2 / (s^2 rad) |>

```

```

<| Jafter ->  $\frac{2.194797400719}{6250000000000} \text{ kg m}^2 / \text{rad}^2$ , Bafter -> 0 kg m^2 / (s rad^2), DeltaAppConst ->  $\frac{3389544870828501}{500000000000000} \text{ kg m}^2 / (\text{s}^2 \text{ rad})$  |>

```

```

aMotor = motorParameters["AM 60 A"];
aMotorWithFlywheel = addMotorLoad[aMotor, aFlywheel] // siUnits;
aUnitlessMotorWithFlywheel = aMotorWithFlywheel // clearUnits // N
aMotorWithFlywheelNoGear = addMotorLoad[motorParameters["AM 60 A", WithGearbox -> False], aFlywheel];
aUnitlessMotorWithFlywheelNoGear = aMotorWithFlywheelNoGear // siUnits // clearUnits // N

```

```

<| R -> 3.3, L -> 0.000694, N -> 60., eta -> 0.9, Ke -> 0.0177667, Kt -> 0.0177667,
B -> 0.000101852, J -> 3.21296 x 10^-9, Jafter -> 0.05, Bafter -> 0., DeltaAppConst -> 0. |>

```

```

<| R -> 3.3, L -> 0.000694, N -> 1., eta -> 1., Ke -> 0.0177667, Kt -> 0.0177667,
B -> 0.000101852, J -> 3.21296 x 10^-9, Jafter -> 0.05, Bafter -> 0., DeltaAppConst -> 0. |>

```

5. Laplace Transforms and Transfer Function Models

5.1. Laplace Transforms

Returning to the differential equations, we form the Laplace Transform of each, then solve the for the various transforms. First, we apply the LaplaceTransform[] function to each equation. It automatically makes use of the linearity and differential properties of the transform, and insinuates itself just around the time dependent parts (ie: the parts dependent on the time variable we told it about, namely t).

```
leqns = LaplaceTransform[#, t, s] &@ ΔdiffEqns;
leqns // prettyPrint
```

$$Kt(\mathcal{L}_t[i(t)](s)) = \mathcal{L}_t[\tau(t)](s)$$

$$\mathcal{L}_t[v_g(t)](s) = Ke(s(\mathcal{L}_t[\theta(t)](s)) - \theta(0))$$

$$R(\mathcal{L}_t[i(t)](s)) + L(s(\mathcal{L}_t[i(t)](s)) - i(0)) + \mathcal{L}_t[v_g(t)](s) = \frac{v_{app0}}{s} + \mathcal{L}_t[\Delta v_{app}(t)](s)$$

$$\frac{\eta(\mathcal{L}_t[\tau(t)](s))N^2 + \frac{r_{app0}N}{s} + (\mathcal{L}_t[\Delta \tau_{app}(t)](s))N - (B\eta N^2 + B_{after})(s(\mathcal{L}_t[\theta(t)](s)) - \theta(0)) - (J\eta N^2 + J_{after})(\mathcal{L}_t[\theta(t)](s))s^2 - \theta(0)s - \theta'(0))}{\eta N} = 0$$

$$\mathcal{L}_t[\tau(t)](s) = \frac{\mathcal{L}_t[\tau_{after}(t)](s)}{\eta N}$$

$$\mathcal{L}_t[\theta_{after}(t)](s) = \frac{\mathcal{L}_t[\theta(t)](s)}{N}$$

$$s(\mathcal{L}_t[\theta_{after}(t)](s)) - \theta_{after}(0) = \frac{s(\mathcal{L}_t[\theta(t)](s)) - \theta(0)}{N}$$

$$(\mathcal{L}_t[\theta_{after}(t)](s))s^2 - \theta_{after}(0)s - \theta_{after}'(0) = \frac{(\mathcal{L}_t[\theta(t)](s))s^2 - \theta(0)s - \theta'(0)}{N}$$

Next, we walk those equations, picking up those insinuations and sowing them to the wind, reaping them on the outside, and finally removing duplicates.

```
allXforms = Reap[Scan[If[MatchQ[#, _LaplaceTransform], Sow[#]] &, leqns, Infinity]][[2, 1]] // Union
```

$$\{\text{LaplaceTransform}[i[t], t, s], \text{LaplaceTransform}[v_g[t], t, s], \text{LaplaceTransform}[\Delta v_{app}[t], t, s], \text{LaplaceTransform}[\Delta \tau_{app}[t], t, s], \\ \text{LaplaceTransform}[\theta[t], t, s], \text{LaplaceTransform}[\theta_{after}[t], t, s], \text{LaplaceTransform}[\tau[t], t, s], \text{LaplaceTransform}[\tau_{after}[t], t, s]\}$$

The voltage transform is input, as is any autonomous externally applied torque; all the others are outputs. Solve for the outputs. Finally substitute what we know about initial conditions, and simplify as much as we can.

```
ΔvappXform = LaplaceTransform[Δvapp[t], t, s];
ΔτappXform = LaplaceTransform[Δτapp[t], t, s];
inputXforms = {ΔvappXform, ΔτappXform}
inputLabels = {"1", "vapp", "τapp"}
(outputXforms = Complement[allXforms, inputXforms]) // ColumnForm
```

$$\{\text{LaplaceTransform}[\Delta v_{app}[t], t, s], \text{LaplaceTransform}[\Delta \tau_{app}[t], t, s]\}$$

```
{1, vapp, τapp}
```

```
LaplaceTransform[i[t], t, s]
LaplaceTransform[v_g[t], t, s]
LaplaceTransform[θ[t], t, s]
LaplaceTransform[θ_{after}[t], t, s]
LaplaceTransform[τ[t], t, s]
LaplaceTransform[τ_{after}[t], t, s]
```



```
(leqnsToSolve = leqns /. (ΔinitialConditions /. Equal → Rule)) // prettyPrint
```

$$\begin{aligned}
 Kt[\mathcal{L}_i[i(t)](s)] &= \mathcal{L}_i[\tau(t)](s) \\
 \mathcal{L}_i[v_g(t)](s) &= Ke s (\mathcal{L}_i[\theta(t)](s)) \\
 R(\mathcal{L}_i[i(t)](s)) + L \left(\frac{-B vapp0 \eta N^2 + Ke rapp0 N - Bafter vapp0}{Ke Kt \eta N^2 + B R \eta N^2 + Bafter R} + s (\mathcal{L}_i[i(t)](s)) \right) + \mathcal{L}_i[v_g(t)](s) &= \frac{vapp0}{s} + \mathcal{L}_i[\Delta vapp(t)](s) \\
 \frac{\eta (\mathcal{L}_i[\tau(t)](s)) N^2 + \frac{rapp0 N}{s} + (\mathcal{L}_i[\Delta rapp(t)](s)) N - s (B \eta N^2 + Bafter R) (\mathcal{L}_i[\theta(t)](s)) - (J \eta N^2 + Jafter) (\mathcal{L}_i[\theta(t)](s)) s^2 + \frac{-Kt vapp0 \eta N^2 - R rapp0 N}{Ke Kt \eta N^2 + B R \eta N^2 + Bafter R}}{\eta N} &= 0 \\
 \mathcal{L}_i[\tau(t)](s) &= \frac{\mathcal{L}_i[rafter(t)](s)}{\eta N} \\
 \mathcal{L}_i[\theta after(t)](s) &= \frac{\mathcal{L}_i[\theta(t)](s)}{N} \\
 s (\mathcal{L}_i[\theta after(t)](s)) &= \frac{s (\mathcal{L}_i[\theta(t)](s))}{N} \\
 (\mathcal{L}_i[\theta after(t)](s)) s^2 + \frac{-Kt vapp0 \eta N^2 - R rapp0}{Ke Kt \eta N^2 + B R \eta N^2 + Bafter R} &= \frac{(\mathcal{L}_i[\theta(t)](s)) s^2 + \frac{-Kt vapp0 \eta N^2 - R rapp0 N}{Ke Kt \eta N^2 + B R \eta N^2 + Bafter R}}{N}
 \end{aligned}$$

```
(rawSolvedXforms = uniqueSolve[leqnsToSolve, outputXforms] // Simplify) // prettyPrint
```

$$\begin{aligned}
 \mathcal{L}_i[i(t)](s) &\rightarrow \frac{Bafter vapp0 + N (B vapp0 \eta N - Ke rapp0)}{s ((Ke Kt + B R) \eta N^2 + Bafter R)} + \frac{((B + J s) \eta N^2 + Bafter + Jafter s) (\mathcal{L}_i[\Delta vapp(t)](s))}{(Ke Kt + (B + J s) (R + L s)) \eta N^2 + Bafter (R + L s) + Jafter s (R + L s)} - \frac{Ke N (\mathcal{L}_i[\Delta rapp(t)](s))}{(Ke Kt + (B + J s) (R + L s)) \eta N^2 + Bafter (R + L s) + Jafter s (R + L s)} \\
 \mathcal{L}_i[v_g(t)](s) &\rightarrow Ke N \left(\frac{Kt vapp0 \eta N + R rapp0}{s ((Ke Kt + B R) \eta N^2 + Bafter R)} + \frac{Kt \eta N (\mathcal{L}_i[\Delta vapp(t)](s))}{(Ke Kt + (B + J s) (R + L s)) \eta N^2 + Bafter (R + L s) + Jafter s (R + L s)} + \frac{(R + L s) (\mathcal{L}_i[\Delta rapp(t)](s))}{(Ke Kt + (B + J s) (R + L s)) \eta N^2 + Bafter (R + L s) + Jafter s (R + L s)} \right) \\
 \mathcal{L}_i[\theta(t)](s) &\rightarrow \frac{N \left(\frac{Kt vapp0 \eta N + R rapp0}{(Ke Kt + B R) \eta N^2 + Bafter R} + \frac{Kt s \eta N (\mathcal{L}_i[\Delta vapp(t)](s))}{(Ke Kt + (B + J s) (R + L s)) \eta N^2 + Bafter (R + L s) + Jafter s (R + L s)} + \frac{s (R + L s) (\mathcal{L}_i[\Delta rapp(t)](s))}{(Ke Kt + (B + J s) (R + L s)) \eta N^2 + Bafter (R + L s) + Jafter s (R + L s)} \right)}{s^2} \\
 \mathcal{L}_i[\theta after(t)](s) &\rightarrow \frac{\frac{Kt vapp0 \eta N + R rapp0}{(Ke Kt + B R) \eta N^2 + Bafter R} + \frac{Kt s \eta N (\mathcal{L}_i[\Delta vapp(t)](s))}{(Ke Kt + (B + J s) (R + L s)) \eta N^2 + Bafter (R + L s) + Jafter s (R + L s)} + \frac{s (R + L s) (\mathcal{L}_i[\Delta rapp(t)](s))}{(Ke Kt + (B + J s) (R + L s)) \eta N^2 + Bafter (R + L s) + Jafter s (R + L s)}}{s^2} \\
 \mathcal{L}_i[\tau(t)](s) &\rightarrow Kt \left(\frac{Bafter vapp0 + N (B vapp0 \eta N - Ke rapp0)}{s ((Ke Kt + B R) \eta N^2 + Bafter R)} + \frac{((B + J s) \eta N^2 + Bafter + Jafter s) (\mathcal{L}_i[\Delta vapp(t)](s))}{(Ke Kt + (B + J s) (R + L s)) \eta N^2 + Bafter (R + L s) + Jafter s (R + L s)} - \frac{Ke N (\mathcal{L}_i[\Delta rapp(t)](s))}{(Ke Kt + (B + J s) (R + L s)) \eta N^2 + Bafter (R + L s) + Jafter s (R + L s)} \right) \\
 \mathcal{L}_i[rafter(t)](s) &\rightarrow -\frac{rapp0}{s} - \frac{(J \eta N^2 + Jafter) (Kt vapp0 \eta N + R rapp0)}{(Ke Kt + B R) \eta N^2 + Bafter R} - \mathcal{L}_i[\Delta rapp(t)](s) + \frac{((B + J s) \eta N^2 + Bafter + Jafter s) ((Ke Kt + (B + J s) (R + L s)) \eta N^2 + Bafter (R + L s) + Jafter s (R + L s)) (Kt v}{s ((Ke Kt + B R) \eta N^2 + Bafter R)}
 \end{aligned}$$

We separate the transforms into applied-voltage- and applied-torque-dependent parts.

```

Clear[separateXforms]
separateXforms[xform_] := Module[{xformVarRules, invertedRules, xformVars, vard, rawCollected, coeffs, const, terms},
  xformVarRules = # -> Unique["laplaceTransform"] & /@ inputXforms;
  invertedRules = Reverse /@ xformVarRules;
  xformVars = xformVarRules[All, 2];
  vard = xform /. xformVarRules;
  rawCollected = Collect[vard, xformVars] /. invertedRules;
  coeffs = Coefficient[rawCollected, #] & /@ inputXforms;
  const = Together[rawCollected /. {# -> 0 & /@ inputXforms}];
  (* we multiply const by s to make the units come out right (?) *)
  const = {const[[1]] * s // Simplify};
  terms = const ~Join~ (#[[1]] * #[[2]] & /@ Transpose[{coeffs, inputXforms}]);
  Plus @@ terms
]
solvedXforms = #[[1]] -> separateXforms[#[[2]]] & /@ rawSolvedXforms;
% // prettyPrint

```

$$\begin{aligned}
 \mathcal{L}_i[i(t)](s) &\rightarrow \frac{Bafter vapp0 + N (B vapp0 \eta N - Ke rapp0)}{(Ke Kt + B R) \eta N^2 + Bafter R} + \frac{((B + J s) \eta N^2 + Bafter + Jafter s) (\mathcal{L}_i[\Delta vapp(t)](s))}{(Ke Kt + (B + J s) (R + L s)) \eta N^2 + Bafter (R + L s) + Jafter s (R + L s)} - \frac{Ke N (\mathcal{L}_i[\Delta rapp(t)](s))}{(Ke Kt + (B + J s) (R + L s)) \eta N^2 + Bafter (R + L s) + Jafter s (R + L s)} \\
 \mathcal{L}_i[v_g(t)](s) &\rightarrow \frac{Ke Kt \eta (\mathcal{L}_i[\Delta vapp(t)](s)) N^2}{(Ke Kt + (B + J s) (R + L s)) \eta N^2 + Bafter (R + L s) + Jafter s (R + L s)} + \frac{Ke (Kt vapp0 \eta N + R rapp0) N}{(Ke Kt + B R) \eta N^2 + Bafter R} + \frac{Ke (R + L s) (\mathcal{L}_i[\Delta rapp(t)](s)) N}{(Ke Kt + (B + J s) (R + L s)) \eta N^2 + Bafter (R + L s) + Jafter s (R + L s)} \\
 \mathcal{L}_i[\theta(t)](s) &\rightarrow \frac{Kt \eta (\mathcal{L}_i[\Delta vapp(t)](s)) N^2}{s ((Ke Kt + B R) \eta N^2 + Bafter R)} + \frac{(Kt vapp0 \eta N + R rapp0) N}{s ((Ke Kt + B R) \eta N^2 + Bafter R)} + \frac{(R + L s) (\mathcal{L}_i[\Delta rapp(t)](s)) N}{s ((Ke Kt + (B + J s) (R + L s)) \eta N^2 + Bafter (R + L s) + Jafter s (R + L s))} \\
 \mathcal{L}_i[\theta after(t)](s) &\rightarrow \frac{Kt vapp0 \eta N + R rapp0}{s ((Ke Kt + B R) \eta N^2 + Bafter R)} + \frac{Kt \eta N (\mathcal{L}_i[\Delta vapp(t)](s))}{s ((Ke Kt + (B + J s) (R + L s)) \eta N^2 + Bafter (R + L s) + Jafter s (R + L s))} + \frac{(R + L s) (\mathcal{L}_i[\Delta rapp(t)](s))}{s ((Ke Kt + (B + J s) (R + L s)) \eta N^2 + Bafter (R + L s) + Jafter s (R + L s))} \\
 \mathcal{L}_i[\tau(t)](s) &\rightarrow \frac{Kt (Bafter vapp0 + N (B vapp0 \eta N - Ke rapp0))}{(Ke Kt + B R) \eta N^2 + Bafter R} + \frac{Kt ((B + J s) \eta N^2 + Bafter + Jafter s) (\mathcal{L}_i[\Delta vapp(t)](s))}{(Ke Kt + (B + J s) (R + L s)) \eta N^2 + Bafter (R + L s) + Jafter s (R + L s)} - \frac{Ke Kt N (\mathcal{L}_i[\Delta rapp(t)](s))}{(Ke Kt + (B + J s) (R + L s)) \eta N^2 + Bafter (R + L s) + Jafter s (R + L s)} \\
 \mathcal{L}_i[rafter(t)](s) &\rightarrow \frac{Kt \eta N (Bafter vapp0 + N (B vapp0 \eta N - Ke rapp0))}{(Ke Kt + B R) \eta N^2 + Bafter R} + \frac{Kt \eta N ((B + J s) \eta N^2 + Bafter + Jafter s) (\mathcal{L}_i[\Delta vapp(t)](s))}{(Ke Kt + (B + J s) (R + L s)) \eta N^2 + Bafter (R + L s) + Jafter s (R + L s)} + \frac{(R + L s) ((B + J s) \eta N^2 + Bafter + Jafter s)}{(Ke Kt + (B + J s) (R + L s)) \eta N^2 + Bafter (R + L s) + Jafter s (R + L s)}
 \end{aligned}$$

5.2. Transfer Function Models

We use the solved transforms to create TransferFunctionModel[]s for every variable of interest.

```

Clear[divideTerms]
divideTerms[sum_, termDivisors_] := Module[{list},
  list = List @@ sum;
  #[[1]] / #[[2]] & /@ Transpose[{list, termDivisors}]
divideTerms[a x + b y, {x, y}]

{a, b}

```

```

Clear[makeModel]
makeModel[var_] := makeModel[var, ToString[var], 1]
makeModel[var_, outputLabel_, factor_] := TransferFunctionModel[
  factor * Function[{sum}, divideTerms[sum, {1} ~Join~ inputXforms]]
  [LaplaceTransform[var[t], t, s] /. solvedXforms],
  s,
  SystemsModelLabels -> {inputLabels, outputLabel}] // FullSimplify // Simplify

```

```

motorPositionModel = makeModel[θ]
motorVelocityModel = makeModel[θ, "ω", s]
motorAccelerationModel = makeModel[θ, "α", s * s]
motorCurrentModel = makeModel[i]
motorEMFModel = makeModel[vg, "vg", 1]
motorTorqueModel = makeModel[τ]

```

$$\left(\begin{array}{c|c|c} 1 & v_{app} & \tau_{app} \\ \hline \theta & \frac{N (Kt v_{app} \eta N + R \tau_{app} \theta)}{(B_{after} R + (Ke Kt + B R) \eta N^2) s} & \frac{Kt \eta N^2}{s (Ke Kt \eta N^2 + (R + L s) (B_{after} + J_{after} s + \eta N^2 (B + J s)))} - \frac{\tau_{app}}{N (R + L s)} \end{array} \right)$$

$$\left(\begin{array}{c|c|c} 1 & v_{app} & \tau_{app} \\ \hline \omega & \frac{N (Kt v_{app} \eta N + R \tau_{app} \theta)}{B_{after} R + (Ke Kt + B R) \eta N^2} & \frac{Kt \eta N^2}{Ke Kt \eta N^2 + (R + L s) (B_{after} + J_{after} s + \eta N^2 (B + J s))} - \frac{\tau_{app}}{N (R + L s)} \end{array} \right)$$

$$\left(\begin{array}{c|c|c} 1 & v_{app} & \tau_{app} \\ \hline \alpha & \frac{N (Kt v_{app} \eta N + R \tau_{app} \theta) s}{B_{after} R + (Ke Kt + B R) \eta N^2} & \frac{Kt \eta N^2 s}{Ke Kt \eta N^2 + (R + L s) (B_{after} + J_{after} s + \eta N^2 (B + J s))} - \frac{N s \tau_{app}}{N s (R + L s)} \end{array} \right)$$

$$\left(\begin{array}{c|c|c} 1 & v_{app} & \tau_{app} \\ \hline i & \frac{B_{after} v_{app} \theta + N (B v_{app} \eta N - Ke \tau_{app} \theta)}{B_{after} R + (Ke Kt + B R) \eta N^2} & \frac{B_{after} + J_{after} s + \eta N^2 (B + J s)}{Ke Kt \eta N^2 + (R + L s) (B_{after} + J_{after} s + \eta N^2 (B + J s))} - \frac{\tau_{app}}{Ke N} \end{array} \right)$$

$$\left(\begin{array}{c|c|c} 1 & v_{app} & \tau_{app} \\ \hline v_g & \frac{Ke N (Kt v_{app} \eta N + R \tau_{app} \theta)}{B_{after} R + (Ke Kt + B R) \eta N^2} & \frac{Ke Kt \eta N^2}{Ke Kt \eta N^2 + (R + L s) (B_{after} + J_{after} s + \eta N^2 (B + J s))} - \frac{Ke N \tau_{app}}{Ke N (R + L s)} \end{array} \right)$$

$$\left(\begin{array}{c|c|c} 1 & v_{app} & \tau_{app} \\ \hline \tau & \frac{Kt (B_{after} v_{app} \theta + N (B v_{app} \eta N - Ke \tau_{app} \theta))}{B_{after} R + (Ke Kt + B R) \eta N^2} & \frac{Kt (B_{after} + J_{after} s + \eta N^2 (B + J s))}{Ke Kt \eta N^2 + (R + L s) (B_{after} + J_{after} s + \eta N^2 (B + J s))} - \frac{\tau_{app}}{Ke Kt N} \end{array} \right)$$

```

motorPositionAfterModel = makeModel[θafter]
motorVelocityAfterModel = makeModel[ωafter, "ωafter", s]
motorAccelerationAfterModel = makeModel[αafter, "αafter", s * s]
motorTorqueAfterModel = makeModel[τafter]

```

$$\left(\begin{array}{c} \theta_{\text{after}} \end{array} \right) \begin{array}{c} \frac{1}{Kt v_{\text{app}} \eta N + R \tau_{\text{app}} \theta} \quad \frac{v_{\text{app}}}{Kt \eta N} \quad \frac{\tau_{\text{app}}}{R + L s} \\ \frac{(B_{\text{after}} R + (Ke Kt + B R) \eta N^2) s}{s (Ke Kt \eta N^2 + (R + L s) (B_{\text{after}} + J_{\text{after}} s + \eta N^2 (B + J s)))} \quad \frac{s (Ke Kt \eta N^2 + (R + L s) (B_{\text{after}} + J_{\text{after}} s + \eta N^2 (B + J s)))}{s (Ke Kt \eta N^2 + (R + L s) (B_{\text{after}} + J_{\text{after}} s + \eta N^2 (B + J s)))} \end{array}$$

$$\left(\begin{array}{c} \omega_{\text{after}} \end{array} \right) \begin{array}{c} \frac{1}{Kt v_{\text{app}} \eta N + R \tau_{\text{app}} \theta} \quad \frac{v_{\text{app}}}{Kt \eta N} \quad \frac{\tau_{\text{app}}}{R + L s} \\ \frac{B_{\text{after}} R + (Ke Kt + B R) \eta N^2}{Ke Kt \eta N^2 + (R + L s) (B_{\text{after}} + J_{\text{after}} s + \eta N^2 (B + J s))} \quad \frac{Ke Kt \eta N^2 + (R + L s) (B_{\text{after}} + J_{\text{after}} s + \eta N^2 (B + J s))}{Ke Kt \eta N^2 + (R + L s) (B_{\text{after}} + J_{\text{after}} s + \eta N^2 (B + J s))} \end{array}$$

$$\left(\begin{array}{c} \alpha_{\text{after}} \end{array} \right) \begin{array}{c} \frac{1}{(Kt v_{\text{app}} \eta N + R \tau_{\text{app}} \theta) s} \quad \frac{v_{\text{app}}}{Kt \eta N s} \quad \frac{\tau_{\text{app}}}{s (R + L s)} \\ \frac{B_{\text{after}} R + (Ke Kt + B R) \eta N^2}{Ke Kt \eta N^2 + (R + L s) (B_{\text{after}} + J_{\text{after}} s + \eta N^2 (B + J s))} \quad \frac{Ke Kt \eta N^2 + (R + L s) (B_{\text{after}} + J_{\text{after}} s + \eta N^2 (B + J s))}{Ke Kt \eta N^2 + (R + L s) (B_{\text{after}} + J_{\text{after}} s + \eta N^2 (B + J s))} \end{array}$$

$$\left(\begin{array}{c} \tau_{\text{after}} \end{array} \right) \begin{array}{c} \frac{1}{Kt \eta N (B_{\text{after}} v_{\text{app}} \theta + N (B v_{\text{app}} \theta \eta N - Ke \tau_{\text{app}} \theta))} \quad \frac{v_{\text{app}}}{Kt \eta N (B_{\text{after}} + J_{\text{after}} s + \eta N^2 (B + J s))} \quad \frac{\tau_{\text{app}}}{Ke Kt \eta N} \\ \frac{B_{\text{after}} R + (Ke Kt + B R) \eta N^2}{Ke Kt \eta N^2 + (R + L s) (B_{\text{after}} + J_{\text{after}} s + \eta N^2 (B + J s))} \quad \frac{Ke Kt \eta N^2 + (R + L s) (B_{\text{after}} + J_{\text{after}} s + \eta N^2 (B + J s))}{Ke Kt \eta N^2 + (R + L s) (B_{\text{after}} + J_{\text{after}} s + \eta N^2 (B + J s))} \end{array}$$

Save the models to a file for later use.

```

saveDefinitions[outputDirectory <> "MotorModels.m",
{diffEqns, ΔdiffEqns, initialConditions, ΔinitialConditions, motorPositionModel,
motorVelocityModel, motorAccelerationModel, motorCurrentModel, motorEMFModel, motorTorqueModel,
motorPositionAfterModel, motorVelocityAfterModel, motorAccelerationAfterModel, motorTorqueAfterModel}]

```

5.3. Units Verification

We do a quick check on one of the transforms to see if units are internally consistent. Mathematica would complain if there were not (and has done so in the past, during development; indeed, that's why this check is here).

```

motorVelocityModel[s][[1]]
modelTest = % /. parameterQuantities /. s -> Quantity[s, "per second"];
inputs = {1, Quantity[vapp, "Volts"], Quantity[τapp, siTorqueUnits]}
modelTest.inputs

```

$$\left\{ \frac{N (Kt v_{\text{app}} \eta N + R \tau_{\text{app}} \theta)}{B_{\text{after}} R + (Ke Kt + B R) \eta N^2}, \frac{Kt \eta N^2}{Ke Kt \eta N^2 + (R + L s) (B_{\text{after}} + J_{\text{after}} s + \eta N^2 (B + J s))}, \frac{(R + L s) N}{Ke Kt \eta N^2 + (R + L s) (B_{\text{after}} + J_{\text{after}} s + \eta N^2 (B + J s))} \right\}$$

$$\{1, v_{\text{app}} V, \tau_{\text{app}} \text{mN/rad}\}$$

$$\left(\frac{Kt v_{\text{app}} \eta N^2}{Ke Kt \eta N^2 + (R + L s) (B_{\text{after}} + J_{\text{after}} s + \eta N^2 (B + J s))} + \frac{(R + L s) N \tau_{\text{app}}}{Ke Kt \eta N^2 + (R + L s) (B_{\text{after}} + J_{\text{after}} s + \eta N^2 (B + J s))} + \frac{N (Kt v_{\text{app}} \eta N + R \tau_{\text{app}} \theta)}{B_{\text{after}} R + (Ke Kt + B R) \eta N^2} \right) \text{rad/s}$$

6. Time Domain Functions

The time domain functions we seek model various states of the motor as a direct function of time. Generally speaking, we may obtain them from the frequency domain by (matrix) multiplying the transfer function by the Laplace transform of the input signal, then taking the inverse Laplace transform of the result. However, if we're starting with the (frequency domain) transfer function and the (time domain) input signal in-hand, we can compute the time domain function more efficiently using the Laplace transform convolution theorem.

6.1. Time Domain Output Functions

6.1.1. makeTimeDomainFunctionConvolve[]

makeTimeDomainFunctionConvolve[] uses convolution and the theorem that the Laplace transform of a convolution is the product of the Laplace transform of the factors. Ref1: <http://mathfaculty.fullerton.edu/mathews/c2003/laplaceconvolutionmod.html>.

With the convolution approach, we can make explicit use of the fact that in the uses we have we know we want to take the inverse Laplace transform of a product, one of whose factors we already know the time domain inverse for, and thus we can avoid having LaplaceTransform[] do as much heavy lifting as we might ask it to do if we were to transform the input, multiply, then inverse-transform the whole result. This convolution approach also has been shown to be more efficient than the OutputResponse[] builtin function.

```
Clear[makeTimeDomainFunctionConvolve, convolve]
parameterAssumptions

convolve[fExpr_, gExpr_, exprVar_, t_] := Module[{τ(*=Unique["τ"]*)},
  Assuming[Union[{t ≥ 0}, parameterAssumptions], Integrate[(fExpr /. exprVar → τ) * (gExpr /. exprVar → t - τ), {τ, 0, t}]]
]

makeTimeDomainFunctionConvolve[model_, tInputFunctions_] := Module[
  {s(*=Unique["s"]*), τ(*=Unique["τ"]*), modelExpr},
  modelExpr = InverseLaplaceTransform[model[s], s, τ][[1]];
  Function[{t}, Module[{convolutions},
    convolutions = convolve[modelExpr[[#]], (tInputFunctions[[#]])[τ, t] &/@ Range[1, Length[tInputFunctions]]];
    {Total[convolutions]} (* put in list to mirror InverseLaplaceTransform[] output structure *)
  ]
]

{Bafter ∈ ℝ, Jafter ∈ ℝ, vapp0 ∈ ℝ, ΔvappConst ∈ ℝ, ΔτappConst ∈ ℝ, τapp0 ∈ ℝ, i[_] ∈ ℝ, vg[_] ∈ ℝ, α[_] ∈ ℝ, Δvapp[_] ∈ ℝ,
Δτapp[_] ∈ ℝ, θ[_] ∈ ℝ, τ[_] ∈ ℝ, τafter[_] ∈ ℝ, ω[_] ∈ ℝ, Ke > 0, Kt > 0, L > 0, R > 0, η > 0, N > 0, B ≥ 0, J ≥ 0, t ≥ 0}
```

6.1.2. makeMotorTimeDomainFunction[]

makeMotorTimeDomainFunction[] takes a transfer function model and returns a function that, when provided with ea and τa time-domain input functions, returns a function of time.

```
Clear[makeMotorTimeDomainFunction]
makeMotorTimeDomainFunction[model_] := makeMotorTimeDomainFunction[model, makeTimeDomainFunctionConvolve]
makeMotorTimeDomainFunction[model_, builder_] := makeMotorTimeDomainFunction[model, builder] = (*cache the result*)
Module[{t(*=Unique["t"]*)},
  Function[{vappfn, τappfn}, Module[{fn},
    fn = Function[{aTime},
      builder[model, {1 &, vappfn[#] &, τappfn[#] &}] [aTime] [[1]] (*unwrap output structure mentioned above*);
    ]
  ]
]
```

6.1.3. Time Domain Output Functions for Motor State

We define time domain functions for all the motor state of interest.

```
Clear[motorPosition, motorVelocity, motorAcceleration, motorCurrent, motorEMF, motorTorque]
```

```
motorPosition = makeMotorTimeDomainFunction[motorPositionModel];
```

```
motorVelocity = makeMotorTimeDomainFunction[motorVelocityModel];
```

```
motorAcceleration = makeMotorTimeDomainFunction[motorAccelerationModel];
```

```
motorCurrent = makeMotorTimeDomainFunction[motorCurrentModel];
```

```
motorEMF = makeMotorTimeDomainFunction[motorEMFModel];
```

```
motorTorque = makeMotorTimeDomainFunction[motorTorqueModel];
```

The *output* too.

```
Clear[motorPositionAfter, motorVelocityAfter, motorAccelerationAfter, motorTorqueAfter]
```

```
motorPositionAfter = makeMotorTimeDomainFunction[motorPositionAfterModel];
```

```
motorVelocityAfter = makeMotorTimeDomainFunction[motorVelocityAfterModel];
```

```
motorAccelerationAfter = makeMotorTimeDomainFunction[motorAccelerationAfterModel];
```

```
motorTorqueAfter = makeMotorTimeDomainFunction[motorTorqueAfterModel];
```

Save these time domain functions so we can load them in later without having to develop them from scratch.

```
saveDefinitions[outputDirectory <> "MotorTimeDomainFunctions.m",
{
  motorPosition, motorVelocity, motorAcceleration, motorCurrent, motorEMF, motorTorque,
  motorPositionAfter, motorVelocityAfter, motorAccelerationAfter,
  makeMotorTimeDomainFunction, makeTimeDomainFunctionConvolve
}]
```

6.2. Time Domain Input Functions

6.2.1. Typical Input Examples

In analogy to aMotor above, we now define some typical inputs. The first (and only, for now) is a constant 12V input with no externally applied torque.

```
applyTwelveVolts = Association[t → Quantity[t, "Seconds"],
  vapp0 → Quantity[0, "Volts"], τapp0 → Quantity[0, siTorqueUnits], ΔvappConst → Quantity[12, "Volts"]]
```

```
<| t → t s, vapp0 → 0 V, τapp0 → 0 mN/rad, ΔvappConst → 12 V |>
```

```
steadyTwelveVolts = Association[t → Quantity[t, "Seconds"],
  vapp0 → Quantity[12, "Volts"], τapp0 → Quantity[0, siTorqueUnits], ΔvappConst → Quantity[0, "Volts"]]
```

```
<| t → t s, vapp0 → 12 V, τapp0 → 0 mN/rad, ΔvappConst → 0 V |>
```

```
brakeFromTwelveVolts = Association[t → Quantity[t, "Seconds"],
  vapp0 → Quantity[12, "Volts"], τapp0 → Quantity[0, siTorqueUnits], ΔvappConst → Quantity[-12, "Volts"]]
```

```
<| t → t s, vapp0 → 12 V, τapp0 → 0 mN/rad, ΔvappConst → -12 V |>
```

6.3. Examples

We try out our time-domain functions on our example data. Note that the time axes in the plots here vary in scale.

```

Clear[stepMotorTimeFunction, stepMotorTimeFunctionGeneric]
stepMotorTimeFunctionGeneric[motorTimeFunction_, t_] := stepMotorTimeFunctionGeneric[motorTimeFunction, t] = Module[{},
  motorTimeFunction[ΔvappConst &, ΔτappConst &][t]]

stepMotorTimeFunction[motorTimeFunction_, motorWithLoad_, input_, t_] :=
  stepMotorTimeFunction[motorTimeFunction, motorWithLoad, input, t] = Module[{generic, withUnits, unitless},
    generic = stepMotorTimeFunctionGeneric[motorTimeFunction, t];
    withUnits = generic /. (input ~Join~ motorWithLoad) // siUnits // N // FullSimplify;
    unitless = withUnits // clearUnits // FullSimplify;
    {generic, withUnits, unitless}]

```


6.3.1. Velocity, after gearbox

```

{velAfterStepGeneric, velAfterUnits, velAfter} =
  stepMotorTimeFunction[motorVelocityAfter, aMotorWithFlywheel, applyTwelveVolts, t];
velAfterUnits // N
Plot[velAfter, {t, 0, 1.0}, AxesLabel → {"Seconds", HoldForm[Radians / Second]}, PlotRange → All, AxesOrigin → {0, 0}]

```

$$2.71828^{-6.86584 t} \left(-10.1885 \text{ rad/s} \right) + 2.71828^{-4748.84 t} \left(0.0147304 \text{ rad/s} \right) + 10.1737 \text{ rad/s}$$




```

{steadyVelAfterStepGeneric, steadyVelAfterUnits, steadyVelAfter} =
  stepMotorTimeFunction[motorVelocityAfter, aMotorWithFlywheel, steadyTwelveVolts, t];
steadyVelAfterUnits // N
Plot[steadyVelAfter, {t, 0, 1.0}, AxesLabel → {"Seconds", HoldForm[Radians / Second]}, PlotRange → All, AxesOrigin → {0, 0}]


```

$$\left(2.71828^{-4748.84 t} + 2.71828^{-6.86584 t} \right) \left(0. \text{ rad/s} \right) + 10.1737 \text{ rad/s}$$



```
{brakeVelAfterStepGeneric, brakeVelAfterUnits, brakeVelAfter} =
  stepMotorTimeFunction[motorVelocityAfter, aMotorWithFlywheel, brakeFromTwelveVolts, t];
brakeVelAfterUnits // N
Plot[brakeVelAfter, {t, 0, 1}, AxesLabel → {"Seconds", HoldForm[Radians / Second]}, PlotRange → All, AxesOrigin → {0, 0}]
```


$$2.71828^{-4748.84 t} \left(-0.0147304 \text{ rad/s} \right) + 7.85934 \times 10^{-14} \text{ rad/s} + 2.71828^{-6.86584 t} \left(10.1885 \text{ rad/s} \right)$$



6.3.2. Velocity, before gearbox

```
{velStepGeneric, velUnits, vel} = stepMotorTimeFunction[motorVelocity, aMotorWithFlywheel, applyTwelveVolts, t];
velUnits // N
Plot[vel, {t, 0, 1.0}, AxesLabel → {"Seconds", HoldForm[Radians / Second]}, PlotRange → All, AxesOrigin → {0, 0}]
```

$$2.71828^{-6.86584 t} \left(-611.308 \text{ rad/s} \right) + 2.71828^{-4748.84 t} \left(0.883825 \text{ rad/s} \right) + 610.424 \text{ rad/s}$$





6.3.3. Torque, before gearbox

Note that the torque first rises, then descends.

```
{torStepGeneric, torUnits, tor} = stepMotorTimeFunction[motorTorque, aMotorWithFlywheel, applyTwelveVolts, t];
torUnits // N
Plot[tor, {t, 0, 1}, AxesLabel → {"Seconds", HoldForm[Netwon Meters / Radian]}, PlotRange → All, AxesOrigin → {0, 0}]
Plot[tor, {t, 0, .01}, AxesLabel → {"Seconds", HoldForm[Netwon Meters / Radian]}, PlotRange → All, AxesOrigin → {0, 0}]
```

$$2.71828^{-4748.84 t} \left(-0.0647752 \text{ kg m}^2 / (\text{s}^2 \text{ rad}) \right) + 0.00621728 \text{ kg m}^2 / (\text{s}^2 \text{ rad}) + 2.71828^{-6.86584 t} \left(0.0585579 \text{ kg m}^2 / (\text{s}^2 \text{ rad}) \right)$$





6.3.4. Torque, after gearbox

Note that the torque first rises, then descends.


```
{torAfterStepGeneric, torAfterUnits, torAfter} =
  stepMotorTimeFunction[motorTorqueAfter, aMotorWithFlywheel, applyTwelveVolts, t];
torAfterUnits // N
Plot[torAfter, {t, 0, 1}, AxesLabel → {"Seconds", HoldForm[Netwon Meters / Radian]}, PlotRange → All, AxesOrigin → {0, 0}]
Plot[torAfter, {t, 0, .01}, AxesLabel → {"Seconds", HoldForm[Netwon Meters / Radian]}, PlotRange → All, AxesOrigin → {0, 0}]
```

$$2.71828^{-4748.84 t} \left(-3.49786 \text{ kg m}^2 / (\text{s}^2 \text{ rad}) \right) + 0.335733 \text{ kg m}^2 / (\text{s}^2 \text{ rad}) + 2.71828^{-6.86584 t} \left(3.16213 \text{ kg m}^2 / (\text{s}^2 \text{ rad}) \right)$$





6.3.5. Current

Note that the current first rises, then descends.

```
{curStepGeneric, curUnits, cur} = stepMotorTimeFunction[motorCurrent, aMotorWithFlywheel, applyTwelveVolts, t];
curUnits // N
Plot[cur, {t, 0, 1}, AxesLabel → {"Seconds", HoldForm[Amps]}, PlotRange → All, AxesOrigin → {0, 0}]
Plot[cur, {t, 0, 0.01}, AxesLabel → {"Seconds", HoldForm[Amps]}, PlotRange → All, AxesOrigin → {0, 0}]
```


$$2.71828^{-4748.84 t} \left(-3.64588 \text{ A} \right) + 0.349941 \text{ A} + 2.71828^{-6.86584 t} \left(3.29594 \text{ A} \right)$$



6.3.6. Back EMF

```
{emfStepGeneric, emfUnits, emf} = stepMotorTimeFunction[motorEMF, aMotorWithFlywheel, applyTwelveVolts, t];
emfUnits // N // simplifyUnits
Plot[{emf, ΔvappConst /. applyTwelveVolts}, {t, 0, 1},
  AxesLabel → {"Seconds", HoldForm[Volts]}, PlotRange → All, AxesOrigin → {0, 0}]
```


$$2.71828^{-6.86584 t} \left(-10.8609 \text{ V} \right) + 2.71828^{-4748.84 t} \left(0.0157026 \text{ V} \right) + 10.8452 \text{ V}$$



6.3.7. Position, before gearbox

```
{posStepGeneric, posUnits, pos} = stepMotorTimeFunction[motorPosition, aMotorWithFlywheel, applyTwelveVolts, t];
posUnits // N
Plot[pos, {t, 0, 1}, AxesLabel → {"Seconds", HoldForm[Radians]}, PlotRange → All, AxesOrigin → {0, 0}]
```


$$2.71828^{-4748.84 t} \left(-0.000186114 \text{ rad} \right) + 2.71828^{-6.86584 t} \left(89.0361 \text{ rad} \right) + (-89.0359 + 610.424 t) \text{ rad}$$



6.3.8. Position, after gearbox

```
{posAfterStepGeneric, posAfterUnits, posAfter} =
stepMotorTimeFunction[motorPositionAfter, aMotorWithFlywheel, applyTwelveVolts, t];
posAfterUnits // N
Plot[posAfter, {t, 0, 1}, AxesLabel → {"Seconds", HoldForm[Radians]}, PlotRange → All, AxesOrigin → {0, 0}]
```



$$2.71828^{-4748.84 t} \left(-3.1019 \times 10^{-6} \text{ rad} \right) + 2.71828^{-6.86584 t} \left(1.48394 \text{ rad} \right) + (-1.48393 + 10.1737 t) \text{ rad}$$



6.3.9. Acceleration, before gearbox



```
{accStepGeneric, accUnits, acc} = stepMotorTimeFunction[motorAcceleration, aMotorWithFlywheel, applyTwelveVolts, t];
accUnits // N
Plot[acc, {t, 0, 1}, AxesLabel → {"Seconds", HoldForm[Radians / Seconds^2]}, PlotRange → All, AxesOrigin → {0, 0}]
Plot[acc, {t, 0, .01}, AxesLabel → {"Seconds", HoldForm[Radians / Seconds^2]}, PlotRange → All, AxesOrigin → {0, 0}]
```

$$2.71828^{-4748.84 t} \left(-4197.14 \text{ rad/s}^2 \right) + 0. \text{ rad/s}^2 + \text{DiracDelta} \left[t \text{ s} \right] \left(0. \text{ rad/s} \right) + 2.71828^{-6.86584 t} \left(4197.14 \text{ rad/s}^2 \right)$$



6.3.10. Acceleration, after gearbox

```
{accAfterStepGeneric, accAfterUnits, accAfter} =
  stepMotorTimeFunction[motorAccelerationAfter, aMotorWithFlywheel, applyTwelveVolts, t];
accAfterUnits // N
Plot[accAfter, {t, 0, 1}, AxesLabel → {"Seconds", HoldForm[Radians / Seconds^2]}, PlotRange → All, AxesOrigin → {0, 0}]
Plot[accAfter, {t, 0, .01}, AxesLabel → {"Seconds", HoldForm[Radians / Seconds^2]}, PlotRange → All, AxesOrigin → {0, 0}]
```

$$2.71828^{-4748.84 t} \left(-69.9523 \text{ rad/s}^2 \right) + 0. \text{ rad/s}^2 + \text{DiracDelta} \left[t \text{ s} \right] \left(0. \text{ rad/s} \right) + 2.71828^{-6.86584 t} \left(69.9523 \text{ rad/s}^2 \right)$$



6.4. Examples with unitless loads

The unitless loads compute faster, but provide less insight.

6.4.1. With gear, using flywheel load


```
applyTwelveVolts // siUnits // clearUnits // N
{velUnits2StepGeneric, velUnits2, vel2} =
  stepMotorTimeFunction[motorVelocity, aUnitlessMotorWithFlywheel, applyTwelveVolts // siUnits // clearUnits // N, t];
Plot[vel2, {t, 0, 1}, AxesLabel → {"Seconds", HoldForm[Radians / Second]}, PlotRange → All, AxesOrigin → {0, 0}]
```

<| t → t, vapp0 → 0., tapp0 → 0., ΔvappConst → 12. |>



6.4.2. Without gear, using flywheel load

Note the time axis.



7. Steady State

7.1. Calculating Steady State Values : The Final Value Theorem

7.1.1. The general case

We explore the steady state behavior, the limit of the time domain functions as time goes to infinity.

One way to calculate said limit is to brute-force calculate the limit in question. While that works, and we can and have previously calculated same (Ref: <https://github.com/rgatkinson/RobotPhysics/blob/master/MotorPhysics.pdf>), it is very slow to do so.

A more efficient way to calculate the steady state behavior is to avail ourselves of the use of the Final Value Theorem (Ref: https://en.wikipedia.org/wiki/Final_value_theorem). The final value theorem tells us that if the time-domain limit exists, it is equal to the limit of the Laplace transform of the time domain function as $s \rightarrow 0$.

```
Clear[ssFinalValueTheorem]
ssFinalValueTheorem[model_, inputs_] := Module[{s(= Unique[ ]*), t(= Unique[ ]*), xformedInputs, expr},
  xformedInputs = LaplaceTransform[#, {t, s}] &@ inputs;
  expr = (model[s].xformedInputs)[[1]];
  (*Limit is really slow if we include assumptions (!?), so don't*)
  Block[{$Assumptions = {}}, Limit[s expr, s -> 0] // FullSimplify]
]
```

```
(ss = {
  ssPos → ssFinalValueTheorem[motorPositionModel, {1 &, ΔvappConst &, ΔτappConst &}] // Factor,
  ssVel → ssFinalValueTheorem[motorVelocityModel, {1 &, ΔvappConst &, ΔτappConst &}] // Factor,
  ssAcc → ssFinalValueTheorem[motorAccelerationModel, {1 &, ΔvappConst &, ΔτappConst &}] // Factor,
  ssEmf → ssFinalValueTheorem[motorEMFModel, {1 &, ΔvappConst &, ΔτappConst &}] // Factor,
  ssCur → ssFinalValueTheorem[motorCurrentModel, {1 &, ΔvappConst &, ΔτappConst &}] // Factor,
  ssTor → ssFinalValueTheorem[motorTorqueModel, {1 &, ΔvappConst &, ΔτappConst &}] // Factor,

  ssPosAfter → ssFinalValueTheorem[motorPositionAfterModel, {1 &, ΔvappConst &, ΔτappConst &}] // Factor,
  ssVelAfter → ssFinalValueTheorem[motorVelocityAfterModel, {1 &, ΔvappConst &, ΔτappConst &}] // Factor,
  ssAccAfter → ssFinalValueTheorem[motorAccelerationAfterModel, {1 &, ΔvappConst &, ΔτappConst &}] // Factor,
  ssTorAfter → ssFinalValueTheorem[motorTorqueAfterModel, {1 &, ΔvappConst &, ΔτappConst &}] // Factor
} // FullSimplify) // prettyPrint
(simpleSs = ss /. {N → 1, η → 1, Bafter → 0, ΔτappConst → 0, τapp0 → 0, vapp0 → 0} // FullSimplify) // prettyPrint
```

```
ssPos → Indeterminate
ssVel →  $\frac{N(Kt(vapp0 + \Delta vappConst) \eta N + R(\Delta \tauappConst + rapp0))}{(Ke Kt + B R) \eta N^2 + Bafter R}$ 
ssAcc → 0
ssEmf →  $\frac{Ke N(Kt(vapp0 + \Delta vappConst) \eta N + R(\Delta \tauappConst + rapp0))}{(Ke Kt + B R) \eta N^2 + Bafter R}$ 
ssCur →  $\frac{Bafter(vapp0 + \Delta vappConst) + N(B(vapp0 + \Delta vappConst) \eta N - Ke(\Delta \tauappConst + rapp0))}{(Ke Kt + B R) \eta N^2 + Bafter R}$ 
ssTor →  $\frac{Kt(Bafter(vapp0 + \Delta vappConst) + N(B(vapp0 + \Delta vappConst) \eta N - Ke(\Delta \tauappConst + rapp0)))}{(Ke Kt + B R) \eta N^2 + Bafter R}$ 
ssPosAfter → Indeterminate
ssVelAfter →  $\frac{Kt(vapp0 + \Delta vappConst) \eta N + R(\Delta \tauappConst + rapp0)}{(Ke Kt + B R) \eta N^2 + Bafter R}$ 
ssAccAfter → 0
ssTorAfter →  $\frac{Kt \eta N(Bafter(vapp0 + \Delta vappConst) + N(B(vapp0 + \Delta vappConst) \eta N - Ke(\Delta \tauappConst + rapp0)))}{(Ke Kt + B R) \eta N^2 + Bafter R}$ 
```

```
ssPos → Indeterminate
ssVel →  $\frac{Kt \Delta vappConst}{Ke Kt + B R}$ 
ssAcc → 0
ssEmf →  $\frac{Ke Kt \Delta vappConst}{Ke Kt + B R}$ 
ssCur →  $\frac{B \Delta vappConst}{Ke Kt + B R}$ 
ssTor →  $\frac{B Kt \Delta vappConst}{Ke Kt + B R}$ 
ssPosAfter → Indeterminate
ssVelAfter →  $\frac{Kt \Delta vappConst}{Ke Kt + B R}$ 
ssAccAfter → 0
ssTorAfter →  $\frac{B Kt \Delta vappConst}{Ke Kt + B R}$ 
```

The “Indeterminate” result regarding position is Mathematica informing us that it has proven to itself that the limit does not exist.

7.1.2. With specific examples

```
ss /. aUnitlessMotorWithFlywheel /. applyTwelveVolts // siUnits // clearUnits
```

```
{ssPos → Indeterminate, ssVel → 610.424, ssAcc → 0, ssEmf → 10.8452, ssCur → 0.349941,
 ssTor → 0.00621728, ssPosAfter → Indeterminate, ssVelAfter → 10.1737, ssAccAfter → 0, ssTorAfter → 0.335733}
```

```
ss /. aUnitlessMotorWithFlywheelNoGear /. applyTwelveVolts // siUnits // clearUnits
```

```
{ssPos → Indeterminate, ssVel → 610.424, ssAcc → 0, ssEmf → 10.8452, ssCur → 0.349941,
 ssTor → 0.00621728, ssPosAfter → Indeterminate, ssVelAfter → 610.424, ssAccAfter → 0, ssTorAfter → 0.00621728}
```

aMotorWithFlywheel

$$\left(\begin{aligned} R &\rightarrow \frac{33}{10} \text{ kg m}^2 / (\text{s}^3 \text{A}^2) , L \rightarrow \frac{347}{500000} \text{ kg m}^2 / (\text{s}^2 \text{A}^2) , N \rightarrow 60 , \eta \rightarrow \frac{9}{10} , K_e \rightarrow \frac{533}{30000} \text{ kg m}^2 / (\text{s}^2 \text{A rad}) , \\ K_t &\rightarrow \frac{533}{30000} \text{ kg m}^2 / (\text{s}^2 \text{A rad}) , B \rightarrow \frac{11}{1080000} \text{ kg m}^2 / (\text{s rad}^2) , J \rightarrow \frac{347}{108000000000} \text{ kg m}^2 / \text{rad}^2 , \\ J_{\text{after}} &\rightarrow \frac{1}{20} \text{ kg m}^2 / \text{rad}^2 , B_{\text{after}} \rightarrow 0 \text{ kg m}^2 / (\text{s rad}^2) , \Delta \tau_{\text{appConst}} \rightarrow 0 \text{ kg m}^2 / (\text{s}^2 \text{rad}) \end{aligned} \right)$$

7.2. On the Applicability of the Final Value Theorem

Can we always use the Final Value Theorem?

No, the Final Value Theorem only applies under certain conditions. Specifically: that all non-zero poles of the transfer function must have negative real parts, and the transfer function must have at most one pole at the origin (Ref: <http://www-personal.umich.edu/~dsbaero/others/39-FVTrevisited.pdf>). The predicate exhibited in the ConditionalExpression of the brute-force result captures exactly these conditions. When using the Final Value Theorem on any concrete system, we will need to test the conditions in any particular application. Lets illustrate by looking at the velocity model.

motorVelocityModel

`(velPoles = TransferFunctionPoles[motorVelocityModel] // Flatten // Union)`

$$\left(\begin{aligned} &\frac{1}{N (K_t v_{\text{app}} \eta N + R \tau_{\text{app}} \theta)} \quad \frac{v_{\text{app}}}{K_t \eta N^2} \quad \frac{\tau_{\text{app}}}{N (R + L s)} \\ \omega &\frac{\text{Bafter } R + (K_e K_t + B R) \eta N^2}{K_e K_t \eta N^2 + (R + L s) (\text{Bafter} + J_{\text{after}} s + \eta N^2 (B + J s))} \quad \frac{K_e K_t \eta N^2 + (R + L s) (\text{Bafter} + J_{\text{after}} s + \eta N^2 (B + J s))}{K_e K_t \eta N^2 + (R + L s) (\text{Bafter} + J_{\text{after}} s + \eta N^2 (B + J s))} \end{aligned} \right)$$

$$\left\{ \begin{aligned} &\frac{(-\text{Bafter } L - J_{\text{after}} R - B L \eta N^2 - J R \eta N^2 - \sqrt{-4 (J_{\text{after}} L + J L \eta N^2) (\text{Bafter } R + K_e K_t \eta N^2 + B R \eta N^2) + (\text{Bafter } L + J_{\text{after}} R + B L \eta N^2 + J R \eta N^2)^2})}{2 (J_{\text{after}} L + J L \eta N^2)} , \\ &\frac{(-\text{Bafter } L - J_{\text{after}} R - B L \eta N^2 - J R \eta N^2 + \sqrt{-4 (J_{\text{after}} L + J L \eta N^2) (\text{Bafter } R + K_e K_t \eta N^2 + B R \eta N^2) + (\text{Bafter } L + J_{\text{after}} R + B L \eta N^2 + J R \eta N^2)^2})}{2 (J_{\text{after}} L + J L \eta N^2)} \end{aligned} \right\}$$

In order to get a feel for where these zeros are in practice, let's look at them for an actual example motor. We compute where the poles occur for one of our examples;

`velPoles /. aMotorWithFlywheel // siUnits // clearUnits // N`

`{-4748.84, -6.86584}`

This, for this example, the Final Value Theorem applies (at least for velocity). As another example, adding a ridiculous additional load moves the poles, but they remain in the LHS of the complex plane.

`velPoles /. addMotorLoad[aMotor, flywheel[Quantity[1000, "kg"], Quantity[1, "m"]]] // siUnits // clearUnits // N`

`{-4755.04, -0.000685831}`

An analogous analytic development awaits a future revision to this document, as do examples that involve gears or externally applied torque.

8. More Administrivia

```
saveDefinitions[outputDirectory <> "Misc.m",
{importMotorData, motorParameters, motorLoad, flywheel, addMotorLoad, gearboxlessEquivalent, massOnPulley,
stepMotorTimeFunctionGeneric, stepMotorTimeFunction, ss, simpleSs,
velAfterStepGeneric, velStepGeneric, torAfterStepGeneric,
torStepGeneric, curStepGeneric, emfStepGeneric, posStepGeneric, posAfterStepGeneric,
accStepGeneric, accAfterStepGeneric
}]
```

9. Revision History

- 2018.07.23 Added acceleration-dependant external torque. Corrected “viscous friction” to “viscous drag”. Clarified usage of units.
- 2018.08.06 Added gearbox support, removed time consuming internal tests. Regularized state variable names. Removed (flawed) “Back EMF vs Applied Voltage” discussion
- 2018.08.06 Renamed variables. Remove setting \$Assumptions (that can slow things down)
- 2018.08.11 Added initial conditions support
- 2018.08.12 Bug fixes, miscellaneous polish
- 2018.08.13 Introduced ‘step’ terminology
- 2018.08.16 Adjusted to new schema in motor parameters spreadsheet
- 2018.08.24 Add gearboxlessEquivalent[]
- 2018.08.27 Add accStepGeneric. Lighten background in output.