

Motor Physics


Robert Atkinson, Maddy Nguyen, Samantha Hordyk, Cole Welch, and John Fraser
19 July 2018

An exploration of the physics of motors. We begin by loading some handy utilities.

```
Get[NotebookDirectory[] <> "Utilities.m"]
```

Core Motor Model

The following gives a conceptual overview of a DC motor (ref: <http://www.egr.msu.edu/classes/me451/jchoi/2008/>).



The angular velocity is the time derivative of the position, so the velocity is redundant. With that considered, we can see that the state of the system can be described by the following variables.

- V , the applied voltage
- i , the armature current
- e , the back EMF
- θ , the angular position

The parameter constants that characterize the motor are the following

- J , Moment of inertia
- b , Motor viscous friction constant
- K_e , Electromotive force constant

- K_t , Motor torque constant
- R , Electric resistance
- L Electric inductance

Further, the moment of inertia J and viscous friction constant b can be decomposed into additive contributions from both the motor and the load; we'll return to that later.

```
motorParameterNames = {J, b, Ke, Kt, R, L}
```

```
{J, b, Ke, Kt, R, L}
```

Differential Equations

Differential Equation Development

The system can be modelled as a set of differential equations

```
diffEqns = {}
```

```
{}
```

Torque is proportional to current.

```
diffEqns = Union[diffEqns, {τ[t] == Kt i[t]}]
```

```
{τ[t] == Kt i[t]}
```

The back EMF is proportional to velocity.

```
diffEqns = Union[diffEqns, {e[t] == Ke θ'[t]}]
```

```
{e[t] == Ke θ'[t], τ[t] == Kt i[t]}
```

Viscous friction reduces the torque, and what's left accelerates the inertial mass.

```
diffEqns = Union[diffEqns, {τ[t] - b θ'[t] == J θ''[t]}]
```

```
{e[t] == Ke θ'[t], τ[t] == Kt i[t], τ[t] - b θ'[t] == J θ''[t]}
```

The voltage drop across the resistive, inductive, and back EMF must equal the applied voltage (due to one of Kirchhoff's laws).

```
diffEqns = Union[diffEqns, {V[t] == R i[t] + L i'[t] + e[t]}]
```

```
{e[t] == Ke θ'[t], V[t] == e[t] + R i[t] + L i'[t], τ[t] == Kt i[t], τ[t] - b θ'[t] == J θ''[t]}
```

```
diffEqns // prettyPrint
```

$$\begin{aligned}e(t) &= K_e \theta'(t) \\ V(t) &= e(t) + R i(t) + L i'(t) \\ \tau(t) &= K_t i(t) \\ \tau(t) - b \theta'(t) &= J \theta''(t)\end{aligned}$$

Initial Conditions

Some of the initial conditions we can intuit intellectually:

```
initialConditions = {
   $\theta[0] == 0,$ 
   $i[0] == 0,$ 
   $e[0] == 0,$ 
   $V[0] == v$ 
};
```

Others we can solve for:

```
others = {i'[0],  $\theta'[0]$ ,  $\theta''[0]$ ,  $\tau[0]$ };
diffEqns /. t -> 0 /. (initialConditions /. Equal -> Rule)
(allInitialConditions =
  (uniqueSolve[%, others] /. Rule -> Equal) ~Join~ initialConditions) // prettyPrint
{ $\theta == K_e \theta'[0]$ ,  $v == L i'[0]$ ,  $\tau[0] == 0$ ,  $\tau[0] - b \theta'[0] == J \theta''[0]$ }
```

$$\begin{aligned}i'(0) &= \frac{v}{L} \\ \theta'(0) &= 0 \\ \theta''(0) &= 0 \\ \tau(0) &= 0 \\ \theta(0) &= 0 \\ i(0) &= 0 \\ e(0) &= 0 \\ V(0) &= v\end{aligned}$$

Motor Parameters

We now take a moment to carefully work out the units of each parameter and variable. Some units are obvious and clear. Let's write those down.

```

parameterUnits = {
  R → parseUnit["Ohms"],
  L → parseUnit["Henrys"],
  θ[t] → parseUnit["Radians"],
  θ'[t] → parseUnit["Radians per Second"],
  θ''[t] → parseUnit["Radians per Second per Second"],
  i[t] → parseUnit["Amperes"],
  i'[t] → parseUnit["Amperes"] / parseUnit["Second"],
  i''[t] → parseUnit["Amperes"] / parseUnit["Second"] / parseUnit["Second"],
  V[t] → parseUnit["Volts"],
  e[t] → parseUnit["Volts"],
  τ[t] → parseUnit["Newton Meters / Radians"]
} // Association
unitsToQuantities[units_] := Module[{rules, makeQuantity},
  rules = Normal[units];
  makeQuantity = Function[{param, unit}, Quantity[param, unit]];
  (#[[1]] → makeQuantity @@ # & /@ rules) // Association
]
parameterQuantities = unitsToQuantities[parameterUnits]

```

$$\left\langle \begin{array}{l}
R \rightarrow \text{Ohms}, L \rightarrow \text{Henries}, \theta[t] \rightarrow \text{Radians}, \theta'[t] \rightarrow \frac{\text{Radians}}{\text{Seconds}}, \\
\theta''[t] \rightarrow \frac{\text{Radians}}{\text{Seconds}^2}, i[t] \rightarrow \text{Amperes}, i'[t] \rightarrow \frac{\text{Amperes}}{\text{Seconds}}, i''[t] \rightarrow \frac{\text{Amperes}}{\text{Seconds}^2}, \\
V[t] \rightarrow \text{Volts}, e[t] \rightarrow \text{Volts}, \tau[t] \rightarrow \frac{\text{Meters Newtons}}{\text{Radians}}
\end{array} \right\rangle$$

$$\left\langle \begin{array}{l}
R \rightarrow R \Omega, L \rightarrow L H, \theta[t] \rightarrow \theta[t] \text{ rad}, \theta'[t] \rightarrow \theta'[t] \text{ rad/s}, \\
\theta''[t] \rightarrow \theta''[t] \text{ rad/s}^2, i[t] \rightarrow i[t] \text{ A}, i'[t] \rightarrow i'[t] \text{ A/s}, \\
i''[t] \rightarrow i''[t] \text{ A/s}^2, V[t] \rightarrow V[t] \text{ V}, e[t] \rightarrow e[t] \text{ V}, \tau[t] \rightarrow \tau[t] \text{ mN/rad}
\end{array} \right\rangle$$

Let's look at how far that takes us:

```
diffEqns /. parameterQuantities // ColumnForm
```

$$\begin{aligned}
e[t] \text{ V} &= K_e \left(\theta'[t] \text{ rad/s} \right) \\
V[t] \text{ V} &= \left(e[t] + R i[t] + L i'[t] \right) \text{ A } \Omega \\
\tau[t] \text{ mN/rad} &= K_t \left(i[t] \text{ A} \right) \\
\tau[t] \text{ mN/rad} + b \left(-\theta'[t] \text{ rad/s} \right) &= J \left(\theta''[t] \text{ rad/s}^2 \right)
\end{aligned}$$

We can see that if we write down the units for either b or J , the remaining units will be fully determined. We choose to specify J .

```
parameterUnits[J] = parseUnit["kg m^2"] / parseUnit["Radians^2"];
parameterQuantities = unitsToQuantities[parameterUnits]
diffEqns /. parameterQuantities // ColumnForm
remainingUnits = uniqueSolve[diffEqns[{{1, 3, 4}}], {Ke, Kt, b}]
```

```
<| R → R Ω, L → L H, θ[t] → θ[t] rad, θ'[t] → θ'[t] rad/s,
  θ''[t] → θ''[t] rad/s^2, i[t] → i[t] A, i'[t] → i'[t] A/s, i''[t] → i''[t] A/s^2,
  V[t] → V[t] V, e[t] → e[t] V, τ[t] → τ[t] mN/rad, J → J kg m^2/rad^2 |>
```

```
e[t] V == Ke (θ'[t] rad/s)
V[t] V == (e[t] + R i[t] + L i'[t]) A Ω
τ[t] mN/rad == Kt (i[t] A)
τ[t] mN/rad + b (-θ'[t] rad/s) == J θ''[t] kg m^2/(s^2 rad)
```

```
{Ke →  $\frac{e[t]}{\theta'[t]}$ , Kt →  $\frac{\tau[t]}{i[t]}$ , b →  $\frac{\tau[t] - J \theta''[t]}{\theta'[t]}$ }
```

Let's check how those work.

```
remainingUnits /. parameterQuantities
```

```
{Ke →  $\frac{e[t]}{\theta'[t]}$  s V/rad, Kt →  $\frac{\tau[t]}{i[t]}$  mN/(A rad), b →  $\frac{\tau[t] - J \theta''[t]}{\theta'[t]}$  m s N/rad^2 }
```

Huzzah! They check out (if the units were incompatible, Mathematica would've told us). Let's the remaining units to our kit.

```
(parameterUnits[#] = QuantityUnit[
  # /. (remainingUnits /. parameterQuantities)]) & /@ remainingUnits[All, 1]
parameterQuantities = unitsToQuantities[parameterUnits]
```

```
{  $\frac{\text{Seconds Volts}}{\text{Radians}}$ ,  $\frac{\text{Meters Newtons}}{\text{Amperes Radians}}$ ,  $\frac{\text{Meters Newtons Seconds}}{\text{Radians}^2}$  }
```

```
<| R → R Ω, L → L H, θ[t] → θ[t] rad, θ'[t] → θ'[t] rad/s,
  θ''[t] → θ''[t] rad/s^2, i[t] → i[t] A, i'[t] → i'[t] A/s,
  i''[t] → i''[t] A/s^2, V[t] → V[t] V, e[t] → e[t] V, τ[t] → τ[t] mN/rad,
  J → J kg m^2/rad^2, Ke → Ke s V/rad, Kt → Kt mN/(A rad), b → b m s N/rad^2 |>
```

Laplace Transforms and Transfer Function Models

Laplace Transforms

Returning to the differential equations, we form the Laplace Transform of each, then solve for the various transforms. First, we apply the `LaplaceTransform[]` function to each equation. It automatically makes use of the linearity of the transform, and insinuates itself just around the time dependent parts (ie: the parts dependent on the time variable we told it about, namely t).

```
leqns = LaplaceTransform[#, t, s] & /@ diffEqns
leqns // prettyPrint

{LaplaceTransform[e[t], t, s] == Ke (s LaplaceTransform[θ[t], t, s] - θ[0]),
 LaplaceTransform[V[t], t, s] = LaplaceTransform[e[t], t, s] +
   R LaplaceTransform[i[t], t, s] + L (-i[0] + s LaplaceTransform[i[t], t, s]),
 LaplaceTransform[τ[t], t, s] == Kt LaplaceTransform[i[t], t, s],
 LaplaceTransform[τ[t], t, s] - b (s LaplaceTransform[θ[t], t, s] - θ[0]) ==
   J (s^2 LaplaceTransform[θ[t], t, s] - s θ[0] - θ'[0]) }
```

$$\begin{aligned}\mathcal{L}_t[e(t)](s) &= Ke (s (\mathcal{L}_t[\theta(t)](s)) - \theta(0)) \\ \mathcal{L}_t[V(t)](s) &= \mathcal{L}_t[e(t)](s) + R (\mathcal{L}_t[i(t)](s)) + L (s (\mathcal{L}_t[i(t)](s)) - i(0)) \\ \mathcal{L}_t[\tau(t)](s) &= Kt (\mathcal{L}_t[i(t)](s)) \\ \mathcal{L}_t[\tau(t)](s) - b (s (\mathcal{L}_t[\theta(t)](s)) - \theta(0)) &= J ((\mathcal{L}_t[\theta(t)](s)) s^2 - \theta(0) s - \theta'(0))\end{aligned}$$

Next, we walk those equations, picking up those insinuations and sowing them to the wind, reaping them on the outside, and finally removing duplicates

```
allXforms =
  Reap[Scan[(If[MatchQ[#, _LaplaceTransform], Sow[#]]) &, leqns, Infinity]][[2, 1]] //
  Union

{LaplaceTransform[e[t], t, s],
 LaplaceTransform[i[t], t, s], LaplaceTransform[V[t], t, s],
 LaplaceTransform[θ[t], t, s], LaplaceTransform[τ[t], t, s]}
```

The voltage transform is input; all the others are outputs. Solve for the outputs. Finally substitute what we know about initial conditions, and simplify as much as we can.

```

voltageXform = LaplaceTransform[V[t], t, s];
outputXforms = Complement[allXforms, {voltageXform}]
solvedXforms = uniqueSolve[leqns, outputXforms]
solvedXforms =
  solvedXforms /. (allInitialConditions /. Equal → Rule) // FullSimplify

```

```

{LaplaceTransform[e[t], t, s], LaplaceTransform[i[t], t, s],
 LaplaceTransform[θ[t], t, s], LaplaceTransform[τ[t], t, s]}

```

```

{LaplaceTransform[e[t], t, s] →
  - ( (-Ke Kt L i[0] - Ke Kt LaplaceTransform[V[t], t, s] - J Ke R θ'[0] - J Ke L s θ'[0]) /
    (Ke Kt + b R + b L s + J R s + J L s^2)),
 LaplaceTransform[i[t], t, s] → - ( (-b L i[0] - J L s i[0] - b LaplaceTransform[V[t], t, s] -
    J s LaplaceTransform[V[t], t, s] + J Ke θ'[0]) / (Ke Kt + b R + b L s + J R s + J L s^2)),
 LaplaceTransform[θ[t], t, s] → - ( (-Kt L i[0] - Kt LaplaceTransform[V[t], t, s] -
    Ke Kt θ[0] - b R θ[0] - b L s θ[0] - J R s θ[0] - J L s^2 θ[0] - J R θ'[0] - J L s θ'[0]) /
    (s (Ke Kt + b R + b L s + J R s + J L s^2))), LaplaceTransform[τ[t], t, s] →
  - ( (-b Kt L i[0] - J Kt L s i[0] - b Kt LaplaceTransform[V[t], t, s] - J Kt s
    LaplaceTransform[V[t], t, s] + J Ke Kt θ'[0]) / (Ke Kt + b R + b L s + J R s + J L s^2))}

```

```

{LaplaceTransform[e[t], t, s] →  $\frac{Ke Kt LaplaceTransform[V[t], t, s]}{Ke Kt + (b + J s) (R + L s)},$ 
 LaplaceTransform[i[t], t, s] →  $\frac{(b + J s) LaplaceTransform[V[t], t, s]}{Ke Kt + (b + J s) (R + L s)},$ 
 LaplaceTransform[θ[t], t, s] →  $\frac{Kt LaplaceTransform[V[t], t, s]}{s (Ke Kt + (b + J s) (R + L s))},$ 
 LaplaceTransform[τ[t], t, s] →  $\frac{Kt (b + J s) LaplaceTransform[V[t], t, s]}{Ke Kt + (b + J s) (R + L s)}\}$ 

```

Transfer Function Models

We use the solved transforms to create TransferFunctionModel[]s for every variable of interest.

```

Clear[makeModel]
makeModel[var_] := makeModel[var, ToString[var], 1]
makeModel[var_, label_, coefficient_] := TransferFunctionModel[
  (LaplaceTransform[var[t], t, s] * coefficient /. solvedXforms) / voltageXform,
  s,
  SystemsModelLabels -> {"V", label}]
motorPositionModel = makeModel[θ]
motorVelocityModel = makeModel[θ, "Ω", s]
motorAccelerationModel = makeModel[θ, "α", s * s]
motorCurrentModel = makeModel[i]
motorEMFModel = makeModel[e]
motorTorqueModel = makeModel[τ]

```

$$\left(\begin{array}{c|c} & V \\ \hline \theta & Kt \\ \hline & s (Ke Kt + (b + J s) (R + L s)) \end{array} \right) \mathcal{T}$$

$$\left(\begin{array}{c|c} & V \\ \hline \Omega & Kt \\ \hline & Ke Kt + (b + J s) (R + L s) \end{array} \right) \mathcal{T}$$

$$\left(\begin{array}{c|c} & V \\ \hline \alpha & Kt s \\ \hline & Ke Kt + (b + J s) (R + L s) \end{array} \right) \mathcal{T}$$

$$\left(\begin{array}{c|c} & V \\ \hline i & b + J s \\ \hline & Ke Kt + (b + J s) (R + L s) \end{array} \right) \mathcal{T}$$

$$\left(\begin{array}{c|c} & V \\ \hline e & Ke Kt \\ \hline & Ke Kt + (b + J s) (R + L s) \end{array} \right) \mathcal{T}$$

$$\left(\begin{array}{c|c} & V \\ \hline \tau & Kt (b + J s) \\ \hline & Ke Kt + (b + J s) (R + L s) \end{array} \right) \mathcal{T}$$

Motor Parameters Redux

On Ke & Kt

It is often remarked that, in compatible units, Ke & Kt have the same magnitude and sign. However,

while often true, this relation does not always hold. Recall our differential equations:

```
diffEqns // ColumnForm
```

```
e[t] == Ke θ'[t]
V[t] == e[t] + R i[t] + L i'[t]
τ[t] == Kt i[t]
τ[t] - b θ'[t] == J θ''[t]
```

In steady state, the current and speed are constant:

```
(ssEqns = diffEqns /. {i'[t] → 0, θ''[t] → 0}) // ColumnForm
(ssEqns = Eliminate[ssEqns, {e[t], b}]) // ColumnForm
```

```
e[t] == Ke θ'[t]
V[t] == e[t] + R i[t]
τ[t] == Kt i[t]
τ[t] - b θ'[t] == 0
```

```
V[t] == R i[t] + Ke θ'[t]
τ[t] == Kt i[t]
```

From a power point of view (ref: <https://bit.ly/2Lxka0Z>), we must have (electrical) power in = (mechanical) power out + (electrical + mechanical) power losses:

```
powerEqns = And @@ {
  pwrIn == V[t] i[t],
  pwrOut == τ[t] θ'[t],
  pwrLosses == i[t]^2 R + pwrMechanicalPowerLosses,
  pwrIn == pwrOut + pwrLosses
};
(ssPowerEqns = ssEqns ~Join~ powerEqns) // prettyPrint
```

```
V(t) = R i(t) + Ke θ'(t)
τ(t) = Kt i(t)
pwrIn = i(t) V(t)
pwrOut = τ(t) θ'(t)
pwrLosses = R i(t)^2 + pwrMechanicalPowerLosses
pwrIn = pwrLosses + pwrOut
```

Let's eliminate a few of the variables

```
kBalanceEqns = Eliminate[ssPowerEqns, {i[t], τ[t], θ'[t], V[t]};
kBalanceEqns // prettyPrint
uniqueSolve[kBalanceEqns[[2]], pwrMechanicalPowerLosses] /. Rule -> Equal
```

```
pwrIn = pwrLosses + pwrOut
Kt pwrMechanicalPowerLosses = (Ke - Kt) pwrOut
```

$$\{ \text{pwrMechanicalPowerLosses} == \frac{(K_e - K_t) \text{pwrOut}}{K_t} \}$$

We can conclude that if the mechanical (and other) power losses are zero, then K_e must equal K_t .

Experimental Motor Data

We have experimental data from several motors (Nguyen, Hordyk, & Fraser, 2017). We load in same.

```
(motorData = Import[
  NotebookDirectory[] <> "Characterized Motors.xlsx", {"Data", 1}]) // TableForm
```

Name	J	b	K	R	L (uH)	L (H)	J
AM 20 A	9.011×10^{-6}	0.0022	0.351	2.3	691.	0.000691	9.011×10^{-6}
AM 20 B	9.011×10^{-6}	0.0025	0.389	1.9	684.	0.000684	9.011×10^{-6}
AM 20 C	8.931×10^{-6}	0.0028	0.385	5.1	717.	0.000717	8.931×10^{-6}
AM 40 A	2.221×10^{-5}	0.2269	0.753	2.5	674.	0.000674	0.00002221
AM 40 B	1.741×10^{-5}	0.56	0.705	3.8	705.	0.000705	0.00001741
AM 40 C	2.471×10^{-5}	0.018	0.763	2.1	716.	0.000716	0.00002471
AM 60 A	1.041×10^{-5}	0.033	1.066	3.3	694.	0.000694	0.00001041
AM 60 B	8.421×10^{-6}	0.02	1.076	5.1	696.	0.000696	8.421×10^{-6}
AM 3.7 A	2.791×10^{-5}	0.00014	0.099	8.9	679.	0.000679	0.00002791
AM 3.7 B	3.151×10^{-5}	0.000176	0.108	2.6	797.	0.000797	0.00003151
AM 3.7 C	3.091×10^{-5}	0.00017	0.105	8.7	880.	0.00088	0.00003091
Matrix A	9.431×10^{-6}	0.00151	0.34	3.8	718.	0.000718	9.431×10^{-6}
Matrix B	7.761×10^{-6}	0.00191	0.363	7.8	777.	0.000777	7.761×10^{-6}
Matrix C	7.231×10^{-6}	0.00186	0.338	20.6	658.	0.000658	7.231×10^{-6}
CoreHex A	7.331×10^{-4}	0.0112	0.822	3.6	1356.	0.001356	0.0007331
CoreHex B	6.551×10^{-4}	0.008	0.858	11.3	1352.	0.001352	0.0006551
CoreHex C	4.541×10^{-4}	0.0078	0.711	5.6	1342.	0.001342	0.0004541

We build a function to retrieve data from the table. We convert from floating point to rational in order to help delay floating point collapse later on down the line. We allow for optional load inertia and viscous friction.

```

siAngularInertialUnits = parseUnit["kg m^2"] / parseUnit["Radians^2"];
siAngularViscousFrictionUnits = parseUnit["N m s"] / parseUnit["Radians^2"];

```

```

Clear[motorParameters]
Options[motorParameters] = {
  JLoad → Quantity[0, siAngularInertialUnits],
  bLoad → Quantity[0, siAngularViscousFrictionUnits]
};
motorParameters[motorName_, opts : OptionsPattern[]] :=
Module[{row, paramValues, quantify, assoc},
  row = Select[motorData, #[[1]] == motorName &][[1]];
  paramValues = #[[1]] → toRational[row[[#[[2]]]]] & /@
    {{J, 8}, {b, 3}, {Ke, 4}, {Kt, 4}, {R, 5}, {L, 7}};
  quantify = Function[{name, value},
    name → ((name /. parameterQuantities) /. name → value)
  ];
  assoc = quantify @@ # & /@ paramValues // Association;
  assoc[J] = assoc[J] + OptionValue[JLoad];
  assoc[b] = assoc[b] + OptionValue[bLoad];
  assoc
];

```

```
motorParameters["AM 60 A", JLoad → Quantity[1, siAngularInertialUnits]]
```

$$\left\langle \begin{array}{l} J \rightarrow \frac{100\,001\,041}{100\,000\,000} \text{ kg m}^2/\text{rad}^2, \quad b \rightarrow \frac{33}{1000} \text{ m s N/rad}^2, \\ \\ Ke \rightarrow \frac{533}{500} \text{ s V/rad}, \quad Kt \rightarrow \frac{533}{500} \text{ m N/(A rad)}, \quad R \rightarrow \frac{33}{10} \Omega, \quad L \rightarrow \frac{347}{500\,000} \text{ H} \end{array} \right\rangle$$

Time Domain Functions

We define a motor with a load that we'll use to illustrate examples. We also define a generic, abstract motor that can help explore things symbolically

```

siAngularInertialUnits
aMotor = motorParameters["AM 60 A", JLoad → Quantity[1, siAngularInertialUnits]]

```

Kilograms Meters²
Radians²

$$\left\langle \begin{array}{l} J \rightarrow \frac{100\,001\,041}{100\,000\,000} \text{ kg m}^2/\text{rad}^2, \quad b \rightarrow \frac{33}{1000} \text{ m s N/rad}^2, \\ \\ Ke \rightarrow \frac{533}{500} \text{ s V/rad}, \quad Kt \rightarrow \frac{533}{500} \text{ m N/(A rad)}, \quad R \rightarrow \frac{33}{10} \Omega, \quad L \rightarrow \frac{347}{500\,000} \text{ H} \end{array} \right\rangle$$

```
genericMotor = # → (# /. parameterQuantities) & /@ motorParameterNames

{J → J kg m2/rad2, b → b m s N/rad2, Ke → Ke s V/rad, Kt → Kt m N/(A rad), R → R Ω, L → L H}
```

We make time domain functions for each of the values of interest

```
Clear[makeTimeDomainFunction, motorPosition,
      motorVelocity, motorCurrent, motorEMF, motorTorque]
makeTimeDomainFunction[parameters_, transferFunctionModel_, stimulus_] := Module[
  {transferExpr, inputExpr, outputExpr, s = Unique[], t = Unique[], tExpr},
  transferExpr = transferFunctionModel[s][[1, 1]] /. parameters // siUnits;
  inputExpr = LaplaceTransform[stimulus[t], t, s];
  outputExpr = transferExpr inputExpr;
  tExpr = InverseLaplaceTransform[outputExpr, s, t];
  tExpr /. t → # &]

motorPosition[parameters_, V_] := motorPosition[parameters, V] =
  makeTimeDomainFunction[parameters, motorPositionModel, V UnitStep[#] &]
motorVelocity[parameters_, V_] := motorVelocity[parameters, V] =
  makeTimeDomainFunction[parameters, motorVelocityModel, V UnitStep[#] &]
motorCurrent[parameters_, V_] := motorCurrent[parameters, V] =
  makeTimeDomainFunction[parameters, motorCurrentModel, V UnitStep[#] &]
motorEMF[parameters_, V_] := motorEMF[parameters, V] =
  makeTimeDomainFunction[parameters, motorEMFModel, V UnitStep[#] &]
motorTorque[parameters_, V_] := motorTorque[parameters, V] =
  makeTimeDomainFunction[parameters, motorTorqueModel, V UnitStep[#] &]
```


We explore same. Notice how the units come out correctly. Huzzah.

```
pos = motorPosition[aMotor, Quantity[12, "Volts"]][Quantity[t, "Seconds"]] // N //
FullSimplify
pos // siUnits
Plot[pos, {t, 0, 100}]
```

$$e^{-4754.7 t} \left(-1.7149 \times 10^{-7} \text{ s}^3 \text{A rad V} / (\text{kg m}^2) \right) +$$

$$e^{-0.377374 t} \left(27.2234 \text{ s}^3 \text{A rad V} / (\text{kg m}^2) \right) + (-27.2234 + 10.2726 t) \text{ s}^3 \text{A rad V} / (\text{kg m}^2)$$

$$e^{-4754.7 t} \left(-1.7149 \times 10^{-7} \text{ rad} \right) + e^{-0.377374 t} \left(27.2234 \text{ rad} \right) + (-27.2234 + 10.2726 t) \text{ rad}$$




```

vel = motorVelocity[aMotor, Quantity[12, "Volts"]][Quantity[t, "Seconds"]] // N //
FullSimplify
vel // siUnits
Plot[vel, {t, 0, 10}]

```

$$e^{-0.377374 t} \left(-10.2734 \text{ s}^2 \text{A rad V} / (\text{kg m}^2) \right) + e^{-4754.7 t} \left(0.000815385 \text{ s}^2 \text{A rad V} / (\text{kg m}^2) \right) + 10.2726 \text{ s}^2 \text{A rad V} / (\text{kg m}^2)$$


$$e^{-0.377374 t} \left(-10.2734 \text{ rad/s} \right) + e^{-4754.7 t} \left(0.000815385 \text{ rad/s} \right) + 10.2726 \text{ rad/s}$$



```
cur = motorCurrent[aMotor, Quantity[12, "Volts"]][Quantity[t, "Seconds"]] // N //
FullSimplify
cur // siUnits // simplifyUnits
Plot[cur, {t, 0, 10}]
```

$$e^{-4754.7 t} \left(-3.63689 \text{ s}^3 \text{ A}^2 \text{ V} / (\text{kg m}^2) \right) + 0.318007 \text{ s}^3 \text{ A}^2 \text{ V} / (\text{kg m}^2) + e^{-0.377374 t} \left(3.31888 \text{ s}^3 \text{ A}^2 \text{ V} / (\text{kg m}^2) \right)$$


$$e^{-4754.7 t} \left(-3.63689 \text{ A} \right) + 0.318007 \text{ A} + e^{-0.377374 t} \left(3.31888 \text{ A} \right)$$



```
emf = motorEMF[aMotor, Quantity[12, "Volts"]][Quantity[t, "Seconds"]] // N //
FullSimplify
emf // siUnits // simplifyUnits
Plot[emf, {t, 0, 10}]
```

$$e^{-0.377374 t} \left(-10.9514 \text{ V} \right) + e^{-4754.7 t} \left(0.000869201 \text{ V} \right) + 10.9506 \text{ V}$$

$$e^{-0.377374 t} \left(-10.9514 \text{ V} \right) + e^{-4754.7 t} \left(0.000869201 \text{ V} \right) + 10.9506 \text{ V}$$




```

tor = motorTorque[aMotor, Quantity[12, "Volts"]][Quantity[t, "Seconds"]] // N //
FullSimplify
tor // siUnits // simplifyUnits
Plot[tor, {t, 0, 10}]

```

$$e^{-4754.7 t} \left(-3.87693 \text{ s AV/rad} \right) + 0.338995 \text{ s AV/rad} + e^{-0.377374 t} \left(3.53793 \text{ s AV/rad} \right)$$

$$e^{-4754.7 t} \left(-3.87693 \text{ kg m}^2 / (\text{s}^2 \text{rad}) \right) + 0.338995 \text{ kg m}^2 / (\text{s}^2 \text{rad}) + e^{-0.377374 t} \left(3.53793 \text{ kg m}^2 / (\text{s}^2 \text{rad}) \right)$$



Steady State

We explore the steady state behavior. We can compute same either using the final value theorem, or directly by taking the limit. The former is definitely more efficient.


```

Clear[ssValue]
ssValue[model_, stimulus_] := Module[{s = Unique[], t = Unique[], expr},
  expr = model[s] * LaplaceTransform[stimulus[t], t, s];
  Limit[s expr, s → 0]
]
ssPos → ssValue[motorPositionModel, √UnitStep[#] &]
ssVel → ssValue[motorVelocityModel, √UnitStep[#] &]
ssAcc → ssValue[motorAccelerationModel, √UnitStep[#] &]
ssEmf → ssValue[motorEMFModel, √UnitStep[#] &]
ssCur → ssValue[motorCurrentModel, √UnitStep[#] &]
ssTor → ssValue[motorTorqueModel, √UnitStep[#] &]

```

```
ssPos → {{Indeterminate}}
```

```
ssVel → {{ $\frac{K_t \sqrt{v}}{K_e K_t + b R}$ }}
```

```
ssAcc → {{0}}
```

```
ssEmf → {{ $\frac{K_e K_t \sqrt{v}}{K_e K_t + b R}$ }}
```

```
ssCur → {{ $\frac{b \sqrt{v}}{K_e K_t + b R}$ }}
```

```
ssTor → {{ $\frac{b K_t \sqrt{v}}{K_e K_t + b R}$ }}
```

For comparison, we also exhibit velocity calculated the hard way.

```

genericVelExpr =
  motorVelocity[genericMotor, Quantity[v, "Volts"]][Quantity[t, "Seconds"]];
genericVelExpr = genericVelExpr // siUnits // clearUnits
parameterAssumptions = # > 0 & /@ Select[Keys[parameterQuantities], unboundQ[#] &]
ssVelHardWay = Limit[genericVelExpr, t -> Infinity, Assumptions -> parameterAssumptions]

```

$$\begin{aligned}
& Kt \left(\frac{1}{Ke Kt + b R} - \left(-b e^{\frac{(-b L - J R - \sqrt{-4 J L (Ke Kt + b R) + (b L + J R)^2}) t}{2 J L}} L + b e^{\frac{(-b L - J R + \sqrt{-4 J L (Ke Kt + b R) + (b L + J R)^2}) t}{2 J L}} L - \right. \right. \\
& e^{\frac{(-b L - J R - \sqrt{-4 J L (Ke Kt + b R) + (b L + J R)^2}) t}{2 J L}} J R + e^{\frac{(-b L - J R + \sqrt{-4 J L (Ke Kt + b R) + (b L + J R)^2}) t}{2 J L}} J R + \\
& e^{\frac{(-b L - J R - \sqrt{-4 J L (Ke Kt + b R) + (b L + J R)^2}) t}{2 J L}} \sqrt{-4 J Ke Kt L + b^2 L^2 - 2 b J L R + J^2 R^2} + \\
& \left. e^{\frac{(-b L - J R + \sqrt{-4 J L (Ke Kt + b R) + (b L + J R)^2}) t}{2 J L}} \sqrt{-4 J Ke Kt L + b^2 L^2 - 2 b J L R + J^2 R^2} \right) / \\
& \left(2 (Ke Kt + b R) \sqrt{-4 J Ke Kt L + b^2 L^2 - 2 b J L R + J^2 R^2} \right) v
\end{aligned}$$

{R > 0, L > 0, J > 0, Ke > 0, Kt > 0, b > 0}

$$\begin{aligned}
& \text{ConditionalExpression} \left[\frac{Kt v}{Ke Kt + b R}, \right. \\
& \left(\frac{1}{\sqrt{-4 J Ke Kt L + b^2 L^2 - 2 b J L R + J^2 R^2}} \mid \sqrt{-4 J Ke Kt L + b^2 L^2 - 2 b J L R + J^2 R^2} \mid \right. \\
& \left. \sqrt{-4 J L (Ke Kt + b R) + (b L + J R)^2} \mid v \right) \in \mathbb{R} \&\& \\
& \sqrt{-4 J Ke Kt L + b^2 L^2 - 2 b J L R + J^2 R^2} > 0 \&\& \sqrt{-4 J Ke Kt L + b^2 L^2 - 2 b J L R + J^2 R^2} < b L + J R \&\& \\
& b L + J R + \sqrt{-4 J Ke Kt L + b^2 L^2 - 2 b J L R + J^2 R^2} > 0 \left. \right]
\end{aligned}$$

Good: the two limits match. Also, the conditional predicate in the hard-way approach illustrates the conditions under which the final value theorem is applicable, namely (a) that all non-zero roots of the denominator of the transform must have negative real parts, and (b) the transform must not have more than one pole at the origin (ref: https://en.wikipedia.org/wiki/Final_value_theorem #Example _where _FVT _holds).